

CAST3M EXAMPLES

E. LE FICHOUX

ANNOTATED TESTING FILES



This document has been initially designed and achieved by Emmanuelle LE FICHOUX
at the French *École Nationale Supérieure des Techniques Avancées* (ENSTA).

Cast3M documentation 2011

<http://www-cast3m.cea.fr>

Cast3M is a general Finite Element computer code for fluid and structural mechanics. Cast3M is developed by the "Département de Modélisation des Systèmes et Structures" (DM2S) of the "Direction de l'Énergie Nucléaire" of the french Alternative Energies and Atomic Energy Commission (CEA).

The development of Cast3M enters within the framework of a research activity in the field of mechanics. The goal is to build a high level instrument, able to be used as a valid support for the design, the dimensioning and the analysis of structures and components.

Accordingly, Cast3M represents a complete system, integrating not only the functions of calculation, but also the functions of model development (pre-processor) and result analysis (post-processor). Cast3M is a « toolbox » program in which the user can develop functions that meet its own needs.

Cast3M is more particularly used in the domain of nuclear energy as a simulation tool, as well as a generic platform for the development of specialized program applications. In particular, Cast3M is used by the "Institut de Radioprotection et de Sûreté Nucléaire" (IRSN) within the framework of safety analysis of the french nuclear facilities.



Contents

1	ELASTICITY	1
1.1	Test elas1 (Description sheet)	2
1.2	Test elas2 (Description sheet)	6
1.3	Test elas3 (Description sheet)	10
1.4	Test elas4 (Description sheet)	14
1.5	Test elas7 (Description sheet)	19
1.6	Test elas8 (Description sheet)	22
1.7	Test elas11 (Description sheet)	26
1.8	Test elas13 (Description sheet)	31
1.9	Test stru2 (Description sheet)	35
1.10	Test stru4 (Description sheet)	41
1.11	Test elas20 (Description sheet)	46
1.12	Test testICQ4 (Description sheet)	50
1.13	Test vsur1 (Description sheet)	53
2	BUCKLING	57
2.1	Test flam1 (Description sheet)	58
2.2	Test four1 (Description sheet)	62
3	PLASTICITY	67
3.1	Test plas1 (Description sheet)	68
3.2	Test plas2 (Description sheet)	74
3.3	Test plas4 (Description sheet)	78
3.4	Test plas5 (Description sheet)	81
3.5	Test plas6 (Description sheet)	85
3.6	Test plas7 (Description sheet)	89
3.7	Test plas8 (Description sheet)	93
3.8	Test plas9 (Description sheet)	96
3.9	Test plas10 (Description sheet)	99
3.10	Test plas11 (Description sheet)	102
3.11	Test plas12 (Description sheet)	106
3.12	Test plas13 (Description sheet)	110
3.13	Test plas14 (Description sheet)	112
4	BEHAVIOUR LAWS	117
4.1	Test norton_cis1 (Description sheet)	118
4.2	Test chaboche2 (Description sheet)	123
4.3	Test ohno2 (Description sheet)	128
4.4	Test guionnet_tra (Description sheet)	133
4.5	Test zerill (Description sheet)	138

4.6	Test preston1 (Description sheet)	142
5	DAMAGE ELASTOPLASTIC MODELS	149
5.1	Test endocp1 (Description sheet)	150
5.2	Test endoaxi1 (Description sheet)	155
5.3	Test endoaxi2 (Description sheet)	159
5.4	Test endoaxi3 (Description sheet)	163
5.5	Test ddi (Description sheet)	168
6	FRACTURE MECHANICS	173
6.1	Test rupt1 (Description sheet)	174
6.2	Test rupt2 (Description sheet)	180
6.3	Test rupt3 (Description sheet)	187
6.4	Test rupt4 (Description sheet)	191
6.5	Test rupt5 (Description sheet)	194
6.6	Test rupt6 (Description sheet)	197
6.7	Test rupt7 (Description sheet)	203
6.8	Test rupt8 (Description sheet)	206
6.9	Test rupt9 (Description sheet)	209
7	COMPOSITE	213
7.1	Test comp1 (Description sheet)	214
7.2	Test comp2 (Description sheet)	218
8	CONCRETE	223
8.1	Test beton (Description sheet)	224
8.2	Test precont1 (Description sheet)	228
8.3	Test precont2 (Description sheet)	233
8.4	Test precont3 (Description sheet)	237
9	POROSITY	241
9.1	Test pore1 (Description sheet)	242
9.2	Test pore2 (Description sheet)	246
9.3	Test pore3 (Description sheet)	252
10	METALLURGY OF POWDERS	257
10.1	Test poudre1 (Description sheet)	258
11	JOINT	263
11.1	Test joi22 (Description sheet)	264
11.2	Test joi41 (Description sheet)	269
11.3	Test joi44 (Description sheet)	273
11.4	Test joi45 (Description sheet)	277
12	DYNAMICS	281
12.1	Test dyna6 (Description sheet)	282
12.2	Test dyna7 (Description sheet)	286
12.3	Test fsi4 (Description sheet)	292
12.4	Test dyna5 (Description sheet)	296
12.5	Test vibr6 (Description sheet)	303
12.6	Test vibr10 (Description sheet)	312
12.7	Test dyna12 (Description sheet)	315



12.8 Test dyna11 (Description sheet)	319
12.9 Test dy_devo5 (Description sheet)	324
12.10 Test dy_devo6 (Description sheet)	329
12.11 Test dy_devo3 (Description sheet)	335
12.12 Test dy_devo4 (Description sheet)	341
12.13 Test dy_devo8 (Description sheet)	347
12.14 Test dyne01 (Description sheet)	353
12.15 Test vibr2 (Description sheet)	358
13 NON LINEAR DYNAMICS	363
13.1 Test dyna_nl1 (Description sheet)	364
13.2 Test dyna_nl3 (Description sheet)	369
14 THERMAL ANALYSIS	375
14.1 Test ther2 (Description sheet)	376
14.2 Test ther7or (Description sheet)	383
14.3 Test ther-perm (Description sheet)	388
15 THERMO-MECHANICS	393
15.1 Test thme1 (Description sheet)	394
15.2 Test thpl1 (Description sheet)	398
16 MAGNETO-STATIC	403
16.1 Test Ac2d93 (Description sheet)	404
16.2 Test Ac3d93 (Description sheet)	408



Chapter 1

ELASTICITY

1.1 Test elas1 (Description sheet)

Test name: elas1

Calculation type: MECHANICS ELASTICITY 2D AXI

Finite element type: COQ2


Topic: A spherical skullcap subjected to different types of load. The spherical skullcap is subjected to an internal pressure, to its own weight and to a radial load. To solve this problem, we use the principle of superposition. The equivalent loads for each load were calculated and then added before the resolution of the system.

Goal: Find the radial displacement at a given point B (1,1).

Reference CASTEM: Test NAFEMS : ME1 : Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



$$\begin{aligned} E &= 2.1 \cdot 10^{11} \text{ Pa} \\ \text{RHO} &= 7.85 \cdot 10^4 \text{ g/m}^3 \\ \text{NU} &= 0.28 \end{aligned}$$




Test elas1 (Results)

RESULTS

	U_r
Reference	$4.677 \cdot 10^{-6} m$
Castem	$4.6685 \cdot 10^{-6} m$

CASTEM FIGURES



A SPHERIC SKULLCAP SOUBMISSIVE TO DIFFERENT CHARGE TYPE

Test elas1 (Listing CASTEM)

```

*****
TITR 'A SPHERIC SKULLCAP SUBJECTED      *
      TO DIFFERENT TYPES OF LOADING';
*****
*
OPTION DIME 2 ELEM SEG2 MODE AXIS ;
OPTI ECHO 1;
*
*----- GEOMETRY -----
*
*
A = 0 0 ;
B = 1 1 ;
O = 0 1 ;
*
L1 = CERC 100 A O B ;
*
TRACE QUAL L1 ;
*
*----- BOUNDARY CONDITIONS -----
*
CL1 = BLOQ B UZ ;
CL2 = BLOQ A UR ;
CL = CL1 ET CL2 ;
*
* DESCRIPTION OF MODEL MATERIAL -----
*
MO = MODE L1 MECANIQUE ELASTIQUE COQ2 ;

MA = MATE MO YOUN 2.1E11 NU 0.3 RHO 7.85E4 ;
CA = CARA MO EPAI 0.02 ;
MA = MA ET CA;
*
*- CALCULATION OF THE RIGIDITY MATRIX ----
*
RI1 = RIGI MO MA ;
RI2 = RI1 ET CL ;
*
*----- LOADS DEFINITION -----
*
*-- A UNIFORMALY INTERNAL PRESSURE ----
*
FO1 = PRES COQU MO -10000 NORM ;
*
*----- ITS OWN WEIGHT -----
*
MAS = MASSE MO MA;
EL1 = CHANGE L1 POI1 ;
POP = MANU CHPO EL1 1 UZ -1 ;
FO2 = MAS*POP ;
*
*-----A RADIAL LOAD -----
*
F=1000*2*PI;
FO3 = FORC FR F B ;
*
FO = FO1 + FO2 + FO3 ;
*
*-----CALCULATION AND OUTPUT -----
*
RE = RESO RI2 FO ;
DR = EXTR RE UR B ;
DR = 1000000*DR ;
SAUT PAGE ;
SAUT 2 LIGN ;
MESS ' RADIAL CALCULATED DISPLACEMENT ' ;
MESS 'UR=4.677 MICRON UR =' DR 'MICRON';
TEMPS ;
*----- GOOD WORKING MESSAGE -----

ERR=100*(ABS(4.677 -DR)/4.677);
SI (ERR < 5);
ERRE 0;
SINON;
ERRE 5;
FINSI;
FIN;

```



Test elas1 (Comments)

1. Stiffness and Mass Matrix

RI1 = RIGI MO MA ;
RI2 = RI1 ET CL ;
MAS = MASSE MO MA;

The operators RIGI and MASS require the mechanical model (here MO) and the material and geometrical characteristics (here MA). For the shell elements it is necessary to define the thickness (EPAI)

2. Pressure

FO1 = PRES COQU MO -10000 NORM ;

The operator PRES(SION) calculates the nodal forces equivalent to a pressure. The keyword COQU indicates that the pressure is applied on shell elements.

The operator requires the model (MO) and the value of the pressure. The keyword NORM(AL) indicates that the pressure is positive when it is carried by a vector perpendicular to the element, oriented positively.

3. Own weight

MAS = MASSE MO MA;
EL1 = CHANGE L1 POI1 ;
POP = MANU CHPO EL1 1 UZ -1 ;
FO2 = MAS*POP ;

There are no operators to calculate the own weight. The solution is to multiply the mass matrix (MAS) by an object of CHPOINT type (POP). The POP object defines the value of gravity (direction UZ and value -1).

4. Force

F=1000*2*PI;
FO3 = FORC FR F B ;

The operator FORCE calculates a force field which is given as a localized force. Here the force is defined by its component FR, its value F and the geometry on which it is applied (point B).

5. Addition of the three loads

FO = FO1 + FO2 + FO3 ;

The three loads are objects of CHPOINT type. It is also possible to add them in order to get the complete load.

1.2 Test elas2 (Description sheet)

Test name: elas2

Calculation type: MECHANICS ELASTICITY 3D

Finite element type: DKT


Topic: A cylindrical solid embedded, subjected to a diametrical load. The cylindrical structure is pinched at one end while the other one is embedded. For symmetry reasons, just 1/4 of this structure is modelled. The boundary conditions should reproduce the symmetry. A diametrical load is applied.

Goal: Find the displacement on X at the point P1 (7,0,0) and the displacement on Z at point P3 (0,70,0) where the diametrical force is applied.

Reference CASTEM: Test NAFEMS : ME6 : Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



$$\begin{aligned}
 E &= 3 \cdot 10^7 \text{ Pa} \\
 \text{RHO} &= 7.85 \cdot 10^4 \text{ g/m}^3 \\
 \text{NU} &= 0.3
 \end{aligned}$$




Test elas2 (Results)

RESULTS

	$U_x (P_1)$	$U_z (P_3)$
Reference	0 cm	-1.14 Cm
Castem	$-2.1247 \cdot 10^{-19}$ Cm	-1.1394 cm

CASTEM FIGURES



GIBI FECIT

Test elas2 (Listing CASTEM)

```

*****
*
*TITRE 'CYLINDER AFFECTED'
*      EMBEDDED IN ONE BOARD'
*
*****
OPTION ECHO 1 DIME 3 ELEM TRI3 ;
TEMPS ;
*-----
* GEOMETRY AND MESHING
*-----
P1=7. 0. 0. ;
P2=0. 0. 7. ;
C1=0. 0. 0. ;
VECT=0. 70. 0. ;
LIG1=C 6 P1 C1 P2 ;
SURF=LIG1 TRAN 30 VECT ;
LIG1 LIG2 LIG3 LIG4=COTE SURF ;
P3=LIG2 POIN FINA ;
*
OEIL1 = 100. 100. 100. ;
TRAC QUAL OEIL1 SURF ;
*-----
* MODEL AND MATERIAL CHARACTERISTICS
*-----
MOD1=MODE SURF MECANIQUE ELASTIQUE DKT;
MAT1=MATE MOD1 YOUN 3.E7 NU 0.3 ;
EPEE=CARA MOD1 EPAI 0.1 ;
MAT1=MAT1 ET EPEE;
*-----
*** RIGIDITY AND MASS;
*-----
*
RIG1=RIGI MOD1 MAT1;
*
*-----
* BOUNDARY CONDITIONS
*-----
ENC1=BLOQ LIG4 UZ ;
ENC2=BLOQ LIG4 RX ;
ENC3=BLOQ LIG4 RY ;
ENC4=BLOQ LIG2 UX ;
ENC5=BLOQ LIG2 RY ;
ENC6=BLOQ LIG2 RZ ;
ENC7=BLOQ LIG3 UY ;
ENC8=BLOQ LIG3 RX ;
ENC9=BLOQ LIG3 RZ ;
ENC10=BLOQ LIG1 DEPL ROTA ;
ENC11=ENC1 ET ENC2 ET ENC3 ET ENC4
      ET ENC5 ET ENC6 ET ENC7 ET ENC8;
ET ENC9 ET ENC10 ;
*-----
* LOADS
*-----
F1=0 0 -2500 ;
EFOR=FORC F1 P3 ;
RIG2=RIG1 ET ENC11;
*-----
* RESOLUTION AND RESULTS
*-----
DE1=RESOU RIG2 EFOR;
D1X = EXTR DE1 UX P1 ;
D3Z = EXTR DE1 UZ P3 ;
SAUT PAGE ;
MESS ' THEORETICAL X DISPLACEMENT ON P1 : 0.00' ;
MESS ' COMPUTED X DISPLACEMENT ON P1 : ' D1X ;
SAUT 2 LIGNE ;
MESS ' THEORETICAL Z DISPLACEMENT ON P3 : -1.140' ;
MESS ' COMPUTED Z DISPLACEMENT ON P3:' D3Z ;
TEMPS ;
*      GOOD WORKING MESSAGE
DZREF=-1.14;
RESI=ABS((D3Z-DZREF)/DZREF);
SI(RESI <EG 1E-2);
      ERRE 0;
SINO;
      ERRE 5;
FINSI;
FIN;
    
```




Test elas2 (Comments)

1. Resolution

DE1 = RESOU RIG2 EFOR;

The operator RESO(UDRE) constructs the solution of the linear system ($Ku=F$) with K equal to RIG2,

EFOR equal to F and DE1 equal to u.

2. Results

D1X = EXTR DE1 UX P1 ;

D3Z = EXTR DE1 UZ P3 ;

The operator EXTR(AIRE) allows to extract a value of one component of an object. Here D1X contains the value of the UX displacement at point P1 and D3Z contains the value of the UZ displacement at point P3.

1.3 Test elas3 (Description sheet)

Test name: elas3

Calculation type: MECHANICS ELASTICITY 3D

Finite element type: TRI3


Topic: The problem is similar to elas2, but the cylinder is free and not embedded. The boundary conditions must reproduce the symmetrical geometry. Just 1/4 of this structure is modelled. A diametrical load is applied.

Goal: Find the bending in the direction Z at a point P3 (0,70,0), where the diametrical load is applied.

Reference CASTEM: Test NAFEMS : ME7 : Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



$$\begin{aligned}
 E &= 3 \cdot 10^7 \text{ Pa} \\
 \text{RHO} &= 7.85 \cdot 10^4 \text{ g/m}^3 \\
 \text{NU} &= 0.3
 \end{aligned}$$




1.3. TEST ELAS3 (DESCRIPTION SHEET)

Test elas3 (Results)


RESULTS

	$U_z (P_3)$
Reference	-0.452 cm
Castem	-0.45214 cm

CASTEM FIGURES



CYLINDER FREE PINCHED AT BOARD



CYLINDER FREE PINCHED AT BOARD

VAL - ISO
>-4.52E-01
< 4.25E-19
A -.45
D -.43
G -.41
J -.39
M -.36
P -.34
S -.32
V -.30
Y -.28
b -.26
= -.24
h -.22
k -.19
n -.17
q -.15
t -.13
w -.11
z -8.83E-02
2 -6.71E-02
5 -4.59E-02
8 -2.47E-02
@ -3.53E-03

AMPLITUDE
DEFORMEE
1.1

Test elas3 (Listing CASTEM)

```

TITR 'CYLINDER FREE PINCHED AT BORDER' ;
OPTION ECHO 1 ;
OPTION DIME 3 ELEM TRI3 ;
DENS 0.75 ;
TEMPS ;
*-----
* GEOMETRY LINES AND SURFACES
*-----
P1=4.953 0 0 ;P2=0 0 4.953 ;C1=0 0 0 ;
VECT=0 5.175 0 ;
LIG1=C P1 C1 P2 ;SURF=LIG1 TRAN VECT;
LIG1 LIG2 LIG3 LIG4=COTE SURF ;
P3=LIG2 POIN FINA ;
*-----
* DRAWINGS
*-----
OEIL1 = 100 100 100;
TRAC QUAL OEIL1 SURF;
*-----
* DESCRIPTION OF MODEL AND MATERIAL
* CHARACTERISTICS
*-----
MOD1=MODEL SURF MECANIQUE ELASTIQUE DKT;
MAT1=MATER MOD1 YOUN 10.5E6 NU 0.3125
RHO 1. EPAI 0.094 ;
*-----
* RIGIDITY AND MASS;
*-----
*
RIG1=RIGI MOD1 MAT1;
*-----
* BOUNDARY CONDITIONS
*-----
ENC1=BLOQ LIG4 UZ ;
ENC2=BLOQ LIG4 RX ;
ENC3=BLOQ LIG4 RY ;
ENC4=BLOQ LIG2 UX ;
ENC5=BLOQ LIG2 RY ;
ENC6=BLOQ LIG2 RZ ;
ENC7=BLOQ LIG3 UY ;
ENC8=BLOQ LIG3 RX ;
ENC9=BLOQ LIG3 RZ ;
ENC10=ENC1 ET ENC2 ET
ENC3 ET ENC4 ET ENC5
ET ENC6 ET ENC7 ET ENC8 ET ENC9 ;
*-----
* LOADS
*-----
F1=0 0 -100 ;
EFOR=FORC F1 P3 ;
RIG2=RIG1 ET ENC10;
*-----
* RESOLUTION
*-----
DE1=RESOU RIG2 EFOR;
D3Z = EXTR DE1 UZ P3;
DZ = EXCO UZ DE1;
*-----
* RESULTS (ISOVALUES DISPLACEMENT ON Z)
*-----
SAUT PAGE ;
SURF1 = DEFO SURF DE1;
OPTI ISOV LIGNE;
TRAC OEIL1 DZ SURF1 ;
*
MESS 'THEORETICAL DISPLACEMENT ON Z':'-0.452';
MESS 'CALCULATED DISPLACEMENT ON Z':'D3Z' ;
TEMPS ;
*
DZREF=-0.452;
RESI=ABS((D3Z-DZREF)/DZREF);
SI(RESI <EG 5E-2);
ERRE 0;
SINO;
ERRE 5;
FINSI;
FIN;
    
```



Test elas3 (Comments)

1. Creation of a new field

DZ = EXCO UZ DE1;

The operator EXCO creates, from the field DE1 a new field of the same type by extracting one or more components. Here, the new field DZ is made up of the UZ component of DE1.

2. Deformed object

SURF1 = DEFO SURF DE1;

The operator DEFO(RME) constructs an object of type DEFORME from an initial geometry (SURF) and a displacement field (DE1).

3. Isovalues

OPTI ISOV LIGNE;

TRAC OEIL1 DZ SURF1 ;

The directive OPTI followed by ISOV LIGNE specifies that the operator TRAC will draw the isovalues with lines. The operator TRAC represents the field DZ on the deformed object SURF1 using lines of isovalues.

1.4 Test elas4 (Description sheet)

Test name: elas4

Calculation type: MECHANICS ELASTICITY 3D

Finite element type: COQ4


Topic: A cantilever beam with cross section is subject to a torsion torque. Consider a simple cantilever beam with a cross section. This structure is subjected to a torsion torque. The value of the torsion torque is 1200mN, obtained from two loads of 600N each. The loads are uniformly applied on each side.

Goal: Find the axial stress at point A(2.5, 0, 0). The analytical value is $S_{mxx} = -108$ MPa.

Reference CASTEM: Test NAFEMS : ME8 : Modélisation des structures élastique dans CASTEM 2000.

Version: 97' customer version

Model description:



$$E = 210 \cdot 10^9 \text{ Pa}$$

$$EPAI = 0.1 \text{ m}$$

$$NU = 0.3$$




Test elas4 (Results)


RESULTS

	$S_{mxx} (A(2.5, 0, 0))$
Reference	-108 MPa
Castem	-112.73 MPa

CASTEM figures



TEST ELAS4 : MAILLAGE



ELAS4 : STRAIN, LOADS, REACTIONS

Test elas4 (Listing CASTEM)

```

TITR 'CANTILEVER SUBJECTED TO TORSTION TORQUE';
OPTI ECHO 1;
OPTI DIME 3;
OPTI ELEM QUA4;
*
*=====
*           GEOMETRY AND MESCHING           *
*=====
*
*POINTS
*
OEIL = 1000 -1000 1000;
P1 = 0 0 0; P21 = (P1 PLUS (10 0 0));
P2 = 0 0 1; P22 = (P2 PLUS (10 0 0));
P3 = 0 2 1; P23 = (P3 PLUS (10 0 0));
P4 = 0 2 2; P24 = (P4 PLUS (10 0 0));
*
P1M = 10 0 .5; P2M = 10 2 1.5;
*
* LINES
N1 = 2;
LIG1 = P1 D N1 P2 D N1 P3 D N1 P4;
*
* MESHING
N2 = 16;
SUR1 = LIG1 TRANS N2 (10 0 0);
*
ELIM(SUR1 ET P21 ET P22 ET P23 ET
      P24 ET P1M ET P2M) 0.001;
*
TITR ' TEST elas4 : MAILLAGE ';
TRAC OEIL QUAL SUR1;
*
*=====
*           MODEL AND MATERIAL CHARACTERISTICS           *
*           RIGIDITY AND BOUNDARY CONDITIONS           *
*=====
*
MODL1 = MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE COQ4;
MATR1 = MATE MODL1 YOUN 210E9 NU 0.3 EPAI 0.1;
RI1 = RIGI MATR1 MODL1;
CL1 = BLOQ DEPL LIG1;
RIG1 = RI1 ET CL1;
*
*=====
*           CHARGEMENT           *
*=====
*
F1 = 0 0 -150E3;
F1M = 0 0 -300E3;
F2 = 0 0 150E3;
F2M = 0 0 300E3;
*
FORC1 = FORC F1 P21;
FORC2 = FORC F1 P22;
FORC1M = FORC F1M P1M;
FORC3 = FORC F2 P23;
FORC4 = FORC F2 P24;
FORC2M = FORC F2M P2M;
*
CHA1 = FORC1 ET FORC2 ET FORC3 ET
      FORC4 ET FORC1M ET FORC2M;
*
*=====
* RESOLUTION : DISPLACEMENT AND STRESS *
*=====
*
DEP1 = RESO RIG1 CHA1;
*
*
* DRAWINGS THE STRAINS AND THE REACTIONS
AMPVEC = .5E-6;
REAL = REAC DEP1 RIG1;
VECT1 = VECT CHA1 (AMPVEC * 20) FX FY FZ ROUGE;
VECT2 = VECT REAL AMPVEC FX FY FZ BLEU;
DEF0 = DEFO 0. DEP1 SUR1;
DEF1 = DEFO 5. DEP1 SUR1 (VECT1 ET VECT2) VERT;
TITR ' elas4 : STRAIN, LOADS, REACTIONS';
TRAC OEIL SUR1 (DEF0 ET DEF1);
*
*=====
* ONE EXTRACTING STRESS AND COMPARISON *
* WITH THEORETICAL SOLUTION *
*=====
*
CHAM1 = SIGM MODL1 MATR1 DEP1;
*
CHAM2 = CALP CHAM1 MATR1 MODL1 MOYEN;
CHAM3 = CHAN NOEUD CHAM2 MODL1;
CHPO3 = CHAN CHPO CHAM3 MODL1;
*
PA = SUR1 POIN PROC (2.5 0 0);
SMXXA2 = EXTR CHPO3 SMYY PA;
SMXXA2 = SMXXA2 / 1.E6;
SMXXA1 = -108;
ERGXXA = 100*(ABS((SMXXA2-SMXXA1)/SMXXA1));
*
*=====
*           RESULTS           *
*=====
*
MESS ' RESULTS TEST elas4 ';
MESS ' ----- ';
SAUT 2 LIGN;
MESS 'WE FIND THE AXIAL STRESS ON THE MOYEN SURFACE';
MESS 'AT THE POINT A';
MESS 'IT IS COMPARED TO THE THEORETICAL VALUE ';
SAUT 1 LIGN;
MESS 'MEC THEORETICAL AXIAL STRESS AT A : ' SMXXA1 'MPa';
MESS 'COMPUTED AXIAL STRESS AT A : ' SMXXA2 'MPa';
MESS 'ERR : ' ERGXXA ' % ';
SAUT 1 LIGN;
*
ELSUR1 = NBEL SUR1;
NOSUR1 = NBNO SUR1;
*
MESS ' FINITE ELEMENT NUMBER : ' ELSUR1;
MESS ' NODES NUMBER : ' NOSUR1;
*
*=====
*           GOOD WORKING MESSAGE           *
*=====
*
SAUT 2 LIGNE;
SI (ERGXXA <EG 5);
ERRE 0;
SINON;
ERRE 5;
FINSI;
*
SAUT 1 LIGN;
TIME;
SAUT 1 LIGN;
*
FIN;

```




Test elas4 (Comments)

1. Strains and reactions

```
AMPVEC = .5E-6;
REA1 = REAC DEP1 RIG1;
VECT1 = VECT CHA1 (AMPVEC * 20) FX FY FZ ROUGE;
VECT2 = VECT REA1 AMPVEC FX FY FZ BLEU;
DEFO = DEFO 0. DEP1 SUR1;
DEF1 = DEFO 5. DEP1 SUR1 (VECT1 ET VECT2) VERT;
TITR ' elas4 : STRAIN, LOADS, REACTIONS';
TRAC OEIL SUR1 (DEFO ET DEF1);
```

The operator REAC(TION) constructs from the solution DEP1 the variation of the second member of the linear system relative to RIG1.

The operator VECT(EUR) constructs an object of VECTEUR type (VECT1 and VECT2) from the components of a field (CHA1 and REA1). The amplification coefficient must be given along with the names of the components (FX, FY, FZ).

The operator DEFO can be used by associating an object of VECTEUR type (VECT1 ET VECT2) with the object of DEFORME type. The result will represent the field of reactions by means of vectors in blue and the field of loads in red on the deformed shape drawing (in green).

2. Stresses

```
CHAM1 = SIGM MODL1 MATR1 DEP1;
CHAM2 = CALP CHAM1 MATR1 MODL1 MOYEN;
CHAM3 = CHAN NOEUD CHAM2 MODL1;
CHPO3 = CHAN CHPO CHAM3 MODL1 ;
PA = SUR1 POIN PROC (2.5 0 0);
SMXXA2 = EXTR CHPO3 SMYY PA;
```

The operator SIGMA calculates a stress field (CHAM1) from a displacement field (DEP1). In the case of thin shell elements it is a matter of generalized stresses which are calculated in the local basis.

The operator CALP (CALcul en Peau) calculates a stress field in terms of continuum mechanics starting from a generalized stress field (CHAM1). In the case of shell elements the calculation is carried out at points obtained by projecting supporting points either onto the shell mid-plane (option chosen here with keyword MOYEN) or onto the top or bottom layer.

The operator CHAN(GER) constructs from a field by elements (CHAM2) and from a model object (MODL1) a new field by elements (CHAM3) of the same kind as CHAM2 but born by the nodes (option taken here with keyword NOEUD).

The operator CHAN has different functions, the keyword CHPO loads it to construct from a field by elements and a model object a new field by points (CHPO3).

Note that the names of the components of the first stress field (CHAM1) are different from the names of components of the other stress fields.

3. Nodes and elements number

```
ELSUR1 = NBEL SUR1;
NOSUR1 = NBNO SUR1;
```

The operator NBEL supplies the number of elements contained in the geometry SUR1. In the same way, the operator NBNO supplies the number of nodes contained in the geometry SUR1.



1.5 Test elas7 (Description sheet)

Test name: elas7

Calculation type: MECHANICS ELASTICITY 2D AXI

Finite element type: COQ2


Topic: solid : cylinder-sphere-cylinder subjected to an internal pressure. Consider a solid structure composed by one cylinder surmounted by one sphere and surmounted by one cylinder. This structure is subjected to an internal pressure ($P = 1 \text{ MPa}$). On grounds of symmetry (load and geometry), we solve this problem in axisymmetrical model.

Goal: Find the axial stress at a point C(0.70, 1). The analytical value is $S_{mzz} = -319.9 \text{ MPa}$.

Reference CASTEM: Test NAFEMS : ME11 : Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer releas



Model description:



$$\begin{aligned}
 E &= 210 \cdot 10^9 \text{ Pa} \\
 \text{EPAI} &= 0.01 \text{ m} \\
 \text{NU} &= 0.3
 \end{aligned}$$

Test elas7 (Results)
RESULTS

	$Sm_{zz}(C(0.70,1))$
Reference	-319.9 MPa
Castem	-312.29 MPa

CASTEM figures




1.5. TEST ELAS7 (DESCRIPTION SHEET)

Test elas7 (Listing CASTEM)

```
OPTI ECHO 1; OPTI DIME 2; OPTI MODE AXIS;
OPTI ELEM SEG2;
*
*=====*
*   GEOMETRY AND MESHING   *
*=====*
*
* POINTS
COS75 = COS 7.5; COS375 = COS 37.5;
COS45 = COS 45; SIN75 = SIN 7.5;
SIN375 = SIN 37.5; SIN45 = SIN 45;
R = 2 ** -.5;
RA = R; ZA = 0.; PA = RA ZA;
RB = 0; ZB = COS45; PB = RB ZB;
RC = R; ZC = 1.; PC = RC ZC;
RD = R; ZD = 2.; PD = RD ZD;
RO = 0; ZO = 1 + COS45; PO=RO ZO;
PA1 = PC PLUS (0 -0.125);
PD1 = PC PLUS (0 0.125);
PB1 = PB PLUS (SIN375 (1-COS375));
* DROITS, CERCULES AND LINES
K = 1; SI (K EGA 1); TITR ' LE9 : COARSE MESHING';
SINON; TITR ' LE9 : FINE MESHING'; FINSI;
N1 = 4 * K; AA1 = PA DROI N1 PA1;
A1C = PA1 DROI N1 PC; CD1 = PC DROI N1 PD1;
D1D = PD1 DROI N1 PD; B1C = PB1 CERC N1 PO PC;
N2 = 5 * K; BB1 = PB CERC N2 PO PB1;
*
SUR1 = BB1 ET B1C; SUR2 = CD1 ET D1D;
SUR3 = AA1 ET A1C; SURF1 = SUR1 ET SUR2 ET SUR3;
TITR 'elas7 : MESHING'; TRAC QUAL SURF1;
ELSURF1 = NBEL SURF1; NOSURF1 = NBNO SURF1;
*
*=====*
*   MODEL AND MATERIAL CHARACTERISTICS   *
*   RIGIDITY AND BOUNDARY CONDITIONS     *
*=====*
*
MODL1 = MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE COQ2;
MODL2 = MODE SUR2 MECANIQUE ELASTIQUE ISOTROPE COQ2;
MODL3 = MODE SUR3 MECANIQUE ELASTIQUE ISOTROPE COQ2;
MATR1 = MATE MODL1 YOUN 210E9 NU 0.3 EPAI 0.01;
MATR2 = MATE MODL2 YOUN 210E9 NU 0.3 EPAI 0.01;
MATR3 = MATE MODL3 YOUN 210E9 NU 0.3 EPAI 0.01;
RI1 = RIGI MODL1 MATR1;
RI2 = RIGI MODL2 MATR2;
RI3 = RIGI MODL3 MATR3;
*
* BOUNDARY CONDITIONS
*
CL1 = BLOQ ROTA DEPL PA;
RIG1 = RI1 ET RI2 ET RI3 ET CL1;
*
*=====*
*   LOADING   *
*=====*
*
* UNIFORM INTERNAL PRESSURE
CH1 = PRES COQU MODL1 -1.E6 NORM;
CH2 = PRES COQU MODL2 -1.E6 NORM;
CHA1 = CH1 ET CH2;
*
*=====*
*   RESOLUTION : FIELD OF DISPLACEMENT   *
*=====*
*
DEP1 = RESO RIG1 CHA1;

*DRAWINGS THE STRAINS AND THE REACTIONS
AMPVEC = .4E-6;
REAL = REAC DEP1 RIG1;
VECT1 = VECT CHA1 (AMPVEC * 20) FR FZ ROUGE;
VECT2 = VECT REAL AMPVEC FR FZ BLEU;
DEF0 = DEFO 0. DEP1 SURF1;
DEF1 = DEFO 40. DEP1 SURF1 (VECT1 ET VECT2) VERT;
TITR 'elas7 : DEFORMEE, CHARGEMENT, REACTIONS';
TRAC SURF1 (DEF0 ET DEF1);
*
*=====*
*   ONE EXTRACTING STRESS AND COMPARISON   *
*   WITH THEORETICAL SOLUTION             *
*=====*
*
DEP2 = REDU DEP1 SUR2;
CHAM1 = SIGM MODL2 MATR2 DEP2;
CHAM2 = CALP CHAM1 MATR2 MODL2 INFE;
*
* INTERPOLATION OF THE STRESS AT THE MESHING KNOTS
CHAM3 = CHAN NOEUD MODL2 CHAM2;
CHPO3 = CHAN CHPO MODL2 CHAM3;
SMZZC = EXTR CHPO3 SMXX PC;
SMZZC = SMZZC / 1.E6;
SMZZC1 = -319.9;
* CALCULATED MARGIN WITH THE REFERRING SOLUTION.
ERGZZC=100*(ABS((SMZZC - SMZZC1) / SMZZC1));
*
*=====*
*   RESULTS   *
*=====*
*
MESS ' RESULTS : TEST elas7 ';
MESS ' ----- ';
SAUT 2 LIGN;
MESS 'THE FINDING VALUE IS THE AXIAL STRESS AT C' ;
MESS
' IT IS COMPARED AT THE ANALYTICAL VALUE.';
SAUT 1 LIGN;
MESS 'ANALYTICAL AXIAL STRESS AT C:'SMZZC1'MPa';
SAUT 1 LIGN;
MESS 'COMPUTED AXIAL STRESS AT C : 'SMZZC'MPa';
SAUT 1 LIGN;
MESS ' WHICH GIVES AN ERROR OF : 'ERGZZC '%';
SAUT 2 LIGN;
SI (K EGA 1); MESS ' COARSE MESHING';
SINON; MESS ' FINE MESHING'; FINSI;
MESS ' FINITE ELEMENTS NUMBER : ' ELSURF1;
MESS ' NODES NUMBER : ' NOSURF1;
MESS ' THEREFORE ' (NOSURF1 * 3) 'd.d.l.';
*
*=====*
*   GOOD MESSAGE WORKING   *
*=====*
*
SAUT 2 LIGN;
SI (ERGZZC <EG 5); ERRE 0;
SINON; ERRE 5; FINSI;
*
*=====*
*   RUNNING CALCULATION TIME AND END   *
*=====*
*
SAUT 1 LIGN;
TIME;
SAUT 1 LIGN;
FIN;
```

1.6 Test elas8 (Description sheet)

Test name: elas8

Calculation type: MECHANICS ELASTICITY 2D PLANE STRESS

Finite element type: QUA8


Topic: solid : ELLIPTICAL MEMBRANE SUBJECTED TO LINEAR LOAD. The model is an elliptical membrane, obtained by a projection of one arc of circle onto one plane. This structure is subjected to linear load in one sides. On grounds of symmetry, we solve this problem in axisymmetrical mode.

Goal: Find the stress at a point D(2, 0; 0). The analytical value is $S_{my} = 92.7$ MPa.

Reference CASTEM: Test NAFEMS : ME12 : Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer releas

Model description:



$$\begin{aligned}
 E &= 210 \cdot 10^9 \text{ Pa} \\
 \nu &= 0.3
 \end{aligned}$$




Test elas8 (Results)


RESULTS

	$S_{myy} (D(2,0,0))$
Reference	92,7 MPa
Castem	93,548 MPa

CASTEM figures



ELAS8 : MESHING



AMPLITUDE

0.00E+00

1.50E+02

COMPOSANTES
VECTEURS

FX FY

FX FY

ELAS8 : DEFORMEE, CHARGEMENT, REACTIONS

Test elas8 (Listing CASTEM)

```

* CAS TEST DU 91/06/13 PROVENANCE : TEST
SAUT PAGE;
*
TITR 'ELLIPTICAL MEMBRANE SUBJECTED
      TO LINEAR LOAD';
*
OPTI ECHO 1;
*
OPTI DIME 3;
OPTI ELEM QUA8;
*
*=====
*   GEOMETRY AND MESHING   *
*=====
*
* POINTS; LINES, CERCULES
*
PB = 0. 2.75 0.;
PB1 = 0. 2.75 (((3.25 ** 2)
              -(2.75 ** 2)) ** 0.5);
PC = 3.25 0. 0.;
*
C1 = PC CERC 6 (0. 0. 0.) PB1;
C2 = C1 PROJ CYLI (0. 0. 1)
      PLAN (0 0 0) (1 0 0) (0 1 0);
*
PA = 0. 1. 0.;
PA1 = 0. 1. (((2.**2)-(1.** 2))** 0.5);
PD = 2. 0. 0.;
*
C3 = PD CERC 6 (0. 0. 0.) PA1;
C4 = C3 PROJ CYLI (0. 0. 1)
      PLAN (0 0 0) (1 0 0) (0 1 0);
*
D1 = PA DROI 2 (0. 1.583 0.) DROI 2 PB;
D3 = PC DROI 2 (2.417 0. 0.) DROI 2 PD;
*
ELIM (D1 ET C2 ET D3 ET C4) 0.0001;
SUR1 = DALL D1 C2 D3 C4 PLAN;
*
TITR 'elas8 : MESHING'; TRAC (0 0 1000) QUAL SUR1;
*
OPTI DIME 2;
OPTI MODE PLAN CONT;
*
*=====
*   MODEL AND MATERIAL CHARACTERISTICS   *
*   RIGIDITY AND BOUNDARY CONDITIONS     *
*=====
*
MODL1=MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE QUA8;
MATR1=MATE MODL1 YOUN 210E9 NU 0.3;
RI1 = RIGI MODL1 MATR1;
*
*BOUNDARY CONDITIONS
*
CL1 = SYMT DEPL ROTA (0 0) (0 1) SUR1 0.0001;
CL2 = SYMT DEPL ROTA (0 0) (1 0) SUR1 0.0001;
*
RIG1 = RI1 ET CL1 ET CL2;
*
*=====
*   LOADING   *
*=====
*
*   UNIFORMLY INTERNAL PRESSURE
CHA1 = PRES MASS MODL1 -10.E6 C2;
*
*=====
*RESOLUTION : FIELD OF DISPLACEMENT *
*=====
*
DEP1 = RESO RIG1 CHA1;
*
* DRAWINGS THE STRAINS AND THE REACTIONS
AMPVEC = 1.E-7;
REAL = REAC DEP1 RIG1;
VECT1 = VECT CHA1 AMPVEC FX FY ROUGE;
VECT2 = VECT REAL AMPVEC FX FY BLEU;
DEFO = DEFO 0. DEP1 SUR1;
DEF1 = DEFO 150. DEP1 SUR1 (VECT1 ET VECT2) VERT;
TITR 'elas8 : DEFORMEE, CHARGEMENT, REACTIONS';
TRAC SUR1 (DEFO ET DEF1);
*
*=====
*   FIELD OF STRESS   *
*=====
*
CHAM1 = SIGM MODL1 MATR1 DEP1;
*INTERPOLATION ON THE MECHING KNOTS
CHAM2 = CHAN NOEUD CHAM1 MODL1;
CHP2 = CHAN CHPO CHAM2 MODL1;
*
*=====
*   ONE EXTRACTING STRESS AND COMPARISON   *
*   WITH THEORETICAL SOLUTION             *
*=====
*
* SOUGHT THE STRESS SMYY AT POINT D.
SMYYD = EXTR CHP2 SMYY PD;SMYYD = SMYYD / 1.E6;
SMYYD1 = 92.7;
ERG = 100 * (ABS ((SMYYD - SMYYD1) / SMYYD1));
*
MESS ' RESULTS ';
MESS ' ----- ';
SAUT 1 LIGN;
MESS ' THEORETICAL STRESS SMYY AT D : ' SMYYD1 'MPA';
MESS ' COMPUTED STRESS SMYY AT D : ' SMYYD 'MPA';
MESS ' WHICH GIVES AN ERROR OF : ' ERG '%';
SAUT 1 LIGN;
*
NBNO SUR1 = NBNO SUR1;
NBEL SUR1 = NBEL SUR1;
MESS ' FINITE ELEMENT NUMBER : 'NBEL SUR1'QUA8';
MESS ' NUMBER OF NODES : ' NBNO SUR1 ;
SAUT 1 LIGN;
*
*=====
*   GOOD MESSAGE WORKING   *
*=====
*
SI (ERG <EG 5.);
ERRE 0;
SINON;
ERRE 5;
FINSI;
*
SAUT 1 LIGN;
TEMPS;
SAUT 1 LIGN;
*
FIN;
    
```




Test elas8 (Comments)

1. Projection

C4 = C3 PROJ CYLI (0. 0. 1) PLAN (0 0 0) (1 0 0) (0 1 0);

The operator PROJ(ETER) constructs the object C4 which results from the projection of the object C3 with respect of the vector (0. 0. 1.). The projection is done onto a plane support defined by 3 points (keyword PLAN followed by 3 points).

2. Boundary conditions of symmetry

CL1 = SYMT DEPL ROTA (0 0) (0 1) SUR1 0.0001;

CL2 = SYMT DEPL ROTA (0 0) (1 0) SUR1 0.0001;

The operator SYMT imposes boundary conditions of symmetry type on the degrees of freedom in rotation (keyword ROTA). The two points define the axis of symmetry, SUR1 is the geometry on which are imposed the conditions and the real 0.0001 is the criterion of selection of the points.

3. Pressure

CHA1 = PRES MASS MODL1 -10.E6 C2;

The operator PRES calculates the nodal forces equivalent to the pressure applied on the model object MODL1. The keyword MASS indicates that the type of the elements are massive (QUA4).

1.7 Test elas11 (Description sheet)

Test name: elas11

Calculation type: MECHANICS ELASTICITY 2D AXI

Finite element type: QUA8

Topic: A solid (cylinder-cone-sphere), subjected to a gradient linear of temperature. The structure is similar to ELAST10, but we solve this problem in plane axisymmetrical mode. The field of temperature is described by:


$$T(R+z) = R+z$$

Goal: Find the stress at a point A(1, 0). The analytical value is $S_{mzz}(A) = 105$ MPa.

Reference CASTEM: Test NAFEMS : LE11 : Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer releas

Model description:



T : Linear Gradient of Temperature
 $T(r,z) = r + z$

$A = (1, 0)$
 $B = (1.4, 0)$
 $D = (1.21, 0.7)$
 $G = (1, 1.39)$
 $I = (1, 1.79)$
 $H = (0.7071, 1.79)$
 $E = (\text{Cos}30, \text{Sin}30)$

$$\begin{aligned}
 E &= 210 \cdot 10^9 \text{ Pa} \\
 \text{NU} &= 0.3
 \end{aligned}$$





Test elas11 (Results)

RESULTS

	$S_{my} (A(1,0))$
Reference	-105 MPa
Castem	-103.34 MPa

CASTEM figures



AMPLITUDE

0.00E+00

1.00E+02

COMPOSANTES
VECTEURS

FR: FZ

Test elas11 (Listing CASTEM)

```

* CAS TEST DU 91/06/13      PROVENANCE : TEST
SAUT PAGE;
*
*
OPTI ECHO 0;
*
OPTI DIME 2;
OPTI MODE AXIS;
OPTI ELEM QUA8;
*
*
*=====*
*      GEOMETRY AND MESHING      *
*=====*
*
* POINTS
*
PO = 0. 0.;
P1 = 0. 1.;
PA = 1. 0.;
PB = 1.4 0.;
PE = (1. * (COS 30)) (1. * (SIN 30));
PD = (1.4 * (COS 30)) 0.7;
PC = (COS 30) (SIN 30);
PE = (COS 45) (SIN 45);
PH = 0.7071 1.79;
PI = 1. 1.79;
PG = 1. 1.39;
*
* DROITS, CIRCLES, LINES
*
* MESHING
*
K = 2;
*
N1 = 1 * K;
AB = PA DROI N1 PB;
GI = PG DROI N1 PI;
EC = PE CERC N1 PO PC;
IH = PI DROI N1 PH;
*
N2 = 2 * K;
BD = PB CERC N2 PO PD;
DG = PD DROI N2 PG;
CA = PC CERC N2 PO PA;
HE = PH DROI N2 PE;
*
* MESHING WITH COARSE FINITE ELEMENT
*
SUR1 = DALL AB (BD ET DG ET GI) IH (HE ET EC ET CA) PLAN;
*
    SI (K EGA 1);
        TITR 'elas11 : COARSE MESHING';
    SINON;
        TITR 'elas11 : FINE MESHING';
    FINSI;
    TRAC QUAL SUR1;
*
ELSUR1 = NBEL SUR1;
NOSUR1 = NBNO SUR1;
*
*=====*
*      MATERIAL AND CHARACTERISTICS AND RIGIDITY      *
*=====*
*
MODE1 = MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE QUA8;
MATE1 = MATE MODE1 YOUN 210E9 NU 0.3 ALPH 2.3E-4;
RI1 = RIGI MODE1 MATE1;
*
*=====*
*      BOUNDARY CONDITIONS      *
*=====*
*
CL1 = BLOQ UZ AB;
CL2 = BLOQ UZ IH;
*
CL11 = CL1 ET CL2;
RIG1 = RI1 ET CL11;
*
*=====*
*      THERMIC LOADING      *
*=====*
*
* CREATION OF THE FIELD OF TEMPERATURE
*
CHPR = COOR 1 SUR1;
CHPZ = COOR 2 SUR1;
*
CHP4 = CHPR + CHPZ;
*
LIS10 = PROG 0. 200.;
LIS20 = PROG 0. 200.;
EVOL1 = EVOL MANU SCAL LIS10 T LIS20;
CHP5 = VARI CHP4 EVOL1 T;
*
* CREATION OF THE EQUIVALENT FIELD OF STRESS
*
CHAM10 = THET MODE1 MATE1 CHP5;
*
* CREATION OF THE EQUIVALENT LOADS FIELD
*
CHAR1 = BSIG MODE1 CHAM10;
*
*=====*
*      RESOLUTION      *
*=====*
*
DEP1 = RESO RIG1 CHAR1;
*
*
    AMPVEC = 1.E-8;
    REAL = REAC DEP1 RIG1;
    VECT1 = VECT REAL AMPVEC FR FZ BLEU;
    DEFO = DEFO 0. DEP1 SUR1;
    DEF1 = DEFO 100. DEP1 SUR1 VECT1 VERT;
    TITR 'elas11 ; DEFORMEE, REACTIONS';
    TRAC SUR1 (DEFO ET DEF1);
*
*=====*
*      ONE EXTRACTING STRESS AND      *
*      COMPARISON WITH THE ANALYTICAL SOLUTION      *
*=====*
*
* SOUGHT THE STRESS SIGZZ AT A
*
* TOTAL STRESS FIELD CALCULATION
*
CHAM1 = SIGM MODE1 MATE1 DEP1;
*
* SOUGHT THE THERMICAL STRESS
*
*
CHAM1 = CHAM1 - CHAM10;
*
    
```



1.7. TEST ELAS11 (DESCRIPTION SHEET)

```
* INTERPOLATION ON THE MESHING KNOTS.
*
CHAM2 = CHAN NOEUD MODE1 CHAM1;
CHPO10 = CHAN CHPO MODE1 CHAM2;
*
*
SMZZA = EXTR CHPO10 SMZZ PA;
SMZZA = SMZZA / 1.E6;
*
SMZZA1 = -105;
*
* DISCARDING CALCULATION VALUE WITH THEORETICAL SOLUTION.
*
ERGZZA = 100 * (ABS ((SMZZA - SMZZA1) / SMZZA1));
*
*=====*
*      RESULTS      *
*=====*
*
MESS ' RESULTS ' ;
MESS ' ----- ' ;
SAUT 1 LIGN;
*
MESS 'THE FINDING VALUE IS AXIAL STRESS AT A, ' ;
MESS
  ' IT IS COMPARED TO THE THEORETICAL VALUE. ' ;
SAUT 1 LIGN;
MESS ' VERTICAL ANALYTICAL STRESS AT A : ' SMZZA1 'MPa' ;
SAUT 1 LIGN;
MESS ' VERTICAL COMPUTED STRESS AT A   : ' SMZZA 'MPa' ;
SAUT 1 LIGN;
MESS 'WITCH YIELD GIVES A ERROR : ' ERGZZA '% ' ;
SAUT 2 LIGN;
*
SI (K EGA 1);
  MESS ' COARSE MESHING ' ;
SINON;
  MESS ' FINE MESHING ' ;
FINSI;
*
MESS '  FINITE ELEMENT NUMBER : ' ELSUR1;
MESS '  NUMBER OF KNOTS   : ' NOSUR1;
MESS '  OR   : ' (NOSUR1 * 3) 'd.d.l.' ;
*
*=====*
*      GOOD WORKING MESSAGE      *
*=====*
*
SAUT 1 LIGNE;
SI (ERGZZA <EG 5); ERRE 0;
SINON; ERRE 5;
FINSI;
*
*=====*
*      RUNNING CALCULATION TIME AND END      *
*=====*
*
SAUT 1 LIGN;
TEMPS;
SAUT 1 LIGN;
*
FIN;
*
```

Test elas11 (Comments)

1. Creation of a variable field

CHP4 = CHPR + CHPZ;

...

EVOL1 = EVOL MANU SCAL LIS10 T LIS20;

CHP5 = VARI CHP4 EVOL1 T;

The operator VARI calculates a variable field from the field CHP4 and the variation law in the form of the function described in EVOL1. T will be the name of the component of the generated field CHP5.

2. Thermal stresses

CHAM10 = THET MODE1 MATE1 CHP5;

CHAR1 = BSIG MODE1 CHAM10;

The operator THET calculates the stresses associated with the temperature field CHP5.

The operator BSIGMA calculates the nodal force field resulting from the integration of the stress field CHAM10. The object CHAR1 represent nodal forces equivalent to the thermic load.



1.8 Test elas13 (Description sheet)

Test name: elas13

Calculation type: MECHANICS ELASTICITY 3D AXI

Finite element type: POUT


Topic: BEAM ON ELASTIC GROUND WITH ARTICULATED EXTRIMITIES. The structure is composed by a simple cantilever beam transported by an elastic ground. This beam is articulated in two extremities.

Goal: Find the Rotation and the Reaction at one extremity, point A(-2.48, 0, 0), and the deflexion at D (0, 0, 0) (center of the beam).

Reference CASTEM: Test VPCS : SSSL16/89 : Modélisation des structures élastiques dans CASTEM 2000.
M. Courtand P. Lebellet Formulaire du béton armé T.2 Paris Eyrolles 1976.

Version: 97' customer releas


Model description:



$$\begin{aligned}
 E &= 2 \cdot 10^{11} \text{ Pa} \\
 \text{NU} &= 0.3 \\
 \text{SECT} &= 1000 \text{ m}^2
 \end{aligned}$$

Test elas13 (Results)
RESULTS

	$Rot_A(Rad)$	$R_A(N)$	$U_Y(D)(m)$	$M_f(D)(N.m)$
Reference	$-3.045E-3$	$1.167E4$	$-4.233E3$	$3.384E4$
Castem	$-3.15250E-3$	11559	$-4.38085E3$	$3.3407E4$

CASTEM figures




1.8. TEST ELAS13 (DESCRIPTION SHEET)

Test elas13 (Listing CASTEM)

```

OPTI ECHO 1; OPTI DIME 3; OPTI ELEM SEG2;
*
*+++++*
* GEOMETRY AND MESHING *
*+++++*
L = PI * (10 ** .5) / 2.;
PA = (-.5 * L) 0. 0.; PB = (0.5 * L) 0. 0.;
SUR1 = PA DROI 8 PB;
PD = SUR1 POIN PROC (0. 0. 0.);
TITR 'elas13 : MESHING';TRAC (0. 0. 1000.) QUAL SUR1;
*+++++*
* MATRIX OF RIGIDITY *
*+++++*
MODL1 = MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE POUT;
MATR1 = MATE MODL1 YOUN 2.E11 NU 0.3 TORS 1000.
          SECT 1000 INRY 1.E-4 INRZ 1.E-4;
RIGI1 = RIGI MODL1 MATR1;
*+++++*
* BOUNDARY CONDITIONS *
*+++++*
CL1 = BLOQ UX (PA ET PB); CL2 = BLOQ UY (PA ET PB);
CL3 = BLOQ UZ SUR1;CL4 = BLOQ RX SUR1;
NEL1 = NBEL SUR1; K = 840000;
KNOEUD = K * L / (2. * NEL1);
I = 0; CL5 = APPUI DEPL 0. SUR1;
REPETER BLOC2 NEL1;
I = I + 1; POI1 = SUR1 POIN I;
POI2 = SUR1 POIN (I + 1);
APP1 = APPUI DEPL KNOEUD POI1;
APP2 = APPUI DEPL KNOEUD POI2;
CL5 = CL5 ET APP1 ET APP2; FIN BLOC2;
RIGI1 = RIGI1 ET CL1 ET CL2 ET CL3 ET CL4 ET CL5;
*+++++*
* MECHANICAL LOADING *
*+++++*
CHA1 = FORC (0. -10000. 0.) PD;
FTOTAL = -5000. * L; FPOINT = FTOTAL / (2. * NEL1);
CHA2 = FORC FY 0. PA; I = 0;
REPETER BLOC1 NEL1; I = I + 1;
POI1 = SUR1 POIN I; POI2 = SUR1 POIN (I + 1);
FOR1 = FORC FY FPOINT POI1;
FOR2 = FORC FY FPOINT POI2;
CHA2 = CHA2 ET FOR1 ET FOR2; FIN BLOC1;
*
CHA3 = (MOMENT MZ -15000 PA) ET (MOMENT MZ 15000 PB);
*
CHAR1 = CHA1 ET CHA2 ET CHA3;
*+++++*
* RESOLUTION : DISPLACEMENT FIELD *
* STRESS FIELD *
*+++++*
DEPL1 = RESO RIGI1 CHAR1;
* DRAWING THE STAINS
REAC1 = REAC RIGI1 DEPL1;
TITR 'elas13 : STRAIN, LOADING, REACTIONS';
VECT1 = VECT CHAR1 3.E-4 FX FY FZ ROUGE;
VECT2 = VECT REAC1 3.E-4 FX FY FZ BLEU;
DEFO = DEFO SUR1 DEPL1 0.;
DEF1 = DEFO SUR1 DEPL1 100. (VECT1 ET VECT2) VERT;
TRAC (0. 0. 1000.) SUR1 (DEFO ET DEF1);
*
SIGM1 = SIGM MODL1 MATR1 DEPL1;
SIGM2 = CHAN NOEUD MODL1 SIGM1;
SIGM3 = CHAN CHPO MODL1 SIGM2;
*+++++*
* RESULTS EXTRACTING *
*+++++*
* ROTATION AT A :
RZA1 = EXTR DEPL1 RZ PA; RZA2 = -3.045E-3;
ERG1 = 100 * (ABS ((RZA2 - RZA1) / RZA2));
*
* REACTION AT A :
VA1 = EXTR REAC1 FY PA; VA2 = 1.1674E4;
ERG2 = 100 * (ABS ((VA2 - VA1) / VA2));
*
* DEFLECTION AT D :
UYD1 = EXTR DEPL1 UY PD; UYD2 = -4.23326E-3;
ERG3 = 100 * (ABS ((UYD2 - UYD1) / UYD2));
*
* MOMENT OF FLEXION AT D :
MD1 = EXTR SIGM3 MOMZ PD;
MD2 = 3.3840E4;
ERG4 = 100 * (ABS ((MD1 - MD2) / MD2));
*+++++*
* RESULTS *
*+++++*
MESS 'RESULTS : TEST elas13';
MESS '-----';
SAUT LIGN;
*
MESS ' ROTATION AT A ';
MESS '-----';
MESS ' THEORETICAL VALUE : ' RZA2 'rad';
MESS ' COMPUTED VALUE : ' RZA1 'rad';
MESS ' ERRE : ' ERG1 '%';
SAUT LIGN;
*
MESS ' REACTION AT A ';
MESS '-----';
MESS ' THEORETICAL VALUE : ' VA2 'N';
MESS ' COMPUTED VALUE : ' VA1 'N';
MESS ' ERRE : ' ERG2 '%';
SAUT LIGN;
*
MESS ' DEFLECTION AT D ';
MESS '-----';
MESS ' THEORETICAL VALUE : ' UYD2 'm';
MESS ' COMPUTED VALUE : ' UYD1 'm';
MESS ' ERRE : ' ERG3 '%';
SAUT LIGN;
*
MESS ' MOMENT OF FLEXION AT D ';
MESS '-----';
MESS ' THEORETICAL VALUE : ' MD2 'N.m';
MESS ' COMPUTED VALUE : ' MD1 'N.m';
MESS ' ERRE : ' ERG4 '%';
SAUT LIGN;
*+++++*
* GOOD WORKING MESSAGE *
*+++++*
ERGMAX = MAXI (PROG ERG1 ERG2 ERG3 ERG4);
SI (ERGMAX <EG 5.); ERRE 0; SINON; ERRE 5;
FINSI; SAUT 1 LIGN ;
TEMPS;
FIN;

```

Test elas13 (Comments)

- **Linear supports**

```
APP1 = APPUI DEPL KNOEUD POI1;
```

```
APP2 = APPUI DEPL KNOEUD POI2;
```

The operator APPUI constructs the rigidity object associated with linear support. The keyword DEPL puts a support on every displacement components. KNOEUD is the value of the stiffness and POI1 or POI2 are the points where the support is imposed.



1.9 Test stru2 (Description sheet)

Test name: stru2

Calculation type: ELASTIC MECHANIC 2D, PLAN STRESS

Finite element type: QUA4

Topic: Embedded orthotropic square plate. The plate is embedded at one side. A strength is applied to the opposite side, it is parallel to the plate.


Goal: Find the horizontal and vertical displacements at the center of the plate (point PE) and at the two free angles (points PA and PD).

Reference CASTEM: Comparison with six others finite element codes.

Version: 97' customer release

Model description:

YOUNG1	=	$2.15737 \cdot 10^{10} Pa$
YOUNG2	=	$7.377 \cdot 10^9 Pa$
YOUNG3	=	$7.377 \cdot 10^9 Pa$
NU12	=	0.31
NU23	=	0.31
NU13	=	0.31
G12	=	$2.319 \cdot 10^9 Pa$



Test stru2 (Results)


RESULTS

(10^{-4} m)	Displacement at PA		Displacement at PD		Displacement at PE	
	Horiz.	Verti.	Horiz.	Verti.	Horiz.	Verti.
Reference	1.09800	-1.30900	1.00300	-0.71340	0.47680	-0.49650
Castem	1.09961	-1.31276	0.99508	-0.71706	0.47668	-0.49812


CASTEM FIGURES



1.9. TEST STRU2 (DESCRIPTION SHEET)




ORTHOTROPIC SQUARE PLATE



ORTHOTROPIC SQUARE PLATE | DEFORMED PLATE


AMPLITUDE
10.



ORTHOTROPIC SQUARE PLATE | DEFORMED PLATE


AMPLITUDE
1.00E-06
0.00E+00

COMPONENTS
VECTORS
PX PY
PX PY




STRESS (MPa)

VAL = 0.00
> 4.48E-05
< -3.88E-06
0.00E+00
1.24E-05
2.48E-05
3.72E-05
4.96E-05
6.20E-05
7.44E-05
8.68E-05
9.92E-05
1.12E-04
1.24E-04
1.36E-04
1.48E-04
1.60E-04
1.72E-04
1.84E-04
1.96E-04
2.08E-04




STRESS (MPa)

VAL = 0.00
> 6.36E-04
< -6.00E-05
0.00E+00
1.20E-04
2.40E-04
3.60E-04
4.80E-04
6.00E-04
7.20E-04
8.40E-04
9.60E-04
1.08E-03
1.20E-03
1.32E-03
1.44E-03
1.56E-03
1.68E-03
1.80E-03
1.92E-03
2.04E-03



STRESS (MPa)

VAL = 0.00
> 3.52E-05
< -6.00E-06
0.00E+00
1.12E-05
2.24E-05
3.36E-05
4.48E-05
5.60E-05
6.72E-05
7.84E-05
8.96E-05
1.01E-04
1.12E-04
1.24E-04
1.36E-04
1.48E-04
1.60E-04
1.72E-04
1.84E-04
1.96E-04
2.08E-04



MIN-MAX STRESS

VAL = 0.00
> 4.27E-05
< -2.00E-06
0.00E+00
1.12E-05
2.24E-05
3.36E-05
4.48E-05
5.60E-05
6.72E-05
7.84E-05
8.96E-05
1.01E-04
1.12E-04
1.24E-04
1.36E-04
1.48E-04
1.60E-04
1.72E-04
1.84E-04
1.96E-04
2.08E-04

Test stru2 (Listing CASTEM)

```

*****
TITR 'AN EMBEDDED ORTHOTROPIC SQUARE PLATE' ;
*****
OPTI DIME 2 ELEM QUA4 MODE PLAN CONT ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
PA = 0.5 0.5 ; PB = -0.5 0.5 ;
PC = -0.5 -0.5 ; PD = 0.5 -0.5 ;
PE = 0. 0. ;
*
D1 = PA DROI 10 PB ; D2 = PB DROI 10 PC ;
D3 = PC DROI 10 PD ; D4 = PD DROI 10 PA ;
*
SUR1 = DALLAGE D1 D2 D3 D4 PLAN ;
SUR = SUR1 ET PE ;
ELIM SUR 0.001 ;
*
TITR 'ORTHOTROPIC SQUARE PLATE' ;
TRAC SUR ;
*
NBNSUR = NBNO SUR1 ;
NBELSUR = NBEL SUR1 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD = MODE SUR1 MECANIQUE ELASTIQUE ORTHOTROPE QUA4 ;
*
DIR1 = (COS 30.) (SIN 30.) ;
*
MAT = MATE MOD DIRE DIR1 YG1 2.5737E10 YG2 7.377E9
      YG3 7.377E9 NU12 0.31 NU23 0.31 NU13 0.31
      G12 2.319E9 ;
*
*----- BOUNDARY CONDITIONS -----*
CL = BLOQ DEPL ROTA D2 ;
*
*----- LOAD DEFINITION -----*
*THE LOAD IS A 1E4 N/m-LINEAR LOAD APPLIED ON AD.
*WITH A THICKNESS OF 0.01 m, THE FORCE BY UNIT PER
  THICKNESS IS 1E6 N/m
FTOT = 1.E6 ;
NBNOD4 = NBNO D4 ;
F5 = FTOT / ((2 * NBNOD4) - 2) ;
F50 = F5 0. ;
I = 1 ;
NBI = NBNOD4 - 1 ;
CHA = FORC (0. 0.) (D4 POIN 1) ;
REPETER BLOC1 NBI ;
  CHA1 = FORC F50 (D4 POIN I) ;
  CHA2 = FORC F50 (D4 POIN (I+1)) ;
  CHA = CHA ET CHA1 ET CHA2 ;
  I = I + 1 ;
FIN BLOC1 ;
*
*----- CALCULATION OF THE RIGIDITY MATRIX -----*
RI = RIGI MOD MAT ;
RIG = RI ET CL ;
*
*----- CALCULATION AND OUTPUT -----*
DEP = RESO RIG CHA ;
*
TITR ' ORTHOTROPIC SQUARE PLATE : DEFORMED PLATE' ;
DEFO = DEFO SUR DEP 0. BLAN ;
DEF1 = DEFO SUR DEP 10. ROUG ;
TRAC (DEFO ET DEF1) ;
*
ERGUA = 100 * (ABS ((UA1 - UA) / UA1)) ;
*
UD = EXTR DEP UX PD ; UD1 = 1.003e-4 ;
ERGUD = 100 * (ABS ((UD1 - UD) / UD1)) ;
*
UE = EXTR DEP UX PE ; UE1 = 4.768E-5 ;
ERGUE = 100 * (ABS ((UE1 - UE) / UE1)) ;
*
VA = EXTR DEP UY PA ; VA1 = -1.309E-4 ;
ERGVA = 100 * (ABS ((VA1 - VA) / VA1)) ;
*
VD = EXTR DEP UY PD ; VD1 = -7.134E-5 ;
ERGVD = 100 * (ABS ((VD1 - VD) / VD1)) ;
*
VE = EXTR DEP UY PE ; VE1 = -4.965E-5 ;
ERGVE = 100 * (ABS ((VE1 - VE) / VE1)) ;
*
SAUT 2 LIGN ;
MESS ' Results : test stru2.dgibi' ;
MESS ' -----' ;
SAUT 1 LIGN ;
MESS ' Displacements' ;
MESS ' -----' ;
SAUT 1 LIGN ;
*
MESS ' Point A : ' ;
MESS ' Theoretical horizontal displacement : 'UA1'm' ;
MESS ' Calculated horizontal displacement : 'UA'm' ;
MESS ' Error : ' ERGUA '%' ;
SAUT 1 LIGN ;
MESS ' Theoretical vertical displacement : 'VA1'm' ;
MESS ' Calculated vertical displacement : 'VA'm' ;
MESS ' Error : ' ERGVA '%' ;
SAUT 1 LIGN ;
*
MESS ' Point D : ' ;
MESS ' Theoretical horizontal displacement : 'UD1'm' ;
MESS ' Calculated horizontal displacement : 'UD'm' ;
MESS ' Error : ' ERGUD '%' ;
SAUT 1 LIGN ;
MESS ' Theoretical vertical displacement : 'VD1'm' ;
MESS ' Calculated vertical displacement : 'VD'm' ;
MESS ' Error : ' ERGVD '%' ;
SAUT 1 LIGN ;
*
MESS ' Point E : ' ;
MESS ' Theoretical horizontal displacement : 'UE1'm' ;
MESS ' Calculated horizontal displacement : 'UE'm' ;
MESS ' Error : ' ERGUE '%' ;
SAUT 1 LIGN ;
MESS ' Theoretical vertical displacement : 'VE1'm' ;
MESS ' Calculated vertical displacement : 'VE'm' ;
MESS ' Error : ' ERGVE '%' ;
SAUT 1 LIGN ;
*
*SAUT 1 LIGN ;
*MESS ' Stress at point E' ;
*SAUT 1 LIGN ;
*
CHAM1 = SIGM MOD MAT DEP ;
CHAM2 = CHAN NOEUD MOD CHAM1 ;
CHPO1 = CHAN CHPO MOD CHAM2 ;
*
SMXXE = EXTR CHPO1 SMXX PE ;
SMXXE = SMXXE / 1.E6 ;

SMXXE1 = 1.004 ;
ERGXX = 100 * (ABS ((SMXXE1 - SMXXE) / SMXXE1)) ;
SMYYE = EXTR CHPO1 SMYY PE ;

```



1.9. TEST STRU2 (DESCRIPTION SHEET)

```
SMYYE = SMYYE / 1.E6 ;
SMXYE = EXTR CHP01 SMXY PE ;
SMXYE = SMXYE / 1.E6 ;
*
*MESS ' Theoretical horizontal stress :' SMXXE1 'MPa';
*MESS ' Calculated horizontal stress :' SMXXE 'MPa';
*MESS '      Error :' ERGXX '%' ;
SAUT 1 LIGN ;
*MESS ' Theoretical vertical stress :' SMYYE 'MPa';
SAUT 1 LIGN ;
*MESS ' Calculated shear stress :' SMXYE 'MPa';
SAUT 1 LIGN ;
*
MESS ' Number of nodes :' NBNOSUR ;
MESS ' Number of elements :' NBELSUR ;
*
AFFICHE RIG CHA SUR 1.E-6 1.E-6 ;

CHAM10 = EXCO SMXX CHAM1 ;
TITR 'STRESS SMXX' ;
TRAC MOD CHAM10 SUR ;
*
CHAM11 = EXCO SMYY CHAM1 ;
TITR 'STRESS SMYY' ;
TRAC MOD CHAM11 SUR ;
*
CHAM12 = EXCO SMXY CHAM1 ;
TITR 'STRESS SMXY' ;
TRAC MOD CHAM12 SUR ;
*
CHAM13 = VMIS MOD CHAM1 ;
TITR 'VON-MISES STRESS' ;
TRAC MOD CHAM13 SUR ;
*
*----- GOOD WORKING MESSAGE -----*
ERGMAX = MAXI ( PROG ERGUA ERGUD ERGUE ERGVA ERGVD ERGVD );
*
SAUT 2 LIGN ;
SI (ERGMAX <EG 5.) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;
*
SAUT 1 LIGN ;
TEMPS ;
SAUT 1 LIGN ;
*
FIN ;
```

Test stru2 (Comments)

1. Construction of a field by element

```
CHAM2 = CHAN NOEUD MOD CHAM1 ;
```

Depending on the data, the functions of this operator are different. Here the key word 'NOEUD' will bring about the CHAN operator to construct from a new field by element CHAM1 (MCHAML type) and from a model object MOD (MCHAML type), a field by element CHAM2 of the same type as that given on input, lying respectively on the nodes, centres of gravity, stiffness integration points, mass integration points, and element stress calculation points. For this purpose, a least squares method is used together with the shape functions of the element.

2. Deformed shape display

```
AFFICHE RIG CHA SUR 1.E-6 1.E-6 ;
```

This procedure displays, on the screen, the deformed shape of a structure subjected to a given loading.

RIG : structure stiffness matrix (RIGIDITE type)

CHA : structure loading (CHPOINT type)

SUR : structure mesh (MAILLAGE type)

1.E-6 : coefficient of amplification of the deformed shape (FLOTTANT type) and of the vectors representing the loading (FLOTTANT type)



1.10 Test stru4 (Description sheet)

Test name: stru4

Calculation type: ELASTIC MECHANIC, 3D

Finite element type: TUYAU

Topic: Embedded curved tube subjected to flexion. The curved tube is embedded at one side. A strength is applied to the free end, it is parallel to the plan of the tube.

Goal: Find the horizontal and vertical displacement at the free end (point PB) and find the rotation of the tube.

Reference CASTEM: Guide VPCS, Test SSSL 06/90

Solution analytique; JS Prezemieniecki;


Theory of matrix structural analysis

New York, McGraw-Hill, 1968

Version: 97' customer release

Model description:

YOUNG	=	$2 \cdot 10^{11} \text{ Pa}$
NU	=	0.3
EPAISSEUR	=	0.002 m
RAYON	=	0.01 m




Test stru4 (Results)

RESULTS

	Horiz. displ. at PB	Vertical displ. at PB	Rotation
Reference	0.37908 m	0.24173 m	0.16539 rad
Castem	0.37871 m	0.24174 m	0.16533 rad

CASTEM FIGURES



AN EMBEDDED CURVED TUBE SUBJECTED TO FLEXION

Test stru4 (Listing CASTEM)

```

*****
TITR 'AN EMBEDDED CURVED TUBE SUBJECTED TO FLEXION' ;
*****
OPTI DIME 3 ELEM SEG2 ;

*----- GEOMETRY -----*
P0 = 0. 0. 0. ; PA = 0. 3. 0. ;
PB = 3. 0. 0. ;
CAB = CERC 20 PA P0 PB ;

OEIL = 0. 0. 1000. ;
TRAC OEIL CAB ;

*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*

MODL1 = MODE CAB MECANIQUE ELASTIQUE TUYAU ;
MAT1 = MATE MODL1 YOUN 2E11 NU 0.3 EPAI 0.002
      RAYO 0.01 VECT OEIL ;

*----- BOUNDARY CONDITIONS -----*

BL1 = BLOQ DEPL ROTA PA ;

*----- LOAD DEFINITION -----*

FOR1 = FORC FX 10. FY 5. PB ;
MOM1 = MOME MZ 8 PB ;
FORT = FOR1 ET MOM1 ;

*----- CALCULATION OF THE RIGIDITY MATRIX -----*

RG1 = RIGI MODL1 MAT1 ;
RGTOT = RG1 ET BL1 ;

*----- CALCULATION -----*

DEP = RESO RGTOT FORT ;
UBX = EXTR DEP PB UX ;
UBY = EXTR DEP PB UY ;
RBZ = EXTR DEP PB RZ ;

F1 = 10. ; F2 = 5. ; MZ = 8. ; E= 2E11 ;
I = 4.637E-9 ; R = 3. ;

UBX1 = (R**2)/4/E/I*((F1*PI*R)+(2*F2*R)+(4*MZ)) ;
UBY1 = (R**2)/4/E/I*((2*F1*R)+((3*PI)-(8.))*F2*R)+
(2*(PI-(2.))*MZ)) ;
RBZ1 = R/4/E/I*((4*R*F1)+(2*(PI-(2.))*F2*R)+(2*PI*MZ)) ;

*----- RESULTS PLACARDING -----*

MESS ' RESULTS ' ;
MESS ' ----- ' ;

SAUT 2 LIGN ;
MESS ' THEORETICAL HORIZONTAL DISPLACEMENT AT PB : ' UBX1 'M' ;
MESS ' CALCULATED HORIZONTAL DISPLACEMENT AT PB : ' UBX 'M' ;
R1= 100*(ABS((UBX-UBX1)/UBX1)) ;
MESS ' ERROR : ' R1 '%' ;
SAUT 2 LIGN ;
MESS ' THEORETICAL VERTICAL DISPLACEMENT AT PB : ' UBY1 'M' ;
MESS ' CALCULATED VERTICAL DISPLACEMENT AT PB : ' UBY 'M' ;
R2= 100*(ABS((UBY-UBY1)/UBY1)) ;
MESS ' ERROR : ' R2 '%' ;
SAUT 2 LIGN ;
MESS ' THEORETICAL ROTATION AT PB : ' RBZ1 'rad' ;
MESS ' CALCULATED ROTATION AT PB : ' RBZ 'rad' ;
R3= 100*(ABS((RBZ-RBZ1)/RBZ1)) ;
MESS ' ERROR : ' R3 '%' ;

*----- GOOD WORKING MESSAGE -----*
    
```



Test stru4 (Comments)

1. GEOMETRICAL CHARACTERISTIC : TUYAU

```
MODE1 = MODE CAB MECANIQUE ELASTIQUE TUYAU ;  
MAT1 = MATE MODL1 YOUN 2E11 NU 0.3 EPAI 0.002  
      RAYO 0.01 VECT OEIL ;
```

The TUYAU element is used in the modeling of parts of straight pipe or elbow which differ only in their radius of curvature. The pipe features are defined in the element local base, in the same way as the POUTRE element.

EPAI : thickness

RAYO : pipe external radius

VECT : key word for defining the local axis Oy. It must be followed by a POINT type object modeling an xOy vector. This is especially required when dealing with an elbow.

2. APPLICATION OF A MOMENT

```
MOM1 = MOME MZ 8 PB ;
```

The MOMENT operator creates a moment field resulting from the application of a moment represented either by the components of a vector, or by component values.

MZ : name of the components (MOT type)

8 : component values (FLOTTANT type) for PLAN CONT or PLAN DEFO

PB : geometrical support (MAILLAGE type)

MOM1 : moment field (CHPOINT type)

1.11 Test elas20 (Description sheet)

Test elas20 (Description sheet)

Test name: elas20

Calculation type: MECHANICS ELASTICITY 3D

Finite element type: TUYAU


Topic: Rubber-tubing subjected to gradient of pressure. The structure is composed by a one rubber-tubing : two linear parts separated by one crank part. The solid is modelised by a TUYAU finite element.

Goal: Find the displacements at a points PB, PBP and PAP. The castem results are compared to the analytical results.

Reference CASTEM: Test VPCS : SSSL08/90 : Modélisation des structures élastiques dans CASTEM 2000. J. S. Przemieniecki, theory of matrix structural analysis (New-York : Mcgraw-Hill Book Company, 1968) pp 155.

Version: 97' customer releas

Model description:




$$\begin{aligned}
 E &= 210 \cdot 10^9 \text{ Pa} \\
 \text{NU} &= 0.3 \\
 \text{SECT} &= 1.13 \cdot 10^{-4} \text{ m}^2
 \end{aligned}$$



Test elas20 (Results) RESULTS

	$U_x (PB)$	$U_y (PB)$	$U_x (PBP)(m)$	$U_y (PBP)(m)$	$U_y (PAP)(m)$
Reference	$8.2200E - 4$	$8.2200E - 4$	$1.37E - 4$	$8.22E - 4$	$6.8500E - 4$
Castem	$822692E - 4$	$8.22692E - 4$	$1.37544E - 4$	$8.22692E - 4$	$6.85148E - 4$

CASTEM figures



Test elas20 (Listing CASTEM)

```

*****
* RUBBER-TUBING SUBJECTED TO GRADIENT OF PRESSURE *
*****
OPTI ECHO 0 ; OPTI DIME 3 ELEM SEG2 ;
*****
*          GEOMETRY AND MESHING          *
*****
OEIL = 0. 0. 10. ;
LDRO1 = 3.048 ; RCO1 = 0.61 ;
LTOT1 = LDRO1 + RCO1 ;
PA = 0. 0. 0. ; PAP = 0. LDRO1 0. ;
PBP = RCO1 LTOT1 0. ; PB = LTOT1 LTOT1 0. ;
LIG1 = PA D 5 PAP ; LIG2 = PBP D 5 PB ;
LCOU1 = CERC 10 PAP ( RCO1 LDRO1 0. ) PBP ;
LIGNE = LIG1 ET LCOU1 ET LIG2 ;
TASS LIGNE ; ELIM LIGNE 1.E-5 ;
'TITRE' 'TUYPRES : MESHING' ;
TRAC LIGNE OEIL QUAL ;
*****
*          MECHANICS MODEL          *
*****
OBLI1 = 'MODL' LIG1 MECANIQUE ELASTIQUE TUYAU ;
OBLI2 = 'MODL' LIG2 MECANIQUE ELASTIQUE TUYAU ;
OBCO1 = 'MODL' LCOU1 MECANIQUE ELASTIQUE TUYAU ;
OBLIGN = OBLI1 ET OBLI2 ET OBCO1 ;
*****
*          VALUE OF THE PRESSURE LOAD          *
*****
PRES1 = 15.E6 ;
*****
*          MATERIAL AND CHARACTERISTICS          *
*****
MATLI1 = 'MATE' ( OBLI1 ET OBLI2 ) 'YOUN' 2.1E11
          'NU' 0.3 ;
MATCO1 = 'MATE' OBCO1 'YOUN' 2.1E11 'NU' 0.3 ;
MATLIGN = MATLI1 ET MATCO1 ;
CARLI1 = 'CARA' OBLI1 'EPAI' 26.187E-3 'RAYO' 203.2E-3
          'VECT' ( 0. 0. -1. ) 'PRES' PRES1 ;
CARLI2 = 'CARA' OBLI2 'EPAI' 26.187E-3 'RAYO' 203.2E-3
          'VECT' ( 0. 0. -1. ) 'PRES' PRES1 ;
CARCO1 = 'CARA' OBCO1 'EPAI' 26.187E-3 'RAYO' 203.2E-3
          'RACO' 0.6 'VECT' ( 0. 0. -1. ) 'PRES' PRES1 ;
CARLIGN = CARLI1 ET CARLI2 ET CARCO1 ;
CARTOT = MATLIGN ET CARLIGN ;
*****
*          BOUNDARY CONDITIONS          *
*****
RIGBLOQ = 'BLOQ' 'DEPL' 'ROTA' PA ;
*****
*          MATRIX OF RIGIDITY          *
*****
RIGLIGN = 'RIGI' OBLIGN CARTOT ;
RIGTOT = RIGLIGN ET RIGBLOQ ;
*****
*          MECHANICS LOADING          *
*****
FPRES1 = 'PRES' 'TUYA' OBLIGN CARLIGN ;
*****
*          RESOLUTION          *
*****
DEPL1 = 'RESO' RIGTOT FPRES1 ;
*****
*          RESULTS EXTRACTING          *
*****
VXB1 = 0.822E-3 ;
UXB1 = 'EXTR' DEPL1 'UX' PB ;
ERG1 = 100. *('ABS'((VXB1 - UXB1) / VXB1)) ;
VYB1 = VXB1 ;
UYB1 = 'EXTR' DEPL1 'UY' PB ;
ERG2 = 100. *('ABS'((VYB1 - UYB1) / VYB1)) ;
VXBP1 = 0.137E-3 ;
UXBP1 = 'EXTR' DEPL1 'UX' PBP ;
ERG3 = 100. *('ABS'((VXBP1 - UXBP1) / VXBP1)) ;
VYBP1 = 0.822E-3 ;
UYBP1 = 'EXTR' DEPL1 'UY' PBP ;
ERG4 = 100. *('ABS'((VYBP1 - UYBP1) / VYBP1)) ;
VYAP1 = 0.685E-3 ;
UYAP1 = 'EXTR' DEPL1 'UY' PAP ;
ERG5 = 100. *('ABS'((VYAP1 - UYAP1) / VYAP1)) ;
*****
*          PLACARDING OF THE RESULTS          *
*****
'MESS' 'RESULTS : TEST TUYPRES' ;
'MESS' '-----' ;
SAUT LIGN ;
'MESS' ' displacement ux (m) of point PB' ;
'MESS' '-----' ;
'MESS' ' theoretical value : ' VXB1 'm' ;
'MESS' ' computed value : ' UXB1 'm' ;
'MESS' ' discarding value : ' ERG1 '%' ;
SAUT LIGN ;
'MESS' ' displacement uy (m) of point PB' ;
'MESS' '-----' ;
'MESS' ' theoretical value : ' VYB1 'm' ;
'MESS' ' computed value : ' UYB1 'm' ;
'MESS' ' discarding value : ' ERG2 '%' ;
SAUT LIGN ;
'MESS' ' displacement ux (m) of point PBP' ;
'MESS' '-----' ;
'MESS' ' theoretical value : ' VXBP1 'm' ;
'MESS' ' computed value : ' UXBP1 'm' ;
'MESS' ' discarding value : ' ERG3 '%' ;
SAUT LIGN ;
'MESS' ' displacement uy (m) of point PBP' ;
'MESS' '-----' ;
'MESS' ' theoretical value : ' VYBP1 'm' ;
'MESS' ' computed value : ' UYBP1 'm' ;
'MESS' ' discarding value : ' ERG4 '%' ;
SAUT LIGN ;
'MESS' ' displacement uy (m) of point PAP' ;
'MESS' '-----' ;
'MESS' ' theoretical value : ' VYAP1 'm' ;
'MESS' ' computed value : ' UYAP1 'm' ;
'MESS' ' discarding value : ' ERG5 '%' ;
ERGMAX = 'MAXI'('PROG' ERG1 ERG2 ERG3 ERG4 ERG5) ;
'SI' ( ERGMAX < EG 5. ) ; ERRE 0 ;
'SINON' ; ERRE 5 ; 'FINSI' ;
SAUT LIGN ; TEMPS ;
FIN ;

```




Test elas20 (Comments)

- **Internal pressure in a TUYAU element**

```
CARLI1 = 'CARA' OBLI1 'EPAI' 26.187E-3 'RAYO' 203.2E-3 'VECT' ( 0. 0. -1. ) 'PRES' PRES1 ;  
CARLI2 = 'CARA' OBLI2 'EPAI' 26.187E-3 'RAYO' 203.2E-3 'VECT' ( 0. 0. -1. ) 'PRES' PRES1 ;  
CARCO1 = 'CARA' OBCO1 'EPAI' 26.187E-3 'RAYO' 203.2E-3 'RACO' 0.6 'VECT' ( 0. 0. -1. )  
'PRES' PRES1 ;
```

The geometrical characteristics of TUYAU elements can be completed by giving the internal pressure (with keyword PRES) and the radius of curvature if it is an elbow (with keyword RACO).

1.12 Test testICQ4 (Description sheet)

Test name: testICQ4

Calculation type: ISOTROPIC ELASTIC MECHANIC, PLAN DEFO

Finite element type: ICQ4

Topic: Square plate subjected to hydrostatic and deviatoric loads. A square plate is embedded at two adjacent sides. Deviatoric and hydrostatic loads are applied to each side.

Goal: Test the ICQ4-incompressible elements.

Reference CASTEM: Rapport DMT sur les elements ICQ4.

Version: 97' customer release

Model description:

YOUNG	=	$2 \cdot 10^{11} Pa$
NU	=	0.499
COEFF. HYDROSTATIC	:	$ALPHA = 1.$
COEFF. DEVIATORIC	:	$BETA = 1.E - 3$



1.12. TEST TESTICQ4 (DESCRIPTION SHEET)

Test testICQ4 (Listing CASTEM)

```

*****
TITR 'ICQ4 INCOMPRESSIBLE ELEMENTS' ;
*****
OPTI DIME 2 MODE PLAN DEFO ELEM QUA4 ;
*----- GEOMETRY -----*
P1 = 0. 0. ; P2 = 100. 0. ;
P3 = 100. 100. ; P4 = 0. 100. ;
*
LI1 = DROI 10 P1 P2 ;
LI2 = DROI 10 P2 P3 ;
LI3 = DROI 10 P3 P4 ;
LI4 = DROI 10 P4 P1 ;
SUR1 = DALL LI1 LI2 LI3 LI4 PLAN ;
*
*----- MECHANIC MODEL -----*
NU1 = 0.499 ;
*
MOD1 = MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE ICQ4 ;
MAT1 = MATE MOD1 YOUNG 2.E11 NU NU1 ;
RIGI1 = RIGI MOD1 MAT1 ;
*
*----- BOUNDARY CONDITIONS -----*
BL1 = BLOQ LI1 UY ;
BL2 = BLOQ LI4 UY ;
BL3 = BLOQ P1 UX ;
*
*----- LOAD DEFINITION -----*
* COEFFICIENT OF LOAD CONTRIBUTION
* (ALPHA --> HYDROSTATIC)
* (BETA --> DEVIATORIC)
*
ALPHA = 1. ;
BETA = 1.E-3 ;
*
*----- DEVIATORIC LOAD -----*
*-----ON LI1-----
CHPX1 = COOR 1 LI1 ;
CHPRESY1 = PRES MASS MOD1 (CHPX1 * (-2.)*1.E5) ;
CHPRES1 = BETA * CHPRESY1 ;
VECFOR1 = VECT CHPRES1 1.E-7 FX FY ROUG ;
*
*-----ON LI2-----
CHPY2 = COOR 2 LI2 ;
CHPRESX2 = PRES MASS MOD1 (200.*1.E5) LI2 ;
CHPRESY2 = PRES MASS MOD1 (CHPY2 * (-2.)*1.E5) ;
CHFORSX2 = CHPRESX2 ;
CHFORY2 = EXCO FX CHPRESY2 FY ;
CHPRES2 = BETA * ( CHFORSX2 ET CHFORY2 ) ;
VECFOR2 = VECT CHPRES2 FX FY 1.E-7 ROUG ;
*
*-----ON LI3-----
CHPX3 = COOR 1 LI3 ;
CHPRESX3 = PRES MASS MOD1 (-200.*1.E5) LI3 ;
CHPRESY3 = PRES MASS MOD1 (CHPX3 * (-2.)*1.E5) ;
CHFORSX3 = EXCO FY CHPRESX3 FX ;
CHFORY3 = CHPRESY3 ;
CHPRES3 = BETA * ( CHFORSX3 ET CHFORY3 ) ;
VECFOR3 = VECT CHPRES3 FX FY 1.E-7 ROUG ;
*
*-----ON LI4-----
CHPY4 = COOR 2 LI4 ;
CHPRESY4 = PRES MASS MOD1 (CHPY4 * (-2.)*1.E5) ;
CHFORY4 = EXCO FX CHPRESY4 FY ;
CHPRES4 = BETA * CHFORY4 ;
VECFOR4 = VECT CHPRES4 1.E-7 FX FY ROUG ;
*
*----- HYDROSTATIC LOAD -----*
*-----ON LI1-----
PRESY1 = PRES MASS MOD1 (1.E5) LI1 ;
PRES1 = ALPHA * PRESY1 ;
VFOR1 = VECT PRES1 1.E-5 FX FY VERT ;
*
*-----ON LI2-----
PRESX2 = PRES MASS MOD1 (1.E5) LI2 ;
PRES2 = ALPHA * PRESX2 ;
VFOR2 = VECT PRES2 FX FY 1.E-5 VERT ;
*
*-----ON LI3-----
PRESY3 = PRES MASS MOD1 (1.E5) LI3 ;
PRES3 = ALPHA * PRESY3 ;
VFOR3 = VECT PRES3 FX FY 1.E-5 VERT ;
*
*-----ON LI4-----
PRESX4 = PRES MASS MOD1 (1.E5) LI4 ;
PRES4 = ALPHA * PRESX4 ;
VFOR4 = VECT PRES4 1.E-5 FX FY VERT ;
*
*----- DISPLACEMENT CALCULATION -----*
* ANALYTIC SOLUTION AT P3 : UYAN.
X3 = COOR 1 P3 ;
Y3 = COOR 2 P3 ;
MU = 1.E11 / ( 1 + nu1 ) ;
UYAN = ( 2 * X3 * Y3 ) / ( 2 * MU ) * 1.E5 * BETA ;
*
* CASTEM SOLUTION AT P3 : UYCAS.
DEPTOT = RESO (RIGI1 ET BL1 ET BL2 ET BL3)
(CHPRES2 ET CHPRES3 ET CHPRES4 ET CHPRES1
ET PRES2 ET PRES3 ET PRES4 ET PRES1) ;
UYCAS = EXTR DEPTOT UY P3 ;
*
* ERROR TEST ON UY DISPLACEMENT (<1%)
ERRY = ABS ( ( UYAN - UYCAS ) / ( UYAN ) ) * 100 ;
SI ( ERRY >EG 1. ) ;
ERREUR 5 ;
FINSI ;
*
*----- STRESS CALCULATION -----*
* ANALYTIC SOLUTION : PRESSURE = ALPHA*1.E5.
*
* CASTEM SOLUTION : MAXIMUM PRESSURE = MAXPRES1.
*
SIG1 = SIGMA MOD1 DEPTOT MAT1 ;
SXX = EXCO SMXX SIG1 P ;
SYY = EXCO SMYY SIG1 P ;
SZZ = EXCO SMZZ SIG1 P ;
SP = (SXX + SYY + SZZ) / 3. ;
MAXPRES1 = MAXI SP ;
*
* ERROR TEST ON PRESSURE (<1%)
ERRPRESS = ABS ((MAXPRES1 + (ALPHA * 1.E5))
/ (ALPHA * 1.E5)) * 100. ;
SI ( ERRPRESS >EG 1. ) ;
ERREUR 5 ;
FINSI ;
*
FIN;

```

Test testICQ4 (Comments)

1. ICQ4 INCOMPRESSIBLE ELEMENT

```
OPTI DIME 2 MODE PLAN DEFO ELEM QUA4 ;
MOD1 = MODE SUR1 MECANIQUE ELASTIQUE ISOTROPE ICQ4 ;
```

The incompressible quadrangle with four nodes ICQ4 is a finite element of the QUA4 geometrical support in MECANIQUE.

2. DEVIATORIC LOAD

```
BETA = 1.E-3 ;
*-----ON LI1-----
CHPX1 = COOR 1 LI1 ;
CHPRESY1 = PRES MASS MOD1 (CHPX1 * (-2.)*1.E5) ;
CHPRES1 = BETA * CHPRESY1 ;
VECFOR1 = VECT CHPRES1 1.E-7 FX FY ROUG ;
```

The COOR operator enables the user to recover the Nith coordinate (here the first, then UX) of the MAILLAGE (LT1) type object.

The PRES operator calculates the nodal forces which are equivalent to a pressure applied on an object. Here this operator is connected by key words MASS referring to the type of element on which is applied. MOD1 is the object on which the pressure is applied (MMODEL type). The pressure (FLOTTANT type) $(-2)*1.E-5$ is negative because it is carried by the negative normal to the element.

The VECT operator constructs a VECTEUR type object (VECTFOR1) from the components of a vector field (CHPRES1), and $1.E-7$ is the coefficient of amplification (FLOTTANT type). The names of the components of the field associated with the directions Ox, Oy (MOT type) are FX and FY.

3. HYDROSTATIC LOAD

```
ALPHA = 1.
PRESX2 = PRES MASS MOD1 (1.E5) LI2 ;
PRES2 = ALPHA * PRESX2 ;
VFOR2 = VECT PRES2 FX FY 1.E-5 VERT ;
```



1.13 Test vsur1 (Description sheet)

Test name: vsur1

Calculation type: ISOTROPIC ELASTIC MECHANIC


Finite element type: COQ2

Topic: VSUR operator test. A circular structure is subjected to a surface vector field, a normal vector field and a field of jacobian.

Goal: Test the VSUR operator with COQ2 elements.

Version: 97' customer release

Model description:



Test vsur1 (Listing CASTEM)

```
*****
TITR 'VSUR OPERATOR TEST WITH COQ2' ;
*****
OPTI ECHO 0 ;
OPTI DIME 2 ELEM SEG2 MODE AXIS ;
*
*----- GEOMETRY -----*
P0 = 0. 0. ; P1 = 10. 10. ; PC = 0. 10. ;
OBJ1 = CERC 10 P0 PC P1 ;
*
*----- MECHANIC MODEL -----*
MOD = MODE OBJ1 MECANIQUE ELASTIQUE ISOTROPE COQ2 ;

*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ DEPL ROTA P1 ;
CL2 = BLOQ UZ P0 ;
* COMPARISON WITH FIELD ISSUED FROM VSUR
* CHE1 : SURFACE VECTORS FIELD
  CHE1 = VSUR MOD ;
* CHN1 : NORMAL VECTOR FIELD
  CHN1 = VSUR MOD NORM ;
* CHJ1 : FIELD OF JACOBIANS
  CHJ1 = JACO MOD ;
CNR1 = EXCO VR CHE1 SCAL ;
CNZ1 = EXCO VZ CHE1 SCAL ;
* FIELD OF NORM
CNN1 = (CNR1*CNR1) + (CNZ1*CNZ1) ;
CJN1 = CHJ1*CHJ1 ;
CZE1 = CNN1 - CJN1 ;
* LIST CZE1 ;
CPZ1 = CHAN CHPO MOD CZE1 ;
RES1 = MAXI CPZ1 ABS ;
* LIST RES1 ;
CNR1 = EXCO VR CHN1 SCAL ;
CNZ1 = EXCO VZ CHN1 SCAL ;
CNN1 = (CNR1*CNR1) + (CNZ1*CNZ1) ;
* LIST CNN1 ;
CPN1 = CHAN CHPO MOD CNN1 ;
RES2 = MINI CPN1 ABS ;
RES3 = MAXI CPN1 ABS ;
* LIST RES2 ;
* LIST RES3 ;
EC1 = RES3 - RES2 ;
*----- GOOD WORKING MESSAGE -----*
SI ((RES1 < 1.e-6) ET (ABS(RES2 - 1.) < 1.E-6)
  ET (EC1 < 1.E-6)) ;
  MESS OPERATEUR <VSUR> ERR 0 ;
  ERRE 0 ;
SINON ;
  MESS OPERATEUR <VSUR> ERR 5 ;
  ERRE 5 ;
FINSI ;
FIN ;
```



Test vsur1 (Comments)

1. VSUR OPERATOR

CHE1 = VSUR MOD ;
CHN1 = VSUR MOD NORM

The VSUR operator enables the user to calculate the surface vectors associated with the integration points of the shell elements, for the other types of elements the vector field is empty. If the NORM key-word is not specified, the norm of the vectors is equal to the jacobians at the considered integration point, otherwise the vector field is the normal vector field at the shell elements. Here MOD is the model object (MMODEL type) and CHE1 and CHN1 are the generated objects (MCHAML type, SCALAIRE subtype).

2. JACOBIEEN OPERATOR

CHJ1 = JACO MOD ;

The JACOBIEEN operator enables the user to calculate the absolute value of Jacobians at the integration points of the elements of the model. Here MOD is the model object (MMODEL type) and CHJ1 is the generated object (MCHAML type, SCALAIRE subtype).



Chapter 2

BUCKLING

2.1 Test flam1 (Description sheet)

Test name: flam1

Calculation type: MECHANICS BUCKLING. 2D PLANE STRESS

Finite element type: COQ2


Topic: Eulerian buckling of a beam clamped at one end. The beam is clamped at one end. A force, parallel to the beam, is applied to the free end.

Goal: Find the strength multiplier M_S .

Reference: Analytical solution.

Version: 97' customer version

Model description:





Test flam1 (Results)

Results

	M_y
Analytic	41.123
Castem	41.123

CASTEM figures



Test flam1 (Listing CASTEM)

```

*****
TITRE ' CLAMPED BEAM BUCKLING ' ;
OPTION ECHO 0 DIME 2 ELEM SEG2
      MODE PLAN CONT ;

TEMPS ;
*
*----- MESHING DEFINITION -----
*
DENSITE 1. ;
P1 = 50. 0. ;
P2 = 50. 10. ;
L1 = P1 D P2 ;
*
TRAC L1 ;
*
*----- CREATION OF THE MODEL -----
*
MOD1 = MODEL L1 MECANIQUE ELASTIQUE COQ2;
*
*----- DEFINITION OF MATERIAL
*----- AND GEOMETRIC CHARACTERISTICS ----
*
MAT1 = MATER MOD1 YOUN 20000. NU 0.3      ;
CAR1 = CARAC MOD1 EPAI 1. ;
MAT1=MAT1 ET CAR1;
*
*----- ELEMENTARY RIGIDITIES
*----- AND BOUNDARY CONDITIONS -----
*
RIG1 = RIGI MOD1 MAT1 ;
BL1  = BLOQUE DEPL ROTA P1 ;
RIG11 = RIG1 ET BL1;
*
*----- LOAD DEFINITION -----
*
FE1 = FORCE ( 0 -1. ) P2 ;
*
*-- RESOLUTION AND STRESSES CALCULATION ---
*
DEP1 = RESO RIG11 FE1 ;
SIG1 = SIGM MOD1 MAT1 DEP1 ;
*
*----- CALCULATION OF INITIAL
*----- STRESSES MATRIX -----
*
KSII1 = KSIGMA MOD1 SIG1 CAR1 'FLAM' ;
*
*----- INITIALISATION OF INPUT TABLE
*----- OF FLAMBAGE SUBROUTINE -----
*
ETAB = TABLE      ;
ETAB.'CLIM' = BL1 ;
*ETAB.'RIGI' = RIG1 ;
*ETAB.'KSIG' = KSII1 ;
ETAB.'SIG1' = SIG1;
ETAB.'MATE' = MAT1;
ETAB.'LAM1' = 0.001 ;
ETAB.'LAM2' = 100. ;
ETAB.'NMOD' = 1    ;
ETAB.'OBJM' = MOD1;
*
*----- FLAMBAGE SUBROUTINE CALL -----
*
SAUT PAGE ;
STAB = FLAMBAGE ETAB ;

```

```

*
*----- RESULTS -----
*
LA1 = STAB . 1 . LAMB ;
SAUT PAGE ;
*
*----- BUCKLING SHAPE DRAWING -----
*
MM1 = STAB . 1 . DEPL ;
DE = DEFO MM1 L1 1. ROUGE ;
DA = DEFO MM1 L1 0. BLEU ;

TRAC ( DE ET DA ) ;

*
MESS ' ' ;
MESS ' ' ;
MESS ' STRENGTH MULTIPLIER ' ;
MESS ' ' ;
MESS ' ANALYTICAL SOLUTION : 41.123 ' ;
MESS ' COMPUTED SOLUTION : ' LA1 ;
*
TEMPS ;
*----- GOOD WORKING MESSAGE -----
ERR=100*(ABS(41.123 - LA1)/41.123);
SI (ERR < 5);
ERRE 0;
SINON;
ERRE 5;
FINSI;

FIN ;

```



Test flam1 (Comments)

1. Computation of initial stresses matrix

KSII = KSIGMA MOD1 SIG1 CAR1 'FLAM' ;

It is necessary to compute the initial stresses matrix in order to make a buckling analysis. The operator KSIGMA builds the geometrical rigidity matrix joined to a field of stress. The keyword FLAM is necessary to specify that the matrix will be used for a buckling analysis, the created object is also a mass matrix (without the keyword FLAM, the object would be a rigidity matrix).

2. Initialisation of input table of the 'flambage' subroutine

ETAB = TABLE ; ETAB.'CLIM' = BL1 ;
*ETAB.'RIGI' = RIG1 ; ETAB.'KSIG' = KSI1 ;
ETAB.'SIG1' = SIG1 ; ETAB.'MATE' = MAT1 ;
ETAB.'LAM1' = 0.001 ; ETAB.'LAM2' = 100. ;
ETAB.'NMOD' = 1 ; ETAB.'OBJM' = MOD1 ;

ETAB is a table which contains the input data for the using of 'flambage' subroutine. This table is defined by different indices. The indices CLIM, LAM1, LAM2, NMOD and OBJM are necessary. The both other indices are either SIG1 and MATE or RIGI and KSIG.

- CLIM : matrix joined to the clampings
- LAM1 : minimal multiplier coefficient
- LAM2 : maximal multiplier coefficient
- NMOD : number of eigen shapes
- OBJM : model object
- SIG1 : yield of stress changing with the strength multiplier
- MATE : material properties
- RIGI : rigidity matrix
- KSIG : geometrical rigidity matrix

3. 'Flambage' subroutine

STAB = FLAMBAGE ETAB ;
LA1 = STAB . 1 . LAMB ;
MM1 = STAB . 1 . DEPL ;

STAB is a table which contains the output data given by the 'flambage' subroutine. The index of STAB square with every mode shape number so there are as many tables as number of computed moden shapes. Each table joined to a mode shape contains four objects which are indexed by :

- LAMB : strength multiplier coefficient
- DEPL : deformed and normed shape
- MODN : number of Fourier harmonic
- MODM : number of azimuth mode shape

2.2 Test four1 (Description sheet)

Test name: four1

Calculation type: MECHANICS BUCKLING. 2D FOURIER

Finite element type: COQ2


Topic: Endless cylinder under external pressure. This test simulates an endless cylinder subjected to an external pressure. The first natural frequency of the structure's second mode shape is determined by a Fourier analysis. This result is used to make a buckling analysis.

Goal: Find the strength multipliers M_{s1} and M_{s2} .

Reference: Analytic solution.

Version: 97' customer release

Model description:



$$\begin{aligned}
 L &= 100mm \\
 R &= 1000mm \\
 Th &= 1mm \\
 P &= 1MPa
 \end{aligned}$$

$$\begin{aligned}
 E &= 20000MPa \\
 \nu &= 0
 \end{aligned}$$



Test four1 (Results)

Results

	M_s1	M_s2
Analytic	$6.66 \cdot 10^{-6}$	$5 \cdot 10^{-6}$
Castem	$6.659 \cdot 10^{-6}$	$4.995 \cdot 10^{-6}$

CASTEM figures

4

3

2

1

ENDLESS CYLINDER UNDER EXTERNAL PRESSURE

Test four1 (Listing CASTEM)

```

*****
TITRE 'ENDLESS CYLINDER UNDER
      EXTERNAL PRESSURE';
OPTI DIME 2 ELEM SEG2 MODE FOUR 0;
TEMPS ;

*
*----- GEOMETRY CONSTRUCTION -----
*
PA = 1000. 0.; PB = 1000. 100.;
PO1= 0. 0.; PO2= 0. 100.;
CYL=PA D 4 PB;
*
TRAC ECLA ELEM CYL;
*
MOD1= MODEL CYL MECANIQUE ELASTIQUE COQ2;

*
*-- DECLARATION OF 'FOURIER NOHARM' ----
*----- FOR OBJECTS USED WITH -----
*----- DIFFERENT HARMONIC NUMBERS -----
*
OPTI MODE FOUR NOHARM;
*
*
*
*--- SYMMETRICAL BOUNDARY CONDITIONS ---
*
SYMB = SYMT CYL DEPL ROTA PA PO1 0.5;
SYMH = SYMT CYL DEPL ROTA PB PO2 0.5;
CDL = SYMB ET SYMH;
*
*----- MATERIAL AND -----
*----- GEOMETRICAL CHARACTERISTICS -----
*
PP = MANU CHPO CYL 1 P 1. ;
MAT = MATER MOD1 YOUN 20000. NU 0.;
CAR = CARA MOD1 EPAI 1.;
MAT = MAT ET CAR;
*
*--- DECLARATION OF 'FOURIER MODE 0' ---
*----- FOR STRESS CALCULATION -----
*
OPTI MODE FOUR 0;

FP = PRES COQU MOD1 1. NORM ;
RIG = RIGI MOD1 MAT;
*
RIGADD = RIGI UT 1.E-4 PA ;
*
AAA = RIG ET CDL ET RIGADD ;
U = RESO AAA FP ;
SIG = SIGMA U MOD1 MAT;
*
*--- DECLARATION OF 'FOURIER MODE 2' ---
*----- FOR BUCKLING ANALYSIS -----
*
OPTI MODE FOUR 2;
*
PP = MANU CHPO CYL 1 P 1. ;
*
MKSI = KSIGMA MOD1 MAT (SIG * -1.) FLAM;
RIG = RIGI MOD1 MAT ;
*

```

```

*----- DETERMINATION OF 1ST -----
*----- NATURAL FREQUENCY -----
*
MODF = VIBR PROC (PROG 0.)
      (RIG ET CDL) MKSI;
LAMBDA1 = ((TIRE MODF FREQ RANG 1)
           * 6.28) ** 2;
*
MKP = KP PP MOD1 FLAM ;
MODF = VIBR PROC (PROG 0.)
      (RIG ET CDL) (MKSI ET MKP);
LAMBDA2 = ((TIRE MODF FREQ RANG 1)
           * 6.28) ** 2;
*
SAUT PAGE ;
SAUT 5 LIGN ;
MESS 'K(SIG) ONLY : WE HAVE TO FIND
      LAMBDA= 6.66E-6' ;
MESS '          CALCULATION GIVES
      LAMBDA=' LAMBDA1;
SAUT 2 LIGN ;
*
MESS 'K(SIG)+K(P) : WE HAVE TO FIND
      LAMBDA=5E-6' ;
MESS '          CALCULATION
      LAMBDA=' LAMBDA2;
SAUT 2 LIGN ;
*
TEMPS ;
*----- GOOD WORKING MESSAGE -----

ERR1=100*(ABS(6.66E-6 - LAMBDA1)/6.66E-6);
ERR2=100*(ABS(5E-6 - LAMBDA2)/5E-6);
ERRMAX=MAXI (PROG ERR1 ERR2);
SI (ERRMAX < 5 );
ERRE 0;
SINON ;
ERRE 5;
FINSI;

FIN ;

```




Test four1 (Comments)

1. Drawing of meshing with dispersed elements

TRAC ECLA ELEM CYL;

The operator TRAC with keyword ECLA allows to represent the meshing with dispersed elements. Each element is represented with a homothetic rate (0.5 by default). The other keyword ELEM allows to indicate the local number of the elementary objects.

2. Computation with FOURIER mode

OPTI MODE FOUR NOHARM;

The computation mode is defined with the operator OPTI. The keyword NOHARM allows to create objects which could be used with any number of harmonic.

OPTI MODE FOUR 0;

All objects which are created after this command will be used only with harmonic 0. When the computation is done with harmonic 2, the previous objects are unavailable.

3. Determination of natural frequencies

MODF = VIBR PROC (PROG 0.) (RIG ET CDL) MKSI;
LAMBDA1 = ((TIRE MODF FREQ RANG 1) * 6.28) ** 2;

The natural mode shapes are calculated which frequency value is close to zero. The type of the object MODF is called SOLUTION. MODF has different components among which the frequency (component FREQ).

In order to obtain the value of the frequency, we use the operator TIRE with keyword FREQ to indicate the type of the desired component and the keyword RANG to indicate the row of the moment. In this case there is only one shape, so the value of the row is one, but if there were two shapes, the first row would correspond to the lower frequency and the second to the higher frequency.

4. Computation of pressure matrix

PP = MANU CHPO CYL 1 P 1. ;
MKP = KP PP MOD1 FLAM ; The operator KP creates the pressure matrix joined to the work of pressure in large displacements. With keyword FLAM, the created object is a mass matrix.



Chapter 3

PLASTICITY

3.1 Test plas1 (Description sheet)

Test name: plas1

Calculation type: MECHANICS PLASTICITY 2D PLANE STRESS

Finite element type: COQ4


Topic: A rectangular-plate subjected to simple bending. The structure is a rectangular-plate subjected to a simple bending. This bending is obtained by the fixed displacement (D) at the tips (see figure), which is a function of time (T). At the first step of time (0.02875), the strain range is elastic, at the second one (0.05), it is plastic. At both time steps, the elastic and the plastic theoretical bending moments are compared to the CASTEM computed values. These values are computed from the reactions.

Goal: Find the computed elastic and plastic bending moment M1 and M2. The theoretical values are :
 $M1 = 805 \text{ kN}$ $M2 = 1074 \text{ kN}$

Reference CASTEM: Test NAFEMS : Analytical-Solution : MP1 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:





3.1. TEST PLAS1 (DESCRIPTION SHEET)


$$E = 2.1 \cdot 10^{11} \text{ Pa}$$

$$NU = 0.28$$

$$SYG = 483E6 \text{ Pa}$$

Test plas1 (Results)
RESULTS

	$M1$	$M2$
Reference	805kN	1074KN
Castem	811.08kN	1085.2kN

CASTEM FIGURES


A RECTANGULAR-PLATE SUBJECTED TO SIMPLE FLEXION



3.1. TEST PLAS1 (DESCRIPTION SHEET)

Test plas1 (Listing CASTEM)

```

TITRE'RECTANGULAR-PLATE SUBJECTED TO SIMPLE FLEXION';
*****
OPTION ECHO 0 ;
OPTION DIME 2 ELEM QUA4 MODE PLAN CONT ;
TEMPS ;
*-----GEOMERTY AND MESHING -----
O = 0 0 ;
A = 12.5E-3 50E-3 ; B = 12.5E-3 -50E-3 ;
C = 0 -50E-3 ; D = 0 50E-3 ;
A1 = 12.5E-3 37.5E-3 ; A2 = 12.5E-3 25E-3 ;
A3 = 12.5E-3 12.5E-3 ; A4 = 12.5E-3 0 ;
A5 = 12.5E-3 -12.5E-3 ; A6 = 12.5E-3 -25E-3 ;
A7 = 12.5E-3 -37.5E-3 ;
L1 = A DROIT 1 A1 ; L2 = A1 DROIT 1 A2 ;
L3 = A2 DROIT 1 A3 ; L4 = A3 DROIT 1 A4 ;
L5 = A4 DROIT 1 A5 ; L6 = A5 DROIT 1 A6 ;
L7 = A6 DROIT 1 A7 ; L8 = A7 DROIT 1 B ;
*
L14 = L1 ET L2 ET L3 ET L4
      ET L5 ET L6 ET L7 ET L8 ;
*
L9 = B DROIT 1 C ; L10 = C DROIT 4 O ;
L11 = O DROIT 4 D ; L12 = L10 ET L11 ;
L13 = D DROIT 1 A ;
*
SU = L14 L9 L12 L13 DALL PLAN ;
ELIM 0.01 SU ; TRACE qual SU ;
*
*----- BOUNDARY CONDITIONS -----
*----- AND FIXED DISPLACEMENT-----
*
CL1 = BLOQ L12 UX ; CL1B = BLOQ O UY ;
CL2 = BLOQ A UX ; CL3 = BLOQ B UX ;
CL = CL1 ET CL1B ET CL2 ET CL3 ;
*
D1 = DEPI CL2 1E-3 ; D2 = DEPI CL3 -1E-3 ;
DEP = D1 ET D2 ;
*-----
* FOR EXSEMPL : AT A1 , D = 7/8 D1 + 1/8 D2
*-----
*---- THE DISPLACEMENT AT EACH NODE -----
*----- IS A LINEAR RELATION OF THE -----
*----- DISPLACEMENTS AT THE EXTREMITIES -----
*
R1 = RELA 8 UX A1 - 7 UX A - 1 UX B ;
R2 = RELA 8 UX A2 - 6 UX A - 2 UX B ;
R3 = RELA 8 UX A3 - 5 UX A - 3 UX B ;
R4 = RELA 8 UX A4 - 4 UX A - 4 UX B ;
R5 = RELA 8 UX A5 - 3 UX A - 5 UX B ;
R6 = RELA 8 UX A6 - 2 UX A - 6 UX B ;
R7 = RELA 8 UX A7 - 1 UX A - 7 UX B ;
R = R1 ET R2 ET R3 ET R4 ET R5 ET R6 ET R7 ;
*
*- DESCRIPTION OF THE MODEL MATERIAL-----
*- OPERATOR NONLIN (NON LINEAIRE) -----
*
MO = MODE SU MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT QUA4;
*----- MATERIAL CHARACTERISTICS -----
*----- AND RIGIDITY MATRIX -----
*
MAT1 = MATEPLAST.tex MO YOUN 2.1E11 SIGY 483E6 NU 0.3 ;
RI1 = RIGI MO MAT1 ;
RIG1 = RI1 ET CL ET R ;
*
*-----LINEAR RELATION WITH DISPLACEMENT AND TIME-----
*----- DEP = F(T)*D -----
*
L11 = PROG 0. 1. ; LI2 = PROG 0. 1. ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR 'DIMP' DEP EV ;
*
LIS1 = PROG 0.02875 0.05 ;
*
*--CALCULATION WITH THE NONLIN OPERATOR-----
*----- PROCEDURE PASAPAS -----
*
TAB1 = TABLE ;
TAB1.'BLOCAGES_MECANIQUES' = CL 'ET' R ;
TAB1.'MODELE' = MO;
TAB1.'CHARGEMENT' = CHA1;
TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'TEMPS_CALCULES' = LIS1;

PASAPAS TAB1 ;
*
*----- RESULTS AND OUTPUT-----
*
REA1 = REAC RIG1 (TAB1.DEPLACEMENTS.1) ;
REA2 = REAC RIG1 (TAB1.DEPLACEMENTS.2) ;
*
R11 = EXTR REA1 FX A ;
R21 = EXTR REA2 FX A ;
R12 = EXTR REA1 FX A1 ;
R22 = EXTR REA2 FX A1 ;
R13 = EXTR REA1 FX A2 ;
R23 = EXTR REA2 FX A2 ;
R14 = EXTR REA1 FX A3 ;
R24 = EXTR REA2 FX A3 ;
*
*----- REACTIONS -----
*
M1 = (2.E-3)*((0.050*R11)+(0.0375*R12)+
              (0.025*R13)+(0.0125*R14)) ;
M2 = (2.E-3)*((0.050*R21)+(0.0375*R22)+
              (0.025*R23)+(0.0125*R24)) ;
*
SAUT PAGE ; SAUT 2 LIGN ;
*
MESS ' THEORETICAL MOMENT   COMPUTED MOMENT' ;
SAUT 1 LIGN ;
MESS 'INCREMENT ELASTIQUE
      M1= 805 KN           M1= 'M1 'KN';
SAUT 1 LIGN ;
MESS 'INCREMENT PLASTIQUE
      M2= 1074 KN         M2= 'M2 'KN';
*
TEMPS ;
*
*-----GOOD WORKING MESSAGE-----
*
MR1=805. ;
MR2=1074. ;
RESI1=ABS((M1-MR1)/MR1);
RESI2=ABS((M2-MR2)/MR2);
SI((RESI1 <EG 5E-2) ET (RESI2 <EG 5E-2));
ERRE 0;
SINO;
ERRE 5;
FINSI;
FIN ;

```

Test plas1 (Comments)

1. BOUNDARY CONDITIONS

```
CL1 = BLOQ L12 UX ; CL1B = BLOQ O UY ; CL2 = BLOQ A UX ; CL3 = BLOQ B UX ;
CL = CL1 ET CL1B ET CL2 ET CL3 ;
```

The BLOQUE operator constructs the RIGIDITE stiffness, which is associated to conditions of imposed values on the unknowns of a discretized problem. Later, this stiffness will have to be added to that of the structure.

The non-zero values to be imposed must also be provided in a CHPOINT type object built by means of the DEPIMP operator. For example here :

```
D1 = DEPI CL2 1E-3 ; D2 = DEPI CL3 -1E-3 ; DEP = D1 ET D2 ;
```

Thus, The DEPI operator specifies the value of certain relationships. CL2 and CL3 are the RIGIDITE, type object, of BLOCAGE subtype, defining the conditions imposed on the degrees of freedom . Here 1E-3 and -1E-3 , are the values to be imposed on certain displacement relations.

```
R1 = RELA 8 UX A1 - 7 UX A - 1 UX B ; R2 = RELA 8 UX A2 - 6 UX A - 2 UX B ;
R3 = RELA 8 UX A3 - 5 UX A - 3 UX B ; R4 = RELA 8 UX A4 - 4 UX A - 4 UX B ;
R5 = RELA 8 UX A5 - 3 UX A - 5 UX B ; R6 = RELA 8 UX A6 - 2 UX A - 6 UX B ;
R7 = RELA 8 UX A7 - 1 UX A - 7 UX B ; R = R1 ET R2 ET R3 ET R4 ET R5 ET R6 ET R7 ;
```

The RELATION operator enables the user to construct the stiffness associated with a linear relation between the unknowns named U_i , weighted by (possible) coefficients, for respective mesh nodes (here it is UX A1, UX B , UX A2, UX B,.....). The DEPIMP operator allows the user to impose a non-zero value on the relation. For example here at a point A1 the displacement D is :

$$D = 7/8D1 + 1/8D2$$

2. CALCULATION WITH THE NONLIN OPERATOR PROCEDURE PASAPAS

```
TAB1 = TABLE ; TAB1.'BLOCAGES_MECANIQUES' = CL 'ET' R ;
TAB1.'MODELE' = MO ; TAB1.'CHARGEMENT' = CHA1 ;
TAB1.'CARACTERISTIQUES' = MAT1 ; TAB1.'TEMPS_CALCULES' = LIS1 ;
PASAPAS TAB1 ;
```

The (NONLIN) PASAPAS procedure enables the user to carry out an incremental non linear calculation. The non linearity can result either from the material (plasticity) or from the large displacements, or else from both. Here the non linearity result from the material.

As input, TAB1 is used to define the options and the calculation parameters. The indices of the TAB1 object are words (written in words), their list is precedently specified.

3. RESULTS AND OUTPUT

```
REA1 = REAC RIG1 (TAB1.DEPLACEMENTS.1) ; REA2 = REAC RIG1 (TAB1.DEPLACEMENTS.2) ;
```




3.1. TEST PLAS1 (DESCRIPTION SHEET)

The reaction operator REAC constructs the reaction force from the solution field TAB1.DEPLACEMENT.1 and TAB1.DEPLACEMENTS.2, of CHPOINT types, resulting from a linear system on the first member of which conditions have been imposed. allowing the verification of these conditions.

Thus, for to obtain the bending moments, these reactions are multiplied by their lever arm :

$$M1 = (2.E-3) * ((0.050*R11) + (0.0375*R12) + (0.025*R13) + (0.0125*R14)) ;$$

$$M2 = (2.E-3) * ((0.050*R21) + (0.0375*R22) + (0.025*R23) + (0.0125*R24)) ;$$

3.2 Test plas2 (Description sheet)

Test name: plas2

Calculation type: MECHANIC PLASTICITY 2D AXI

Finite element type: COQ4


Topic: A Spheric-Solid, subjected to an internal pressure. The structure is a Hollow-Spheric-Solid subjected to an internal pressure. The geometry is axisymetrical, the inner radius is $R1 = 1\text{mm}$ and the outer radius is $R2 = 2\text{mm}$. The internal pressure P varies as a function of time: at 0s $P = 100\text{MPa}$ and at 1000s $P = 358.9\text{MPa}$. This non linear evolution (resulting from the material) responds in a plastic manner up to $R = 1.5\text{mm}$. time (T).

Goal: Find the computed radial displacements $U_r(A)_{t0}$ and $U_r(A)_{t1000}$ The theoretical values are :
 $U_r(A)_{t0} = 0.4\mu\text{m}$ $U_r(A)_{t1000} = 2.83\mu\text{m}$

Reference CASTEM: Test NAFEMS : Analytical-Solution : MP2 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



$R1 = 1\text{mm}$
 $R2 = 2\text{mm}$
 $A = (1\text{E-}3, 0)$
 $B = (2\text{E-}3, 0)$
 $C = (0, 2\text{E-}3)$
 $D = (0, 1\text{E-}3)$

$E = 2.1 \cdot 10^{11} \text{ Pa}$
 $\text{NU} = 0.3$
 $\text{SYG} = 300\text{E}6 \text{ Pa}$




Test plas2 (Results)

RESULTS

	$U_r(A)_{t0}$	$U_r(A)_{t1000}$
Reference	$0.4 \mu m$	$2.83 \mu m$
Castem	$0.39821 \mu m$	$1085.2 \mu m$

CASTEM FIGURES



A SOLID-SPHERIC SUBJECTED TO INTERNAL PRESSURE

Test plas2 (Listing CASTEM)

```

TITR 'A SOLID-SPHERIC SUBJECTED TO INTERNAL PRESSURE';
OPTION ECHO 1;
OPTION DIME 2 ELEM QUA4 MODE AXIS ;
TEMPS ;
*
*-----GEOMERTY-----
*
O = 0 0 ; A = 1E-3 0 ;
B = 2E-3 0 ; C = 0 2E-3 ;
D = 0 1E-3 ; X = 1.1E-3 0 ;
Y = 0 1.1E-3 ;
*
*-----MESHING-----
*
L1 = A DROIT 5 X ; L2 = X DROIT 9 B ;
LR = L1 ET L2 ; CE = CERC 16 B O C ;
L3 = C DROIT 9 Y ; L4 = Y DROIT 5 D ;
LZ = L3 ET L4 ; CI = CERC 16 D O A ;
*
SU = LR CE LZ CI DALL PLAN ; TRACE QUAL SU ;
*
*----- BOUNDARY CONDITIONS -----
*
CL1 = BLOQ LR UZ ;
CL2 = BLOQ LZ UR ;
CL = CL1 ET CL2 ;
*
*----- MATERIAL CHARACTERISTICS -----
*
MO = MODE SU MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT QUA4;
*
MAT1 = MATE MO YOUN 2E11 SIGY 300E6 NU 0.3;
*
*-----LOAD DEFINITION INTERNAL PRESSURE-----
*
FO = PRES MASS MO 1E6 CI ;
*
*AT THE STEP OF T-TIME OF THE INTERNAL PRESSURE
*      IS F(T)*FO
*
LI1 = PROG 0. 1000. ;
LI2 = PROG 0. 1000. ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR MECA FO EV ;
*
**-----CALCULATION WITH THE NONLIN OPERATOR -----
*-----PROCEDURE PASAPAS -----
*
LIS1 = PROG 100. 358.9 ;
TAB1 = TABLE ;
TAB1.'BLOCAGES_MECANIQUES' = CL;
TAB1.'MODELE' = MO;
TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'CHARGEMENT' = CHA1;
TAB1.'TEMPS_CALCULES' = LIS1;
*
PASAPAS TAB1 ;
*
*-----RESULTS AND OUTPUT -----
*
RE1 = (TAB1.DEPLACEMENTS.1) ;
RE2 = (TAB1.DEPLACEMENTS.2) ;
*
V1 = EXTR RE1 UR A ;
V1 = 1E6*V1 ;
V2 = EXTR RE2 UR A ;
V2 = 1E6*V2 ;
*
SAUT PAGE ; SAUT 2 LIGN ;
*
MESS ' EN R=1MM ANALYTICAL VALUE COMPUTED VALUE' ;
SAUT 2 LIGN ;
MESS 'ELASTIC INCREMENT UR= 0.40 MICRONS UR= 'V1 'MICRON';
SAUT 1 LIGN ;
MESS 'PLASTIC INCREMENT UR= 2.83 MICRONS UR= 'V2 'MICRON';
TEMPS ;
*----- GOOD WORKING MESSAGE -----
ERR1=100*(ABS(0.4 - V1)/0.4);
ERR2=100*(ABS(2.83 - V2)/2.83);
LIST1=PROG ERR1 ERR2;
ERRMAX=MAXI (PROG ERR1 ERR2);
SI (ERRMAX < 5);
ERRE 0;
SINON;
ERRE 5;
FINSI;
FIN;

```



Test plas2 (Comments)

1. LOAD DEFINITION RESULTING FROM AN INTERNAL PRESSURE

```
FO = PRES MASS MO 1E6 CI ;
```

The PRES operator calculates the forces which are equivalent to a pressure applied on an object. Here this operator is connected to key word MASS, referring to the type of solid element on which the pressure is applied.

MO is the object on which the pressure is applied (MMODEL type) and 1E6 is a pressure algebraic value (FLOTTANT type). CI is the mesh on which the pressure is applied (MAILLAGE type).

For solid elements, such as those used in this problem, the pressure is assumed to be directed towards the inside of the solid. In case of a depression, it is required that a field of a negative pressure points have to be supplied.

```
LI1 = PROG 0. 1000. ;  
LI2 = PROG 0. 1000. ;  
EV = EVOL MANU T LI1 F(T) LI2 ;  
CHA1 = CHAR MECA FO EV ;
```

EV is time description of the loading (EVOLUTION type) function containing in abscissa the times (in chronological order) and in ordinate the values F(T) of the function F for each value of the times T.

Thus, The CHAR operator constructs a CHAR1 object of CHARGEMENT type, of FORCE subtype which contains the space and time description of the equivalent loading.

3.3 Test plas4 (Description sheet)

Test name: plas4

Calculation type: MECHANICS PLASTICITY 3D

Finite element type: CUB8

Topic:


A Cubical solid, subjected to tensile test. The structure is a cubical-Solid composed by two element. It is subjected to a tensile test. For this test, The Y and Z displacement values are fixed at zero value and the X displacement value is imposed on the tips : SU1 and SU2 surfaces.

Goal: Find the computed stress S_{mxx} , S_{myy} and S_{mzz} at 4 time steps (T1, T2, T3 and T4) on the first Gaussian point of the first cubical element. These CASTEM computed values are compared to the theoretical values.

Reference CASTEM: Test NAFEMS : Analytical-Solution : MP4 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



Cubic Solid subjected to Tensile Test Samples
 $U_x = f(t)$, $U_y = U_z = 0$

$P1 = (1\ 0\ 0)$
 $P2 = (1\ 1\ 0)$
 $P3 = (1\ 1\ 1)$
 $P4 = (1\ 0\ 1)$
 $Q1 = (0.5\ 0\ 0)$
 $Q2 = (0.5\ 1\ 0)$
 $Q3 = (0.5\ 1\ 1)$
 $Q4 = (0.5\ 0\ 1)$
 $R1 = (0\ 0\ 0)$
 $R2 = (0\ 1\ 0)$
 $R3 = (0\ 1\ 1)$
 $R4 = (0\ 0\ 1)$

$E = 2.5\ 10^{11}\ Pa$
 $NU = 0.25$
 $SYG = 5E6\ Pa$




Test plas4 (Results)

RESULTS

Evolution Parameter	T1	T2	T3	T4
Reference	$T_{xx} = 7.5E6$ $T_{yy} = 2.5E6$ $T_{zz} = 2.5E6$	$T_{xx} = 11.66E6$ $T_{yy} = 6.66E6$ $T_{zz} = 6.66E6$	$T_{xx} = 4.16E6$ $T_{yy} = 4.16E6$ $T_{zz} = 4.16E6$	$T_{xx} = -3.33E6$ $T_{yy} = 1.66E6$ $T_{zz} = 1.66E6$
Castem	$T_{xx} = 7.5E6$ $T_{yy} = 2.5E6$ $T_{zz} = 2.5E6$	$T_{xx} = 11.66E6$ $T_{yy} = 6.66E6$ $T_{zz} = 6.66E6$	$T_{xx} = 4.16E6$ $T_{yy} = 4.16E6$ $T_{zz} = 4.16E6$	$T_{xx} = -3.33E6$ $T_{yy} = 1.6E6$ $T_{zz} = 1.6E6$

CASTEM FIGURES



CUBE EN TRACTION UNIAXIALE

Test plas4 (Listing CASTEM)

```

TITRE 'CUBIC-SOLID SUBJECTED TO TENSILE TEST';
*
OPTION DIME 3 ECHO 0; TEMPS ; DENS 1;
*-----
*                GEOMETRY
*-----
A = 1.; A1 = -1.; A2 = .5; B = 1.; C = 1.;
P1 = A 0 0; Q1 = A2 0 0; R1 = 0 0 0;
P2 = A B 0; Q2 = A2 B 0; R2 = 0 B 0;
P3 = A B C; Q3 = A2 B C; R3 = 0 B C;
P4 = A 0 C; Q4 = A2 0 C; R4 = 0 0 C;
TR = A1 0 0;
OPTION ELEM SEG2;
LI1 = P1 DROIT 1 P2; LI2 = P2 DROIT 1 P3;
LI3 = P3 DROIT 1 P4; LI4 = P4 DROIT 1 P1;
CO1 = LI1 ET LI2 ET LI3 ET LI4;
LJ1 = Q1 DROIT 1 Q2; LJ2 = Q2 DROIT 1 Q3;
LJ3 = Q3 DROIT 1 Q4; LJ4 = Q4 DROIT 1 Q1;
CO2 = LJ1 ET LJ2 ET LJ3 ET LJ4;
LK1 = R1 DROIT 1 R2; LK2 = R2 DROIT 1 R3;
LK3 = R3 DROIT 1 R4; LK4 = R4 DROIT 1 R1;
CO3 = LK1 ET LK2 ET LK3 ET LK4;
OPTION ELEM QUA4;
SU1 = SURFACE CO1 PLANE; SU2 = SURFACE CO2 PLANE;
SU3 = SURFACE CO3 PLANE; OPTION ELEM CUB8;
VOL1 = SU1 VOLU 1 SU2; VOL2 = SU2 VOLU 1 SU3;
VOL = VOL1 ET VOL2; ELIM .001 VOL;
OEIL = (30 20 20); TRAC QUAL OEIL VOL;
*-----
***** BOUNDARY CONDITIONS
*-----
CL1 = BLOQUE SU3 UX; CL2 = BLOQUE VOL UY UZ;
CL3 = BLOQUE SU2 UX; CL4 = BLOQUE SU1 UX;
CL = CL1 ET CL2 ET CL3 ET CL4;
*-----
* DISPLACEMENT VALUE IMPOSED
*-----
EPS0 = .25E-4; COEF1 = A * EPS0 * .5;
COEF2 = A * EPS0; D1 = DEPI CL3 COEF1;
D2 = DEPI CL4 COEF2; DEP = D1 ET D2;
*-----
* MODEL AND CHARACTERISTICS
*-----
MO = MODL VOL MECANIQUE ELASTIQUE
    PLASTIQUE PARFAIT CUB8;
MAT1 = MATR MO YOUN 2.5E11 SIGY 5.0E6 NU .25;
*-----
* NON LINEAR CALCULATION PROCEDURE
*-----
LP1 = PROG 0. 1. 2. 3. 4.;
LP2 = PROG 0. 1. 2. 1. 0.;
EV = EVOL MANU T LP1 F(T) LP2;
CHA1 = CHAR DIMP DEP EV;
LIS1 = PROG 1. 2. 3. 4.;
TAB1 = TABLE;
TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'CHARGEMENT' = CHA1;
TAB1.'MODELE' = MO;
TAB1.'TEMPS_CALCULES' = LIS1;
TAB1.'BLOCAGES_MECANIQUES' = CL;
PASAPAS TAB1;
*-----
*                RESULTS
*-----
SI1 = TAB1.CONTRAINTE.1; SI2 = TAB1.CONTRAINTE.2;
SI3 = TAB1.CONTRAINTE.3; SI4 = TAB1.CONTRAINTE.4;
SXX1 = EXTR SI1 SMXX 1 1 1;
SYY1 = EXTR SI1 SMYY 1 1 1;
SZZ1 = EXTR SI1 SMZZ 1 1 1;
SXX2 = EXTR SI2 SMXX 1 1 1;
SYY2 = EXTR SI2 SMYY 1 1 1;
SZZ2 = EXTR SI2 SMZZ 1 1 1;
SXX3 = EXTR SI3 SMXX 1 1 1;
SYY3 = EXTR SI3 SMYY 1 1 1;
SZZ3 = EXTR SI3 SMZZ 1 1 1;
SXX4 = EXTR SI4 SMXX 1 1 1;
SYY4 = EXTR SI4 SMYY 1 1 1;
SZZ4 = EXTR SI4 SMZZ 1 1 1;
*-----
*                RESULTS PLACARDING
*-----
SAUT PAGE;
*///// PARAMETER OF EVOLUTION T : 1 ///////////////
*-----
MESS '                T = 1 ' ;
MESS ' THEORETICAL VALUE : SIGMAXX = 7.5 E 6 ' ;
MESS '                SIGMAYY = 2.5 E 6 ' ;
MESS '                SIGMAZZ = 2.5 E 6 ' ;
MESS '                ' ;
MESS ' COMPUTED VALUE : SIGMAXX = 'SXX1 ;
MESS '                SIGMAYY = 'SYY1 ;
MESS '                SIGMAZZ = 'SZZ1 ;
SAUT 1 LIGNE ;
*///// PARAMETER OF EVOLUTION T : 2 ///////////////
*-----
MESS '                T = 2 ' ;
MESS ' THEORETICAL VALUE : SIGMAXX = 11.66 E 6 ' ;
MESS '                SIGMAYY = 6.66 E 6 ' ;
MESS '                SIGMAZZ = 6.66 E 6 ' ;
MESS '                ' ;
MESS ' COMPUTED VALUE : SIGMAXX = 'SXX2 ;
MESS '                SIGMAYY = 'SYY2 ;
MESS '                SIGMAZZ = 'SZZ2 ;
SAUT 1 LIGNE ;
*///// PARAMETER OF EVOLUTION T : 3 ///////////////
*-----
MESS '                T = 3 ' ;
MESS ' THEORETICAL VALUE : SIGMAXX = 4.16 E 6 ' ;
MESS '                SIGMAYY = 4.16 E 6 ' ;
MESS '                SIGMAZZ = 4.16 E 6 ' ;
MESS '                ' ;
MESS ' COMPUTED VALUE : SIGMAXX = 'SXX3 ;
MESS '                SIGMAYY = 'SYY3 ;
MESS '                SIGMAZZ = 'SZZ3 ;
SAUT 1 LIGNE ;
*///// PARAMETER OF EVOLUTION T : 4 ///////////////
*-----
MESS '                T = 4 ' ;
MESS ' THEORETICAL VALUE : SIGMAXX = -3.33 E 6 ' ;
MESS '                SIGMAYY = 1.66 E 6 ' ;
MESS '                SIGMAZZ = 1.66 E 6 ' ;
MESS '                ' ;
MESS ' COMPUTED VALUE : SIGMAXX = 'SXX4 ;
MESS '                SIGMAYY = 'SYY4 ;
MESS '                SIGMAZZ = 'SZZ4 ;
MESS '                ' ;
TEMPS ;
*                GOOD MESSAGE WORKING
SREF=1.66E6;RESI=ABS((SYY4-SREF)/SREF);
SI(RESI <EG 5E-2);
ERRE 0; SINO; ERRE 5;
FINSI;
FIN;
    
```




3.4 Test plas5 (Description sheet)

Test name: plas5

Calculation type: MECHANICS PLASTICITY 2D PLANE STRESSE

Finite element type: QUA4


Topic: A solid plate, subjected to imposed displacement. The structure is a solid plate, it is subjected to imposed displacement from two sides and blocked from the two others (see figure). For this test, the goal is to study the DRUCKER-PRAGER elastic-plastic behaviour.

Goal: Find the computed stress S_{mxx} , and S_{myy} on the first Gaussian point of the first finite element. These CASTEM computed values are compared to the theoretical values.

Reference CASTEM: Test NAFEMS : Analytical-Solution : MP4 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version


Model description:



$$\begin{aligned} E &= 2.5 \cdot 10^{11} \text{ Pa} \\ \text{NU} &= 0.25 \\ \text{SYG} &= 5E6 \text{ Pa} \end{aligned}$$

Test plas5 (Results)
RESULTS

DRUCKER-PRAGER STRESSES	S_{mx}	S_{my}
Reference	1.8717Pa	1.8717Pa
Castem	1.8717Pa	1.8717Pa

CASTEM FIGURES




3.4. TEST PLAS5 (DESCRIPTION SHEET)

Test plas5 (Listing CASTEM)

```
TITRE 'A PLATE WITH CROSS SECTIONAL SUBJECTED
      TO BI-TENSILE TEST ' ;
OPTION ECHO 0 ;
OPTION DIME 2 ELEM QUA4 MODE PLAN CONT ;
TEMPS ;
*-----
*          GEOMETRY AND MESHING
*-----
A = 0. 0. ;   B = 1.5 0. ;
C = 1.5 1.5 ; D = 0. 1.5 ;
*
L1 = A DROIT 1 B ; L2 = B DROIT 1 C ;
L3 = C DROIT 1 D ; L4 = D DROIT 1 A ;
LL = L1 ET L2 ET L3 ET L4 ;
SU = SURF LL PLAN ;
TRAC QUAL SU ;
*-----
*          BOUNDARY CONDITIONS
*          AND IMPOSED DISPLACEMENTS
*-----
CL4 = BLOQ L4 UX ; CL2 = BLOQ L2 UX ;
CL3 = BLOQ L3 UY ; CL1 = BLOQ L1 UY ;
CL = CL1 ET CL2 ET CL3 ET CL4 ;
*
D3 = DEPI CL3 1.5E-4 ;
D2 = DEPI CL2 1.5E-4 ;
DEP = D2 ET D3 ;
*-----
*          MODEL AND CHARACTERISTICS
*-----
MO = MODE SU MECANIQUE ELASTIQUE PLASTIQUE
    DRUCKER_PRAGER QUA4 ;

KLL= 3.56 ;   ETT= 0.441; MUU= 1.02 ;
MAT1 = MATE MO YOUN 40000 NU 0.21
ALFA 0.441 BETA 1.02 K 4.1 GAMM 0.441
DELT 1.02 ETA ETT MU MUU
KL KLL H 0. ;
*
*-----DISPLACEMENT VALUE IMPOSED-----
*--AT THE T TIME THE DISPLACEMENT IS F(T)*D-----
*
LI1 = PROG 0. 1. 2. 3. 4. 5. ;
LI2 = PROG 0. 1. 2. 3. 4. 5. ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR DIMP DEP EV ;
*-----
*          NON LINEAR CALCULATION PASAPAS PROCEDURE
*-----
LIS1 = PROG 0. 1. 2.;
TAB1 = TABLE ;
TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'CHARGEMENT' = CHA1;
TAB1.'MODELE' = MO;
TAB1.'TEMPS_CALCULES' = LIS1;
TAB1.'BLOCAGES_MECANIQUES' = CL;
*
PASAPAS TAB1;
*
*          FIND THE STRESSES WITH PECHE PROCEDURE
*
SIG1 = PECHE TAB1 CONTRAINTES 1.;
SIG2 = PECHE TAB1 CONTRAINTES 2.;
*
SXX = EXTR SIG2 'SMXX' 1 1 1 ;
syy = EXTR sig2 'SMYY' 1 1 1 ;
*
* THEORETICAL VALUE CALCULATION
*
VALTH = KLL / ( ( 2. * ETT ) + MUU );
SAUT PAGE ; SAUT 2 LIGN ;
*
MESS ' THEORETICAL VALUE          COMPUTED VALUE ' ;
SAUT 1 LIGN ;
MESS ' SIG = ' VALTH ' SMXX=' SXX ' SMYY=' syy ;
*
TEMPS ;
*          GOOD MESSAGE WORKING
RESI1=ABS((SXX-VALTH)/VALTH);
RESI2=ABS((SYY-VALTH)/VALTH);
SI((RESI1 <EG 5E-2) ET (RESI2 <EG 5E-2));
    ERRE 0;
SINO;
    ERRE 5;
FINSI;
FIN;
```

Test plas5 (Comments)

1. MODEL AND CHARACTERISTICS

```

MO = MODE SU MECANIQUE ELASTIQUE PLASTIQUE
      DRUCKER_PRAGER QUA4 ;
KLL= 3.56 ; ETT= 0.441; MUU= 1.02 ;
MAT1 = MATE MO YOUN 40000 NU 0.21
ALFA 0.441 BETA 1.02 K 4.1 GAMM 0.441
DELT 1.02 ETA ETT MU MUU
KL KLL H 0. ;
    
```

The DRUCKER PRAGER model enables the user to perform a calculation with non-associated yield and with stress hardening. The equations of this model are as follows :

initial criterion :

$$ALFA * Tr(S) + BETA * Seq = K$$

ultime criterion :

$$ETA * Tr(S) + MU * Seq = KL$$

hardening :

$$dK = H * dp \quad (H \text{ in algebraic value})$$

Flow potential :

$$GAMM * Tr(S) + DELT * Seq$$

For this model the parameters to define are : ALFA, BETA, K, ETA, MU, KL, H, GAMM, and DE.

S is stress tensor
 Seq is Von Mises equivalent stress
 p is cumulated equivalent plastic strain

2. FIND THE STRESSES WITH PECHE PROCEDURE

```

SIG1 = PECHE TAB1 CONTRAINTES 1.;
SIG2 = PECHE TAB1 CONTRAINTES 2.;
    
```

The PECHE procedure enables the user to retrieve for a given time the results of a calculation carried out using the PASAPAS procedure. Contents :

TAB1 : table used in PASAPAS (TABLE type)

CONTRAINTES : key word (MOT type) corresponding to the requested stress index

1. (or 2.) : time (FLOTTANT type) for which the results (stresses) are requested.

By default, the results of the last calculated time will be retrieved.

SIG1 (or SIG2) : generated stress field (CHPOINT type)



3.5 Test plas6 (Description sheet)

Test name: plas6

Calculation type: MECHANICS ELASTIC PLASTIC - PLANE STRESS

Finite element type: QUA4

Topic: Rectangle subjected to pure bending. A rectangle is embedded from two adjacent sides. Linear imposed displacements are applied to one side. Calculations are performed for plane stresses and with a pure plastic model.

Goal; Calculate the pure bending moment along the side where the displacements are applied.


Reference CASTEM: R. Hill, Mathematical theory of plasticity, Oxford, Oxford Clavendon Press, 1950.
Test V.P.C.S. numero SSNP11/89

Version: 97' customer version

Model description:


Elastic and pure plastic material

YOUNG = $2 \cdot 10^{11}$ Pa
NU = 0.3
SIGY = 483 MPa




Geometry of the rectangle
thickness: $a = 12.5$ mm
height: $b = 100.0$ mm

Test plas6 (Results) CASTEM FIGURES





PLAS6: MESHING



PLAS6: STRESS (MXX)

VAL = S00
 7 -6.376E+08
 6 6.376E+08
 4 -6.376E+08
 3 -5.68E+08
 2 -5.68E+08
 1 -5.68E+08
 10 -3.29E+08
 9 -3.29E+08
 8 -2.69E+08
 7 -2.69E+08
 6 -1.49E+08
 5 -8.9E+07
 4 -8.9E+07
 3 2.69E+07
 2 8.9E+07
 1 1.49E+08
 10 2.69E+08
 9 3.29E+08
 8 3.29E+08
 7 4.48E+08
 6 5.68E+08
 5 5.68E+08
 4 6.376E+08

AMPLITUDE
 0.00E+00
 2.0





3.5. TEST PLAS6 (DESCRIPTION SHEET)

Test plas6 (Listing CASTEM)

```

TITR 'RECTANGLE SUBJECTED TO BENDING' ;
OPTI ECHO 1 ;
OPTI DIME 2 MODE PLAN CONT ELEM QUA4 ;
OPTI ISOV LIGNE ;
TEMPS ;
*****
*          GEOMETRY AND MESHING          *
*****
PA = 12.5E-3  50.E-3; PB = 12.5E-3 -50.E-3;
PC = 0.      -50.E-3; PD = 0.      50.E-3;
NB1 = 32 ; NB2 = 4 ; D1 = PA DROI NB1 PB ;
D2 = PB DROI NB2 PC ; D3 = PC DROI NB1 PD ;
D4 = PD DROI NB2 PA ;
SUR1 = DALL D1 D2 D3 D4 PLAN ;
TRAC QUAL SUR1 'TITR' 'plas6 : MESHING' ;
NBNO1 = NBNO SUR1 ; NBEL1 = NBEL SUR1 ;
*****
*          MODEL AND MATERIAL CHARACTERISTICS          *
*****
MOD1 = MODE SUR1 MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT QUA4 ;
MAT1 = MATE MOD1 YOUN 2.1E11 NU 0.3
      SIGY 483.E6 ;
*****
*          BOUNDARY CONDITIONS          *
*****
CL1 = BLOQ UX D3 ;
CL2 = BLOQ UY (SUR1 POIN PROC (0. 0.)) ;
CL3 = BLOQ UX D1; CL = CL1 ET CL2 ET CL3;
* Creation of the chpoint of variable
* imposed displacement: imposed
* displacement with a linear variation
* from u to -u, between A and B; u
* increases up to 0.5 mm
I = 0. ;
LISD1 = PROG 1. ;
REPETER BLOC10 NB1 ;
  I = I + 1. ;
  YY = 1. - (2. * (I / NB1)) ;
  LISD1 = LISD1 ET (PROG YY) ;
FIN BLOC10 ;

LISD1 = LISD1 * 1.E-3 ;
CHPO1 = MANU CHPO D1 1 UX LISD1 ;
CHAL = DEPI CL3 CHPO1;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
TAB1 = TABLE;
* Take the displacement u as a list of
* pseudo-time
LREEL1 = PROG 0.02875 0.05 0.1 0.15 0.2
        0.3 0.4 0.5 1.0;

* LOAD DEFINITION
EVOL1 = EVOL MANU UA LREEL1 COEFF LREEL1;
CHAR1 = CHAR DIMP CHAL EVOL1;

TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'CHARGEMENT' = CHAR1;
TAB1.'MODELE' = MOD1;
TAB1.'TEMPS_CALCULES' = LREEL1;
TAB1.'BLOCAGES_MECANIQUES' = CL;
PASAPAS TAB1;
*****
* RESULTS          *
*****

* SETUP
MOM1 = TABLE;
LISD2 = LISD1 * 50.;
MOM2 = TABLE;
ERG = TABLE;

MOM2.1 = 805.0; MOM2.2 = 1074. ;
MOM2.3 = 1174.; MOM2.4 = 1193.;
MOM2.5 = 1199.; MOM2.6 = 1204.;
MOM2.7 = 1205.; MOM2.8 = 1206.;
MOM2.9 = 1207.5;

* Loop for each step of load: function of UA
I = 0;
REPETER BLOC1 9;
  I = I + 1;
  CONT1 = TAB1.CONTRAINTE.S.I;
  DEPL1 = TAB1.DEPLACEMENTS.I;
  CHPO10 = CHAN CHPO MOD1 CONT1;

* Drawing of the stress distribution SMXX
SI (I EGA 9);
  TITR 'plas6: STRESS SMXX';
  BB = EXCO 'SMXX' CONT1;
  TRAC MOD1 BB SUR1;
FINSI;

* Drawing of the deformed (checking of load)
SI (I EGA 9);
  TITR 'plas6: DEFORMED';
  DEFO = DEFO SUR1 DEPL1 0.;
  DEF1 = DEFO SUR1 DEPL1 2. VERT;
  TRACE SUR1 (DEFO ET DEF1);
FINSI;

* Calculation of the bending moment
* (stress SMXX along AB)
LIST2 = PROG;
J = 0;
NB = NB1 + 1;
REPETER BLOC4 NB;
  J = J + 1;
  D = EXTR LISD2 J;
  POI1 = SUR1 POIN PROC (12.5E-3 D);
  SM = EXTRA CHPO10 SMXX POI1;
  LIST2 = LIST2 ET (PROG SM);
FIN BLOC4;

LIST3 = LIST2 * LISD2;
TITR 'STRESS SMXX';
ESMXX = EVOL MANU Y LISD2 SMXX LIST2;
TITR 'PRODUCT Y * SMXX';
EYSMXX = EVOL MANU Y LISD2 YSMXX LIST3;

SI (I EGA 9);
  DESS ESMXX;
  DESS EYSMXX;
FINSI;
MOM1.I = SOMME EYSMXX;
MOM1.I = EXTR MOM1.I 1;
MOM1.I = -1.E-3 * MOM1.I;
* Difference between theory and calculation
ERG.I = 100. * (ABS ((MOM1.I - MOM2.I)
                    / MOM2.I));

FIN BLOC1;
*****
* RESULTS PLACARDING          *
*****

```

```

MESS 'RESULTS : plas6';
MESS '-----';
SAUT LIGN;

MESS ' Bending moment (stress SMXX along AB';
MESS ' -----';
SAUT LIGN;

MESS ' UAmx = 0.02875 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.1 'kN.m';
MESS ' Calculated moment : ' MOM2.1 'kN.m';
MESS ' Error : ' ERG.1 '%';
SAUT LIGN;

MESS ' UAmx = 0.05 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.2 'N.m';
MESS ' Calculated moment : ' MOM2.2 'N.m';
MESS ' Error : ' ERG.2 '%';
SAUT LIGN;

MESS ' UAmx = 0.1 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.3 'N.m';
MESS ' Calculated moment : ' MOM2.3 'N.m';
MESS ' Error : ' ERG.3 '%';
SAUT LIGN;

MESS ' UAmx = 0.15 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.4 'N.m';
MESS ' Calculated moment : ' MOM2.4 'N.m';
MESS ' Error : ' ERG.4 '%';
SAUT LIGN;

MESS ' UAmx = 0.2 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.5 'N.m';
MESS ' Calculated moment : ' MOM2.5 'N.m';
MESS ' Error : ' ERG.5 '%';
SAUT LIGN;

MESS ' UAmx = 0.3 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.6 'N.m';
MESS ' Calculated moment : ' MOM2.6 'N.m';
MESS ' Error : ' ERG.6 '%';
SAUT LIGN;

MESS ' UAmx = 0.4 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.7 'N.m';
MESS ' Calculated moment : ' MOM2.7 'N.m';
MESS ' Error : ' ERG.7 '%';
SAUT LIGN;

MESS ' UAmx = 0.5 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.8 'N.m';
MESS ' Calculated moment : ' MOM2.8 'N.m';
MESS ' Error : ' ERG.8 '%';

SAUT LIGN;

MESS ' UAmx = 1.0 mm :';
MESS ' -----';
MESS ' Theoretical moment : ' MOM1.9 'N.m';
MESS ' Calculated moment : ' MOM2.9 'N.m';
MESS ' Error : ' ERG.9 '%';
SAUT LIGN;

MESS 'Detail on the mesh';
MESS '-----';
MESS ' Number of elements : ' NBEL1 'QUA4';
MESS ' Number de nodes : ' NBN01 ;
SAUT LIGN;
*****
* GOOD WORKING MESSAGE *
*****
ERGMX = MAXI (PROG ERG.1 ERG.2 ERG.3 ERG.4 ERG.5 ERG.6
ERG.7 ERG.8 ERG.9);

SI (ERGMX <EG 5.);
ERRE 0;
SINON;
ERRE 5;
FINSI;
SAUT LIGN;

TEMPS;

FIN;
    
```




3.6 Test plas7 (Description sheet)

Test name: plas7

Calculation type: MECHANICS ELASTIC PLASTIC - PLANE STRAIN

Finite element type: QUA44

Topic: Rectangle subjected to pure bending. A rectangle is embedded from two adjacent sides. Linear imposed displacements are applied to one side. Calculations are made for plan strains and with a pure plastic model.

Goal: Calculate the pure bending moment along the side where the displacements are applied.


Reference CASTEM: R. Hill, Mathematical theory of plasticity, Oxford, Oxford Clavendon Press, 1950.
Test V.P.C.S. numero SSNP11/89

Version: 97' customer version




Model description:

Elastic and pure plastic material

$$\begin{aligned} \text{YOUNG} &= 2 \cdot 10^{11} \text{ Pa} \\ \text{NU} &= 0.3 \\ \text{SIGY} &= 483 \text{ MPa} \end{aligned}$$





Test plas7 (Results) CASTEM FIGURES



VAL = SDO
 -2.11E+09
 -1.01E+09
 A -9.96E+08
 D -9.01E+08
 G -8.06E+08
 F -7.11E+08
 H -6.16E+08
 P -5.22E+08
 S -4.27E+08
 Y -3.32E+08
 T -2.37E+08
 B -1.42E+08
 -5.0E+07
 N 4.74E+07
 K 1.42E+08
 R 2.37E+08
 Q 3.32E+08
 L 4.27E+08
 W 5.22E+08
 Z 6.16E+08
 2 7.11E+08
 5 8.06E+08
 6 9.01E+08
 8 9.96E+08

AMPLITUDE
 0.00E+00
 2.0





3.6. TEST PLAS7 (DESCRIPTION SHEET)

Test plas7 (Listing CASTEM)

```

TITR 'RECTANGLE SUBJECTED TO BENDING' ;
OPTI ECHO 1; OPTI DIME 2;
OPTI MODE PLAN DEFO; OPTI ELEM QUA4;
OPTI ISOV LIGNE;
TEMPS;
*****
*          GEOMERTY AND MESHING          *
*****
PA = 12.5E-3  50.E-3; PB = 12.5E-3 -50.E-3;
PC = 0.      -50.E-3; PD = 0.      50.E-3;
NB1 = 32; NB2 = 4;
D1 = PA DROI NB1 PB; D2 = PB DROI NB2 PC;
D3 = PC DROI NB1 PD; D4 = PD DROI NB2 PA;
SUR1 = DALL D1 D2 D3 D4 PLAN;
TITR 'plas7: MESH';
TRAC QUAL SUR1;
NBNO1 = NBNO SUR1; NBEL1 = NBEL SUR1;
*****
*          MODEL AND MATERIAL CHARACTERISTICS          *
*****
MOD1 = MODE SUR1 MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT QUA4;
MAT1 = MATE MOD1 YOUN 2.1E11 NU 0.3
      SIGY 483.E6;
*****
*          BOUNDARY CONDITIONS          *
*****
CL1 = BLOQ UX D3;
CL2 = BLOQ UY (SUR1 POIN PROC (0. 0.));
CL3 = BLOQ UX D1;
CL  = CL1 ET CL2 ET CL3;
* Creation of the variable imposed chpoint
* of displacement: imposed displacement
* with a linear variation from u to -u,
* between A and B; u increases up to 0.5 mm
I = 0.;
LISD1 = PROG 1.;
REPETER BLOC10 NB1;
  I = I + 1.;
  YY = 1. - (2. * (I / NB1));
  LISD1 = LISD1 ET (PROG YY);
FIN BLOC10;

LISD1 = LISD1 * 1.E-3;
CHPO1 = MANU CHPO D1 1 UX LISD1;
CHA1 = DEPI CL3 CHPO1;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
TAB1 = TABLE;
* Take the displacement u as a list of
* pseudo-time
LREEL1 = PROG 0.029435 0.1 0.2 0.3 0.4 0.5 1.0;
* LOAD DEFINITION
EVOL1 = EVOL MANU UA LREEL1 COEFF LREEL1;
CHAR1 = CHAR DIMP CHA1 EVOL1;

TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'CHARGEMENT' = CHAR1;
TAB1.'MODELE' = MOD1;
TAB1.'TEMPS_CALCULES' = LREEL1;
TAB1.'BLOCAGES_MECANIQUES' = CL;
PASAPAS TAB1;
*****
*          RESULTS          *
*****
* SETUP
MOM1 = TABLE;
LISD2 = LISD1 * 50.;

MOM2 = TABLE;
ERG = TABLE;
MOM2.1 = 90.57; MOM2.2 = 135.2;
MOM2.3 = 138.4; MOM2.4 = 139.0;
MOM2.5 = 139.2; MOM2.6 = 139.3;
MOM2.7 = 139.4; MOM2.8 = 543.4;
MOM2.9 = 557.7;
* Loop for each step of load: function of UA
I = 0;
REPETER BLOC1 7;
  I = I + 1;
  CONT1 = TAB1.CONTRAINTE.S.I;
  DEPL1 = TAB1.DEPLACEMENTS.I;
  CHPO10 = CHAN CHPO MOD1 CONT1;
* Drawing of the stress distribution SMXX
SI (I EGA 7);
  TITR 'plas7: STRESS SMXX';
  BB = EXCO 'SMXX' CONT1;
  TRAC MOD1 BB SUR1;
FINSI;
* Drawing of the deformed (checking of load)
SI (I EGA 7);
  TITR 'plas7: DEFORMED';
  DEFO = DEFO SUR1 DEPL1 0.;
  DEF1 = DEFO SUR1 DEPL1 2. VERT;
  TRACE SUR1 (DEFO ET DEF1);
FINSI;
* Calculation of the bending moment
* (stress SMXX along AB)
LIST2 = PROG;
J = 0;
NB = NB1 + 1;
REPETER BLOC4 NB;
  J = J + 1;
  D = EXTR LISD2 J;
  POI1 = SUR1 POIN PROC (12.5E-3 D);
  SM = EXTRA CHPO10 SMXX POI1;
  LIST2 = LIST2 ET (PROG SM);
FIN BLOC4;
LIST3 = LIST2 * LISD2;
TITR 'SRESS SMXX';
ESMXX = EVOL MANU Y LISD2 SMXX LIST2;
TITR 'PRODUCT Y * SMXX';
EYSMXX = EVOL MANU Y LISD2 YSMXX LIST3;
SI (I EGA 7);
  DESS ESMXX;
  DESS EYSMXX;
FINSI;
MOM1.I = SOMME EYSMXX;
MOM1.I = EXTR MOM1.I 1;
MOM1.I = -1.E-4 * MOM1.I;
* Difference between theory and calculation
ERG.I = 100. * (ABS ((MOM1.I - MOM2.I) /
  MOM2.I));
* Extraction of the maximum stress SMXX for
* steps 1 and 7
SI (I EGA 1);
MOM1.8 = EXTR CHPO10 SMXX PA;
MOM1.8 = MOM1.8 / 1.E6;
ERG.8 = 100. * (ABS ((MOM1.8 - MOM2.8) /
  MOM2.8));
FINSI;
SI (I EGA 7);
MOM1.9 = EXTR CHPO10 SMXX (D1 POIN 2);
MOM1.9 = MOM1.9 / 1.E6;
ERG.9 = 100. * (ABS ((MOM1.9 - MOM2.9) /
  MOM2.9));
FINSI;
FIN BLOC1;

```

```

*****
* RESULTS PLACARDING *
*****
MESS 'RESULTS : plas7';
MESS '-----';
SAUT LIGN;

MESS ' Bending moment (stress SMXX along AB';
MESS '-----';
SAUT LIGN;

MESS ' UAmix = 0.029435 mm :';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.1 'kN.m';
MESS ' Calculated moment : ' MOM1.1 'kN.m';
MESS ' Error : ' ERG.1 '%';
SAUT LIGN;

MESS ' UAmix = 0.1 mm :';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.2 'kN.m';
MESS ' Calculated moment : ' MOM1.2 'kN.m';
MESS ' Error : ' ERG.2 '%';
SAUT LIGN;

MESS ' UAmix = 0.2 mm :';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.3 'kN.m';
MESS ' Calculated moment : ' MOM1.3 'kN.m';
MESS ' Error : ' ERG.3 '%';
SAUT LIGN;

MESS ' UAmix = 0.3 mm :';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.4 'kN.m';
MESS ' Calculated moment : ' MOM1.4 'kN.m';
MESS ' Error : ' ERG.4 '%';
SAUT LIGN;

MESS ' UAmix = 0.4 mm :';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.5 'kN.m';
MESS ' Calculated moment : ' MOM1.5 'kN.m';
MESS ' Error : ' ERG.5 '%';
SAUT LIGN;

MESS ' UAmix = 0.5 mm :';
MESS '-----';

MESS ' Theoretical moment : ' MOM2.6 'kN.m';
MESS ' Calculated moment : ' MOM1.6 'kN.m';
MESS ' Error : ' ERG.6 '%';
SAUT LIGN;

MESS ' UAmix = 1.0 mm :';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.7 'kN.m';
MESS ' Calculated moment : ' MOM1.7 'kN.m';
MESS ' Error : ' ERG.7 '%';
SAUT LIGN;

MESS 'Maximum stress SMXX :';
MESS '----- :';
SAUT LIGN;
MESS ' UAmix = 0.029435 mm';
MESS '-----';
MESS ' Theoretical moment : ' MOM2.8 'MPa';
MESS ' Calculated moment : ' MOM1.8 'MPa';
MESS ' Error : ' ERG.8 '%';
SAUT LIGN;
MESS ' UAmix = 1.0 mm';

MESS '-----';
MESS ' Theoretical moment : ' MOM2.9 'MPa';
MESS ' Calculated moment : ' MOM1.9 'MPa';
MESS ' Error : ' ERG.9 '%';
SAUT LIGN;

MESS 'Detail on the mesh';
MESS '-----';
MESS ' Number of elements : ' NBEL1 'QUA4';
MESS ' Number de nodes : ' NBN01 ;
SAUT LIGN;
*****
* GOOD WORKING MESSAGE *
*****
ERGMIX = MAXI (PROG ERG.1 ERG.2 ERG.3 ERG.4
ERG.5 ERG.6 ERG.7 ERG.8 ERG.9);

SI (ERGMIX <EG 5.);
ERRE 0;
SINON;
ERRE 5;
FINSI;
SAUT LIGN;

TEMPS;

FIN;
    
```



3.7 Test plas8 (Description sheet)

Test name: plas8

Calculation type: MECHANICS ELASTIC PLASTIC

Finite element type: TRI3

Topic: Punctual strength as a function of the deflection. A beam is embedded at one side and at the free side we impose a vertical displacement (e.g. a deflection).

Goal: Calculate the evolution of the reaction strength as a function of the deflection. According to the pure elastoplastic law the strength should converge to an asymptotic value given by the following relation:

$$F_U = (SIGY * H^2 / 4) / L$$

Where:

SIGY = stress at the yield point

H = thickness of the beam

L = length of the beam

Version: 97' customer version


Model description:

Elastic and pure plastic material

YOUNG = $2.1 \cdot 10^{11}$ Pa



NU = 0.

SIGY = 100. MPa



Test plas8 (Results)

CASTEM FIGURES





3.7. TEST PLAS8 (DESCRIPTION SHEET)

Test plas8 (Listing CASTEM)

```
TITRE 'PUNCTUAL STRENGTH AS A FUNCTION OF DEFLECTION';
COMPLET = FAUX; COMM * full calculation if vrai
OPTI ECHO 0;
OPTI DIME 3 ELEM TRI3 ECHO 0 DENS 0.5 ;
*****
*          GEOMERTY AND MESHING          *
*****
B= 1. ; L=10. ;
A1 = 0. 0. 0. ;A2=0. B 0.0;
B1 = L 0. 0. ; B2 = L B 0.0 ;
NEL = 20;
A2A1 =DROI 2 A2 A1 ; A1B1 =DROI NEL A1 B1 ;
B1B2 =DROI 2 B1 B2 ; B2A2 =DROI NEL B2 A2 ;
ST = DALL A2A1 A1B1 B1B2 B2A2 PLAN ;
OEIL = 0 0 1000 ;
TRACE OEIL ST ;
*****
*          SETUP          *
*****
XABS = PROG ;
FONC1 = PROG;
EP = L/10 ;
*****
*          MODEL AND MATERIAL CHARACTERISTICS          *
*****
MOD1 = MODE ST MECANIQUE ELASTIQUE PLASTIQUE
      PARFAIT DKT 'INTEGRE' 7 ;
E = 2.1E11 ; SY = 100.E6;
MAT1 = MATE MOD1 YOUN E NU 0. SIGY SY
      EPAI EP ALFA 1. ;
*****
*          BOUNDARY CONDITIONS AND RIGIDITY          *
*****
R11=RIGI MOD1 MAT1 ;
CL1=BLOQ UX UY UZ RY RZ A2A1 ;
CLDEP=BLOQ UZ B1B2 ;
CL = CL1 ET CLDEP;
RIT1 = R11 ET CL;
*****
*          IMPOSED DISPLACEMENT          *
*****
FDEP = DEPI CLDEP 0.2;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
LI1 = PROG 0. 1. ;
LI2 = PROG 0. 1. ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR DIMP FDEP EV ;

SI COMPLET;
  LIS1 = PROG PAS 0.03333 0.1
        PAS 0.02 0.3 PAS 0.01 0.4 ;
SINON;
  LIS1 = PROG PAS 0.03333 0.1 0.12
        0.14 0.16 ;

FINSI;
TAB1 = TABLE ;
TAB1.'CARACTERISTIQUES' = MAT1;
TAB1.'CHARGEMENT' = CHA1;
TAB1.'MODELE' = MOD1;
TAB1.'TEMPS_CALCULES' = LIS1;
TAB1.'BLOCAGES_MECANIQUES' = CL;

PASAPAS TAB1;
TEMPS ;

*****

* RESULTS *
*****
ITER=0 ;
OPTI ECHO 1;
NB=DIME LIS1 ;
XABS=INSER XABS 1 (0.0) ;
FONC1=INSER FONC1 1 (0.0);
REPETE BOUC1 NB ;
ITER=ITER+1 ;
ITER1=ITER+1 ;
TT=EXTR LIS1 ITER ;
DEP=PECHE TAB1 'DEPLACEMENTS' TT ;
FLECH=EXTR DEP UZ B1;
FF=REAC DEP RIT1 ;
FFF=RESUL(REDU FF B1B2) ;
FORC=EXTR FFF 'FZ' B1 ;
XABS=INSER XABS ITER1 FLECH ;
FONC1=INSER FONC1 ITER1 FORC;
FIN BOUC1 ;

* Drawing of curve of strength as a
* function of deflection
EV1=EVOL ROUGE MANU 'DEFLECTION' XABS
      'STRENGTH' FONC1 ;
DESSIN EV1 ;
TEMPS ;
*****
* ANALYTIC RESULT OF THE ASYMPTOTIC VALUE*
*****
SI COMPLET;
  VAL1=SY*(EP**2.) ;FU=(VAL1/4)/L ;
SINON;
  FU = 1.679E+06;
FINSI;
*****
* GOOD WORKING MESSAGE *
*****
FC=RESUL(REDU FF B1B2) ;
FCC=EXTR FC 'FZ' B1 ;
RESI=ABS((FCC-FU)/FU) ;
SI (RESI <EG 5E-2);
  ERRE 0 ;
SINO;
  ERRE 5 ;
FINSI;
FIN ;
```

3.8 Test plas9 (Description sheet)

Test name: plas9

Calculation type: MECHANICS ELASTIC PLASTIC - DRUCKER-PRAGER'S MATERIAL

Finite element type: CUB8

Topic: Unidirectionnal tensile-test on a parallelipiped.


An isostatic embedded parallelipiped is subjected to an imposed displacement on one of its faces. It follows to Drucker-Prager model (negative strain-hardening)

Goal: Compare the curves of unidirectionnal behavior with the exact solution.

Version: 97' customer version

Model description:

	Elastoplastic Mechanic Drucker-Prager	Elastic mechanic
YOUNG	40000	2.D-3
NU	0.21	0.




Geometry of the parallelipiped:
 $1.5*1.5*1$ m³




Test plas9 (Results)

CASTEM FIGURES



PARALLELEPIPEDE EN TRACTION UNIAXIALE



PARALLELEPIPEDE EN TRACTION UNIAXIALE

X1.E-4

Test plas9 (Listing CASTEM)

```

TITRE 'ONE-AXIS TENSILE-TEST ON A PARALLELIPIPED' ;
OPTI DIME 3 ELEM CUB8 MODE TRID ;
OPTI ECHO 0 ;
*****
*      GEOMERTY AND MESHING      *
*****
A1= 0. 0. 0.; B1= 1.5 0. 0.;
C1= 1.5 1.5 0.; D1= 0. 1.5 0.;
A2= 0. 0. 1.; B2= 1.5 0. 1.;
C2= 1.5 1.5 1.; D2= 0. 1.5 1.;
*
L1 = A1 DROI 1 B1 ; L2 = B1 DROI 1 C1 ;
L3 = C1 DROI 1 D1 ; L4 = D1 DROI 1 A1 ;
LL = L1 ET L2 ET L3 ET L4 ;
SU1= SURF LL PLAN ;
M1 = A2 DROI 1 B2 ; M2 = B2 DROI 1 C2 ;
M3 = C2 DROI 1 D2 ; M4 = D2 DROI 1 A2 ;
MM = M1 ET M2 ET M3 ET M4 ;
SU2= SURF MM PLAN ;
SU = SU1 VOLU 1 SU2 ;
TRAC (888 1000 2000 ) SU ;
*****
* MODEL AND MATERIAL CHARACTERISTICS *
*****
MO1= MODE SU MECANIQUE ELASTIQUE PLASTIQUE
      DRUCKER_PRAGER CONS CONS1 ;
MO2= MODE SU MECANIQUE ELASTIQUE
      CONS CONS2;
MO = MO1 ET MO2 ;
VALK = 2. ;
VALALF =0.441 ;
VALBE = 1.02 ;
VALH = -10000. ;
YOYO = 40000. ;
MAT1 = MATE MO1 YOUN YOYO NU 0.21
      ALFA VALALF BETA VALBE K VALK
      GAMM VALALF DELT VALBE ETA VALALF
      MU VALBE KL VALK H VALH ;
MAT2 = MATE MO2 YOUN 2.D-3 NU 0. ;
MAT = MAT1 ET MAT2 ;
*****
* BOUNDARY CONDITIONS AND RIGIDITY *
*****
CL4 = (BLOQ L4 UX) ET ( BLOQ M4 UX) ;
CL2 = (BLOQ L2 UX) ET ( BLOQ M2 UX) ;
CL1 = (BLOQ L1 UY) ET (BLOQ M1 UY);
CL5 = BLOQ SU1 UZ ;
CL = CL1 ET CL2 ET CL4 ET CL5 ;
*****
* IMPOSED DISPLACEMENT *
*****
DEP= DEPI CL2 1.5E-4 ;

* SETUP
LI1 = PROG 0. PAS 1. 1000. ;
LI2 = LI1 ;
EV = EVOL MANU T LI1 F(T) LI2 ;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
* The displacement is F(T)*D at the T time
CHA1 = CHAR DIMP DEP EV ;
LIS1 = PROG 0. PAS 0.1 4. ;
TAB1 = TABLE ;
TAB1.'CARACTERISTIQUES' = MAT;
TAB1.'CHARGEMENT' = CHA1;

TAB1.'TEMPS_CALCULES' = LIS1;
TAB1.'BLOCAGES_MECANIQUES' = CL;
TAB1.FTOL=0.000001;
PASAPAS TAB1;
*****
* RESULTS *
*****
MU = ((2. * ( VALALF ** 2)) +
      ( VALBE ** 2) ) ** 0.5) /
      ( VALALF + VALBE) ;
VALHMU = VALH * MU ;
PGX = PROG 0. ;
PGY = PROG 0. ;
PGT = PROG 0. ;
TT = TAB1 . CONTRAINTES ;
NN = (DIME TT) - 1 ;
SI (( TAB1 . TEMPS . NN) < 3.99) ;
      ERRE 5 ;
      FINSI;

REPE BOU NN;
FF = EXTR ( TAB1 . CONTRAINTES . (&BOU) )
      SMXX 1 1 1 ;
EPS = EPSI MO1 ( TAB1 . DEPLACEMENTS . (&BOU) ) ;
GG = EXTR EPS EPXX 1 1 1 ;
ETH = VALK / ( ( VALALF + VALBE ) * YOYO ) ;
SI ( GG < ETH ) ;
THE = YOYO * GG ;
SINON;
THE = ( VALK + ( VALHMU * GG ) ) /
      ( (VALALF + VALBE) + ( VALHMU / YOYO ) ) ;
SI ( THE < 0. ) ; THE = 0. ; FINSI ;
FINSI ;
PGX = PGX ET ( PROG GG ) ;
PGY = PGY ET ( PROG FF ) ;
PGT = PGT ET ( PROG THE ) ;
FIN BOU ;
EV1= EVOL ROUG MANU ABS PGX ORD PGY ;
EV2= EVOL TURQ MANU ABS PGX ORD PGT ;
EV = EV1 ET EV2;
SI(NEG GRAPH 'N');
DESS EV;
FINSI ;

TEMPS ;
*****
* GOOD WORKING MESSAGE *
*****
RESI1=(MAXI ABS (PGY - PGT ))/ VALK ;
SI(RESI1 <EG 5E-5);
MESS ' ' ;
MESS ' ' ;
MESS ' ' ;
MESS ' ' ;

MESS ' COMPARISON BETWEEN THEORETICAL
      AND ANALYTIC RESULTS' ;
MESS ' ' ;
MESS ' UP TO TIME 4.: PRECISION
      INFERIOR TO 5E-5' ;
SINON;
      ERRE 5;
FINSI;
FIN;

```



3.9 Test plas10 (Description sheet)

Test name: plas10

Calculation type: MECHANICS ELASTIC PLASTIC - CAM-CLAY'S MATERIAL


Finite element type: CUB8

Topic: Tensile-test on a parallelepiped. An isostatic embedded parallelepiped is subjected to imposed displacement on three of its faces. It follows the Cam-Clay model.

Goal: Compare the hydrostatic compressive curve with the exact solution.


Version: 97' customer version

Model description:




Test plas10 (Results)

CASTEM FIGURES



PARALLELEPIPEDE EN COMPRESSION HYDROSTATIQUE



PARALLELEPIPEDE EN COMPRESSION HYDROSTATIQUE

x1.E-3



3.9. TEST PLAS10 (DESCRIPTION SHEET)

Test plas10 (Listing CASTEM)

```

TITR 'PARALLELEPIPED IN HYDROSTATIC COMPRESSION' ;
OPTI ECHO 0 ;
OPTI DIME 3 ELEM CUB8 MODE TRID ;
*****
*          GEOMERTY AND MESHING          *
*****
A1= 0. 0. 0.; B1= 1. 0. 0.; C1= 1. 1. 0.;
D1= 0. 1. 0.; A2= 0. 0. 1.; B2= 1. 0. 1.;
C2= 1. 1. 1.; D2= 0. 1. 1.;

L1 = A1 DROI 1 B1 ; L2 = B1 DROI 1 C1 ;
L3 = C1 DROI 1 D1 ; L4 = D1 DROI 1 A1 ;
LL = L1 ET L2 ET L3 ET L4 ;
SU1= SURF LL PLAN ;
M1 = A2 DROI 1 B2 ; M2 = B2 DROI 1 C2 ;
M3 = C2 DROI 1 D2 ; M4 = D2 DROI 1 A2 ;
MM = M1 ET M2 ET M3 ET M4 ;
SU2= SURF MM PLAN ;
SU = SU1 VOLU 1 SU2 ;
TRAC (888 1000 2000 ) SU ;
*****
*          MODEL AND MATERIAL CHARACTERISTICS          *
*****
MO = MODE SU MECANIQUE ELASTIQUE
      PLASTIQUE CAM_CLAY ;
XL = 0.1 ; CC = 0.5 ; XK = 0.01 ;
EE = 2. ; PI0 = 5.;

MAT = MATE MO YOUN 1000. NU 0.2
      E0 EE COHE CC M 1.2
      P0 PI0 LAMD XL KAPA XK
      G1 416.666667 ;
*****
*          BOUNDARY CONDITIONS AND RIGIDITY          *
*****
BLX = BLOQ UX ( A1 ET A2 ET D1 ET D2 );
BLY = BLOQ UY ( A1 ET B1 ET B2 ET A2 );
BLZ = BLOQ UZ SU1 ;
BL = BLX ET BLY ET BLZ ;

CLX = BLOQ UX ( B1 ET B2 ET C1 ET C2 );
CLY = BLOQ UY ( D1 ET C1 ET C2 ET D2 );
CLZ = BLOQ UZ SU2 ;
CL = CLX ET CLY ET CLZ ;
*****
*          IMPOSED DISPLACEMENT          *
*****
VAL = -2.664328E-5 ;
DEPX = DEPI CLX VAL ;
DEPY = DEPI CLY VAL ;
DEPZ = DEPI CLZ VAL ;

LITX = PROG 0. PAS 1. 300. ;
LIX = PROG 0. PAS 1. 300. ;
EVX = EVOL MANU TEMP LITX FX LIX ;
CHA1=CHAR DIMP (DEPX ET DEPY ET DEPZ) EVX;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
* The displacement is F(T)*D at the T time
TAB1 = TABLE ;
LIS1 = PROG 0. PAS 10. 200. ;
TAB1.'MODELE' = MO;
TAB1.'CARACTERISTIQUES' = MAT;
TAB1.'CHARGEMENT' = CHA1;
TAB1.'TEMPS_CALCULES' = LIS1;

TAB1.'BLOCAGES_MECANIQUES' = CL ET BL;

PASAPAS TAB1 ;
*****
* RESULTS *
*****
NT = (DIME (TAB1.DEPLACEMENTS)) - 1 ;
IT = 0 ;
PRD = PROG 0. ;
PRF = PROG 0. ;
REPETER BOPI NT;
IT = IT + 1 ;
DD= TAB1.DEPLACEMENTS.IT;
RR = REAC CLY DD;
RRR = RESU RR;
LEP = ( EXTR RRR MAIL NOMU ) POIN 1 ;
PRD = PRD ET (PROG ( EXTR DD UY D1));
PRF = PRF ET (PROG ( EXTR RRR FY LEP) ) ;
FIN BOPI ;
FV = EVOL ROSE MANU 'D' PRD 'F' PRF;
DESS FV ;
VV = PECHE TAB1 VARIABLES_INTERNES 200. ;
SS = PECHE TAB1 CONTRAINTES 200. ;
PCAL = EXTR SS SMXX 1 1 1 ;
PTHE = -1. * ( PI0 + CC ) *
      (EXP ( ( -300. * VAL ) *
      ( 1.+EE ) / XL ) );
PTHE = PTHE + CC ;
EVPCAL = EXTR VV TRAP 1 1 1 ;
EVPTHE = (300. * VAL ) * ( XL - XK ) / XL ;
SAUTER 3 LIGNE ;
MESS ' THEORETICAL MEAN PRESSURE' PTHE '
      CALCULATED' PCAL;
MESS ' THEORETICAL PLASTIC PLOTTING '
      EVPTHE ' CALCULATED' EVPCAL;
*
RESI1 = ABS ( (PCAL-PTHE)/PTHE );
RESI2 = ABS ( (EVPCAL-EVPTHE)/EVPTHE);
SAUTER 3 LIGNE ;
MESS ' RELATIVE ERRORS ' RESI1 RESI2 ;
*****
* GOOD WORKING MESSAGE *
*****
SI (RESI1 <EG 1.E-6) ;
SI (RESI2 <EG 1.E-6) ;
ERRE 0;
SINO;
ERRE 5;
FINSI;
SINON;
ERRE 5;
FINSI;
FIN;

```

3.10 Test plas11 (Description sheet)

Test name: plas11

Calculation type: MECHANIC ELASTIC PLASTIC - GENERALISED PLANE STRAINS

Finite element type: QUA8, COQ2, POI1

Topic: Beam subjected to a rotation. A rectangular cross-section beam is subjected to an imposed rotation RX. The beam is modelled with a massive piece (1x1), with a COQ2 element piece (length: .5 thickness: 2.) and with a piece made of POI1 point elements (4 points, 2 in the upper piece and 2 in the lower one's).


Goal: Study the behavior of the resulting moment of this imposed rotation up to the yield point.

Version: 97' customer version

Model description:

	Massive piece	COQ2 elements	POI1 point elements
YOUNG	$2 \cdot 10^{11}$ Pa	$2 \cdot 10^{11}$ Pa	$2 \cdot 10^{11}$ Pa
NU	0.	0.	0.
SIGY	$4.83 \cdot 10^8$ Pa	$4.83 \cdot 10^8$	$4.83 \cdot 10^8$

Imposed rotation




Geometry of the rectangular cross-section beam
 thickness: 2.
 heighth: 4.



Test plas11 (Results)

CASTEM FIGURES



Test plas11 (Listing CASTEM)

```

TITRE 'RECTANGULAR CROSS-SECTION OF BEAM
      SUBJECTED TO AN IMPOSED ROTATION';
OPTI ECHO 0;
TEMPS;
OPTI DIME 2 ELEM QUA8;
* we define the support point of general
* plane strains
PG = 0. 0.;

OPTI MODE PLAN GENE PG ;
* initialization of a few constants
I = 1;
EVL1 = PROG 0. ; EVL2 = PROG 0. ;
EVL3 = PROG 0. ;
*****
* CONSTANTS FOR ANALYTIC CALCULATION *
*****
H = 2.; COMM .5*height of the section;
B = 1.; COMM .5*width of the section;
* Young's modulus      yield point
E1 = 2E11      ; SILIM = 4.83E8;
* moment of inertia with respect to X-axis
IX = (2.*B)*((2.*H)**3)/12.;
* maximum value of the elastic strain
EPSEL = SILIM/E1;
* maximum value of the elastic moment
MOMEL = E1*IX*EPSEL/H;
* value of the ultima plastic moment
MOLIM = 3*MOMEL/2.;
*****
*      GEOMERTY AND MESHING      *
*****
* meshing of the massive piece
P1 = 1. -1. ; P2 = 1. 1.;
P3 = -1. 1. ; P4 = -1. -1.;
N = 4 ; M = 1;
D1 = P1 DROI N P2 ; D2 = P2 DROI M P3 ;
D3 = P3 DROI N P4 ; D4 = P4 DROI M P1 ;
SURF1 = DALL D1 D2 D3 D4;
* meshing of the coq2 elements
OPTI ELEM SEG2 ;
P5 = 0. 1. ; P6 = 0. 1.5 ;
P7 = 0. -1. ; P8 = 0. -1.5 ;
D5 = P5 D 2 P6 ; D6 = P7 D 2 P8;
COQU1 = D5 ET D6 ;
* meshing of the poil elements
P9 = 0. 1.625 ; P10 = 0. 1.875;
P11 = 0. -1.625 ; P12 = 0. -1.875;
PO11 = MANU POI1 P9;
PO12 = MANU POI1 P10;
PO13 = MANU POI1 P11;
PO14 = MANU POI1 P12;
PO1 = PO11 ET PO12 ET PO13 ET PO14;

SURTOT = SURF1 ET COQU1 ET PO1 ;
TITR 'plas12 : MESHING';
TRAC QUAL SURTOT;

ELTOT = NBEL SURTOT; NOTOT = NBNO SURTOT;
*****
*      MODEL AND MATERIAL CHARACTERISTICS *
*****
* bulk piece
MO1 = MODE SURF1 MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT QUA8;
MA1 = MATE MO1 YOUN E1 NU 0. SIGY SILIM;

* coq2 elements
MO2 = MODE COQU1 MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT COQ2;
MAT2 = MATE MO2 YOUN E1 NU 0. SIGY SILIM;
CAR2 = CARA MO2 EPAI 2. ;
MA2 = MAT2 ET CAR2 ;

*poil element
MO3 = MODE PO1 MECANIQUE ELASTIQUE
      PLASTIQUE PARFAIT POI1;
MAT3 = MATE MO3 YOUN E1 NU 0. SIGY SILIM;
CAR3 = CARA MO3 SECT .5 ;
MA3 = MAT3 ET CAR3;

MODE1 = MO1 ET MO2 ET MO3;
MATE1 = MA1 ET MA2 ET MA3;

RI1 = RIGI MODE1 MATE1;
*****
* BOUNDARY CONDITIONS AND RIGIDITY *
*****
P15 = SURF1 POINT PROCHE (0. 0.);
P16 = SURF1 POINT PROCHE (0. 1.);
B1 = BLOQ DEPL P15;
B2 = BLOQ UX P16;
B3 = BLOQ DEPL PO1 ;
B4 = BLOQ UX COQU1 ;
B5 = BLOQ UY (P5 ET P7) ;
B6 = BLOQ UZ RY PG;
B7 = BLOQ RX PG;
BTOT = B1 ET B2 ET B3 ET B4 ET B5 ET B6
      ET B7;
RI = RI1 ET BTOT;
*****
*      DEFINITION LOAD      *
*****
TEMP1= (PROG 4. PAS 2. 30.) ET (PROG 50.);
LAM1 = (PROG 4. PAS 2. 30.) ET (PROG 50.);
FAT1 = .0002;
TITRE 'plas11: LOAD';
EV1= EVOL MANU 'TIME' TEMP1 'RX(T)' LAM1 ;
DE1= DEPI B7 FAT1 ;
CHRIX = CHAR 'DIMP' DE1 EV1 ;
DESS EV1 ;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
TAB1 = TABLE;
TAB1.'BLOCAGES_MECANIQUES' = BTOT;
TAB1.'MODELE' = MODE1;
TAB1.'CARACTERISTIQUES' = MATE1;
TAB1.'CHARGEMENT' = CHRIX;
TAB1.'TEMPS_CALCULES' = TEMP1;
PASAPAS TAB1 ;
*****
* POST-TREATMENT: PLOTTING OF A CURVE AND*
* COMPARISON WITH THA ANALYTIC SOLUTION *
*****
DIM1 = DIMENSION TEMP1;

REPETER BLOC1 DIM1;
TPS = EXTR TEMP1 I;
CH1 = PECHE TAB1 DEPLACEMENTS TPS;
MOIMP = REAC RI CH1;
* Extraction of the MX moment
* corresponding to the RX rotation

MMX = EXTR MOIMP MX PG;
RRX = EXTR CH1 RX PG;
EVL1 = EVL1 ET (PROG MMX);
EVL2 = EVL2 ET (PROG RRX);
    
```




3.10. TEST PLAS11 (DESCRIPTION SHEET)

```
* Theoretical solution
EPSN = 2.*RRX ;
* When we are in the plastic domain:
SI (EPSN > EPSEL) ;
  MOMN = (1 - ((( EPSEL / EPSN) ** 2) / 3)) * MOLIM ;
  EVL3 = EVL3 ET (PROG MOMN);
* When we are in the elastic domain:
SINON;
  MOMN = E1 * IX * RRX ;
  EVL3 = EVL3 ET (PROG MOMN) ;
FINSI;

I = I+1;
FIN BLOC1;

COURB = TABLE;
COURB.1 = 'MARQ ETOI NOLI' ;

titre 'plas11 : COURBE MX EN FONCTION DE RX ' ;
EV1 = EVOL MANU 'RX' EVL2 'MX' EVL1 ;
EV2 = EVOL MANU 'RX' EVL2 'MX' EVL3 ;
* theoretical curve:unbroken drawing
* calculated point: star points
DESS (EV1 ET EV2) COURB LOGO;
* calculation of the difference with the reference solution
ERG1 = 100*(ABS((MOMN-MMX)/MOMN));
*****
* RESULTS PLACARDING *
*****
MESS ' RESULTS : TEST plas11 ' ;
MESS '-----';
SAUT 2 LIGN;
MESS 'We search the value of the calculated';
MESS ' moment in plastic domaine';
MESS 'It is compared with the analytic solution';
SAUT 1 LIGN;
MESS ' Theoretical moment:' MOMN 'MPa';
SAUT 1 LIGN;
MESS ' Calculated moment:' MMX 'MPa';
SAUT 1 LIGN;
MESS ' Relative error:' ERG1 '%';
SAUT 2 LIGN;
*
MESS '   Number of element:' ELTOT;
MESS '   Number of nodes:' NOTOT;
MESS '   e.G. ' (NOTOT * 3) 'd.o.f.';
*****
* GOOD WORKING MESSAGE *
*****
* The maximum difference between the
* theoretical value and the calculated's
* one have to be inferior to valeur 0.5%
SI (ERG1 <EG 0.5);
  ERRE 0;
SINON;
  ERRE 5;
FINSI;

TEMPS;

FIN;
```

3.11 Test plas12 (Description sheet)

Test name: plas12

Calculation type: MECHANICS ELASTIC PLASTIC

Finite element type: TUYA

Topic: Pipe subjected to thermal shock. A pipe is embedded at its two ends and subjected to a thermic shock.

Goal: Test the PASAPAS procedure with a material depending on the temperature. Compare the Castem's result for the stress with the analytical solution:


$\text{EPSILLON} = \text{Alpha} * (\text{Delta T})$

$\text{SIGMA} = \text{Interpolation of the EPSILLON value on the tensile curve}$

Version: 97' customer version

Model description:

Material properties	
YOUNG	function of the temperature
NU	0.3
Thermic expansion coefficient	function of the temperature
Tensile curve	function of the temperature




Geometry of the pipe:
 length: 1000 mm
 external diameter: 100 mm
 thickness: 10 mm




3.11. TEST PLAS12 (DESCRIPTION SHEET)


Test plas12 (Results) CASTEM FIGURES




Tensile curve as a function of temperature




Themic expansion as a function of the temperature



Modulus of Young as a function of the temperature



Themic expansion as a function of the temperature



Themic expansion as a function of the temperature

Test plas12 (Listing CASTEM)

```

*****
* PROCEDURE: CALCULATION OF THE YOUNG'S *
* WITH THE TENSILE CURVE *
*****
      DEBP CALYO EV1*EVOLUTION;
      X1=EXTR EV1 ABSC;
      Y1=EXTR EV1 ORDO;
      YOU1=(EXTR Y1 2)/(EXTR X1 2);
      FINP YOU1;
*****

TITR 'Embedded piping subjected to
      thermic shock';
opti dime 3 elem seg2;
* Geometrical data
l=1000.; e = 10; r = 100.;
* Data of the material characteristics
** Tensile curve as a function of
** the temperature
TITR 'Tensile curve as a function of
      temperature';
progx = prog 0. 0.001785 0.00386 0.00701
      0.00984 0.0132 0.0167
      0.0259 0.0491 0.0994;
progy = prog 0. 350.23 470.05 553.73
      610.14 635.94 663.59 717.97
      795.39 877.42;
ev1 = evol manu eps progx sig (progy*1.0);
ev2 = evol manu eps progx sig (progy*0.8);
ev3 = evol manu eps progx sig (progy*0.6);
ev4 = evol manu eps progx sig (progy*0.4);
ev5 = evol manu eps progx sig (progy*0.2);
tra = nuage comp T 0. 200. 400. 600. 800.
      comp TRAC ev1 ev2 ev3 ev4 ev5;
TAB1=TABLE;
TAB1.1 = 'MARQ LOSA TITR T=800.C';
TAB1.2 = 'MARQ TRIB TITR T=600.C';
TAB1.3 = 'MARQ CROI TITR T=400.C';
TAB1.4 = 'MARQ TRIA TITR T=200.C';
TAB1.5 = 'MARQ CARR TITR T=0.C';
DESS (ev5 et ev4 et ev3 et ev2 et ev1)
      lege tab1;
** Young's modulus as a function
** of the temperature
TITR 'Modulus of Young as a function of
      the temperature';
y1=calyo ev1; y2=calyo ev2;
y3=calyo ev3; y4=calyo ev4;
y5=calyo ev5;
evy = evol manu T (prog 0. 200. 400. 600. 800.)
      YOUN (prog y1 y2 y3 y4 y5);
DESS evy;
** Alpha as a function of the temperature
TITR 'Themic expansion as a function
      of the temperature';
eva = evol manu T (prog 0. 200. 400. 600. 800.)
      ALPH ((prog 10.4 11.9 13.0 13.8
      14.25)*8E-6);
DESS eva;
** Cross-section of the pipe
s = pi*((r*r) - ((r - e)*(r - e)));
*****
* GEOMERTY AND MESHING *
*****
p1 = 0 0 0; p2 = L 0 0;
d1= pl d 10 p2;

TRAC D1;

*****
* TEMPARATURE OF REFERENCE *
*****
TREFE = MANU CHPO d1 1 T 0.;
*****
* MODEL AND MATERIAL CHARACTERISTICS *
*****
modti=mode d1 mecanique elastique
      plastique isotrope tuyau;
mater=mate modti YOUN evy NU 0.3 ALPH eva
      TRAC tra;
carac=carb modti EPAI e RAYO r CISA 1. ;
mc=mater et carac;
*****
* BOUNDARY CONDITIONS *
*****
bloqtot = bloq depl rota (p1 et p2) ;
*****
* LOAD DEFINITION *
*****
* Thermic load
TBTEMPE = TABLE;
TBTEMPE.0 =TREFE;
TBTEMPE.1 =MANU CHPO d1 1 T 200.;
TBTEMPE.2 =MANU CHPO d1 1 T 400.;
TBTEMPE.3 =MANU CHPO d1 1 T 600.;
TBTEMPE.4 =MANU CHPO d1 1 T 800.;
TEMP = TABLE;
TEMP.0 = 0.; TEMP.1 = 2.;
TEMP.2 = 4.;TEMP.3 = 6.;
TEMP.4 = 8.;
EVT = EVOL MANU TEMPS
      (prog 0. 2. 4. 6. 8.)
      T(prog 0. 200. 400. 600. 800.);
CH2 = CHAR T TEMP TBTEMPE;
* Mechanic load
f1 = forc (0. 0. 0.) p1;
ch1 = char meca f1 (evol manu (prog 0. 10.)
      (prog 0. 0.));
chat = ch2 et ch1;
*****
*NON LINEAR CALCULATION PASAPAS PROCEDURE*
*****
TAB1 = TABLE;
TAB1.'TEMPERATURE_REFERENCE'=TREFE;
TAB1.'MODELE' = modti;
TAB1.'CARACTERISTIQUES' = MC;
TAB1.'BLOCAGES_MECANIQUES' = BLOQTOT;
TAB1.'TEMPS_CALCULES'=(prog 0. pas 1. 8.);
TAB1.'CHARGEMENT' = CHAT;
PASAPAS TAB1;
*****
* COMPARISON *
*****
TBEVOL=TABLE ; TBEVOL.1=(EV1 + EV2)/2.;
TBEVOL.2=EV2; TBEVOL.3=(EV2 + EV3)/2.;
TBEVOL.4=EV3; TBEVOL.5=(EV3 + EV4)/2.;
TBEVOL.6=EV4; TBEVOL.7=(EV4 + EV5)/2.;
TBEVOL.8=EV5; TBCONT=TAB1.CONTRAINTES;
PCALCUL=PROG; PANALYT=PROG; PEPSI=PROG;
REPETER B1 ((DIME TBCONT) - 1);
V1 = (EXTR (TBCONT.(&B1)) EFX 1 1 1)/S;
PCALCUL=PCALCUL ET (PROG (ABS V1));
T1=IPOL(EXTR EVT ABSC)(EXTR EVT ORDO)
      (TAB1.TEMPS.&B1);

Y1=IPOL(EXTR EVY ABSC)(EXTR EVY ORDO)T1;
A1=IPOL(EXTR EVA ABSC)(EXTR EVA ORDO)T1;
PEPSI=PEPSI ET (PROG (A1*T1));

```



3.11. TEST PLAS12 (DESCRIPTION SHEET)

```
EVC=TBEVOL.&B1;
V2=IPOL(EXTR EVC ABSC)(EXTR EVC ORDO)
(A1*T1);
PANALYT=PANALYT ET (PROG V2);
FIN B1;
ECALCUL=EVOL ROUG MANU 'Epsilon' PEPSI
'Sigma(Mpa)' PCALCUL;
EANALYT=EVOL TURQ MANU 'Epsilon' PEPSI
'Sigma(Mpa)' PANALYT;
TAB1 = TABL;
TAB1.1 = 'MARQ LOSA TITR CASTEM2000';
TAB1.2 = 'MARQ TRIB TITR Analytic ';
DESS (ECALCUL et EANALYT) lege tabl;
ENEANA=EXTR 1 (SOMM EANALYT);
ENEK20=EXTR 1 (SOMM ECALCUL);
ERR1=ABS ((ENEANA - ENEK20)/ENEANA);
mess '*****';
mess 'Energy of cumulated strain';
mess 'k2000,theory,error:'ENEK20 ENEANA ERR1;
mess '*****';
*****
* GOOD WORKING MESSAGE *
*****
si (ERR1 > 5.5d-3);
erre 5;
sinon;
erre 0;
finsi;

fin;
```

3.12 Test plas13 (Description sheet)

Test name: plas13

Calculation type: MECHANICS ELASTIC PLASTIC

Finite element type: POUT

Topic: Material of Ambrois's post type. A post is subjected to a 2-tun load. The material is an Ambrois's post type. It is a global plasticity model for reinforced concrete beams which makes it possible to take into account non linear behavior laws according to the type of loading (axial, bending and shear).


Goal: Study an Ambroy's material with GLOBAL OLARIU elements.

Version: 97' customer version

Model description:

Material properties	
YOUNG	35510 10 ⁶
NU	0.2
RHO	2516.

CASTEM FIGURES





3.12. TEST PLAS13 (DESCRIPTION SHEET)

Test plas13 (Listing CASTEM)

```

TITR 'AMBROIS MATERIAL POST UNDER 2 T';
OPTI ECHO 0;
OPTI DIME 3 ELEM SEG2 MODE TRID;
*-----*
* MATERIAL OF TYPE: AMBROIS POST *
*-----*
SIGY = PROG -21.410E6 -25.208E6
      -22.904E6 -1.4202E6 0. 1.4202E6
      22.904E6 25.208E6 21.410E6 ;
EPSY = PROG -1.80E-2 -8.50E-3 -3.0E-3
      -0.40E-4 0. 0.40E-4 3.0E-3
      8.50E-3 1.80E-2 ;
TITRE 'AMBROIS MATERIAL POST UNDER 2 T' ;
EVT = EVOL ROUGE MANU EPS EPSY IG SIGY ;
*****
* GEOMERTY AND MESHING *
*****
DENS 1 ; A0 = 0. 0. 0. ; B0 = 0. 0. 1.50 ;
*****
* MODEL AND MATERIAL CHARACTERISTICS *
*****
* The posts
POT = A0 D 1 B0 ;
MOPOTA = MODE POT MECANIQUE ELASTIQUE
        PLASTIQUE GLOBAL POUTRE ;
* Prestress post under 2 tons
MOM = PROG -21.401E3 -18.120E3 -0.5961E3
      0. 0.5961E3 18.120E3 21.401E3 ;
Dplasp = PROG -3.08E-2 -2.455E-2 -1.764E-4
      0. 1.764E-4 2.455E-2 3.08E-2 ;
TITRE 'Law of bending-global/2T';
TC=TABLE;
EVE=EVOL BLAN MANU Dplasp Dplasp FLXZ MOM;
TC.1 = 'MARQ ETOI';
DESS EVE CARR LEGE MIMA TC;
*-----*
* MATERIAL OF GLOBAL TYPE *
*-----*
MAPOTA = MATE MOPOTA 'SECT'0.0289
        'INRY' 7.137E-5 'INRZ' 7.137E-5
        'TORS' 1.178E-4 'SECY' 0. 'SE CZ' 0.
        'DY' 0.085 'DZ' 0.085
        'YOUNG' 35510.E6 'NU' 0.2 'RHO' 2516.
        FLXZ EVE 'VECT' ( 0. -1. 0. );
*****
* BOUNDARY CONDITIONS AND RIGIDITY *
*****
RIG = RIGI MOPOTA MAPOTA ;
BL11 = BLOQ DEPL ROTA ( A0 ) ;
BL12 = BLOQ UX UZ RY RZ ( B0 ) ;
BL1 = BL11 ET BL12 ;

RIGV = RIG ET BL1 ;

MAS = MASSE MOPOTA MAPOTA ;
MAS1 = MASSE UY 2.0E3 (B0);
MAS= MAS ET MAS1;
*****
* CALCULATION OF THE FIRST NODES *
*****
*PREMOD1=VIBRE PROC (PROG 0.) RIGV MAS TBAS;
PREMOD2 = VIBRE 'SIMULTANE' 0. 2 RIGV2 MASTOT2 TBAS;
*LIST PREMOD1.MODES . 1 ;
*****
* DEFINITION LOAD *
*****
V1=MANU CHPO POT 1 UZ -9.81 NATURE DIFFUS;
F01 = MAS * V1 ;
VALTEM1 = PROG 0. 1.E-2 ;

VALPOI = PROG 1. 1. ;
EV1= EVOL MANU TEMPS VALTEM1 POIDS VALPOI;
CHA1 = CHAR MECA F01 EV1 ;
*****
* EFFECT OF THE STATIC LOAD IN ORDER TO *
* FIND THE LAW OF COMPORTEMENT OF THE *
* GLOBAL ELEMENT *
*****
V2 = FORCE FY 1. B0 ;
AMPLIF1 = PROG 0. PAS 0.05 0.1;
AMPLIF2 = PROG 0.1 PAS 0.5 1.4266;
AMPLIF = AMPLIF1 ET AMPLIF2;
AMPLIF = AMPLIF * 1.E4 ;
Q = DIME AMPLIF ;
VALTEM2 = PROG 1.E-2 PAS 1.E-2 (Q*1.E-2) ;
EV2=EVOL MANU TEMPS VALTEM2 CHARGEMENT AMPLIF ;
CHA2 = CHAR MECA V2 EV2 ;
*****
* STATIC CALCULATION ONLY WITH WEIGTH *
*****
TAB = TABLE ;
TAB.'CARACTERISTIQUES' = MAPOTA;
TAB.'MODELE' = MOPOTA;
TAB.'BLOCAGES_MECANIQUE' = BL1;
TAB.'CHARGEMENT' = CHA1;
TAB.'TEMPS_CALCULES' = VALTEM1;
PASAPAS TAB ;
* WE ADD THE LOAD
VALPOI = PROG Q * 1. ;
EV3=EVOL MANU TEMPS VALTEM2 POIDS VALPOI ;
CHA3 = CHAR MECA F01 EV3 ;
CHA = CHA3 ET CHA2 ;

TAB . 'DYNAMIQUE' = FAUX ;
TAB.FTOL = 1.E-3;
TAB.MTOL = 1.E-3;
TAB.PRECISION = 1.E-5;
TAB . 'MASSE' = MAS ;
TPCAL = 1.E-2 ;
TEMPCAL = PROG 0. PAS TPCAL ((Q-1)*1.E-2);
TAB.'CHARGEMENT' = CHA;
TAB.'TEMPS_CALCULES' = VALTEM2;
PASAPAS TAB ;
*****
* RECOVERY OF THE RESULTS *
*****
tabd = TAB.DEPLACEMENTS ;
tabc = TAB.CONTRAINTE ;
* Number of the step
time = TAB.TEMPS ; ni = (DIME time) - 1;
RE = REAC BL1 (tabd.ni);
MOMAX = EXTR RE MX A0;
TEXT'The reached maximum moment is'NOMAX'N.m';
PRECIS = 1.E-3;
BOOL1 = (MOMAX <EG ((1 + PRECIS)*21.401E3));
BOOL2 = (((1 - PRECIS)*21.401E3) <EG MOMAX );
BOOL = BOOL1 ET BOOL2;
M1 = (MOT 'PROBLEM IN THE GLOBAL ELEMENT');
M2 = (MOT 'TEST OK');
SI (BOOL EGA FAUX ) ;
MESS M1 ;
SINON;
MESS M2;
FINSI;
FIN ;

```

3.13 Test plas14 (Description sheet)

Test name: plas14

Calculation type: MECHANICS ELASTIC PLASTIC ACIER_UNI AND BETON_UNI
 MECHANICS ELASTIC FIBRE_NL FERAILLE AND FIBRE_NL BETON

Finite element type: QUAS, TRIS

Topic: Example of a concrete section with four 20-mm steel bars. The ACIER_UNI (steel) model is a modified Menegotto-Pinto uni-axial model, which accounts for the steel buckling. The BETON_UNI (concrete) model is a Hognestad-like model, accounting for the confinement. The geometry of the concrete is as follows:

Concrete dimensions:

All section $(0.25*0.25) m^2$

Core section $(0.20*0.20) m^2$

Transversal reinforcement:

10 mm Bars / 7.0 cm

Version: 97' customer version

Model description:





3.13. TEST PLAS14 (DESCRIPTION SHEET)

<i>ACIER_UNI (Steel) model</i>		
Young's modulus		2.03 10 ⁵
Poisson's coefficient		0.30
Yielding stress		440.0
Ultimate strain		.090
Ultimate stress		760.0
Hardening strain		0.030
Ratio between the length of the longitudinal bars between stirrups and its diameter		4.375
A6 factor		620.0
C factor		0.5
RO factor		20.0
Ratio between the hardening stiffness in cyclic behavior and the elastic one		0.010
A1 factor		18.5
A2 factor		0.15
<i>BETON_UNI (concrete) model</i>	Unconfined concrete	Confined concrete
Young's modulus	0.30 10 ⁵	0.2254 10 ⁵
Poisson's coefficient	0.20	0.25
Maximum compression strength	30.0	30.0
Strain corresponding to 'STFC'	.002	.002
Maximum tension strength	3.0	3.0
Confinment parameter	.22687	.22687
Confinment parameter	.32912	.32912
Slope of the descending part of the compression curve	100.0	0.0
Stress defining the compression plateau	.0	6.0
Factor defining the softening behavior of the tension curve	3.0	3.0

Test plas14 (Results)

CASTEM FIGURES





3.13. TEST PLAS14 (DESCRIPTION SHEET)

Test plas14 (Listing CASTEM)

```

*****
* EXAMPLE OF A CONCRETE SQUARE SECTION *
* WITH 4 STEEL 20mm BARS *
* - CONCRETE DIMENSIONS: *
* (0.25*0.25)m2 - ALL SECTION *
* (0.20*0.20)m2 - CORE SECTION *
* - TRANSVERSAL REINFORCEMENT: *
* 10mm Bars / 7.0cm *
* NOVEMBER 1993 *
*****

opti echo 1;
*----- Quadrangrular Elements
opti dime 2 elem qua4 echo 0 ;
*----- Triangrular Elements
* opti dime 2 elem tri3;
*****
* GEOMERTY AND MESHING *
*****
* DEFINITION OF THE STEEL GEOMETRY
ps1 = -9.386d-2 9.386d-2;
ps2 = -9.386d-2 7.614d-2;
ps3 = 1.772d-2 0.0 ;
cars1 = ps1 d 1 ps2 tran 1 ps3;

cars2 = cars1 plus ( 17.0e-2 0.0);
cars3 = cars2 plus ( 0.0 -17.0e-2);
cars4 = cars3 plus (-17.0e-2 0.0);

meshf = cars1 et cars2 et cars3 et cars4;
meshf = coul meshf bleu;
*****
* DEFINITION OF THE CONCRETE GEOMETRY *
* IN THE UNCONFINED CASE *
*****
* verd - vertical number of divisions
* hord - horizontal number of divisions
verd = 1;
hord = 8;

divu = 2;
pcl = -12.5d-2 12.5d-2;
pc2 = -12.5d-2 10.0d-2;
pc3 = 2.5d-2 0.0d-2;
carc1 = pcl d (divu) pc2 tran (divu) pc3;

carc2 = carc1 plus ( 22.5d-2 0.0);
carc3 = carc2 plus ( 0.0 -22.5d-2);
carc4 = carc3 plus (-22.5d-2 0.0);

pc4 = -10.0d-2 12.5d-2;
pc5 = -10.0d-2 10.0d-2;
pc6 = 20.0d-2 0.0d-2;
carc5 = pc4 d (divu) pc5 tran (verd) pc6;

carc6 = carc5 plus ( 0.0 -22.5d-2);

pc7 = -12.5d-2 10.0d-2;
pc8 = -10.0d-2 10.0d-2;
pc9 = 0.0d-2 -20.0d-2;
carc7 = pc7 d (divu) pc8 tran (hord) pc9;

carc8 = carc7 plus ( 22.5d-2 0.0);

meshu = carc1 et carc2 et carc3 et carc4 et
carc5 et carc6 et carc7 et carc8;
meshu = coul meshu jaune;
*****

* DEFINITION OF THE CONCRETE GEOMETRY *
* IN THE CONFINED CASE *
*****
pc10 = -10.0d-2 10.0d-2;
pc11 = 10.0d-2 10.0d-2;
pc12 = 0.0d-2 -20.0d-2;
meshc = pc10 d (verd) pc11 tran (hord) pc12;
meshc = coul meshc rouge;
elim (meshf et meshu et meshc) 0.001;

titr 'Section:blue=steel,yellow=unconfined
concrete,red=confined concrete';
trac (meshf et meshu et meshc);

opti dime 3;
pp0 = 0.0 .0 .0;
pp1 = 1.0 .0 .0;
llb = pp0 d 1 pp1;
elim (llb et pp0 et pp1) 0.001;
*****
* CHARACTERIZATION OF THE STEEL AND *
* CONCRETE MODELS *
*****
*----- Quadrangrular Elements
modf = modl meshf mecanique elastique
PLASTIQUE ACIER_UNI quas;
modu = modl meshu mecanique elastique
PLASTIQUE BETON_UNI quas;
modc = modl meshc mecanique elastique
PLASTIQUE BETON_UNI quas;
*----- Triangrular Elements
* modf = modl meshf mecanique elastique
* fibre_nl ferraille tris;
* modu = modl meshu mecanique elastique
* fibre_nl beton tris;
* modc = modl meshc mecanique elastique
* fibre_nl beton tris;
*----- Steel
matf = mate modf 'YOUN' 2.03e5 'NU' 0.30
'STSY' 440.0 'EPSU' .090 'STSU' 760.0
'EPSH' 0.030 'FALD' 4.375 'A6FA' 620.0
'CFAC' 0.5 'AFAC' 0.008 'ROFA' 20.0
'BFAC' 0.010 'A1FA' 18.5 'A2FA' 0.15;
carf = carb modf 'ALPY' 1. 'ALPZ' 1.;
*----- Unconfined concrete
matu = mate modu 'YOUN' 0.30e5 'NU' .20
'STFC' 30.0 'EZER' .002 'STFT' 3.0
'ALF1' .22687 'OME1' .32912
'ZETA' 100.0 'ST85' .0 'TRAF' 3.0;
caru = carb modu 'ALPY' .66 'ALPZ' .66;
*----- Confined concrete
* Initial concrete Young modulus
* = 2 * STIFC / ( BETA * EZERO )
matc = mate modc 'YOUN' 0.2254e5 'NU' .25
'STFC' 30.0 'EZER' .002 'STFT' 3.0
'ALF1' .22687 'OME1' .32912
'ZETA' 0.0 'ST85' 6.0 'TRAF' 3.0;
carc = carb modc 'ALPY' .66 'ALPZ' .66;

modq = modf et modu et modc;
macq = matf et matu et matc et carf et
caru et carc;

*****
*USE OF MOMCUR PROCEDURE FOR THE ANALYSIS*
* OF THE PLASTIC BEHAVIOUR SECTION *
*****
* CHARACTERIZATION OF THE ACTION *
* (CURVATURES ALONG OY AXIS AND CONSTANT *

```

```

* AXIAL FORCE) *
*****
eppl = 440.0/2.03e5;
cy = prog 0 pas .0005 .005 pas .005 .138;
ncur = dime cy;
cz = prog ncur * .00;
fa = prog ncur * -.25;
*****
* RESOLUTION *
*****
my mz ea = mocu cy cz fa modq macq
          (1.d-6*eppl);
*****
* OUTPUT DIAGRAMS *
*****
c1 = evol rouge manu Curvature cy Moment
    (my*1.d3);
*****
* TRILINEAR CURVE FOR A TAKEDA MODEL FOR *
* THE SAME SECTION *
*****
abstak=prog 0. 2.03791E-03 1.85207E-02
           1.38834E-01;
ordtak=prog 0. 2.05353E+01 7.10923E+01
           7.06633E+01;
albnl=evol vert manu Curvature abstak
        Moment ordtak;
*****
* PLOT *
*****
tt = table;
tt.1 = 'MARQ CARR';
tt.2 = '';
titre 'curve: mocu (rouge) and takeda (vert)';
dess (albnl et c1) tt;

finsi;
*****
* ERREUR *
*****
ordtak=ipol cy abstak ordtak;
errlis=ordtak - (my*1.d3);
errea=((l1l errlis errlis)**0.5) /
      (dime ordtak);
denom=((l1l ordtak ordtak)**0.5) /
      (dime ordtak);
errel=errea/denom;
mess 'relative erreur= ' errel '(+/-3.5%)';
si (errel > 4.d-2); erre 5;
sinon; erre 0; finsi;
fin;
    
```



Chapter 4

BEHAVIOUR LAWS

4.1 Test norton_cis1 (Description sheet)

Test name: norton_cis1

Calculation type: ISOTROPIC ELASTIC MECHANICS - 2D - PLANE STRAIN

Finite element type: QUA4


Topic: Norton creep model for shear stresses. The structure is a square plate subjected to shear stress (imposed load). It is embedded at each side. The plate follows Norton creep model.

Goal: Test the Norton model for solid plates.

Version: 97' customer version

Model description:

E	=	$2.1 \cdot 10^{11}$
NU	=	0.3
AF1	=	$1. \cdot 10^{-45}$
AF2	=	5.
AF1	=	1.2
SMAX	=	$2.1 \cdot 10^8$






4.1. TEST NORTON_CIS1 (DESCRIPTION SHEET)

Test norton_cis1 (Results)

CASTEM FIGURES




STRESS IN THE X-DIRECTION





FINAL STRAIN


AMPLITUDE
0.000e+00
1.0




STRESS IN THE Y-DIRECTION




COMPARISON BETWEEN CALCULATION & THEORY



SHEAR STRESS



EVOLUTION OF ERRORS IN TERMS OF %



COMPARISON BETWEEN CALCULATION THEORY, STRAIN-TIME

Test norton_cis1 (Listing CASTEM)

```

*****
TITR 'CREEP MODEL OF NORTON' ;
*****
OPTI DIME 2 ELEM QUA4 ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
P1 = 0. 0.      ; P2 = 1. 0. ;
P3 = 1. 1.      ; P4 = 0. 1. ;
LI1 = D P1 1 P2 ; LI2 = D P2 1 P3 ;
LI3 = D P3 1 P4 ; LI4 = D P4 1 P1 ;
MAIL1 = SURF ( LI1 ET LI2 ET LI3 ET LI4 ) PLAN ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
AF1 = 1.E-45 ; AF2 = 5.      ; AF3 = 1.2 ;
SMAX = 2.1E8 ; YOUN0 = 2.1E11 ; NU0 = .3 ;

MOD0 = MODE MAIL1 MECANIQUE ELASTIQUE ISOTROPE
      FLUAGE NORTON ;
MAT0 = MATE MOD0 YOUN YOUN0 NU NU0 'AF1' AF1
      'AF2' AF2 'AF3' AF3 'SMAX' SMAX ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ MAIL1 UY ; CL2 = BLOQ LI1 UX ;
CL0 = CL1 ET CL2 ;
*
*----- LOAD DEFINITION -----*
DEPMAX = 7.E7 ;
DEP1 = FORCE LI1 FX ( -1. * DEPMAX ) FY 0. ;
DEP2 = FORCE LI3 FX DEPMAX FY 0. ;
DEP0 = DEP1 ET DEP2 ;
EVO = EVOL MANU TEMPS ( PROG 0. 1.E-4 1000. )
      Y ( PROG 0. 1. 1. ) ;
CHA0 = CHAR MECA DEP0 EVO ;

DT0 = .020      ; DT1 = 2. * DT0 ;
DT2 = 5. * DT1 ; DT3 = 5. * DT2 ;
TFIN0 = 20. ;
*
*----- CALCULATION: PASAPAS PROCEDURE -----*
TA1 = 'TABLE' ;
TA1 .'MODELE' = MOD0 ;
TA1 .'CARACTERISTIQUES' = MAT0 ;
TA1 .'CHARGEMENT' = CHA0 ;
TA1 .'BLOCAGES_MECANIQUES' = CL0 ;
TA1 .'TEMPS_CALCULES' =
      ( PROG 0. PAS 1.E-4 2.E-4 ) ET
      ( PROG 3.E-4 PAS DT0 DT0 ) ET
      ( PROG ( DT0 + DT0 ) PAS DT0 DT1 ) ET
      ( PROG ( DT1 + DT0 ) PAS DT1 DT2 ) ET
      ( PROG ( DT2 + DT1 ) PAS DT2 DT3 ) ET
      ( PROG ( DT3 + DT2 ) PAS DT3 TFIN0 ) ;

PASAPAS TA1 ;
*
*----- CALCULATION AND OUTPUT -----*
DIME = DIME TA1 .'TEMPS' ;
EPS10 = PROG 0. ;
SIGM10 = PROG 0. ;
DEF10 = PROG 0. ;
DEF_TH2 = PROG 0. ;
SIG_TH2 = PROG 0. ;
ERR_E2 = PROG 0. ;
ERR_S2 = PROG 0. ;
TEMP2 = PROG 0. ;
RACO = ( 3. ) ** ( .5 ) ;

H0 = YOUN0 / ( 2. * ( 1. + NU0 ) ) ;

*
PROB0 = VRAI ;
REPETER BLOC0 ( DIM0 - 1 ) ;
      I1 = &BLOC0 ;
*
*----- CALCULATED RESULTS -----*
DEF0 = TA1 .'DEFORMATIONS_INELASTIQUES'.I1 ;
DEPL0 = TA1 .'DEPLACEMENTS'.I1 ;
SIGM0 = TA1 .'CONTRAINTE'.I1 ;
TEMP0 = TA1 .'TEMPS'.I1 ;
VAR0 = TA1 .'VARIABLES_INTERNES'.I1 ;
EPS0 = EPSI MOD0 DEPL0 ;
EPS1 = EXTR(EXCO EPS0 GAXY SCAL )SCAL 1 1 1 ;
SIGM1 = EXTR(EXCO SIGM0 SMXY SCAL )SCAL 1 1 1 ;
DEF1 = EXTR(EXCO DEF0 GAXY SCAL )SCAL 1 1 1 ;
EPS10 = EPS10 ET ( PROG EPS1 ) ;
SIGM10 = SIGM10 ET ( PROG SIGM1 ) ;
SIGEQ0 = VMIS SIGM0 MOD0 ;
SIGEQ1 = EXTR SIGEQ0 SCAL 1 1 1 ;
DEF10 = DEF10 ET ( PROG DEF1 ) ;
TEMP2 = TEMP2 ET ( PROG TEMP0 ) ;
VAR1 = EXTR(EXCO VAR0 EPSE SCAL )SCAL 1 1 1 ;
SIG0 = CHAN CHPO EPS0 MOD0 ;
SIGX = EXCO SIG0 EPXX ;
SIGY = EXCO SIG0 EPYY ;
SIGXY = EXCO SIG0 GAXY ;
*TITR ' STRESSES IN THE X DIRECTION' ;
*TRACE SIGX MAIL1 ;
*TITR ' STRESSES IN THE Y DIRECTION' ;
*TRACE SIGY MAIL1 ;
*TITR ' SHEAR STRESSES' ;
*TRACE SIGXY MAIL1 ;
*
*----- ANALYTIC RESULTS -----*
Y2 = ( ( RACO * DEPMAX ) ** ( AF2 ) ) ;
Y2 = Y2 * ( TEMP0 ** AF3 ) ;
X2 = Y2 * AF1 * RACO ;
*
DEF_TH1 = X2 ;
SIG_TH1 = H0 * ( EPS1 - X2 ) ;
*
DEF_TH2 = DEF_TH2 ET ( PROG DEF_TH1 ) ;
SIG_TH2 = SIG_TH2 ET ( PROG SIG_TH1 ) ;
*MESS ' CALCULATED STRESS = ' SIGM1 ;
*MESS ' THEORETICAL STRESS = ' SIG_TH1 ;
*MESS ' CALCULATED STRAIN = ' DEF1 ;
*MESS ' THEORETICAL STRAIN = ' DEF_TH1 ;
*
*----- GOOD WORKING MESSAGE -----*
SI ( DEF1 > ( 1.E-1 * EPS1 ) ) ;
      ERR_E1 = ( ABS ( DEF1 - DEF_TH1 ) ) / DEF1 ;
      ERR_E1 = ERR_E1 * 100. ;
SINON ;
      ERR_E1 = 0. ;
FINSI ;
SI ( SIGM1 > 0. ) ;
      ERR_S1 = ( ABS ( SIGM1 - SIG_TH1 ) ) / SIGM1 ;
      ERR_S1 = ERR_S1 * 100. ;
SINON ;
      ERR_S1 = 0. ;
FINSI ;
SI ( ( ERR_E1 > .25 ) OU ( ERR_S1 > .25 ) ) ;
      PROB0 = FAUX ;
      MESS 'ERR_E1 ERR_S1' ERR_E1 ERR_S1 ;
FINSI ;

ERR_E2 = ERR_E2 ET ( PROG ERR_E1 ) ;
ERR_S2 = ERR_S2 ET ( PROG ERR_S1 ) ;
MESS 'END OF CALCULATION OF RETURN AT TIME' TEMP0 ;

```




4.1. TEST NORTON_CIS1 (DESCRIPTION SHEET)

```
FIN BLOC0 ;
*
DEFO0 = DEFO MAIL1 DEPL0 0. ;
DEFO1 = DEFO MAIL1 DEPL0 1. ROUG ;
TITR 'FINAL STRAIN' ;
TRACE ( DEFO0 ET DEFO1 ) ;
*
SI ( PROB0 EGA FAUX ) ;
  ERRE 5 ;
SINON ;
  ERRE 0 ;
FINSI ;
*
*----- PLOTTING -----*
EVO1 = EVOL MANU EPSP DEF10 SIGM SIGM10 ;
EVO2 = EVOL MANU EPSP DEF_TH2 SIGM SIG_TH2 ;
EVO3 = EVOL MANU EPSP DEF10 'ERROR_%' ERR_E2 ;
EVO4 = EVOL MANU EPSP DEF10 'ERROR_%' ERR_S2 ;
EVO5 = EVOL MANU TEMPS TEMP2 EPSP DEF10 ;
EVO6 = EVOL MANU TEMPS TEMP2 EPSP DEF_TH2 ;
TAB = 'TABLE' ;
TAB. 1 = 'MARQ CROI TITR CALCULATION' ;
TAB. 2 = 'MARQ PLUS TITR THEORY' ;
DESS ( EVO1 ET EVO2 )
TITR ' COMPARISON BETWEEN CALCULATION & THEORY'
LEGE TAB ;
TAB = 'TABLE' ;
TAB. 1 = 'MARQ CROI TITR STRAIN' ;
*TAB. 2 = 'MARQ PLUS TITR STRESS' ;
DESS ( EVO3 )
TITR ' EVOLUTION OF ERRORS IN TERM OF % '
  LEGE TAB ;
TAB = 'TABLE' ;
TAB. 1 = 'MARQ CROI TITR CALCULATION' ;
TAB. 2 = 'MARQ PLUS TITR THEORY' ;
DESS ( EVO5 ET EVO6 )
TITR ' COMPARISON CALCULATION THEORY,
  AND STRAIN-TIME' LEGE TAB ;
*
FIN ;
```

Test norton_cis1 (Comments)

- The Norton's creep model

```

MOD0 = MODE MAIL1 MECANIQUE ELASTIQUE ISOTROPE
      FLUAGE NORTON ;
MAT0 = MATE MOD0 YOUN YOUN0 NU NU0 'AF1' AF1
      'AF2' AF2 'AF3' AF3 'SMAX' SMAX ;
    
```

Several creep models are available in CASTEM2000. In each case we have to specify the elastic behavior, the equivalent creep deformation (ϵ_f), the equivalent stress (s) and the time (t).

The parameters to be input in Norton's creep model are AF1, AF2, AF3, as well as a SMAX stress reference (equal to Young's modulus times 1.E-3 by default).

The constitutive equation Norton creep model is:

$$\epsilon_f = AF1 * (s^{**AF2}) * (t^{**AF3})$$



4.2 Test chaboche2 (Description sheet)

Test name: chaboche2

Calculation type: ISOTROPIC VISCOPLASTIC MECHANICS - 2D AXIS

Finite element type: COQ8


Topic: Unified viscoplastic model of Chaboche. The structure is an embedded square plate. It is subjected to imposed displacements at the upper surface. The plate follows the Chaboche's viscoplastic law model.

Goal: Test the Chaboche model for solid structure.

Version: 97' customer version


Model description:




$$\begin{aligned} E &= 7.34 \cdot 10^{10} \\ \nu &= 0.33 \\ \rho &= 7.8 \cdot 10^3 \\ \alpha &= 1 \cdot 10^{-5} \end{aligned}$$



Test chaboche2 (Results)

CASTEM FIGURES


 COMPOSANTES
VECTEURS
PX PY PZ

 AMPLITUDE
0.00E+00
1.0




4.2. TEST CHABOCHE2 (DESCRIPTION SHEET)

Test chaboche2 (Listing CASTEM)

```
*****
TITR 'VISCOPLASTIC MODEL OF CHABOCHE' ;
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
OPTI DIME 3 ELEM QUA8 ;
OPTI ECHO 0 ;
OEIL= 100. 0. 0. ;
*
*----- GEOMETRY -----*
EPAISO = .0005 ; RAYONO = .05 ;
L10 = .1 ; N10 = 1 ;
P10 = RAYONO 0. 0. ; P20 = RAYONO L10 0. ;
P11 = RAYONO 0. L10 ; P12 = RAYONO L10 L10 ;
LI1 = DROI P10 N10 P20 ; LI2 = DROI P20 N10 P12 ;
LI3 = DROI P12 N10 P11 ; LI4 = DROI P11 N10 P10 ;
MAIL1 = SURF ( LI1 ET LI2 ET LI3 ET LI4 ) PLAN ;
TITR 'MESHING' ;
TRACE OEIL MAIL1 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD00 = MODE MAIL1 MECANIQUE ELASTIQUE ISOTROPE
        VISCOPLASTIQUE CHABOCHE COQ8 ;
MAT00 = MATE MOD00 YOUN 7.34E10 NU 0.33
        RHO 7.8E3 N 24
        KK 10.E6 K0 116.E6
        ALFK 1.5 ALFR 0.35 ALF 2.E6
        A1 67.5E6 C1 1300 BET1 4807.E6 R1 4
        A2 80.E6 C2 45 BET2 58480.E6 R2 4 PHI 1.
        B 12 GAMA 2.E-7 M 2
        QMAX 455.E6 QSTA 200.E6 MU 19 ETA 0.50 ;
CAR00 = CARA MOD00 EPAI EPAISO ;
*
*----- BOUNDARY CONDITIONS -----*
CL10 = BLOQ MAIL1 UX ; CL20 = BLOQ LI1 UZ ;
CL30 = BLOQ LI3 UZ ; CL40 = BLOQ LI4 UY ;
CL00 = CL10 ET CL20 ET CL30 ET CL40 ;
*
*----- LOAD DEFINITION -----*
DEPMAX = .0050 ;
DEP10 = DEPI CL30 DEPMAX ;
EV00 = EVOL MANU TEMPS ( PROG 0. 1000. )
        Y ( PROG 0. 1. ) ;
CHA00 = CHAR DIMP DEP10 EV00 ;
*
* TIME OF CALCULATION
DT00 = 5. ;
SI COMPLET;
TFIN00 = 200. ;
SINON;
TFIN00 = 25. ;
FINSI ;
*
*----- CALCULATION: PASAPAS PROCEDURE -----*
TA10 = 'TABLE' ;
TA10 .'MODELE' = MOD00 ;
TA10 .'CARACTERISTIQUES' = MAT00 ET CAR00 ;
TA10 .'BLOCAGES_MECANIQUES' = CL00 ;
TA10 .'CHARGEMENT' = CHA00 ;
TA10 .'TEMPS_CALCULES' = PROG 0. PAS DT00 TFIN00 ;

PASAPAS TA10 ;
*
*----- CALCULATION AND OUTPUT -----*
DIM10 = DIME TA10 .'TEMPS' ;
DEPL120 = PROG 0. ;

REACO = TA10 .'REACTIONS'.( DIM10 - 1 ) ;
VEC2 = VECT REACO 1.E-2 FX FY FZ ROUG ;
TITR ' REACTIONS ' ;
TRACE OEIL VEC2 MAIL1 ;
DEFO0 = DEFO MAIL1 DEPLO 0. ;
DEFO1 = DEFO MAIL1 DEPLO 1. ROUG ;
TITR ' DEFORMATION ' ;
TRACE OEIL ( DEFO0 ET DEFO1 ) ;

REPETER BLOC0 ( DIM10 - 1 ) ;
I10 = &BLOC0 ;
REACO = TA10 .'REACTIONS'.I10 ;
DEPL10 = TA10 .'DEPLACEMENTS'.I10 ;
TEMP10 = TA10 .'TEMPS'.I10 ;
DEPL11 = EXTR DEPL10 UY P11 ;
SIGM0 = TA10 .'CONTRAINTE'.I10 ;
SIGM0 = RTENS SIGM0 MOD00 ( 0 1 0 ) CAR00 ;
SIGM1 = CHAN CHPO SIGM0 MOD00 ;
SI ( I10 EGA ( DIM10 - 1 ) ) ;
TITR ' SHEAR STRESS ' ;
TRACE OEIL ( EXCO SIGM1 SMST ) MAIL1 ;
TITR ' STRESS IN THE Y DIRECTION ' ;
TRACE OEIL ( EXCO SIGM1 SMSS ) MAIL1 ;
TITR ' STRESS IN THE Z DIRECTION ' ;
TRACE OEIL ( EXCO SIGM1 SMTT ) MAIL1 ;
FINSI ;
REACO = REDU REACO LI3 ;
REACO = EXCO REACO FZ SCAL ;
CHP1 = MANU CHPO LI3 1 SCAL 1. ;
DEPL11 = XTY REACO CHP1 ( MOTS SCAL )
        ( MOTS SCAL ) ;
DEPL120 = DEPL120 ET ( PROG DEPL11 ) ;
FIN BLOC0 ;

FIN ;
```

Test chaboche2 (Comments)

- The Chaboche's viscoplastic model

```

MOD00 = MODE MAIL1 MECANIQUE ELASTIQUE ISOTROPE
        VISCOPLASTIQUE CHABOCHE COQ8 ;
MAT00 = MATE MOD00 YOUN 7.34E10 NU 0.33
        RHO 7.8E3 N 24
        KK 10.E6 K0 116.E6
        ALFK 1.5 ALFR 0.35 ALF 2.E6
        A1 67.5E6 C1 1300 BET1 4807.E6 R1 4
        A2 80.E6 C2 45 BET2 58480.E6 R2 4 PHI 1.
        B 12 GAMA 2.E-7 M 2
        QMAX 455.E6 QSTA 200.E6 MU 19 ETA 0.50 ;
    
```

The equations for the Chaboche's model are as follows:

→ *Criterion:* $F = J2(S-X) - RR - KK:$

→ *Strain hardening:* $dX_i = 2/3 * A_i * C_i * dEP - C_i * X_i * p' * dp$
 $- (J2(X_i)/BET_i)^{(R_i-1)} * X_i * dt$ (i=1 or 2)
 if BET_i is non zero, else:

$dX_i = 2/3 * A_i * C_i * dEP - C_i * X_i * p' * dp'$ (i=1 or 2)

with : $p' = PHI + (1-PHI) * e^{(-B * p)}$

$dp = < F/K >^{*N} * e^{(ALF * < F/K >^{*(N+1)})}$

$dR = B * (Q - R) * dp + GAMA * Sign(QR-R) * |QR-R|^{*M} * dt$

→ *Flow rule:* $dEP = 3/2 * dp * < S-X > / J2(S-X)$

→ *Memory of inelastic strain* $G = I2(EP-Y) - q < 0$

$dQ = 2 * MU * (QMAX-Q) * dq$

$dq = ETA * < n.nn > * dp$

$dY = (3/2 * 0.5) * (1-ETA) * < n.nn > * nn * dp$

$QR = Q - QSTA * (1 - ((QMAX-Q)/QMAX)^2)$

The initial value of Q is Q0 and must be initialized: $Q = Q0 = 30$ MPa (steel 316). For this purpose, a field of internal variables must be created, with a component named 'QQ', its value being Q0. This field will be passed into the table of PASAPAS.

The parameter values for steel 316L at 600°C are given for information only. The data to be input are the following:

<i>Threshold evolution law:</i>	Values for Steel 316

'N' : viscoplasticity threshold exponent	24
'KK' : initial value for elastic limit	10 MPa
'K0' : initial factor for the threshold normalization	116 MPa
'ALFK' : K isotropic evolution coefficient	1.5
'ALFR' : threshold isotropic evolution coefficient	0.35
'ALF' : viscosity coefficient	2.E6



4.2. TEST CHABOCHE2 (DESCRIPTION SHEET)

	Values for Steel 316
<i>Evolution law for the centres X1 and X2:</i>	
'A1' :	plastic strain coefficient 67.5 MPa
'C1' :	reference term coefficient 1300
'BET1' :	normalization factor for restoration 4807 MPa
'R1' :	restoration term exponent 4
'A2' :	plastic strain coefficient 80 MPa
'C2' :	reference term coefficient 45
'BET2' :	normalization factor for restoration 5840 MPa
'R2' :	restoration term exponent 4
'PHI' :	reference term multiplying coefficient 1.

<i>Evolution law for isotropic stress hardening:</i>	
'B' :	isotropic stress hardening coefficient 12
'GAMA' :	restoration effect coefficient 2.E-7
'M' :	restoration term exponent 2

<i>Evolution law for the plastic deformation memory:</i>	
'QMAX' :	Q maximal value 455 MPa
'QSTA' :	Q stabilized value 200 MPa
'MU' :	Q evolution law coefficient 19
'ETA' :	factor linking q with the plastic strain 0.06

The following parameters will have to be specified with the elastic behavior.

- > *Notation:*
- S stress tensor
 - X_i kinematic stress hardening variables ($i=1$ ou 2)
 - EP inelastic strain tensor
 - p accumulated equivalent inelastic strain
 - q isotropic variable for the memory surface in deformation
 - Y kinematic variable for the memory surface in deformation
 - I2 tensor second invariant
 - J2 deviatoric tensor second invariant
 - n normal from F criterion
 - nn normal from G threshold
 - t time
 - $\langle a \rangle$ positive part of a
 - x.y scalar product of x by y
 - X = X1 for a single centre
 - X1+X2 for two centres
 - RR = ALFR * R
 - K = K0 + ALFK * R
 - Q(0)=Q0

For further details, see D. NOUAILHAS's report:

"A viscoplastic modelling applied to stainless steel behavior", Second Inter. Conf. on Constitutive Laws for Engineering Materials,
University of Arizona, Tucson, 1987

4.3 Test ohno2 (Description sheet)

Test name: ohno2

Calculation type: ISOTROPIC VISCOPLASTIC MECHANICS - 2D AXIS

Finite element type: QUA8


Topic: Ohno -modified Chaboche-model for viscoplastic behaviour. The structure is an axisymmetrical hollow cylinder embedded at the lower surface. It is subjected to imposed load at the upper surface: the radial strength outwards. The cylinder follows Ohno's viscoplastic law.

Goal: Test the Ohno's model for solid structures.

Version: 97' customer version

Model description:


$$\begin{aligned}
 E &= 7.34 \cdot 10^{10} \\
 \text{NU} &= 0.33 \\
 \text{RHO} &= 7.8 \cdot 10^3 \\
 \text{ALPHA} &= 1. \cdot 10^{-5}
 \end{aligned}$$





Test ohno2 (Results)

CASTEM FIGURES



Test ohno2 (Listing CASTEM)

```

*****
TITR 'VISCOPLASTIC MODEL OF OHNO' ;
*****
OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
RAYON0 = .05          ; EPAISO = .0005 ;
L1 = EPAISO + RAYON0 ; L2 = .1 ;
N1 = 1                ; N2 = 200 ;
P1 = RAYON0 0.        ; P2 = L1 0. ;
P3 = L1 L2            ; P4 = RAYON0 L2 ;
D1 = D P1 N1 P2      ; D2 = D P2 N2 P3 ;
D3 = D P3 N1 P4      ; D4 = D P4 N2 P1 ;
VOL1 = DALL D1 D2 D3 D4 ;
TITR 'MESHING' ;
TRAC VOL1 ;
*
*---- DESCRIPTION OF THE MODEL OF MATERIAL ----*
MOD0 = MODE VOL1 MECANIQUE ELASTIQUE ISOTROPE
      VISCOPLASTIQUE OHNO ;
MAT0 = MATE MOD0 YOUN 7.34E10 NU 0.33
      RHO 7.8E3 ALPHA 1.E-5 N 24
      KK 10.E6 K0 116.E6
      ALFK 1.5 ALFR 0.35 ALF 2.E6
      A1 67.5E6 C1 1300 BET1 4807.E6 R1 4
      A2 80.E6 C2 45 BET2 58480.E6 R2 4 PHI 1.
      B 12 GAMA 2.E-7 M 2
      QMAX 455.E6 QSTA 200.E6 MU 19 ETA 0.06
      EXP1 2. EXP2 2. ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ D1 UR UZ RT RZ ;
CL0 = CL1 ;
*
*----- LOAD DEFINITION -----*
PRES0 = 1.E4 ;
P01 = ( RAYON0 + ( EPAISO / 2. ) ) L2 ;
P00 = VOL1 POIN PROC P01 ;
DEP1 = FORCE FR PRES0 P00 ;
EV0 = EVOL MANU TEMPS ( PROG 0. 10000. )
      Y ( PROG 0. 1. ) ;
CHA0 = CHAR MECA DEP1 EV0 ;
VEC1 = VECT DEP1 1.E-5 FR FZ ROUG ;
TITR 'APPLIED STRENGTH' ;
TRAC VEC1 VOL1 ;

*TIME OF CALCULATION
DT0 = 20. ;
TFIN0 = 700. ;
*
*----- RESOLUTION BY PASAPAS PROCEDURE -----*
TA1 = 'TABLE' ;
TA1 .'MODELE' = MOD0 ;
TA1 .'CARACTERISTIQUES' = MAT0 ;
TA1 .'BLOCAGES_MECANIQUES' = CL0 ;
TA1 .'CHARGEMENT' = CHA0 ;
TA1 .'TEMPS_CALCULES' = PROG 0. PAS DT0 TFIN0 ;

PASAPAS TA1 ;
*
*-----CALCULATION AND OUTPUT -----*
DIM0 = DIME ( TA1 .'TEMPS' ) ;
CHAR2 = PROG 0. ;
DEPL2 = PROG 0. ;
DEFI2 = PROG 0. ;

CONT1 = PROG 0. ;

REACO = TA1 .'REACTIONS'.( DIM0 - 1 ) ;
DEPLO = TA1 .'DEPLACEMENTS'.( DIM0 - 1 ) ;
VEC2 = VECT REACO 1.E-2 FR FZ ROUG ;
DEFO0 = DEFO VOL1 DEPLO 0. ROUG ;
DEFO1 = DEFO VOL1 DEPLO 1. ;
REAC1 = REDU REACO P1 ;
REAC2 = REDU REACO P2 ;
VEC01 = VECT REAC1 1.E-2 FR FZ ROUG ;
VEC02 = VECT REAC2 1.E-2 FR FZ VERT ;
TITR 'FINAL STRAIN' ;
TRAC ( DEFO0 ET DEFO1 ) ;

REPETER BLOC0 ( DIM0 - 1 ) ;
I0 = &BLOC0 ;
REACO = TA1 .'REACTIONS'.I0 ;
DEPLO = TA1 .'DEPLACEMENTS'.I0 ;
SIGMO = TA1 .'CONTRAINTE'.I0 ;
DEFI0 = TA1 .'VARIABLES_INTERNES'.I0 ;
TEMPO = TA1 .'TEMPS'.I0 ;
DEPL1 = EXTR DEPLO UR P3 ;
DEPL2 = DEPL2 ET ( PROG DEPL1 ) ;
DEFI1 = MAXI ( EXCO DEFI0 EPSE ) ;
DEFI2 = DEFI2 ET ( PROG DEFI1 ) ;
CHAR1 = PRES0 * TEMPO / TFIN0 ;
CHAR2 = CHAR2 ET ( PROG CHAR1 ) ;
FIN BLOC0 ;

*
FIN ;

```



Test ohno2 (Comments)

- The Ohno viscoplastic model

```

MOD0 = MODE VOL1 MECANIQUE ELASTIQUE ISOTROPE
      VISCOPLASTIQUE OHNO ;
MAT0 = MATE MOD0
      YOUN 7.34E10  NU 0.33
      RHO 7.8E3    ALPHA 1.E-5
      N 24        KK 10.E6    K0 116.E6
      ALFK 1.5     ALFR 0.35   ALF 2.E6
      A1 67.5E6   C1 1300    BET1 4807.E6  R1 4
      A2 80.E6    C2 45      BET2 58480.E6 R2 4
      PHI 1.      B 12      GAMA 2.E-7 M 2
      QMAX 455.E6 QSTA 200.E6 MU 19
      ETA 0.06    EXP1 2.     EXP2 2. ;
  
```

The Ohno model is close to the Chaboche viscoplastic one. The equations for this model are as follows:

→ *Criterion:* $F = J2(S-X) - RR - KK:$

→ *Strain hardening:*

$$dX_i = \frac{2}{3} * A_i * C_i * dEP - C_i * X_i * p' * < dEP : K_i > * (J2(X_i) / (J2(X_i) / BET_i) ** (R_i - 1) * X_i * dt \text{ (i=1 or 2)})$$

if BET_i is non zero, else:

$$dX_i = \frac{2}{3} * A_i * C_i * dEP - C_i * X_i * dp' \text{ (i=1 or 2)}$$

with : $p' = PHI + (1 - PHI) * e ** (-B * p)$

$$LIM_i = A_i / p' \text{ (i=1 ou 2)}$$

$$K_i = X_i / J2(X_i)$$

$$dp = < F / K > ** N * e ** (ALF * < F / K > ** (N + 1))$$

$$dR = B * (Q - R) * dp + GAMA * Sign(QR - R) * |QR - R| ** M * dt$$

→ *Flow rule:*

$$dEP = \frac{3}{2} * dp * < S - X > / J2(S - X)$$

→ *Memory of inelastic strain*

$$G = I2(EP - Y) - q < 0$$

$$dQ = 2 * MU * (QMAX - Q) * dq$$

$$dq = ETA * < n.nn > * dp$$

$$dY = (3/2 ** 0.5) * (1 - ETA) * < n.nn > * nn * dp$$

$$QR = Q - QSTA * (1 - ((QMAX - Q) / QMAX) ** 2)$$

The initial value of Q is Q0 and must be initialized: $Q = Q0 = 30$ MPa (steel 316). For this purpose, a field of internal variables must be created, with a component named 'QQ', its value being Q0. This field will be passed into the table of PASAPAS.

The parameter values for the steel 316L at 600°C are given for information only. The data to be input are the following:

	Values for Steel 316

<i>Threshold evolution law:</i>	
'N' :	viscoplasticity threshold exponent 24
'KK' :	initial value for elastic limit 10 MPa
'K0' :	initial factor for the threshold normalization 116 MPa
'ALFK' :	K isotropic evolution coefficient 1.5
'ALFR' :	threshold isotropic evolution coefficient 0.35
'ALF' :	viscosity coefficient 2.E6

	Values for Steel 316
<i>Evolution law for the centres X1 and X2:</i>	
'A1' :	plastic strain coefficient 67.5 MPa
'C1' :	reference term coefficient 1300
'BET1' :	normalization factor for restoration 4807 MPa
'R1' :	restoration term exponent 4
'A2' :	plastic strain coefficient 80 MPa
'C2' :	reference term coefficient 45
'BET2' :	normalization factor for restoration 5840 MPa
'R2' :	restoration term exponent 4
'PHI' :	reference term multiplying coefficient 1.

<i>Evolution law for isotropic stress hardening:</i>	
'B' :	isotropic stress hardening coefficient 12
'GAMA' :	restoration effect coefficient 2.E-7
'M' :	restoration term exponent 2

<i>Evolution law for the plastic deformation memory:</i>	
'QMAX' :	Q maximal value 455 MPa
'QSTA' :	Q stabilized value 200 MPa
'MU' :	Q evolution law coefficient 19
'ETA' :	factor linking q with the plastic strain 0.06

The following parameters will have to be specified with the elastic behavior.

- > *Notation:*
- S stress tensor
 - X_i kinematic stress hardening variables ($i=1$ ou 2)
 - EP inelastic strain tensor
 - p accumulated equivalent inelastic strain
 - q isotropic variable for the memory surface in deformation
 - Y kinematic variable for the memory surface in deformation
 - I2 tensor second invariant
 - J2 deviatoric tensor second invariant
 - n normal from F criterion
 - nn normal from G threshold
 - t time
 - $\langle a \rangle$ positive part of a
 - x.y scalar product of x by y
 - X = X1 for a single centre
 - X1+X2 for two centres
 - RR = ALFR * R
 - K = K0 + ALFK * R
 - Q(0)=Q0



4.4 Test guionnet_tra (Description sheet)

Test name: guionnet_tra

Calculation type: ISOTROPIC VISCOPLASTIC MECHANICS

Finite element type: CUB8


Topic: Guionnet model. The structure is a square plate embedded at the lower surface and at the upper surface. It is subjected to imposed displacements at the upper surface: the tensile strength. The plate follows a viscoplastic law according the Guionnet's model.

Goal: Test the Guionnet model for solid structures.

Version: 97' customer version

Model description:

$$\begin{aligned} E &= 192500 \\ \text{NU} &= 0.33 \end{aligned}$$





4.4. TEST GUIONNET_TRA (DESCRIPTION SHEET)

Test guionnet_tra (Listing CASTEM)

```
*****
TITR 'VISCOPLASTIC MODEL OF GUIONNET' ;
*****
OPTI DIME 3 ELEM CUB8 ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
EP0 = .01 ;
L1 = 1. ; L2 = 1. ;
P1 = 0. 0. 0. ; P2 = EP0 0. 0. ;
P3 = EP0 L2 0. ; P4 = 0. L2 0. ;
P10 = 0. 0. L1 ; P20 = EP0 0. L1 ;
P30 = EP0 L2 L1 ; P40 = 0. L2 L1 ;

LI1 = DROI P1 1 P2 ; LI2 = DROI P2 1 P3 ;
LI3 = DROI P3 1 P4 ; LI4 = DROI P4 1 P1 ;
LI10 = DROI P10 1 P20 ; LI20 = DROI P20 1 P30 ;
LI30 = DROI P30 1 P40 ; LI40 = DROI P40 1 P10 ;

SUR1 = DALL LI1 LI2 LI3 LI4 ;
SUR10 = DALL LI10 LI20 LI30 LI40 ;
MAIL1 = SUR1 VOLU 1 SUR10 ;
TITR 'MESHING' ;
TRACE OEIL MAIL1 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD0 = MODE MAIL1 MECANIQUE ELASTIQUE ISOTROPE
      VISCOPLASTIQUE GUIONNET ;
MAT0 = MATE MOD0 'YOUN' 192500. 'NU' 0.3
      'N' 24 'K' 200 'CK' 0.
      'R0' 25. 'CD' 0. 'M' 0.8
      'A' 24000. 'A1' 3.3E5 'C' 180.
      'C1' 40. 'G' 0. 'R' 0.
      'NN' 0.005 'C2' 40. 'Q' 1.466
      'G1' 0. 'R1' 0. 'BETA' 1.1
      'C0' 180. 'P1M0' 0.00333 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ SUR1 UZ ;
CL2 = BLOQ ( LI1 ET LI10 ) UY ;
CL3 = BLOQ ( LI2 ET LI20 ) UX ;
CL4 = BLOQ SUR10 UZ ;
CL0 = CL1 ET CL2 ET CL3 ET CL4 ;
*
*----- LOAD DEFINITION -----*
DEPMAX = .1 ;
DEP0 = DEPI CL4 DEPMAX ;
EVO = EVOL MANU TEMPS ( PROG 0. 1000. )
      Y ( PROG 0. 1. ) ;
CHA0 = CHAR MECA DEP0 EVO ;

DT0 = 10. ;
TFINO = 100. ;
PRO0 = PROG 0. PAS DT0 TFINO ;
*
*----- CALCULATION: PASAPAS PROCEDURE -----*
TA1 = 'TABLE' ;
TA1 .'MODELE' = MOD0 ;
TA1 .'CARACTERISTIQUES' = MAT0 ;
TA1 .'CHARGEMENT' = CHA0 ;
TA1 .'BLOCAGES_MECANIQUES' = CL0 ;
TA1 .'TEMPS_CALCULES' = PRO0 ;

PASAPAS TA1 ;
*
*----- CALCULATION AND OUTPUT -----*

DIM0 = 'DIME' TA1 .'TEMPS' ;

REACO = TA1 .'REACTIONS'.( DIM0 - 1 ) ;
DEPLO = TA1 .'DEPLACEMENTS'.( DIM0 - 1 ) ;
TEMPO = TA1 .'TEMPS'.( DIM0 - 1 ) ;
SIGMO = TA1 .'CONTRAINTE'.(DIM0 - 1 ) ;
EPS0 = 'EPSI' MOD0 DEPLO ;

DEFO0 = 'DEFO' MAIL1 DEPLO 0. ;
DEFO1 = 'DEFO' MAIL1 DEPLO 10. 'ROUG' ;
TITR ' FINAL DEFORMATION ' ;
TRACE ( DEFO0 ET DEFO1 ) ;

SIGMO = CHAN CHPO SIGMO MOD0 ;
TITR ' STRESS IN THE X DIRECTION' ;
TRACE ( EXCO SIGMO 'SMXX' ) MAIL1 ;
TITR ' STRESS IN THE Y DIRECTION' ;
TRACE ( EXCO SIGMO 'SMYY' ) MAIL1 ;
TITR ' STRESS IN THE Z DIRECTION' ;
TRACE ( EXCO SIGMO 'SMZZ' ) MAIL1 ;
TITR ' SHEAR STRESS IN THE YZ DIRECTION' ;
TRACE ( EXCO SIGMO 'SMYZ' ) MAIL1 ;
REACO = REDU REACO SUR10 ;
REACO = EXCO REACO FZ SCAL ;
CHP1 = MANU CHPO SUR10 1 SCAL 1. ;
REAC1 = XTY REACO CHP1 ( MOTS SCAL ) ( MOTS SCAL ) ;

MESS ' REACTION AT THE UPPER SURFACE = ' REAC1 ;
*
FIN ;
```

Test guionnet_tra (Comments)

- The Guionnet viscoplastic model

```

MOD0 = MODE MAIL1 MECANIQUE ELASTIQUE ISOTROPE
      VISCOPLASTIQUE GUIONNET ;
MAT0 = MATE MOD0 'YOUN' 192500.      'NU ' 0.3
      'N ' 24      'K ' 200      'CK ' 0.
      'R0 ' 25.    'CD ' 0.      'M ' 0.8
      'A ' 24000.  'A1 ' 3.3E5   'C ' 180.
      'C1 ' 40.    'G ' 0.      'R ' 0.
      'NN ' 0.005  'C2 ' 40.    'Q ' 1.466
      'G1 ' 0.    'R1 ' 0.      'BETA' 1.1
      'C0 ' 180.  'P1M0' 0.00333 ;
    
```

The equations for the Guionnet model are as follows:

→ *Criterion*: $F = J2(S-X) - R - KK$:

→ *Stress hardening*: $dX = M * p1^{**}(M-1) * [2/3 *(A*a2 + a1* EP.n) - (C1 - C*a2)* X.n] * dEP - (C0 * < pI - P1M0 > + C*a4) * X * dp - G * J2(X)**R * X * dt / 3 * Ai * Ci * dEP$

and yield $R = R0*(1-CD) + R0*CD*a2$
 $KK = K *(1+CK*a2)$
 $dEP = 3/2 * < F/KK > **N * n$

if $p1 > pI$: $da2 = da4 = NN * p1^{**}(NN-1) * dp$
 if $p1 < pI$: $da2 = C2*(Q * p1^{**}NN - a2) * dp - G1 * a2^{**}R1 * dt$
 if $p1 < pI$: $da4 = - BETA*a4*dp/p$

The parameter values for the steel 316L at 600°C are given for information only. The data to be input are the following:

	Values for Steel 316

<i>Threshold evolution law:</i>	
'N ' : viscoplasticity threshold exponent	20
'K ' : initial factor for threshold normalization	10 MPa
'CK ' : constant in K evolution law	3.87
'R0 ' : initial value for the elastic limit	80 MPa
'CD ' : constant in R evolution law	0.
 <i>Evolution law for the X centre:</i>	
'A ' : ALPHA2 coefficient	15000 MPa
'M ' : plastic strain exponent (<1)	0.8
'A1 ' : plastic strain coefficient	200000 MPa
'C ' : ALPHA4 coefficient	40
'C1 ' : reference term coefficient	180
'C0 ' : tuning coefficient for ratchetting	0.
'P1M0' : threshold for tuning	0.
'G ' : restoration term exponent	1.5E-10 /se
'R ' : restoration term exponent	4.



4.4. TEST GUIONNET_TRA (DESCRIPTION SHEET)

Evolution law for the ALPHA2 and ALPHA4 internal variables

'NN ' : plastic strain exponent	0.075
'C2 ' : plastic strain coefficient	4.
'Q ' : plastic strain coefficient	3.43
'G1 ' : restoration terme coefficient	1.5E-6
'R1 ' : ALPHA2 exponent	4.
'BETA' : ALPHA2 coefficient	0.4

The following parameters will have to be specified with the elastic behavior.

-> *Notation:* S stress tensor
X kinematic stress hardening variable
ai internal variables (i=2, 4)
EP inelastic strain tensor
p accumulated equivalent inelastic strain
p1 dp1 = dp but p1=0 at each load reversal
pI value of p1 at load reversal
J2 second invariant of deviatoric tensor
n normal from F criterion
t time
< a > positive part of a
x.y scalar product of x by y

For further details, refer to the CEA -N-2612 note.

4.5 Test zerill1 (Description sheet)

Test name: zerill1

Calculation type: ISOTROPIC ELASTOPLASTIC MECHANICS

Finite element type: CUB8



Topic: Zerilli-Armstrong model. The structure is a beam embedded at the lower and at the upper surface. It is subjected to tensile strength (imposed displacements) at the end. The beam follows an elastoplastic law according to the Zerilli-Armstrong's model.

Goal: Test the Zerilli-Armstrong model for BCC structure.

Version: 97' customer version

Model description:

$$\begin{aligned}
 E &= 2.1 \cdot 10^{11} \\
 \nu &= 0.3 \\
 \rho &= 7.8 \cdot 10^3 \\
 \alpha &= 1 \cdot 10^{-5}
 \end{aligned}$$





4.5. TEST ZERIL1 (DESCRIPTION SHEET)

Test zeril1 (Listing CASTEM)

```

*****
TITR 'ELASTOPLASTIC MODEL OF ZERILLI-ARMSTRONG' ;
*****
* APPLICATION TO BODY-CENTERED STRUCTURE
OPTI DIME 3 ELEM CUB8 ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
L1 = .5 ; L2 = .01 ; L3 = .01 ;
N1 = 50 ; N2 = 1 ; N3 = 1 ;
P1 = 0. 0. 0. ; P2 = L1 0. 0. ; P3 = L1 L2 0. ;
P4 = 0. L2 0. ; P5 = 0. 0. L3 ; P6 = L1 0. L3 ;
P7 = L1 L2 L3 ; P8 = 0. L2 L3 ;
D1 = DROI P1 N3 P5 ; D2 = DROI P5 N2 P8 ;
D3 = DROI P8 N3 P4 ; D4 = DROI P4 N2 P1 ;
SUR1 = DALL D1 D2 D3 D4 ;
D5 = DROI P2 N3 P6 ; D6 = DROI P6 N2 P7 ;
D7 = DROI P7 N3 P3 ; D8 = DROI P3 N2 P2 ;
SUR2 = DALL D5 D6 D7 D8 ;
VOL1 = SUR1 VOLU N1 SUR2 ;
TRAC VOL1 ;
*
*---- DESCRIPTION OF THE MODEL OF MATERIAL ----*
MOD0 = MODE VOL1 MECANIQUE ELASTIQUE ISOTROPE
        PLASTIQUE ZERILLI ;
MAT0 = MATE MOD0 YOUN 2.1E11 NU 0.3
        RHO 7.8E3 ALPHA 1.E-5 DYG 46.5E6
        C1 1033.E6 C2 890.E6 C3 (300. * 698.E-5)
        C4 ( 300. * 415.E-6 ) C5 266.E6
        N 0.289 K 6.957E5 L 1E-3 TYPE 1. ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ SUR1 UX ;
CL2 = BLOQ SUR2 UX ;
CL3 = BLOQ P1 UX UY UZ ;
CL4 = BLOQ P5 UX UY ;
CL0 = CL1 ET CL2 ET CL3 ET CL4 ;
*
*----- LOAD DEFINITION -----*
DEP1 = DEPI CL2 .1 ;
EV0 = EVOL MANU TEMPS ( PROG 0. .01 )
        Y ( PROG 0. 1. ) ;
CHA0 = CHAR DIMP DEP1 EV0 ;
*
* TIME OF CALCULATION
DT0 = .00005 ;
TFINO = .0005 ;
*
*----- CALCULATION: PASAPAS PROCEDURE -----*
TA1 = 'TABLE' ;
TA1 .'MODELE' = MOD0 ;
TA1 .'CARACTERISTIQUES' = MAT0 ;
TA1 .'BLOCAGES_MECANIQUES' = CL0 ;
TA1 .'CHARGEMENT' = CHA0 ;
TA1 .'TEMPS_CALCULES' = PROG 0. PAS DT0 TFINO ;
PASAPAS TA1 ;
*
*----- CALCULATION AND OUTPUT -----*
DIM0 = DIME TA1 .'TEMPS' - 1 ;
YOUNG0 = EXCO MAT0 YOUNG SCAL ;
YOUNG1 = MAXI YOUNG0 ;
DYG0 = EXCO MAT0 DYG SCAL ;
DYG1 = MAXI DYG0 ;
C1 = EXCO MAT0 C1 SCAL ; C11 = MAXI C1 ;
C2 = EXCO MAT0 C2 SCAL ; C21 = MAXI C2 ;
C3 = EXCO MAT0 C3 SCAL ; C31 = MAXI C3 ;
C4 = EXCO MAT0 C4 SCAL ; C41 = MAXI C4 ;
C5 = EXCO MAT0 C5 SCAL ; C51 = MAXI C5 ;
XK0 = EXCO MAT0 K SCAL ; XK1 = MAXI XK0 ;
XL0 = EXCO MAT0 L SCAL ; XL1 = MAXI XL0 ;
XN0 = EXCO MAT0 N SCAL ; XN1 = MAXI XN0 ;
TYPE0 = EXCO MAT0 TYPE SCAL ;
TYPE1 = MAXI TYPE0 ;
EPS_EL0 = ( DYG1 + ( XK1 * ( XL1 ** (-.5) ) ) ) /
        YOUNG1 ;
ERR_E1 = 0. ;
ERR_S1 = 0. ;
REPETER BLOC1 DIM0 ;
I1 = &BLOC1 ;
DEPL1 = EXCO TA1 .'DEPLACEMENTS'.I1 UX SCAL ;
DEPL0 = EXCO TA1 .'DEPLACEMENTS'.(I1 - 1 )
        UX SCAL ;
DEP1 = MAXI DEPL1 ;
EPS1 = DEP1 / L1 ;
DEP0 = MAXI DEPL0 ;
SIGM0 = TA1 .'CONTRAINTES'.I1 ;
DEPEQ0 = TA1 .'VARIABLES_INTERNES'.I1 ;
DEF0 = TA1 .'DEFORMATIONS_INELASTIQUES'.I1 ;
T1 = TA1 .'TEMPS'.I1 ;
T0 = TA1 .'TEMPS'.(I1 - 1) ;
DT0 = T1 - T0 ;
DT0 = 0. ;
SI ( ( ABS DT0 ) < 1.E-10 ) ;
DEPS0 = 0. ;
SINON ;
DEPS0 = ( DEP1 - DEP0 ) / ( DT0 * L1 ) ;
FINSI ;
SIGX0 = MAXI ( VMIS MOD0 SIGM0 ) ;
SIGX00 = MINI ( VMIS MOD0 SIGM0 ) ;
DEPX0 = MAXI DEPEQ0 ;
DEPX00 = MINI DEPEQ0 ;
SIG_00 = EXCO SIGM0 SMXX SCAL ;
SIG_0 = MAXI SIG_00 ;
SIG_Y = MAXI ( EXCO SIGM0 SMYY SCAL ) ;
SIG_Z = MAXI ( EXCO SIGM0 SMZZ SCAL ) ;
DEF_00 = EXCO DEF0 EPXX ;
DEF_0 = MAXI DEF_00 ;
* BODY-CENTERED STRUCTURE
SI ( TYPE1 EGA 0. ) ;
SI ( SIGX0 >EG ( DYG1 + ( XK1 *
        ( XL1 ** (-.5) ) ) ) ) ;
SI ( ( ABS DEPS0 ) < 1.E-10 ) ;
B0 = 0. ;
SINON ;
B0 = C21 * ( EXP ( ( C41 * ( LOG DEPS0 ) ) -
        C31 ) ) ;
FINSI ;
A0 = YOUNG1 ;
C0 = DYG1 + ( XK1 * ( XL1 ** (-.5) ) ) -
        ( YOUNG1 * EPS1 ) ;
DELTA = ( B0 * B0 ) - ( 4. * A0 * C0 ) ;
X1 = ( -1. * B0 - ( ( DELTA ) ** ( 0.5 ) ) ) /
        ( 2. * A0 ) ;
X2 = ( -1. * B0 + ( ( DELTA ) ** ( 0.5 ) ) ) /
        ( 2. * A0 ) ;
EPSP0 = X2 * X2 ;
SIG0 = YOUNG1 * ( EPS1 - EPSP0 ) ;
SINON ;
EPSP0 = 0. ;
SIG0 = SIG_0 ;
FINSI ;

```

```

SINON ;
* BODY-CENTERED STRUCTURE
LIM0 = DYG1 + ( XK1 * ( XL1 ** (-.5) ) ) ;
SI ( ( ABS DEPS0 ) < 1.E-10 ) ;
LIM0 = LIM0 + 0. ;
SINON ;
LIM0 = LIM0 +
( C11 * ( EXP ( ( C41 * ( LOG DEPS0 ) ) - C31 ) ) ) ;
FINSI ;
SI ( SIG_0 > LIM0 ) ;
A0 = YOUNG1 ;
SI ( ( ABS DEPS0 ) < 1.E-10 ) ;
B0 = 0. ;
SINON ;
B0 = C11 * ( EXP ( ( C41 * ( LOG DEPS0 ) )
- C31 ) ) ;
FINSI ;
B0 = B0 + DYG1 + ( XK1 * ( XL1 ** (-.5) ) )
- ( YOUNG1 * EPS1 ) ;
C0 = C51 ;

X0 = 0. ;
X2 = X0 ;
Y2 = A0 * X2 ;
Y2 = Y2 + ( C0 * ( X2 ** XN1 ) ) ;
Y2 = Y2 + B0 ;

X1 = EPS1 ;
X2 = X1 ;
Y2 = A0 * X2 ;
Y2 = Y2 + ( C0 * ( X2 ** XN1 ) ) ;
Y2 = Y2 + B0 ;

REPETER BLOC2 ;

X2 = ( X0 + X1 ) / 2. ;
Y2 = A0 * X2 ;
Y2 = Y2 + ( C0 * ( X2 ** XN1 ) ) ;
Y2 = Y2 + B0 ;

SI ( Y2 > 1.E-5 ) ;
X1 = X2 ;
SINON ;
SI ( Y2 < -1.E-5 ) ;
X0 = X2 ;
SINON ;
EPSP0 = X2 ;
SIG0 = YOUNG1 * ( EPS1 - X2 ) ;
QUITTER BLOC2 ;
FINSI ;
FINSI ;

FIN BLOC2 ;

SINON ;
EPSP0 = 0. ;
SIG0 = SIG_0 ;
FINSI ;

FINSI ;

ERR_S0 = ABS ( SIG0 - SIG_0 ) ;
ERR_S0 = ERR_S0 / SIG0 ;
ERR_S0 = ERR_S0 * 100. ;
ERR_E0 = ABS ( EPSP0 - DEF_0 ) ;
SI ( DEF_0 > ( 1.E-1 * EPS_EL0 ) ) ;
ERR_E0 = ERR_E0 / DEF_0 ;
SINON ;

ERR_E0 = 0. ;
FINSI ;
ERR_E0 = ERR_E0 * 100. ;
*----- GOOD WORKING MESSAGE -----*
SI ( ERR_E0 > 1. ) ;
ERRE 5 ;
ERR_E1 = 100. ;
SINON ;
ERR_E1 = ERR_E1 + 0. ;
FINSI ;
SI ( ERR_S0 > 1. ) ;
ERRE 5 ;
ERR_S1 = 100. ;
SINON ;
ERR_S1 = ERR_S1 + 0. ;
'FINSI' ;

FIN BLOC1 ;

SI ( ( ERR_E1 < 1. ) ET ( ERR_S1 < 1. ) ) ;
ERRE 0 ;
FINSI ;

FIN ;

```

**Test zerill1 (Comments)**

- The Zerilli-Armstrong elastoplastic model

```

MOD0 = MODE VOL1 MECANIQUE ELASTIQUE ISOTROPE
      PLASTIQUE ZERILLI ;
MAT0 = MATE MOD0 YOUN 2.1E11 NU 0.3
      RHO 7.8E3 ALPHA 1.E-5 DYG 46.5E6
      C1 1033.E6 C2 890.E6 C3 (300. * 698.E-5)
      C4 ( 300. * 415.E-6 ) C5 266.E6
      N 0.289 K 6.957E5 L 1E-3 TYPE 1. ;

```

The Zerilli-Armstrong model proposes a mathematical formulation of the Von Mises's yielding stress based on the theory of dislocations. The constitutive equations of this model are as follow:

-Yielding limit Y for Face Centered Cubic materials (F.C.C.):

$$Y = DYG + C2' \cdot \sqrt{P'} \cdot \exp(-C3' \cdot T + C4' \cdot T \cdot \ln(EPT)) + K \cdot L^{**(-1/2)}$$

-Yielding limit Y for Body Centered Cubic materials (B.C.C.):

$$Y = DYG + C1' \cdot \exp(-C3' \cdot T + C4' \cdot T \cdot \ln(EPT)) + C5' \cdot (P')^{**N} + K \cdot L^{**(-.5)}$$

with:

T: temperature

P': equivalent plastic strain: $P' = \sqrt{2/3 \cdot EP:EP}$

EP: plastic strains

EPT: equivalent strain rate: $EPT = \sqrt{2/3 \cdot ET:ET}$

ET: strain rate

The parameters to be input in the Zerilli-Armstrong model are DYG, Ci, N, K, L as well as TYPE:

```

'DYG'   : DYG term
'C1'    : C1' coefficient
'C2'    : C2' coefficient
'C3'    : C3'.T term (C3' multiplied by T) entered as an EVOLUTION object
         (the parameter is the temperature)
'C4'    : C4'.T term (C4' multiplied by T) entered as an EVOLUTION object
         (the parameter is the temperature)
'C5'    : C5' coefficient
'N'     : N coefficient
'K'     : K coefficient
'L'     : average grain diameter
'TYPE'  : material structure
         if the material is FCC: TYPE=0.
         if the material is BCC: TYPE=1.

```

4.6 Test preston1 (Description sheet)

Test name: preston1

Calculation type: ISOTROPIC ELASTOPLASTIC MECHANICS

Finite element type: CUB8


Topic: Preston Tonks Wallace model. The structure is a beam embedded at the lower and upper surface. It is subjected to tensile strength (imposed displacements) at the end. The beam follows an elastoplastic law according to the Preston-Tonks-Wallace model.

Goal: Test the Preston Tonks Wallace model for BCC 3 structure.

Version: 97' customer release

Model description:

$$\begin{aligned}
 E &= 2.1 \cdot 10^{11} \\
 \nu &= 0.3 \\
 \rho &= 7.8 \cdot 10^3 \\
 \tau &= 0.025
 \end{aligned}$$



MODEL OF PRESTON TONKS WALLACE



4.6. TEST PRESTON1 (DESCRIPTION SHEET)

Test preston1 (Listing CASTEM)

```

*****
TITR 'MODEL OF PRESTON TONKS WALLACE' ;
*****
* APPLICATION TO BODY-CENTERED STRUCTURE
COMPLET = FAUX ; COMM * FULL CALCULATIONIF VRAI ;
OPTI DIME 3 ELEM CUB8 ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
L1 = .5 ; L2 = .001 ; L3 = .001 ;
N1 = 50 ; N2 = 1 ; N3 = 1 ;
P1 = 0. 0. 0. ; P2 = L1 0. 0. ; P3 = L1 L2 0. ;
P4 = 0. L2 0. ; P5 = 0. 0. L3 ; P6 = L1 0. L3 ;
P7 = L1 L2 L3 ; P8 = 0. L2 L3 ;
D1 = DROI P1 N3 P5 ; D2 = DROI P5 N2 P8 ;
D3 = DROI P8 N3 P4 ; D4 = DROI P4 N2 P1 ;
SUR1 = DALL D1 D2 D3 D4 ;
D5 = DROI P2 N3 P6 ; D6 = DROI P6 N2 P7 ;
D7 = DROI P7 N3 P3 ; D8 = DROI P3 N2 P2 ;
SUR2 = DALL D5 D6 D7 D8 ;
VOL1 = SUR1 VOLU N1 SUR2 ;
TRAC VOL1 ;
*
*----- DESCRIPTION OF MODEL MATERIAL -----*
MOD0 = MODE VOL1 MECANIQUE ELASTIQUE ISOTROPE
PLASTIQUE PRESTON ;
MAT0 = MATE MOD0 YOUN 2.1E11 NU 0.3
RHO 7.8E3 TAU 0.025
P 0. S0 0.0085 SINF 0.00055
K 0. G 0.00001
Y0 0.0001 YINF 0.0001 Y1 0.094 Y2 0.575
BETA 0.25 GP 3.00 GT 0. MU 1.97
TMO 1220. ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ SUR1 UX ; CL2 = BLOQ SUR2 UX ;
CL3 = BLOQ P1 UX UY UZ ; CL4 = BLOQ P5 UX UY ;
CL0 = CL1 ET CL2 ET CL3 ET CL4 ;
*
*----- LOAD DEFINITION -----*
DEP1 = DEPI CL2 .1 ;
EVO = EVOL MANU TEMPS ( PROG 0. 5.E-4 )
Y ( PROG 0. 1. ) ;
CHA0 = CHAR DIMP DEP1 EVO ;
*
* TIME OF CALCULATION
DT0 = 1.E-5 ;
SI COMPLET ;
TFINO = 5.E-4 ;
SINON ;
TFINO = 4.E-5 ;
FINSI ;
*
*----- CALCULATION: PASAPAS PROCEDURE -----*
TA1 = 'TABLE' ;
TA1 .'MODELE' = MOD0 ;
TA1 .'CARACTERISTIQUES' = MAT0 ;
TA1 .'BLOCAGES_MECANIQUES' = CL0 ;
TA1 .'CHARGEMENT' = CHA0 ;
*TA1 .'PRECISION' = 1.E-8 ;
TA1 .'TEMPS_CALCULES' = PROG 0. PAS DT0 TFINO ;
PASAPAS TA1 ;
*
*----- CALCULATION AND OUTPUT -----*
SIG_3DY = PROG 0. ; SIG_3DZ = PROG 0. ;
SIGX1 = PROG 0. ; SIGX2 = PROG 0. ;
DEPX1 = PROG 0. ; DEPX2 = PROG 0. ;
SIG_1 = PROG 0. ; DEF_1 = PROG 0. ;
EPSP1 = PROG 0. ;
ERR1 = PROG 0. ; ERR2 = PROG 0. ;
SIG1 = PROG 0. ;
TEMP1 = PROG 0. ;
EPST1 = PROG 0. ;
DIM0 = DIME TA1 .'TEMPS' - 1 ;
MOD0 = TA1 .'MODELE' ;
YOUNG0 = EXCO MAT0 YOUNG SCAL ;
YOUNG1 = MAXI YOUNG0 ;
XNU0 = EXCO MAT0 NU SCAL ;
XNU1 = MAXI XNU0 ;
RHO0 = MAXI ( EXCO MAT0 RHO SCAL ) ;
TAU0 = EXCO MAT0 TAU SCAL ;
TAU1 = MAXI TAU0 ;
P1 = EXCO MAT0 P SCAL ;
P11 = MAXI P1 ;
S0 = EXCO MAT0 S0 SCAL ;
S01 = MAXI S0 ;
SINF0 = EXCO MAT0 SINF SCAL ;
SINF1 = MAXI SINF0 ;
K0 = EXCO MAT0 K SCAL ;
K1 = MAXI K0 ;
G0 = EXCO MAT0 G SCAL ;
G1 = MAXI G0 ;
Y00 = EXCO MAT0 Y0 SCAL ;
Y01 = MAXI Y00 ;
YINF0 = EXCO MAT0 YINF SCAL ;
YINF1 = MAXI YINF0 ;
Y10 = EXCO MAT0 Y1 SCAL ;
Y11 = MAXI Y10 ;
Y20 = EXCO MAT0 Y2 SCAL ;
Y21 = MAXI Y20 ;
BETA0 = EXCO MAT0 BETA SCAL ;
BETA1 = MAXI BETA0 ;
GP0 = EXTR ( EXCO MAT0 GP SCAL ) SCAL 1 1 1 ;
GT0 = EXTR ( EXCO MAT0 GT SCAL ) SCAL 1 1 1 ;
A1 = 2. * ( 1. + XNU1 ) ;
B1 = A1 / ( 3. * ( 1. - ( 2. * XNU1 ) ) ) ;
REPETER BLOC1 DIM0 ;
I1 = &BLOC1 ;
DEPL0 = TA1 .'DEPLACEMENTS'.I1 ;
TEMPO = TA1 .'TEMPS'.I1 ;
TEMP1 = TEMP1 ET ( PROG TEMPO ) ;
DEX1 = EXCO DEPL0 UX SCAL ;
DEX0 = EXCO TA1 .'DEPLACEMENTS'.( I1 - 1 )
UX SCAL ;
REACO = TA1 .'REACTIONS'.I1 ;
REAC1 = MAXI ( EXCO REACO FX SCAL ) ;
DEP0 = MAXI DEX0 ;
DEP1 = MAXI DEX1 ;
EPS1 = EPSI MOD0 DEPL0 ;
EPSX1 = EXCO EPS1 EPXX SCAL ;
EPSY1 = EXCO EPS1 EPYY SCAL ;
EPSZ1 = EXCO EPS1 EPZZ SCAL ;
TREPS0 = MAXI ( EPSX1 + EPSY1 + EPSZ1 ) ;
EPSX10 = MAXI EPSX1 ;
SIGM0 = TA1 .'CONSTRAINTES'.I1 ;
DEPEQ0 = TA1 .'VARIABLES_INTERNES'.I1 ;
DEF0 = TA1 .'DEFORMATIONS_INELASTIQUES'.I1 ;
T1 = TA1 .'TEMPS'.I1 ;
T0 = TA1 .'TEMPS'.(I1 - 1) ;
DT0 = T1 - T0 ;
DT0 = 0. ;
SI ( ( ABS DT0 ) < 1.E-10 ) ;
DEPS0 = 0. ;
SINON ;
DEPS0 = ( DEP1 - DEP0 ) / ( DT0 * L1 ) ;
FINSI ;

```

```

SIGX0 = MAXI ( VMIS MOD0 SIGM0 ) ;
SIGX00 = MINI ( VMIS MOD0 SIGM0 ) ;
DEPX0 = MAXI ( EXCO DEPEQ0 EPSE SCAL ) ;
DEPX00 = MINI ( EXCO DEPEQ0 EPSE SCAL ) ;
SIG_00 = EXCO SIGM0 SMXX SCAL ;
SIG_0 = MAXI SIG_00 ;
EPS_EL0 = SIG_0 / YOUNG1 ;
SIG_Y = MAXI ( EXCO SIGM0 SMYY SCAL ) ;
SIG_Z = MAXI ( EXCO SIGM0 SMZZ SCAL ) ;
DEF_00 = EXCO DEF0 EPXX ;
DEF_0 = MAXI DEF_00 ;
DEF1 = MAXI ( EXCO DEF0 EPXX ) ;
DEF2 = MAXI ( EXCO DEF0 EPYY ) ;
DEF3 = MAXI ( EXCO DEF0 EPZZ ) ;
DEF4 = MAXI ( EXCO DEF0 GAXY ) ;
DEF5 = MAXI ( EXCO DEF0 GAXZ ) ;
DEF6 = MAXI ( EXCO DEF0 GAYZ ) ;
DEFP1 = ( DEF1 * DEF1 ) + ( DEF2 * DEF2 ) +
( DEF3 * DEF3 ) ;
DEFP1 = DEFP1 - ( DEF1 * DEF2 ) -
( DEF2 * DEF3 ) - ( DEF1 * DEF3 ) ;
DEFP2 = ( DEF4 * DEF4 ) + ( DEF5 * DEF5 ) +
( DEF6 * DEF6 ) ;
DEFP0 = DEFP1 + ( 3. * DEFP2 ) ;
DEFP0 = ( DEFP0 ** .5 ) * 2. / 3. ;
SIGX1 = SIGX1 ET ( PROG SIGX0 ) ;
SIGX2 = SIGX2 ET ( PROG SIGX00 ) ;
DEPX1 = DEPX1 ET ( PROG DEPX0 ) ;
DEPX2 = DEPX2 ET ( PROG DEPX00 ) ;
SIG_1 = SIG_1 ET ( PROG SIG_0 ) ;
DEF_1 = DEF_1 ET ( PROG DEF_0 ) ;
SIG_3DY = SIG_3DY ET ( PROG ( SIG_Y / SIG_0 ) ) ;
SIG_3DZ = SIG_3DZ ET ( PROG ( SIG_Y / SIG_0 ) ) ;
EPST1 = EPST1 ET ( PROG EPSX10 ) ;

* DENSITY RHO1
RHO1 = RHO0 / ( 1. + TREPS0 ) ;
ETA = RHO1 / RHO0 ;

* SHEAR MODULUS G00
G00 = YOUNG1 / ( 2. * ( 1. + XNU1 ) ) ;

* DEBYE'S PULSATION W0
W0 = ( G00 / RHO1 ) ** ( .5 ) ;
DTETA0 = ( 4. / 3.1416 ) ** ( .5 ) ;
DTETA0 = DTETA0 * W0 / 6. ;
DTETA0 = DEPS0 / DTETA0 ;

* STRESS OF SATURATION
SI ( ( ABS DT0 ) < 1.E-10 ) ;
YS1 = SINFL1 ;
SINON ;
YS1 = ERF ( K1 * ( LOG ( G1 / DTETA0 ) ) ) ;
YS1 = S01 - ( ( S01 - SINFL1 ) * YS1 ) ;
FINSI ;
YS2 = S01 * ( ( DTETA0 / G1 ) ** ( BETA1 ) ) ;
YS0 = ( ABS ( YS1 - YS2 ) ) + YS1 + YS2 ;
YS0 = YS0 / 2. ;

* YIELD POINT
SI ( ( ABS DT0 ) < 1.E-10 ) ;
YL1 = YINFL1 ;
SINON ;
YL1 = ERF ( K1 * ( LOG ( G1 / DTETA0 ) ) ) ;
YL1 = Y01 - ( ( Y01 - YINFL1 ) * YL1 ) ;
FINSI ;
YL2 = Y11 * ( ( DTETA0 / G1 ) ** ( Y21 ) ) ;
YL3 = ( ABS ( YL2 - YS2 ) ) - YL2 - YS2 ;
YL3 = -1. * YL3 / 2. ;
YL0 = ( ABS ( YL1 - YL3 ) ) + YL1 + YL3 ;
YL0 = YL0 / 2. ;

SI ( SIG_0 > ( YL0 * G00 ) ) ;

* BODY CENTERED MATERIAL
SI ( P11 EGA 0. ) ;
X0 = 0. ;
X2 = X0 ;
TR1 = TREPS0 - ( A1 * ( EPSX10 - X2 ) /
( 3. * B1 ) ) ;
TR1 = TR1 / 3. ;
Y2 = EXP ( -1. * TAU1 * ( X2 - TR1 ) /
( YS0 - YL0 ) ) ;
Y2 = YS0 - ( ( YS0 - YL0 ) * Y2 ) ;
Y2 = Y2 * G00 ;
Y2 = Y2 + ( YOUNG1 * X2 ) ;
Y2 = Y2 - ( YOUNG1 * EPSX10 ) ;
X1 = EPSX10 ;
X2 = X1 ;
TR1 = TREPS0 - ( A1 * ( EPSX10 - X2 ) /
( 3. * B1 ) ) ;
TR1 = TR1 / 3. ;
Y2 = EXP ( -1. * TAU1 * ( X2 - TR1 ) /
( YS0 - YL0 ) ) ;
Y2 = YS0 - ( ( YS0 - YL0 ) * Y2 ) ;
Y2 = Y2 * G00 ;
Y2 = Y2 + ( YOUNG1 * X2 ) ;
Y2 = Y2 - ( YOUNG1 * EPSX10 ) ;

REPETER BLOC2 ;
X2 = ( X1 + X0 ) / 2. ;
TR1 = TREPS0 - ( A1 * ( EPSX10 - X2 ) /
( 3. * B1 ) ) ;
TR1 = TR1 / 3. ;
Y2 = EXP ( -1. * TAU1 * ( X2 - TR1 ) /
( YS0 - YL0 ) ) ;
Y2 = YS0 - ( ( YS0 - YL0 ) * Y2 ) ;
Y2 = Y2 * G00 ;
Y2 = Y2 + ( YOUNG1 * X2 ) ;
Y2 = Y2 - ( YOUNG1 * EPSX10 ) ;
SI ( Y2 > 1.E-5 ) ;
X1 = X2 ;
SINON ;
SI ( Y2 < -1.E-5 ) ;
X0 = X2 ;
SINON ;
EPSP0 = X2 ;
EPSP1 = EPSP1 ET ( PROG EPSP0 ) ;
DEN0 = A1 * GP0 * ( EPSX10 - X2 ) / 3. ;
DEN0 = DEN0 / ( ETA ** ( 1. / 3. ) ) ;
SUP0 = A1 * ( G00 + GT0 ) *
( EPSX10 - X2 ) ;
SIG0 = SUP0 / ( 1. + DEN0 ) ;
SIG1 = SIG1 ET ( PROG SIG0 ) ;
QUITTER BLOC2 ;
FINSI ;

FINSI ;
FIN BLOC2 ;
SINON ;

* OTHERS MATERIALS
X0 = 0. ;
X2 = X0 ;
EX0 = P11 * ( YS0 - YL0 ) / ( S01 - YL0 ) ;
EX1 = ( S01 - YL0 ) * ( EXP EX0 - 1. ) ;
TR1 = TREPS0 - ( A1 * ( EPSX10 - X2 ) /
( 3. * B1 ) ) ;
TR1 = TR1 / 3. ;
Y2 = EXP ( -1. * P11 * TAU1 *
( X2 - TR1 ) / EX1 ) ;
Y2 = ( 1. - ( EXP ( -1. * EX0 ) ) ) * Y2 ;
Y2 = LOG ( 1. - Y2 ) ;

```




4.6. TEST PRESTON1 (DESCRIPTION SHEET)

```
Y2 = ( ( S01 - YL0 ) * Y2 / P11 ) + YS0 ;
Y2 = Y2 * G00 ;
Y2 = Y2 + ( YOUNG1 * X2 ) ;
Y2 = Y2 - ( YOUNG1 * EPSX10 ) ;
Y0 = Y2 ;

X1 = EPSX10 ;
X2 = X1 ;
TR1 = TREPS0 - ( A1 * ( EPSX10 - X2 ) /
  ( 3. * B1 ) ) ;
TR1 = TR1 / 3. ;
Y2 = EXP ( -1. * P11 * TAU1 *
  ( X2 - TR1 ) / EX1 ) ;
Y2 = ( 1. - ( EXP (-1. * EX0) ) ) * Y2 ;
Y2 = LOG ( 1. - Y2 ) ;
Y2 = ( ( S01 - YL0 ) * Y2 / P11 ) + YS0 ;
Y2 = Y2 * G00 ;
Y2 = Y2 + ( YOUNG1 * X2 ) ;
Y2 = Y2 - ( YOUNG1 * EPSX10 ) ;
Y1 = Y2 ;

REPETER BLOC2 ;
X2 = ( X0 + X1 ) / 2. ;
TR1 = TREPS0 - ( A1 * ( EPSX10 - X2 ) /
  ( 3. * B1 ) ) ;
TR1 = TR1 / 3. ;
Y2 = EXP ( -1. * P11 * TAU1 *
  ( X2 - TR1 ) / EX1 ) ;
Y2 = ( 1. - ( EXP (-1. * EX0) ) ) * Y2 ;
Y2 = LOG ( 1. - Y2 ) ;
Y2 = ( ( S01 - YL0 ) * Y2 / P11 ) + YS0 ;
Y2 = Y2 * G00 ;
Y2 = Y2 + ( YOUNG1 * X2 ) ;
Y2 = Y2 - ( YOUNG1 * EPSX10 ) ;

SI ( Y2 > ( 1.E-8 * ( ABS(Y1 - Y0) ) ) ) ;
  X1 = X2 ;
SINON ;
SI ( Y2 < (-1.E-8 * ( ABS(Y0 - Y1) ) ) ) ;
  X0 = X2 ;
SINON ;
  EPSP0 = X2 ;
  EPSP1 = EPSP1 ET ( PROG EPSP0 ) ;
  DEN0 = A1 * GP0 * ( EPSX10 - X2 ) / 3. ;
  DEN0 = DEN0 / ( ETA ** ( 1. / 3. ) ) ;
  SUP0 = A1 * ( G00 + GT0 ) *
    ( EPSX10 - X2 ) ;
  SIG0 = SUP0 / ( 1. + DEN0 ) ;
  SIG1 = SIG1 ET ( PROG SIG0 ) ;

QUITTER BLOC2 ;
  FINSI ;
  FINSI ;
FIN BLOC2 ;
FINSI ;

SINON ;
  EPSP0 = 0. ;
  EPSP1 = EPSP1 ET ( PROG EPSP0 ) ;
  SIG0 = SIG_0 ;
  SIG1 = SIG1 ET ( PROG SIG0 ) ;
FINSI ;

ERR_S0 = ABS ( SIG0 - SIG_0 ) ;
ERR_S0 = ERR_S0 / SIG0 ;
ERR_S0 = ERR_S0 * 100. ;
ERR1 = ERR1 ET ( PROG ERR_S0 ) ;
ERR_E0 = ABS ( EPSP0 - DEF_0 ) ;
SI ( DEF_0 > ( 1.E-1 * EPS_E0 ) ) ;
  ERR_E0 = ERR_E0 / DEF_0 ;
SINON ;

ERR_E0 = 0. ;
FINSI ;
ERR_E0 = ERR_E0 * 100. ;
SI ( ERR_E0 > 5.E-2 ) ;
  ERRE 5 ;

FINSI ;
SI ( ERR_S0 > 5.E-2 ) ;
  ERRE 5 ;

FINSI ;
ERR2 = ERR2 ET ( PROG ERR_E0 ) ;

FIN BLOC1 ;

FIN ;
```

Test preston1 (Comments)

- The Preston-Tonks-Wallace elastoplastic model

```

MOD0 = MODE VOL1 MECANIQUE ELASTIQUE ISOTROPE
      PLASTIQUE PRESTON ;
MAT0 = MATE MOD0 YOUN 2.1E11 NU 0.3
      RHO 7.8E3 TAU 0.025
      P 0. S0 0.0085 SINP 0.00055
      K 0. G 0.00001
      Y0 0.0001 YINF 0.0001 Y1 0.094 Y2 0.575
      BETA 0.25 GP 3.00 GT 0. MU 1.97
      TMO 1220. ;
    
```

Preston, Tonks and Wallace propose a mathematical formulation of the Von Mises's yielding stress based on the theory of the dislocations and the dimensional analysis. The constitutive equations of the model are:

-shear modulus G:

$$G = G_0 + G_P' \cdot P / (\text{ETA}^{**}(1/3)) + G_T' \cdot (T - 300)$$

$$G_0 = \text{YOUNG}_0 / (2 \cdot (1 + \text{NU}_0))$$

where: YOUNG0 is the initial Young modulus

NU0 is the Poisson's ratio (stay constant)

G0 is the initial shear modulus

ETA is the compression rate: $\text{ETA} = \text{RHO} / \text{RHO}_0$

$$\text{TM} = \text{TMO} \cdot \text{EXP}(2 \cdot \text{MU}_0 \cdot (1 - 1/\text{ETA})) / (\text{ETA}^{**}(2/3))$$

and: if $T > \text{TM}$:

$$G = Y = 0 \quad (Y \text{ is the yield limit})$$

then, the volumetric strain is fully elastic and the deviatoric strain is fully plastic

-Dimensionless terms:

$$Y' = Y/G$$

$$T' = T/\text{TM}$$

$$\text{EPT}' = \text{EPT}/X$$

with:

Y yielding stress

G shear modulus

T temperature

TM melting temperature

$$X = 1/6 \cdot (4/\text{PI})^{**}(.5) \cdot \text{OMEGA}$$

OMEGA: Debye's pulsation $\text{OMEGA} = (G/\text{RHO})^{**}(.5)$

RHO: material's density

$$\text{EPT} = \text{sqrt}(2/3 \cdot \text{ET} \cdot \text{ET})$$

ET: total strain rate

-Dimensionless term for the saturation stress YS:

$$S1 = S_0 - (S_0 - \text{SINF}) \cdot \text{erf}(K' \cdot T' \cdot \ln(g/\text{EPT}'))$$

$$S2 = S_0 \cdot (\text{EPT}'/g)^{**}\text{BETA}$$

$$YS = \max(S1, S2)$$

-Dimensionless term for the yielding limit YL:

$$L1 = Y_0 - (Y_0 - \text{YINF}) \cdot \text{erf}(K' \cdot T' \cdot \ln(g/\text{EPT}'))$$

$$L2 = Y1 \cdot (\text{EPT}'/g)^{**}Y2$$

$$YL = \max(L1, \min(L2, S2))$$

-Dimensionless term for the yielding stress for Body Centered Cubic materials (B.C.C.): $P=0$



4.6. TEST PRESTON1 (DESCRIPTION SHEET)

$$Y' = YS - (YS - YL) \cdot \exp(-TAU \cdot EP / (YS - YL))$$

-Dimensionless term for the yielding stress for the other materials: P differs from 0

$$\text{Coeff1} = (S0 - YL) \cdot (\exp(P \cdot (YS - YL) / (S0 - YL)) - 1)$$

$$\text{Coeff2} = 1 - \exp(-P \cdot (YS - YL) / (S0 - YL))$$

$$Y' = YS + (S0 - YL) / P \cdot \ln(1 - \text{Coeff2} \cdot \exp(-P \cdot TAU \cdot EP / \text{Coeff1}))$$

with:

EP: equivalent plastic strain

$$EP = \sqrt{2/3} \cdot EPS : EPS$$

EPS: plastic strain

The shear modulus G is expressed that way:

$$\text{STRESS} = \text{HOOK} \cdot \text{STRAIN} \text{ i.e. } D_STRESS = \text{HOOK} \cdot D_STRAIN + D_HOOK \cdot \text{STRAIN}$$

- 'RHO' : initial density
- 'TAU' : dimensionless parameter TAU used for the calculation of the hardening law
- 'P' : dimensionless parameter P
 - if P=0, the microstructure of the material is BCC
 - else, the microstructure is NOT BCC
- 'S0' : dimensionless parameter S0
 - it gives the saturation stress at $T = 0^\circ K$
- 'SINF' : dimensionless parameter SINF
 - it gives the saturation stress at $T(^{\circ}K)$ infinite
- 'K' : dimensionless term $K' \cdot T$ used for the calculation of the yielding limit. It is entered as an EVOLUTION object (the parameter is the temperature T)
- 'G' : dimensionless parameter g used for the calculation of the saturation stress and the yielding limit
- 'Y0' : dimensionless parameter Y0
 - it gives the yielding limit at $T = 0^\circ K$
- 'YINF' : dimensionless parameter YINF
 - it gives the yielding limit at $T(^{\circ}K)$ infinite
- 'Y1' : dimensionless Y1 used for the calculation of the yielding limit
- 'Y2' : dimensionless Y2 used for the calculation of the yielding limit
- 'BETA' : dimensionless parameter BETA used for the calculation of the yielding limit and the saturation stress
- 'GP' : rate of the evolution of the shear modulus with the pressure (no unity): GP'
- 'GT' : temperature-dependant term correcting the shear modulus (homogeneous to the shear modulus) One enters here the term: $GT' \cdot (T - 300)$ with the use of an EVOLUTION object (the parameter of this EVOLUTION is the temperature).
- 'TMO' : melting temperature for $\text{ETA} = 1$
- 'MU' : coefficient MU0 used for the calculation of the melting temperature



Chapter 5

DAMAGE ELASTOPLASTIC MODELS

5.1 Test endocp1 (Description sheet)

Test name: endocp1

Calculation type: DAMAGE MODEL FOR ELASTOPLASTIC MATERIAL

Finite element type: QUA8

Topic: Tensile test on a damage elastoplastic material. The structure is an embedded test piece subjected to a tensile stress (imposed displacements). The structure obeys a law according to a damage elastoplastic model.

Goal: Test the model of damage material for MODE PLAN CONT and compare Castem's results with the analytical solution obtained by algorithm. We also compute:
 The cumulated plastic strain (EPSE component in VARIABLES_INTERNES).
 The damage variable D in VARIABLES_INTERNES.
 SIGMA_{yy} (in CONTRAINTES).


Reference: Analytical solution.

Tests carried out by Mr STRUB (rapport DMT/91 .558).

Version: 97' customer version

Model description:


E	=	74020 10 ⁶
NU	=	0.3
RHO	=	7800.
ALPH	=	0.
DC	=	0.23
EPSD	=	0.03
EPSR	=	0.25






Test endocp1 (Results)

CASTEM FIGURES



DAMAGE ELASTOPLASTIC MODEL



DAMAGE ELASTOPLASTIC MODEL

Test endocp1 (Listing CASTEM)

```

*****
TITR 'DAMAGE ELASTOPLASTIC MODEL' ;
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
GRAPH = N ;
OPTI DIME 2 ELEM QUA8 MODE PLAN CONT ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
P1 = 0      0      ; P2 = 5E-3  0      ;
P3 = 5E-3  30E-3  ; P4 = 0      30E-3  ;
L1 = P1 P2 DROI 1 ; L2 = P2 P3 DROI 2 ;
L3 = P3 P4 DROI 1 ; L4 = P4 P1 DROI 2 ;
EPROU = L1 L2 L3 L4 DALL PLAN ;
TRAC EPROU ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ L1 UY ; CL2 = BLOQ L3 UY ;
CL3 = BLOQ L4 UX ;
CL = CL1 ET CL2 ET CL3 ;

DEP = DEPI CL2 1E-3 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO = MODE EPROU MECANIQUE ELASTIQUE PLASTIQUE
      ENDOMMAGEABLE;

PROGEPS = PROG 0. .47285E-2 .76066E-2
      .89226E-2 1.0533E-2 1.4011E-2 1.9096E-2 2.6081E-2
      3.4654E-2 4.5828E-2 6.0813E-2 7.2835E-2 8.3328E-2
      9.5526E-2 10.691E-2 11.808E-2 12.784E-2 13.891E-2
      14.936E-2 15.700E-2 16.746E-2 17.720E-2 18.946E-2
      19.860E-2 20.703E-2 21.607E-2 22.802E-2 23.685E-2
      24.919E-2 26.083E-2 ;
PROGSIG = PROG 0. 350.E06 370.59E06
      379.18E06 387.37E06 400.52E06 414.22E06 427.94E06
      440.69E06 453.63E06 467.26E06 476.30E06 483.24E06
      490.46E06 496.56E06 502.06E06 506.53E06 511.28E06
      515.51E06 518.45E06 522.29E06 525.71E06 529.81E06
      532.73E06 535.33E06 538.03E06 541.47E06 543.92E06
      547.23E06 550.24E06 ;

CTRAC = EVOL MANU EPS PROGEPS SIG PROGSIG ;
DESS CTRAC;

MAT = MATE MO YOUN 74020.E06 NU 0.3 RHO 7800.
      ALPH 0. TRAC CTRAC EPSD 0.03
      DC 0.23 EPSR 0.25 ;
*
*----- LOAD DEFINITION -----*
LI1 = PROG 0. 30. ; LI2 = PROG 0. 30. ;
EV = EVOL MANU T LI1 DEP(T) LI2 ;
CHA = CHAR 'DIMP' DEP EV ;
*
*----- LISTING OF TIME STEPS -----*
*          0.13 MM ELASTIC INCREMENT
*          0.16 MM PLASTIC INCREMENT
SI COMPLET ;
      LIS = PROG 0.13 0.16 0.33 0.5 0.75 1. 1.25 1.5
      1.75 2. 2.25 2.5 2.75 3. 3.25 3.5
      3.75 4. 4.25 4.5 4.75 5. 5.25 5.5
      5.75 6. 6.25 6.5 6.75 7. 7.25 7.5
      7.75 8. ;
SINON ;
      LIS = PROG 0.13 0.16 0.33 0.5 ;
FINSI ;

TAB = TABLE ;
TAB.'BLOCAGES_MECANIQUES' = CL ;

TAB.'MODELE' = MO ;
TAB.'CARACTERISTIQUES' = MAT ;
TAB.'TEMPS_CALCULES' = LIS ;
TAB.'CHARGEMENT' = CHA ;
TAB.'PRECISION' = 1E-08 ;

PASAPAS TAB ;
*
*----- GOOD WORKING TEST -----*
DPC0 = EXTR TAB.VARIABLES_INTERNES.2 EPSE 1 1 1 ;
DPC1 = EXTR TAB.VARIABLES_INTERNES.4 EPSE 1 1 1 ;
DPC0T = 5.46E-4 ; DPC1T = 1.12694E-02 ;
SI COMPLET ;
DPC2 = EXTR TAB.VARIABLES_INTERNES.10 EPSE 1 1 1 ;
DPC4 = EXTR TAB.VARIABLES_INTERNES.18 EPSE 1 1 1 ;
DPC6 = EXTR TAB.VARIABLES_INTERNES.26 EPSE 1 1 1 ;
DPC75 = EXTR TAB.VARIABLES_INTERNES.32 EPSE 1 1 1 ;
DPC2T = 6.03E-2 ; DPC4T = 12.65E-2 ;
DPC6T = 19.29E-2 ; DPC75T = 24.27E-2 ;
FINSI ;

D0 = EXTR TAB.VARIABLES_INTERNES.2 'D ' 1 1 1 ;
D1 = EXTR TAB.VARIABLES_INTERNES.4 'D ' 1 1 1 ;
D0T = 0. ; D1T = 0.0 ;
SI COMPLET ;
D2 = EXTR TAB.VARIABLES_INTERNES.10 'D ' 1 1 1 ;
D4 = EXTR TAB.VARIABLES_INTERNES.18 'D ' 1 1 1 ;
D6 = EXTR TAB.VARIABLES_INTERNES.26 'D ' 1 1 1 ;
D75 = EXTR TAB.VARIABLES_INTERNES.32 'D ' 1 1 1 ;
D2T = 0.0257 ; D4T = 0.0901 ;
D6T = 0.1622 ; D75T = 0.2199 ;
FINSI ;

SZZ0 = EXTR TAB.CONTRAINTE.2 SMYY 1 1 1 ;
SZZ1 = EXTR TAB.CONTRAINTE.4 SMYY 1 1 1 ;
SZZ0T = 354.3E6 ; SZZ1T = 4.07984E+08 ;
SI COMPLET ;
SZZ2 = EXTR TAB.CONTRAINTE.10 SMYY 1 1 1 ;
SZZ4 = EXTR TAB.CONTRAINTE.18 SMYY 1 1 1 ;
SZZ6 = EXTR TAB.CONTRAINTE.26 SMYY 1 1 1 ;
SZZ75 = EXTR TAB.CONTRAINTE.32 SMYY 1 1 1 ;
SZZ2T = 459.3E6 ; SZZ4T = 461.4E6 ;
SZZ6T = 443.3E6 ; SZZ75T = 422.4E6 ;
FINSI ;

EDPC0 = ABS( (DPC0-DPC0T) / DPC0T ) ;
EDPC1 = ABS( (DPC1-DPC1T) / DPC1T ) ;
SI COMPLET ;
EDPC2 = ABS( (DPC2-DPC2T) / DPC2T ) ;
EDPC4 = ABS( (DPC4-DPC4T) / DPC4T ) ;
EDPC6 = ABS( (DPC6-DPC6T) / DPC6T ) ;
EDPC75 = ABS( (DPC75-DPC75T) / DPC75T ) ;
FINSI ;
ED0 = ABS( (D0-D0T) / (D0T+0.001) ) ;
ED1 = ABS( (D1-D1T) / (D1T+0.001) ) ;
SI COMPLET ;
ED2 = ABS( (D2-D2T) / D2T ) ;
ED4 = ABS( (D4-D4T) / D4T ) ;
ED6 = ABS( (D6-D6T) / D6T ) ;
ED75 = ABS( (D75-D75T) / D75T ) ;
FINSI ;
ESZZ0 = ABS( (SZZ0-SZZ0T) / SZZ0T ) ;
ESZZ1 = ABS( (SZZ1-SZZ1T) / SZZ1T ) ;
SI COMPLET ;
ESZZ2 = ABS( (SZZ2-SZZ2T) / SZZ2T ) ;

ESZZ4 = ABS( (SZZ4-SZZ4T) / SZZ4T ) ;
ESZZ6 = ABS( (SZZ6-SZZ6T) / SZZ6T ) ;
ESZZ75 = ABS( (SZZ75-SZZ75T) / SZZ75T ) ;

```




5.1. TEST ENDOCP1 (DESCRIPTION SHEET)

```
FINSI ;
SI COMPLET ;
LERR = PROG EDPC0 EDPC1 EDPC2 EDPC4 EDPC6
      EDPC75 ED0 ED1 ED2 ED4 ED6 ED75
      ESZZ0 ESZZ1 ESZZ2 ESZZ4 ESZZ6 ESZZ75 ;
SINON ;
LERR = PROG EDPC0 EDPC1 ED0 ED1 ESZZ0 ESZZ1 ;
FINSI ;
LIST LERR ;
ERMAX = MAXI LERR ;
LIST ERMAL ;

TEMPS ;
SI ( ERMAL <EG 0.05 ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;

FIN ;
```

Test endocpl (Comments)

- The damage model for elastoplastic material (Lemaitre-Chaboche)

```

MO = MODE EPROU MECANIQUE ELASTIQUE PLASTIQUE
      ENDOMMAGEABLE ;
MAT = MATE MO YOUN 74020.E06 NU 0.3 RHO 7800.
      ALPH 0. TRAC CTRAC EPSD 0.03
      DC 0.23 EPSR 0.25 ;
  
```

The Lemaitre-Chaboche model assume that stress hardening and damage are isotropic. The Von Mises' criterion is coupled with damage. The following parameters will be specified BESIDES those relative to the elastic behavior.

- 'TRAC' : Key word followed by:
- NOMTRAC : Tension curve consisting of an EVOLUTIO type object with the strains in abscissa and the stresses in ordinate. It must contain the origin. It may be drawn by means of the DESSINE instruction.
- 'EPSD' : Damage threshold: it is the plastic deformation above which the material gets damaged.
- 'DC ' : Critical value of the D variable describing damage. DC features the material rupture.
- 'EPSR' : Plastic deformation corresponding to material rupture.



5.2 Test endoaxi1 (Description sheet)

Test name: endoaxi1

Calculation type: DAMAGE MODEL FOR ELASTOPLASTIC MATERIAL

Finite element type: QUA8

Topic: Tensile test on a damage elastoplastic material. The structure is an embedded test piece subjected to a tensile stress (imposed displacements). The structure obeys a law according to a damage elastoplastic model.

Goal: Test the model of damage material for MODE AXIS and compare Castem's results with the analytical solution obtained by algorithm.

We also compute:

The cumulated plastic strain (EPSE component in VARIABLES_INTERNES).

The damage variable D in VARIABLES_INTERNES.

SIGMAzz (in CONTRAINTES).


Reference: Analytical solution.

Tests carried out by Mr STRUB (rapport DMT/91 .558).

Version: 97' customer version


Model description:

E	=	74020 10 ⁶
NU	=	0.3
RHO	=	7800.
ALPH	=	0.
DC	=	0.23
EPSD	=	0.03
EPSR	=	0.25




Test endoaxi1 (Results)

CASTEM FIGURES



DAMAGE ELASTOPLASTIC MODEL



DAMAGE ELASTOPLASTIC MODEL



5.2. TEST ENDOAXI1 (DESCRIPTION SHEET)

Test endoaxil (Listing CASTEM)

```

*****
TITR 'DAMAGE ELASTOPLASTIC MODEL' ;
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
GRAPH = N ;
OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
OPTI TRAC BENS ;
*
*----- GEOMETRY -----*
P1 = 0      0      ; P2 = 3E-3  0      ;
P3 = 3E-3  30E-3  ; P4 = 0      30E-3  ;
L1 = P1 P2 DROI 1 ; L2 = P2 P3 DROI 3 ;
L3 = P3 P4 DROI 1 ; L4 = P4 P1 DROI 3 ;
EPROU = L1 L2 L3 L4 DALL PLAN ;
TRAC EPROU ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ L1 UZ ; CL2 = BLOQ L3 UZ ;
CL3 = BLOQ L4 UR ;
CL  = CL1 ET CL2 ET CL3 ;

DEP = DEPI CL2 1E-3 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO = MODE EPROU MECANIQUE ELASTIQUE PLASTIQUE
      ENDOMMAGEABLE;

PROGEPS = PROG 0. .47285E-2 .76066E-2
.89226E-2 1.0533E-2 1.4011E-2 1.9096E-2 2.6081E-2
3.4654E-2 4.5828E-2 6.0813E-2 7.2835E-2 8.3328E-2
9.5526E-2 10.691E-2 11.808E-2 12.784E-2 13.891E-2
14.936E-2 15.700E-2 16.746E-2 17.720E-2 18.946E-2
19.860E-2 20.703E-2 21.607E-2 22.802E-2 23.685E-2
24.919E-2 26.083E-2 ;
PROGSIG = PROG 0. 350.E06 370.59E06
379.18E06 387.37E06 400.52E06 414.22E06 427.94E06
440.69E06 453.63E06 467.26E06 476.30E06 483.24E06
490.46E06 496.56E06 502.06E06 506.53E06 511.28E06
515.51E06 518.45E06 522.29E06 525.71E06 529.81E06
532.73E06 535.33E06 538.03E06 541.47E06 543.92E06
547.23E06 550.24E06 ;

CTRAC = EVOL MANU EPS PROGEPS TRAC PROGSIG ;
DESS CTRAC ;

MAT = MATE MO YOUN 74020.E06 NU 0.3 RHO 7800.
      ALPH 0. TRAC CTRAC EPSD 0.03
      DC 0.23 EPSR 0.25 ;
*
*----- LOAD DEFINITION -----*
LI1 = PROG 0. 30. ; LI2 = PROG 0. 30. ;
EV = EVOL MANU T LI1 DEP(T) LI2 ;
CHA = CHAR 'DIMP' DEP EV ;
*
*----- LISTING OF TIME STEPS -----*
*
*          0.13 MM ELASTIC INCREMENT
*          0.16 MM PLASTIC INCREMENT
SI COMPLET ;
      LIS = PROG 0.13 0.16 0.5 1. 1.25 1.5 2. 2.5 3.
              3.5 4. 4.5 5. 5.5 6. 6.5 7. 7.5 8. ;
SINON ;
      LIS = PROG 0.13 0.16 0.5 ;
FINSI ;

TAB = TABLE ;

TAB.'BLOCAGES_MECANIQUES' = CL ;
TAB.'MODELE' = MO ;

```

```

TAB.'CARACTERISTIQUES' = MAT ;
TAB.'TEMPS_CALCULES' = LIS ;
TAB.'CHARGEMENT' = CHA ;
TAB.'PRECISION' = 1E-08 ;

PASAPAS TAB ;
*
*----- GOOD WORKING TEST -----*
DPC0 = EXTR TAB.VARIABLES_INTERNES.2 EPSE 1 1 1 ;
DPC1 = EXTR TAB.VARIABLES_INTERNES.3 EPSE 1 1 1 ;
DPC0T = 5.46E-4 ; DPC1T = 1.12694E-02;
SI COMPLET ;
DPC2 = EXTR TAB.VARIABLES_INTERNES.7 EPSE 1 1 1 ;
DPC4 = EXTR TAB.VARIABLES_INTERNES.11 EPSE 1 1 1 ;
DPC6 = EXTR TAB.VARIABLES_INTERNES.15 EPSE 1 1 1 ;
DPC75 = EXTR TAB.VARIABLES_INTERNES.18 EPSE 1 1 1 ;

DPC2T = 6.03E-2 ; DPC4T = 12.65E-2 ;
DPC6T = 19.29E-2 ; DPC75T = 24.27E-2 ;
FINSI ;
D0 = EXTR TAB.VARIABLES_INTERNES.2 'D ' 1 1 1 ;
D1 = EXTR TAB.VARIABLES_INTERNES.3 'D ' 1 1 1 ;
D0T = 0. ; D1T = 0.0 ;
SI COMPLET ;
D2 = EXTR TAB.VARIABLES_INTERNES.7 'D ' 1 1 1 ;
D4 = EXTR TAB.VARIABLES_INTERNES.11 'D ' 1 1 1 ;
D6 = EXTR TAB.VARIABLES_INTERNES.15 'D ' 1 1 1 ;
D75 = EXTR TAB.VARIABLES_INTERNES.18 'D ' 1 1 1 ;
D2T = 0.0257 ; D4T = 0.0901 ;
D6T = 0.1622 ; D75T = 0.2199 ;
FINSI ;

SZZ0 = EXTR TAB.CONTRAINTE.2 SMZZ 1 1 1 ;
SZZ1 = EXTR TAB.CONTRAINTE.3 SMZZ 1 1 1 ;
SZZ0T = 354.3E6 ; SZZ1T = 4.07984E+08 ;
SI COMPLET ;
SZZ2 = EXTR TAB.CONTRAINTE.7 SMZZ 1 1 1 ;
SZZ4 = EXTR TAB.CONTRAINTE.11 SMZZ 1 1 1 ;
SZZ6 = EXTR TAB.CONTRAINTE.15 SMZZ 1 1 1 ;
SZZ75 = EXTR TAB.CONTRAINTE.18 SMZZ 1 1 1 ;

SZZ2T = 459.3E6 ; SZZ4T = 461.4E6 ;
SZZ6T = 443.3E6 ; SZZ75T = 422.4E6 ;
FINSI ;

EDPC0 = ABS( (DPC0-DPC0T) / DPC0T ) ;
EDPC1 = ABS( (DPC1-DPC1T) / DPC1T ) ;
SI COMPLET ;
EDPC2 = ABS( (DPC2-DPC2T) / DPC2T ) ;
EDPC4 = ABS( (DPC4-DPC4T) / DPC4T ) ;
EDPC6 = ABS( (DPC6-DPC6T) / DPC6T ) ;
EDPC75 = ABS( (DPC75-DPC75T) / DPC75T ) ;
FINSI ;
ED0 = ABS( (D0-D0T) / (D0T+0.001) ) ;
ED1 = ABS( (D1-D1T) / (D1T+0.001) ) ;
SI COMPLET ;
ED2 = ABS( (D2-D2T) / D2T ) ;
ED4 = ABS( (D4-D4T) / D4T ) ;
ED6 = ABS( (D6-D6T) / D6T ) ;
ED75 = ABS( (D75-D75T) / D75T ) ;
FINSI ;

ESZZ0 = ABS( (SZZ0-SZZ0T) / SZZ0T ) ;
ESZZ1 = ABS( (SZZ1-SZZ1T) / SZZ1T ) ;

SI COMPLET ;
ESZZ2 = ABS( (SZZ2-SZZ2T) / SZZ2T ) ;
ESZZ4 = ABS( (SZZ4-SZZ4T) / SZZ4T ) ;
ESZZ6 = ABS( (SZZ6-SZZ6T) / SZZ6T ) ;
ESZZ75 = ABS( (SZZ75-SZZ75T) / SZZ75T ) ;

FINSI ;

```

```
SI COMPLET ;
LERR = PROG EDPC0 EDPC1 EDPC2 EDPC4 EDPC6
        EDPC75 ED0 ED1 ED2 ED4 ED6 ED75
        ESZZ0 ESZZ1 ESZZ2 ESZZ4 ESZZ6 ESZZ75 ;
SINON ;
LERR = PROG EDPC0 EDPC1 ED0 ED1 ESZZ0 ESZZ1 ;
FINSI ;
LIST LERR ;

ERMAX = MAXI LERR ;
LIST ERMAS ;
TEMPS ;
SI ( ERMAS <EG 0.05 ) ;
    ERRE 0 ;
SINON ;
    ERRE 5 ;
FINSI ;

FIN ;
```



5.3 Test endoaxi2 (Description sheet)

Test name: endoaxi2

Calculation type: DAMAGE MODEL FOR ELASTOPLASTIC MATERIAL

Finite element type: QUA8

Topic: Tensile test and thermal load on a damage elastoplastic material. The structure is an embedded test piece subjected to a tensile stress (imposed displacements) and a thermal load. The structure obeys a law according to a damage elastoplastic model. Furthermore the behaviour of the material depends on the temperature through its tensile curve and thermal expansion coefficient.


Goal: Test the damage material model for MODE AXIS and compare Castem's results with the analytical solution obtained by algorithm. We also compute:
The cumulated plastic strain (EPSE component in VARIABLES_INTERNES).
The damage variable D in VARIABLES_INTERNES.
SIGMAzz (in CONTRAINTES).

Reference: Analytical solution.

Version: 97' customer version



Model description:

E	=	74020 10 ⁶
NU	=	0.3
RHO	=	7800.
ALPH	=	variable
DC	=	0.23
EPSD	=	0.
EPSR	=	0.25




Test endoaxi2 (Results)

CASTEM FIGURES






5.3. TEST ENDOAXI2 (DESCRIPTION SHEET)



DAMAGE ELASTOPLASTIC MODEL



DAMAGE ELASTOPLASTIC MODEL

Test endoaxi2 (Listing CASTEM)

```

*****
TITR 'DAMAGE ELASTOPLASTIC MODEL' ;
*****
OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
P1 = 0 0 ; P2 = 3E-3 0 ;
P3 = 3E-3 30E-3 ; P4 = 0 30E-3 ;
L1 = P1 P2 DROI 1 ; L2 = P2 P3 DROI 1 ;
L3 = P3 P4 DROI 1 ; L4 = P4 P1 DROI 1 ;
EPROU = L1 L2 L3 L4 DALL PLAN ;
TRAC EPROU ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO = MODE EPROU MECANIQUE ELASTIQUE PLASTIQUE
      ENDOMMAGEABLE ;
* Tangent modulus equal to 783 MPa
PROGEPS1=PROG 0. (350./74020.) 0.3 0.6 ;
PROGSIG1=PROG 0. 350.E06 581.1976E06 816.0976E06 ;
CTRAC1 = EVOL MANU EPS PROGEPS1 SIG PROGSIG1 ;
* Tangent modulus equal to 491 MPa
PROGEPS2=PROG 0. (350./74020.) 0.3 0.6 ;
PROGSIG2=PROG 0. 350.E06 494.9783E06 642.2783E06 ;
CTRAC2 = EVOL MANU EPS PROGEPS2 SIG PROGSIG2 ;
* Tangent modulus equal to 199 MPa
PROGEPS3=PROG 0. (350./74020.) 0.3 0.6 ;
PROGSIG3=PROG 0. 350.E06 408.7590E06 468.4590E06 ;
CTRAC3 = EVOL MANU EPS PROGEPS3 SIG PROGSIG3 ;
* variable tangent modulus
CTRAC = NUAGE T*FLOTTANT TRAC*EVOLUTION
      0. CTRAC1 100. CTRAC2 200. CTRAC3 ;
DESS CTRAC1 ;
DESS CTRAC2 ;
DESS CTRAC3 ;
* variable thermic expansion coefficient
PROGTEMP = PROG 0. 100. 200. ;
PROGALFA = PROG 0. 1E-3 2E-3 ;
EVALPH = EVOL MANU T PROGTEMP ALPH PROGALFA ;
DESS EVALPH ;

MATVAR = MATE MO YOUN 74020.E06 NU 0.3 RHO 7800.
      ALPH EVALPH TRAC CTRAC EPSD 0. DC 0.23
      EPSR 0.25 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ L1 UZ ; CL2 = BLOQ L3 UZ ;
CL3 = BLOQ L4 UR ;
CL = CL1 ET CL2 ET CL3 ;
DEP = DEPI CL2 1E-3 ;
*
*----- LOAD DEFINITION -----*
LI1 = PROG 0. 30. ; LI2 = PROG 0. 30. ;
EV = EVOL MANU T LI1 DEP(T) LI2 ;
CHA1 = CHAR DIMP DEP EV ;
TEMP0 = MANU CHPO EPROU 1 T 0. ;
TEMP200 = MANU CHPO EPROU 1 T 200. ;
T0 = 0. ; T20 = 20. ;
TEMP = TABLE; TEMPERA = TABLE ;
TEMP.0 = T0 ; TEMP.1 = T20 ;
TEMPERA.0 = TEMP0; TEMPERA.1 = TEMP200 ;

CHA2 = CHAR T TEMP TEMPERA ;
CHA = CHA1 ET CHA2 ;
*
*----- LISTING OF TIME STEPS -----*
* 0.14 MM ELASTIC INCREMENT
* 0.15 MM PLASTIC INCREMENT

```

```

LIS = PROG 0.14 0.15 0.2 0.4 0.7 1
      PAS .3 3.1 3.3 PAS .2 4.1 ;

TAB = TABLE ;
TAB.'PRECISION' = 1E-3 ;
TAB.'BLOCAGES_MECANIQUE' = CL ;
TAB.'CARACTERISTIQUES' = MATVAR ;
TAB.'CHARGEMENT' = CHA ;
TAB.'TEMPS_CALCULES' = LIS ;
TAB.'MODELE' = MO ;
PASAPAS TAB ;
*
*----- GOOD WORKING TEST -----*
*----- COMPARISON WITH THE ANALYTIC SOLUTION -----*
REF_D =PROG 2.197E-2 3.417E-2 6.337E-2 8.131E-2 ;
REF_P =PROG 3.646E-2 5.443E-2 9.296E-2 1.143E-1 ;
REF_S =(PROG 368.9 376.0 386.4 389.2)*1E6 ;

D1 = EXTR (TAB.VARIABLES_INTERNES.7) 'D ' 1 1 1 ;
D2 = EXTR (TAB.VARIABLES_INTERNES.9) 'D ' 1 1 1 ;
D3 = EXTR (TAB.VARIABLES_INTERNES.14) 'D ' 1 1 1 ;
D4 = EXTR (TAB.VARIABLES_INTERNES.18) 'D ' 1 1 1 ;

P1 = EXTR (TAB.VARIABLES_INTERNES.7) 'EPSE' 1 1 1 ;
P2 = EXTR (TAB.VARIABLES_INTERNES.9) 'EPSE' 1 1 1 ;
P3 = EXTR (TAB.VARIABLES_INTERNES.14) 'EPSE' 1 1 1 ;
P4 = EXTR (TAB.VARIABLES_INTERNES.18) 'EPSE' 1 1 1 ;

S1 = EXTR (TAB.CONTRAINTE.7) 'SMZZ' 1 1 1 ;
S2 = EXTR (TAB.CONTRAINTE.9) 'SMZZ' 1 1 1 ;
S3 = EXTR (TAB.CONTRAINTE.14) 'SMZZ' 1 1 1 ;
S4 = EXTR (TAB.CONTRAINTE.18) 'SMZZ' 1 1 1 ;

EF_D = PROG D1 D2 D3 D4 ;
EF_P = PROG P1 P2 P3 P4 ;
EF_S = PROG S1 S2 S3 S4 ;

PROG_ERD = MAXI (ABS ((EF_D - REF_D) / REF_D)) ;
PROG_ERP = MAXI (ABS ((EF_P - REF_P) / REF_P)) ;
PROG_ERS = MAXI (ABS ((EF_S - REF_S) / REF_S)) ;

ERR = MAXI (PROG PROG_ERD PROG_ERP PROG_ERS) ;

TEMPS ;
SI ( ERR <EG 0.05 ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;
FIN ;

```



5.4 Test endoaxi3 (Description sheet)

Test name: endoaxi3

Calculation type: DAMAGE MODEL FOR ELASTOPLASTIC MATERIAL

Finite element type: QUA8

Topic: Tensile test and thermal load on a damage elastoplastic material. The structure is an embedded test piece subjected to a tensile stress (imposed displacements) and a thermal load. The structure obeys a law according to a damage elastoplastic model. Furthermore the behaviour of the material depends on temperature through its tensile curve and thermal expansion coefficient.

Reference temperature	=	30°C
Reference temperature for the thermic expansion coefficients	=	10°C
Initial temperature	=	50°C

Goal: Test the damage material model for MODE AXIS and compare Castem's results with the analytical solution obtained by algorithm. We also compute:

The cumulated plastic strain (EPSE component in VARIABLES_INTERNES).

The damage variable D in VARIABLES_INTERNES.


SIGMAzz (in CONTRAINTES).

Reference: Analytical solution.

Version: 97' customer version



Model description:

E	=	74020 10 ⁶
NU	=	0.3
RHO	=	7800.
ALPH	=	variable
DC	=	0.23
EPSD	=	0.
EPSR	=	0.25





Test endoaxi3 (Results)

CASTEM FIGURES





5.4. TEST ENDOAXI3 (DESCRIPTION SHEET)



Test endoaxi3 (Listing CASTEM)

```

*****
TITR 'DAMAGE ELASTOPLASTIC MODEL' ;
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
GRAPH = N ;
OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
P1 = 0 0 ; P2 = 3E-3 0 ;
P3 = 3E-3 30E-3 ; P4 = 0 30E-3 ;
L1 = P1 P2 DROI 1 ; L2 = P2 P3 DROI 1 ;
L3 = P3 P4 DROI 1 ; L4 = P4 P1 DROI 1 ;
EPROU = L1 L2 L3 L4 DALL PLAN ;
TRAC EPROU ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO = MODE EPROU MECANIQUE ELASTIQUE PLASTIQUE
      ENDOMMAGEABLE ;

* Tangent modulus equal to 783 MPa
PROGEPS1=PROG 0. (350./74020.) 0.3 0.6 ;
PROGSIG1=PROG 0. 350.E06 581.1976E06 816.0976E06 ;
CTRAC1 = EVOL MANU EPS PROGEPS1 SIG PROGSIG1 ;

* Tangent modulus equal to 491 MPa
PROGEPS2=PROG 0. (350./74020.) 0.3 0.6 ;
PROGSIG2=PROG 0. 350.E06 494.9783E06 642.2783E06 ;
CTRAC2 = EVOL MANU EPS PROGEPS2 SIG PROGSIG2 ;

* Tangent modulus equal to 199 MPa
PROGEPS3=PROG 0. (350./74020.) 0.3 0.6 ;
PROGSIG3=PROG 0. 350.E06 408.7590E06 468.4590E06 ;
CTRAC3 = EVOL MANU EPS PROGEPS3 SIG PROGSIG3 ;

* variable tangent modulus
CTRAC = NUAGE T*FLOTTANT TRAC*EVOLUTION
      0. CTRAC1 100. CTRAC2 200. CTRAC3 ;
DESS CTRAC1 ; DESS CTRAC2 ; DESS CTRAC3 ;

* variable thermic expansion coefficient
PROGTEMP = PROG 0. 100. 200. ;
PROGALFA = PROG 0. .25E-3 .5E-3 ;
EVALPH = EVOL MANU T PROGTEMP ALPH PROGALFA ;
DESS EVALPH ;

MATVAR = MATE MO YOUN 74020.E06 NU 0.3 RHO 7800.
      ALPH EVALPH TRAC CTRAC EPSD 0.
      DC 0.23 EPSR 0.25 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ L1 UZ ; CL2 = BLOQ L3 UZ ;
CL3 = BLOQ L4 UR ;
CL = CL1 ET CL2 ET CL3 ;
DEP = DEPI CL2 -1E-3 ;
*
*----- LOAD DEFINITION -----*
LI1 = PROG 0. 1 31. ; LI2 = PROG 0. 0. 30. ;
EV = EVOL MANU T LI1 DEP(T) LI2 ;
CHA1 = CHAR DIMP DEP EV ;
TEMPREF = MANU CHPO EPROU 1 T 30. ;
TEMPO = MANU CHPO EPROU 1 T 50. ;
TEMP200 = MANU CHPO EPROU 1 T 200. ;
T0 = 0. ; T1 = 1. ; T20 = 20. ;
TEMP = TABLE ; TEMPERA = TABLE ;
TEMP.0 = T0 ; TEMP.1 = T1 ; TEMP.2 = T20 ;

TEMPERA.0 = TEMPREF ; TEMPERA.1 = TEMPO ;

TEMPERA.2 = TEMP200 ;
CHA2 = CHAR T TEMP TEMPERA ;
CHA = CHA1 ET CHA2 ;
*
*----- LISTING OF TIME STEPS -----*
* 1.035 MM FIRST PLASTIC INCREMENT
SI COMPLET ;
  LIS = PROG 1.035 1.1 PAS 0.1 7 ;
SINON ;
  LIS = PROG 1.035 1.1 PAS 0.1 1.6 ;
FINSI ;

TAB = TABLE ;
TAB.'PRECISION' = 1E-3 ;
TAB.'BLOCAGES_MECANIQUES' = CL ;
TAB.'CARACTERISTIQUES' = MATVAR ;
TAB.'CHARGEMENT' = CHA ;
TAB.'TEMPS_CALCULES' = LIS ;
TAB.'MODELE' = MO ;
TAB.'TALPHA_REFERENCE' = 10. ;
TAB.'TEMPERATURE_REFERENCE' = TEMPREF ;
PASAPAS TAB ;
*
*----- GOOD WORKING TEST -----*
*----- COMPARISON WITH THE ANALYTIC SOLUTION -----*
SI COMPLET ;
  TA = 4. ; TB = 7. ;
  REF_D = PROG 7.5E-2 17.7E-2 ;
  REF_P = PROG 10.3E-2 20.9E-2 ;
  REF_S = (PROG -376.9 -369.1)*1E6 ;
SINON ;
  TA = 1.2 ; TB = 1.6 ;
  REF_D = PROG 3.52404E-03 1.24384E-02 ;
  REF_P = PROG 5.75205E-03 1.97372E-02 ;
  REF_S = PROG -3.52473E+08 -3.57838E+08 ;
FINSI ;

D1 = EXTR (PECHE TAB VARIABLES_INTERNES TA)
  'D ' 1 1 1 ;
D2 = EXTR (PECHE TAB VARIABLES_INTERNES TB)
  'D ' 1 1 1 ;

P1 = EXTR (PECHE TAB VARIABLES_INTERNES TA)
  'EPSE' 1 1 1 ;
P2 = EXTR (PECHE TAB VARIABLES_INTERNES TB)
  'EPSE' 1 1 1 ;

S1 = EXTR(PECHE TAB CONTRAINTES TA)'SMZZ' 1 1 1 ;
S2 = EXTR(PECHE TAB CONTRAINTES TB)'SMZZ' 1 1 1 ;

EF_D = PROG D1 D2 ;
EF_P = PROG P1 P2 ;
EF_S = PROG S1 S2 ;

PROG_ERD = MAXI (ABS ((EF_D - REF_D) / REF_D)) ;
PROG_ERP = MAXI (ABS ((EF_P - REF_P) / REF_P)) ;
PROG_ERS = MAXI (ABS ((EF_S - REF_S) / REF_S)) ;
ERR = MAXI (PROG PROG_ERD PROG_ERP PROG_ERS) ;
TEMPS ;
SI ( ERR <EG 0.05 ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;
FIN ;

```



Test endoaxi3 (Comments)

- The NUAGE operator

```
CTRAC = NUAGE T*FLOTTANT TRAC*EVOLUTION  
0. CTRAC1 100. CTRAC2 200. CTRAC3 ;
```

The NUAGE operator is to define an object of 'NUAGE' type (SET). Here a 'NUAGE' is a set of one couples. The unique couple is 0. CTRAC1 100. CTRAC2 200. CTRAC3 and the two components are T and TRAC.

To define tensile curves versus temperature, we should define three EVOLUTIO objects (CTRAC1 CTRAC2 and CTRAC3) which represent a tensile curve for three different temperatures 0. 100. and 300.°C.

5.5 Test ddi (Description sheet)

Test name: ddi

Calculation type: DAMAGE MODEL FOR VISCOPLASTIC MATERIAL

Finite element type: QUA8

Topic: Two inelastic strains model. Tensile test on a damage viscoplastic material. The structure is an embedded cylindrical test piece subjected to a tensile stress (imposed displacements increasing monotonously). The structure obeys a law according to a damage viscoplastic model.

Goal: Test the two inelastic strain model for MODE AXIS and compare Castem's results with the analytical solution.


Version: 97' customer version

Model description:

Zirconium alpha 200°C

$$E = 8.2 \cdot 10^{10}$$



$$\text{NU} = 0.36$$





Test ddi (Results)

CASTEM FIGURES



Test ddi (Listing CASTEM)

```

*****
TITR 'DAMAGE VISCOPLASTIC MODEL' ;
*****
OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
P1 = 0 0 ; P2 = 5E-3 0 ;
P3 = 5E-3 50E-3 ; P4 = 0 50E-3 ;

L1 = P1 P2 DROIT 1 ; L2 = P2 P3 DROIT 1 ;
L3 = P3 P4 DROIT 1 ; L4 = P4 P1 DROIT 1 ;

EPROU = L1 L2 L3 L4 DALL PLAN ;
TRAC EPROU ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ L1 UZ ;
CL2 = BLOQ L3 UZ ;
CL3 = BLOQ L4 UR ;
CL = CL1 ET CL2 ET CL3 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO = MODE EPROU MECANIQUE ELASTIQUE ISOTROPE
VISCOPLASTIQUE DDI ;
* Material: Zirconium alpha 200°C
MA = MATE MO YOUN 8.2D10 NU 0.36
'RP0' 135.D6 'QP' -60.D6 'BP' 120
'CP1' 34000.D6 'DP1' 250.
'CP2' 60000.D6 'DP2' 3000. 'KS' 960.D6 'N' 3.4
'RV0' 70.D6 'QV' -20.D6 'BV' 10.
'CV1' 24000.D6 'DV1' 300.
'CV2' 9000.D6 'DV2' 3000. 'CVP1' 0. 'CVP2' 0.;

* Time
EPSMAX = 0.01 ;
EPSPT = 0.001 ;
NCALC = 20 ;
TFIN = EPSMAX / EPSPT ;
DT0 = TFIN / NCALC ;
*
*----- LOAD DEFINITION -----*
DEP0 = DEPI CL2 ( EPSMAX * 0.05 ) ;
LI1 = PROG 0. tfin ;
LI2 = PROG 0. 1. ;
EV = EVOL MANU T LI1 LOAD LI2 ;
CHA = CHAR DIMP DEP0 EV ;
DESS EV ;
*
*--- CALCULATION WITH THE PASAPAS PROCEDURE ---*
TAB = TABLE ;
TAB.'BLOCAGES_MECAIQUES' = CL ;
TAB.'CARACTERISTIQUES' = MA ;
TAB.'MODELE' = MO ;
TAB.'CHARGEMENT' = CHA ;
LIS = PROG 0. PAS DT0 TFIN ;
TAB.'TEMPS_CALCULES' = LIS ;
PASAPAS TAB ;
*
*----- COMPARISON WITH REFERENCE VALUES -----*
DIM0 = DIME ( TAB.'TEMPS' ) - 1 ;
VAR0 = TAB.'VARIABLES_INTERNES'.DIM0 ;
SIGM0 = TAB.'CONTRAINTE'.DIM0 ;
P0 = MAXI ( EXCO VAR0 P ) ;
V0 = MAXI ( EXCO VAR0 V ) ;
SIG0 = MAXI ( EXCO SIGM0 SMZZ ) ;

P_TH = 4.17896E-03 ;
V_TH = 3.14432E-03 ;
SIG_TH = 2.19491E+08 ;
ERRP = 100. * (ABS((P_TH - P0)/P_TH)) ;
ERRV = 100. * (ABS((V_TH - V0)/V_TH)) ;
ERRSIG = 100. * (ABS((SIG_TH - SIG0)/SIG_TH)) ;

SI ( (ERRP > 0.1) OU (ERRV > 0.1) OU
(ERRSIG > 0.1) ) ;
ERRE 5 ;
SINON ;
ERRE 0 ;
FINSI ;

FIN ;

```

**Test ddi (Comments)**

- Two inelastic strain viscoplastic model (DDI)

```

MO = MODE EPROU MECANIQUE ELASTIQUE ISOTROPE
      VISCOPLASTIQUE DDI ;
MA = MATE MO YOUN 8.2D10 NU 0.36
      'RP0' 135.D6 'QP' -60.D6 'BP' 120
      'CP1' 34000.D6 'DP1' 250.
      'CP2' 60000.D6 'DP2' 3000. 'KS' 960.D6 'N' 3.4
      'RV0' 70.D6 'QV' -20.D6 'BV' 10.
      'CV1' 24000.D6 'DV1' 300.
      'CV2' 9000.D6 'DV2' 3000. 'CVP1' 0. 'CVP2' 0.;

```

The equations for the two inelastic strain model are as follows:

- Notation:
- | | |
|--------------|--|
| S | stress tensor |
| EP | plastic strain tensor |
| p | accumulated equivalent plastic strain |
| EV | creep strain tensor |
| v | accumulated equivalent creep strain |
| Xpi, Xvi | kinematic stress hardening variables (i=1,2) |
| Rp, Rv | isotropic stress hardening variables |
| J2 | deviatoric tensor second invariant |
| np | normal from Fp criterion |
| nv | normal from Fv criterion |
| < a > | positive part of a |
| Xp = Xp1 | |
| Xv = Xv1 | for a single centre |
| Xp = Xp1+Xp2 | |
| Xv = Xv1+Xv2 | for two centres |
- Criteria:
- | |
|--------------------|
| Fp = J2(S-Xp) - Rp |
| Fv = J2(S-Xv) - Rv |
- Stres shardenig:
- | |
|------------------------------------|
| Xp1 = CP1*ALPHAp1 + CVP1*ALPHAv1 |
| Xp2 = CP2*ALPHAp2 + CVP2*ALPHAv2 |
| Xv1 = CV1*ALPHAv1 + CVP1*ALPHAp1 |
| Xv2 = CV2*ALPHAv2 + CVP2*ALPHAp2 |
| dALPHApi = dEp - DPi*dp*np (i=1,2) |
| dALPHAvi = dEv - DVi*dv*nv (i=1,2) |
| Rp = RP0 + QP*(1-exp(-BP*p)) |
| Rv = RV0 + Qv*(1-exp(-BV*v)) |
- Flow rule:
- | |
|--------------------------------------|
| dp verifies dFp=0 |
| dEp = 3/2 * dp * < S-Xp > / J2(S-Xp) |
| dV = (< Fv > / KS) ** n |
| dEv = 3/2 * dv * < S-Xv > / J2(S-Xv) |

Data to be input:

Evolution law for the centres X_{pi} and X_{vi} :

- 'CP1' : ALPHAp1 coefficient
- 'CP2' : ALPHAp2 coefficient
- 'CV1' : ALPHAv1 coefficient
- 'CV2' : ALPHAv2 coefficient
- 'CVP1' : Visco-plastic coupling coefficient
- 'CVP2' : Visco-plastic coupling coefficient

Evolution law for the ALPHApi et ALPHAvi internal variables:

- 'DP1' : ALPHAp1*dp coefficient
- 'DP2' : ALPHAp2*dp coefficient
- 'DV1' : ALPHAv1*dv coefficient
- 'DV2' : ALPHAv2*dv coefficient

Evolution law for isotropic stress hardening:

- 'BP' : P coefficient
- 'QP' : plastic strain isotropic hardening coefficient
- 'RP0' : initial plastic threshold value
- 'BV' : V coefficient
- 'QV' : creep strain isotropic hardening coefficient
- 'RV0' : initial creep threshold value

Evolution law for the creep strain:

- 'KS' : creep threshold normalization coefficient
- 'N' : creep law exponent



Chapter 6

FRACTURE MECHANICS

6.1 Test rupt1 (Description sheet)

Test name: rupt1

Calculation type: MECHANICS ELASTIC-PLASTIC AXI

Finite element type: QUA8

Topic: STRESS INTENSITY FACTOR IN LINEAR MECHANICS ELASTIC OF A SOLID CYLINDER WITH A CIRCUMFERENCIAL CRACK. The structure is a solid cylinder with a circumferencial crack subjected to tensile stress. The goal is to calculate the Crack Opening Displacement (COD) at the crack tip and the stress factor in linear elastic mechanics of this structure. The theoretical values are given by the relationships:

$$K = \sigma_t \sqrt{\pi a}$$

$$\sigma_t = p / 2\pi R_m t \quad R_m \text{ is the mean radius}$$

$$F = 1.1 + A(1.948(a/t)^{0.5} + 0.3342(a/t)^{4.2})$$

$$A = (0.125(R_i/t) - 0.25)^{0.25}$$


For this problem the CASTEM solution for the stress factor K is given by the relation ship :

$$K = (\sqrt{EG/1 - v^2}) \quad G \text{ is the energy release rate}$$

Reference CASTEM: Test NAFEMS : Ductile Fracture Handbook, A. ZAHOOR, EPRI, 1990 : rupt1
Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



$$\begin{aligned} E &= 2.1 \cdot 10^{11} \text{ Pa} \\ \nu &= 0.3 \\ a &= 0.05\text{m} \end{aligned}$$




6.1. TEST RUPT1 (DESCRIPTION SHEET)

Test rupt1 (Results)


RESULTS

	$K1(MPam^{1/2})$
Reference	1.0848
Castem	1.0855


CASTEM FIGURES



MESH OF THE CYLINDER




MESH OF THE CYLINDER



AMPLITUDE
1.00E+04
0.00E+00

deformation of the cylinder subjected to tensile stress



AMPLITUDE
1.00E+04
0.00E+00

deformation of the cylinder subjected to tensile stress



6.1. TEST RUPT1 (DESCRIPTION SHEET)

Test rupt1 (Listing CASTEM)

```

OPTION ECHO 0;
*****
*
* STRESS INTENSITY FACTOR IN MECHANICS LINER
* ELASTIC OF A SOLID CYLINDER WITH A CRACK TIP
*
*****
*
TA = 1.E6;
R = 1.;
B = 0.1;
A = B / 2.;
L = R * 4.;
MYOU = 2.0E11;
***
*** GEOMETRY AND MESHING
***
OPTION DIME 2 ELEM QUA8 MODE AXIS;
t = a / 100.; densite t; pf = (a 0.);
c1 = (c ( pf moin (t 0.)) pf ( pf plus (0. t)))
      c pf ( pf plus (t 0.));
trac c1 ;
sf = cout pf c1;
trac sf ;
r1 = t ; rrl = t;
repeater bhomo 7;
ri = r1 + ( 0.3 * r1 );
rri = rrl + ri; dens ri;
ci = (c (pf moin (rri 0.)) pf
      ( pf plus (0. rri)))
      c pf (pf plus (rri 0.));
sf = sf et (cout c1 ci);
c1 = ci ; r1 = ri ; rrl = rri;
fin bhomo;
*
dens (a / 3.);
p0 = (0. 0.) ; p1 = (b 0.);
p2 = p0 plus (0. a) ; p3 = p1 plus (0. a);
pi1 = ci poin 1 ; l1 = pi1 d p0 ;
n = (nbel l1) * -1;
pi2 = ci poin 4 ; l2 = pi2 d n p2;
pi3 = ci poin 10 ; l3 = pi3 d n p3;
pi4 = ci poin 13 ; l4 = pi4 d p1;
ci = inve ci ; ligh = p2 d p3;
sc1 = dall l1 (p0 d p2) (inve l2)
      (ci comp pi2 pi1);
sc2 = dall l2 ligh (inve l3)
      (ci comp pi3 pi2);
sc3 = dall l3 (p3 d p1) (inve l4)
      (ci comp pi4 pi3);
sc = sc1 et sc2 et sc3;
dens (a / 2.);
mrest1 = ligh tran (0. (2.*a))
        dini (40*t) dfin (50*t);
l1 = mrest1 cote 3;
YY1 = coor 2 (point l1 init);
l2 = D 3 (B (YY1 + (0.7*a)))
      (0. (yy1 + (0.7*a)));
S1 = COUT l1 l2;
YY1 = COOR 2 (l2 point init);
mrest2 = l2 tran (0. (L - YY1))
        dini (80*t) dfin (150*t);
sut = sf et sc et mrest1 et S1 et mrest2;
ELIM 1.E-8 SUT;
DEPL PLUS SUT (R 0.);
L1 = (CONT SUT) ELEM APPU (SUT POIN
      DROI (R 0.) ((B + R) 0.) 1.E-8);
L2 = (CONT SUT) ELEM APPU (SUT POIN
      DROI ((B + R) 0.) ((B + R) L) 1.E-8);
L3 = (CONT SUT) ELEM APPU (SUT POIN
      DROI ((B + R) L) (R L) 1.E-8);
L4 = (CONT SUT) ELEM APPU (SUT POIN
      DROI (R L) (R 0.) 1.E-8);
L5 = (CONT SUT) COMP P1 PF;
      TITR 'MESH OF THE CYLINDER';
      TRAC sut;
*****
***** RESOLUTION IN MECHANICS ELASTICITY
*****
M01 = MODE SUT MECANIQUE ELASTIQUE
      PLASTIQUE ISOTROPE;
M01 = MATE M01 YOUNG MYOU NU 0.3 TRAC COUTRAC;
RI = (BLOQ UZ L5) ET (RIGI M01 M01);
AIR1 = PI*((R + B)**2.) - (R*R);
FOR1 = PRES 'MASS' M01 (0. - (TA/AIR1)) L3;
DEP1 = RESO FOR1 RI;
SIG1 = SIGMA M01 M01 DEP1;
TITR 'deformation of the cylinder
      subjected to tensile stress';
TRAC ((DEFO SUT DEP1 10000. ROUGE)
      ET (DEFO SUT DEP1 0. BLANC));
*
* Solution COD(mm) and K (MPa M^0.5) of castem
*
COD_CAL = (EXTR DEP1 'UZ' P1)*2000.;
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'J';
SUPTAB.'LEVRE_SUPERIEURE' = l1 diff l5;;
SUPTAB.'FRONT_FISSURE' = PF ;
SUPTAB.'MODELE' = M01;
SUPTAB.'CARACTERISTIQUES' = mal;
SUPTAB.'SOLUTION_RESO' = dep1;
SUPTAB.'CHARGEMENTS_MECANIQUES' = for1;
SUPTAB.'COUCHE' = 5;
G_THETA SUPTAB;
K_CAL = (MYOU*(SUPTAB.'RESULTATS')
        /(1 - (0.3**2)))*0.5;
K_CAL = K_CAL*1.E-6;
***
*** Analytical Solution (Zahoor)
***
RAP1 = R / B;
SI (RAP1 < 10.);
GRANDA = ((0.125*RAP1) - 0.25)**0.25;
FINSI;
SI (RAP1 >EG 10.);
GRANDA = ((0.4*RAP1) - 3.00)**0.25;
FINSI;
F = (1.9480*((A/B)**1.5)) +
    (0.3342*((A/B)**4.2));
F = 1.1 + (GRANDA*F);
SIGT = TA / AIR1;
K_ZAH = SIGT*((PI*A)**0.5)*F;
K_ZAH = K_ZAH*1.E-6;
ERR1 = abs ((K_CAL - K_ZAH)/K_ZAH);
MESS 'K Factor CASTEM
      =' K_CAL '(Mpa.M^0.5)';
MESS 'K Factor ZAHOOR
      =' K_ZAH '(Mpa.M^0.5)';
MESS 'Erreur relative =' ERR1;
SI (ERR1 < 1.E-2); ERRE 0;
SINO ; ERRE 5; FINSI;
FIN;

```

Test rupt1 (Comments)

1. GEOMETRY AND MESHING (OPERATOR COUTURE)

To account for singular displacement and stress distribution at the crack tip the mesh must be finer at the crack tip. Here the mesh is obtain with the COUT operator:

For example:

```
t = a / 100.; densite t; pf = (a 0.);
c1 = (c ( pf moin (t 0.)) pf ( pf plus (0. t)))
      c pf ( pf plus (t 0.));
trac c1 ;
sf = cout pf c1;
```

Thus, the COUT operator constructs a surface which connects two lines or a line and point by means of triangles. here pf is a point (POINT type) to be connected to the line c1 (line between which the surface is generated (MAILLAGE type)).

2. RESOLUTION IN ELASTIC MECHANICS (G_THETA PROCEDURE)

```
SUPTAB = TABLE ;
SUPTAB.'LEVRE_SUPERIEURE' = l1 diff l5;
SUPTAB.'MODELE' = MO1;
SUPTAB.'SOLUTION_RESO' = dep1;
SUPTAB.'COUCHE' = 5;
K_CAL = (MYOU*(SUPTAB.'RESULTATS')/(1 - (0.3**2)))*0.5;
K_CAL = K_CAL*1.E-6;
SUPTAB.'OBJECTIF' = MOT 'J';
SUPTAB.'FRONT_FISSURE' = PF ;
SUPTAB.'CARACTERISTIQUES' = mal;
SUPTAB.'CHARGEMENTS_MECANIQUES' = for1;
G_THETA SUPTAB;
```

The G_THETA procedure has two objectives :

(a) calculate line integrals of fracture mechanics as follows :

- the J line integral of an isotropic material, characterizing the crack tip fields in elasto-plasticity. In 3D massive element case, composite materials are not yet authorised.
- the J line dynamic integral of an isotropic material, characterizing the crack tip fields for elasto-dynamic problems. In 3D massive element case, composite materials are not yet authorised.
- the C* line integral of an isotropic material, characterizing the crack tip fields in the case of secondary creep problems. Applied loadings must be mechanical and composite materials are not yet authorised in 2D or 3D configurations.
- the C*(h) line integral of an isotropic material, characterizing the crack tip fields in the case of primary or tertiary creep problems. Applied loadings must be mechanics and composite materials are not yet acceptable in 2D configurations, nor in 3D configurations.
- the derivative dJ/da (a : crack length) line integral of an isotropic homogeneous material, required in the stability analysis of a single crack or a system of interacting cracks. Only the massive elements (2D or 3D) are allowed and composite materials are not yet available for calculating this derivative integral.



6.1. TEST RUPT1 (DESCRIPTION SHEET)

- (b) separate the stress intensity factors K1, K2 (and K3 in 3D) for elastic problems in 2D, 3D (massive elements only) or axisymmetrical configurations. For separating a mixed mode, the current version of G_THETA procedure allows only isotropic homogeneous materials.

For this problem the goal is to compute the parameter $G = J$ in elasticity which represents the energy release rate.

Therefore : Input data of the procedure is a TABLE object type, SUPTAB, with indexes are objects of MOT type, written in capital and described as follows:

- SUPTAB.'OBJECTIF' = MOT type, specifying the calculation aim, equal to 'J' for calculating the J line integral, characteristics in elasto-plasticity.
- SUPTAB.'LEVRE_SUPERIEURE' = following the definition convention, this object (MAILLAGE type) gives the crack's superior lip.
- SUPTAB.'FRONT_FISSURE' = POINT type in 2D or 3D shell element case, MAILLAGE type in 3D massive element case. This object gives the tip of the crack.
- SUPTAB.'MODELE' = modele object (type MMODEL) covering the entire structure
- SUPTAB.'CARACTERISTIQUES' = MCHAML object of CARACTERISTIQUES subtype In the case of calculations with shell elements, this object includes material's properties (Young's modulus, Poisson's coefficient ...) and geometrical parameters like shell thickness, generated by MATE or CARA operator.
- SUPTAB.'SOLUTION_RESO' = displacement solution (CHPOINT type) resulting from RESO operator.
- SUPTAB.'CHARGEMENTS_MECANIQUES' = all external mechanical loadings (CHPOINT type) if they exist.
- SUPTAB.'COUCHE' = ENTIER type. It is the number of layers of elements surrounding the crack tip in 2D (or surrounding a point on the crack front in 3D), that support the virtual crack extension. If COUCHE = 0, the THETA field is equal to 1 only at the crack tip (in 2D), or to 1 for all points on the crack front (in 3D). In general, the bigger the number of layers of elements, the more accurate the value calculated by the present G_THETA procedure. But the elements included in the number of layers should not reach the mesh border.

The CASTEM solution for the stress factor K is given by the relation ship :

$$K = (\sqrt{EG/1 - \nu^2}) :$$

$$K_CAL = (MYOU*(SUPTAB.'RESULTATS'))/(1 - (0.3**2))**0.5;$$

$$K_CAL = K_CAL*1.E-6;$$

6.2 Test rupt2 (Description sheet)

Test name: rupt2

Calculation type: ELASTIC MECHANICS

Finite element type: QUA8 PLANE STRAIN MODE

Topic: THE USE OF G-INTEGRAL IN THERMAL STRESS FOR AN EDGE CRACKED STRIP. The structure is a solid edge cracked strip which ends are constrained. It is subject to a linear temperature gradient through the thickness with zero temperature at mid-thickness and temperature T_0 at the right edge ($x = w/2$). Due to symmetry only one half of the strip is analysed. The CASTEM solution for the stress intensity factor K is given by the relationship:

$$K = (\sqrt{EG/1 - \nu^2}) \quad G \text{ is the energy release rate}$$

Finally the K values for 7 different temperatures $\theta(\theta_1, \theta_2, \theta_3 \dots \theta_7)$ are computed and compared with the results obtained with WILSON and YU relationship:


$$K = (E\alpha T_0 / 1 - \nu) F \sqrt{\Pi a} \quad F = 0.514$$

Reference CASTEM: Test NAFEMS : The use of J-integral in thermal stress crack problems International Journal of Fracture:

rupt2 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



Edge cracked strip subject to thermal loading

GEOMETRY :

Width of the edge : $W = 200 \text{ mm}$
 Length of the edge : $2L = 800 \text{ mm}$
 Length of the crack : $a = 100 \text{ mm}$

MATERIAL CHARACTERISTICS :

$E = 20000 \text{ daN mm}^{-2}$
 $\alpha = 5.10E-6 \text{ } ^\circ\text{C}^{-1}$
 $\nu = 0.3$

LOADING AND BOUNDARY CONDITIONS :

- 1- Linear Temperature gradient with zero temperature at mid thickness and temperature T_0 at the right edge
 $T(x) = 2T_0 \cdot x/W \quad [-w/2 < x < w/2]$
 $T_0 = 100^\circ\text{C}$
- 2- The vertical displacement (U_y) of the upper and lower extremity is blocked





6.2. TEST RUPT2 (DESCRIPTION SHEET)

Test rupt2 (Results)

RESULTS


	$K1(daNmm^{-3/2})$
Reference	130.15
Castem	125.88

CASTEM FIGURES







6.2. TEST RUPT2 (DESCRIPTION SHEET)




GIBI FEClT

VAL - ISO
> -2.50E+01
< 8.22E+01



GIBI FEClT

VAL - ISO
> -2.50E+01
< 8.22E+01



Test rupt2 (Listing CASTEM)

```

*****
*
* THE USE OF G-INTEGRAL IN THERMAL STRESS
* FOR AN EDEDGE CRACKED STRIP
*
*****
OPTI ECHO 0 ;
OPTION DIME 2 ELEM QUA8 MODE PLAN DEFO ;
*
*-----GEOMERTY AND MESHING -----*
a = 100. ; b = 200. ; h = 400. ; t = a / 100. ;
densite t ; pf = (a 0.) ;
cl = (c ( pf moin (t 0.)) pf ( pf plus (0. t)))
      c pf ( pf plus (t 0.)) ;
sf = cout pf cl;
r1 = t ; rrl = t;
reperter bhomo 7;
ri = r1 + ( 0.3 * r1 ) ;
rri = rrl + ri ;
dens ri ;
ci = ( c ( pf moin (rri 0.))
      pf ( pf plus (0. rri)))
      c pf ( pf plus (rri 0.)) ;
sf = sf et ( cout cl ci ) ;
cl = ci ; r1 = ri ; rrl = rri ;
fin bhomo ;
dens (a / 3.) ;
p0 = (0. 0.) ; p1 = (b 0.) ;
p2 = p0 plus (0. a) ; p3 = p1 plus (0. a) ;
pi1 = ci poin 1 ;
l1 = pi1 d p0 ; n = (nbel l1) * (-1) ;
pi2 = ci poin 4 ; l2 = pi2 d n p2 ;
pi3 = ci poin 10 ; l3 = pi3 d n p3 ;
pi4 = ci poin 13 ; l4 = pi4 d p1 ;
ci = inve ci ; ligh = p2 d p3 ;
sc1 = dall l1 (p0 d p2)
      (inve l2) (ci comp pi2 pi1) ;
sc2 = dall l2 ligh (inve l3)
      (ci comp pi3 pi2) ;
sc3 = dall l3 (p3 d p1)
      (inve l4) (ci comp pi4 pi3) ;
sc = sc1 et sc2 et sc3 ;
dens (a / 2.) ;
mrest = ligh tran (0.(h - a)) dini (a / 2.) ;
lihaut = mrest cote 3 ;
phd = lihaut poin init ;
ccp = sf et sc et mrest ;
elim ccp 0.001 ;
cccp = cont ccp ;
lifis = cccp comp pf p0;
libas = cccp comp p1 pf;
*
*---- MODEL AND MATERIAL CHARACTERISTICS-----*
objaf = mode ccp mecanique elastique isotrope ;
*
mat = mate objaf YOUN 2.e4 NU 0.3 ALPH 5.e-6 ;
*
*----- MATRIX OF RIGIDITY-----*
*----- BOUNDARY CONDITIONS -----*
rig = rigi objaf mat ;
cdl1 = bloq Uy libas ;
cdl2 = bloq Uy lihaut ;
cdl3 = bloq Ux phd ;
*
*--DEFINITION OF THE LINEAR TEMPERATURE GRADIENT--*
chx = coor 1 ccp ;
cha = manu chpo ccp 1 scal a ;
chx = nomc 'T' (chx - cha) ;

cht = chx * (100. / a) ;
ch0 = 0 * cht ;
sigth = thet MAT objaf cht ;
*
*--RESOLUTION - COMPONENT OF THE STRESS FIELD---*
*
U = reso (rig et cdl1 et cdl2 et cdl3 )
          (bsig objaf sigth) ;
SIG = sigm mat objaf U ;
SIG = SIG - sigth ;
sigy = exco SIG smyy ;
TRAC qual CCP ;
TRAC SIGY OBJAF CCP ;
*
*---- G_THETA PROCEDURE -----*
*---- G ENERGY RELEASE RATE VALUE-----*
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'J';
SUPTAB.'LEVRE_SUPERIEURE' = lifis;
SUPTAB.'FRONT_FISSURE' = PF ;
SUPTAB.'MODELE' = objaf;
SUPTAB.'CARACTERISTIQUES' = mat;
SUPTAB.'SOLUTION_RESO' = u;
SUPTAB.'TEMPERATURES' = cht;
SUPTAB.'COUCHE' = 4;
*
SAUT PAGE ;
G_THETA suptab ;
*
*RESULTS AND COMPARAISON WITH THEORITECALS VALUES*
*
G = suptab.'RESULTATS' ;
K = (20000. * G / (1 - (0.3**2)))**0.5 ;
F = K * (1 - 0.3) / ( 20000 *
                    5.e-6 * 100 * ((100*pi)**0.5)) ;
mess 'ENERGY RELEASE RATE G :' G ;
mess 'STRESS FACTOR K :' K ;
mess 'COMPUTED FORM FACTOR Fcal :' F ;
mess 'ANALYTICAL FORM FACTOR Fthe :' 0.514 ;
*
*-----GOOD WORKING MESSAGE ----- ;
RESI = abs (( F - 0.514 ) / 0.514 ) ;
SI (RESI < 5E-2) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
fin;

```


**Test rupt2 (Comments)**

1. DEFINITION OF THE LINEAR TEMPERATURE GRADIENT

```

chx = coor 1 ccp          ;
cha = manu chpo ccp 1 scal a ;
chx = nomc 'T' (chx - cha) ;
cht = chx * (100. / a)    ;
ch0 = 0 * cht            ;
sigth = thet MAT objaf cht ;

```

- COOR OPERATOR :

The COOR operator enables the user to recover the Nith coordinate (for this example it is the first coordinate $N = 1$) of a POINT, MAILLAGE, CHPOINT or MCHAML object type (here CCP is a MAILLAGE (MESH) type).

For a CCP mesh object type, the operator supplies the coordinates of the nodes supporting the mesh.

- MANU OPERATOR :

The MANU operator enables the user to simply create objets of the following types : MAILLAGE, CHPOINT, SOLUTION, RIGIDITE. here the MANU operator constructs a field by point by means of the key word 'CHPO' . :

CCP is the geometrical support (MAILLAGE type)

1 is the number of components of the field by points (ENTIER type)

SCAL is a name of components (MOT type) not exceeding 4 characters

a is a list of the value allocated to each node of CCP.

- NOMC OPERATOR :

The NOMC operator enables the user either to change the component na of a field by points or to replace a sublist of components by a second one :

(chx - cha) is CHPO1 field by points (CHPOINT type) and 'T' (Temperature) is a new name allocated to the component (MOT type) . CHX is the generated object (CHPOINT type) .

NOTE :

The field by points CHPO1 must be composed of a single component which name is arbitrary.

- THET OPERATORS :

The THETA operator calculates the stresses associated with a temperature field, i.e. :

$$SIGMA = HOOK * EPSTHER$$

in which HOOK is Hooke's matrix and EPSTHER corresponds to the strains of thermal origin :

```
sigth = thet MAT objaf cht
```

MAT is a field of material and geometrical properties (MCHAML type, CARACTERISTIQUES subtype)

OBJAF is a model object (MMODEL type)

CHT is a temperature field (CHPOINT type).

SIGTH is a stress field (MCHAML type, CONTRAINTES subtype).

Note :

For the shell elements, the temperature field must have three components with the following names : TINF ,T, and TSUP which define respectively the temperature on the bottom , mid-surface and top layers. For the other elements, the temperature field must have one component with the following name : T. You may recover the thermal strains from the thermal stresses using the ELAS operator.

2. RESOLUTION G_THETA PROCEDURE

The CASTEM solution for the stress intensity factors K is given by the G_THETA PROCEDURE from the relationship :

$$K = (\sqrt{EG/1 - \nu^2}) \quad G \text{ is the energy release rate}$$



6.3 Test rupt3 (Description sheet)

Test name: rupt3

Calculation type: MECHANICS ELASTIC

Finite element type: QUA8 PLANE STRAIN MODE

Topic: THE USE OF G-INTEGRAL IN TENSILE STRESS TEST FOR A SIMPLE PLATE WITH A CENTRAL CRACK. The structure is a solid plate containing a central crack . This structure is subject to a uniform load by a simple tensile stress test. Due to symmetry only 1/4 of the structure is analysed. The CASTEM solution for the stress intensity factors K is given by the relationship:

$$K = (\sqrt{EG/1 - \nu^2}) \quad G \text{ is the energy release rate}$$

Finally this CASTEM K value for the stress intensity factors is compared with the result obtained with the analytical formula of ROOKE and CARTWRIGHT from the relationship:


$$\frac{K}{K_0} = \frac{1 - 0.5\frac{a}{b} + 0.326(\frac{a}{b})^2}{(1 - \frac{a}{b})^{1/2}}$$

$$K_0 = \sigma\sqrt{\pi a}$$

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Compendium of Intensity Factors HMSO (1976): rupt3 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:


GEOMETRY :

 Width of the edge : $2b = 400 \text{ mm}$

 Length of the edge : $2h = 1200 \text{ mm}$

 Length of the crack : $2a = 200 \text{ mm}$
MATERIAL CHARACTERISTICS :
 $E = 20000 \text{ daN mm}^{-2}$
 $\nu = 0.3$
LOADING CONDITIONS :

1- Uniform load (Tensile stress)


 $\text{Sigma} = 1 \text{ daN mm}^{-2}$
Test rupt3 (Results)
RESULTS

	$K1(\text{daNmm}^{-3/2})$
Reference	20.84
Castem	21.02


CASTEM FIGURES



6.3. TEST RUPT3 (DESCRIPTION SHEET)



GIBI FECIT



GIBI FECIT

Test rupt3 (Listing CASTEM)

```

*****
*
*   THE USE OF G-INTEGRAL IN TENSILE STRESS TEST   *
*   FOR A SIMPLE PLATE WITH A CENTRAL CRACK       *
*
*****
*
OPTI ECHO 0      ;
OPTION DIME 2 ELEM QUA8 MODE PLAN DEFO      ;
*
*----- GEOMERTY AND MESHING -----*
*
a = 100. ; b = 200. ; h = 600. ;
t = a / 100. ;
densite t ; pf = (a 0.)      ;
c1 = (c ( pf moin (t 0.)) pf
( pf plus (0. t))) c pf ( pf plus (t 0.));
sf = cout pf c1; r1 = t ; rrl = t      ;
repeteur bhomo 7;
ri = r1 + ( 0.3 * r1 )      ;
rri = rrl + ri      ;
dens ri      ;
ci = ( c ( pf moin (rri 0.)) pf
( pf plus (0. rri)))
c pf ( pf plus (rri 0.)) ;
sf = sf et ( cout cl ci ) ;
c1 = ci ; r1 = ri ; rrl = rri ;
fin bhomo      ;
dens (a / 3.) ;
p0 = (0. 0.) ; p1 = (b 0.) ;
p2 = p0 plus (0. a) ;
p3 = p1 plus (0. a) ;
pil = ci poin 1 ; l1 = pil d p0 ;
n = (nbcl l1) * -1 ;
pi2 = ci poin 4 ; l2 = pi2 d n p2 ;
pi3 = ci poin 10 ; l3 = pi3 d n p3 ;
pi4 = ci poin 13 ; l4 = pi4 d p1 ;
ci = inve ci ;ligh = p2 d p3 ;
sc1 = dall l1 (p0 d p2) (inve l2)
(c1 comp pi2 pil) ;
sc2 = dall l2 ligh (inve l3)
(c1 comp pi3 pi2) ;
sc3 = dall l3 (p3 d p1)
(inve l4) (ci comp pi4 pi3) ;
sc = sc1 et sc2 et sc3 ;
dens (a / 2.) ;
mrest = ligh tran (0. (h - a))
dini (a / 2.) ;
lihaut = mrest cote 3 ;
phd = lihaut poin init ;
ccp = sf et sc et mrest ;
elim ccp 0.001 ;
cccp = cont ccp ;
lifis = cccp comp pf p0 ;
libas = cccp comp p1 pf ;
licot = ccp poin droi (0. 0.)
(0. 100.) 0.001 ;
licot = cccp elem appuy stric licot ;
TRAC CCP ;
*
*--MODEL AND MATERIAL CHARACTERISTICS ---*
*
objaf = mode ccp mecanique elastique isotrope ;
*
mat = matr objaf YOUN 2.e4 NU 0.3 ;
*
*----- MATRIX OF RIGIDITY-----*
*-- BOUNDARY CONDITIONS -----*
*
rig = rigi objaf mat ;
cdl1 = bloq Ux licot ;
cdl2 = bloq Uy libas ;
*
F = pres mass objaf -1. lihaut ;
*
*--RESOLUTION - COMPONENT OF THE STRESS FIELD--*
*---- G_THETA PROCEDURE -----*
*---- G ENERGY RELEASE RATE VALUE-----*
*
U = reso (rig et cdl1 et cdl2 ) F ;
SIG = sigm mat objaf U ;
*
*
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'J';
SUPTAB.'LEVRE_SUPERIEURE' = lifis;
SUPTAB.'FRONT_FISSURE' = PF ;
SUPTAB.'MODELE' = objaf;
SUPTAB.'CARACTERISTIQUES' = mat;
SUPTAB.'SOLUTION_RESO' = u;
SUPTAB.'CHARGEMENTS_MECANIQUES' = F;
SUPTAB.'COUCHE' = 4;
*
SAUT PAGE ;
G_THETA suptab ;
*
*RESULTS AND COMPARAISON WITH THEORITICALS VALUES*
*
G = suptab.'RESULTATS' ;
Kth = 20.8426 ;
K = (20000. * G / (1 - (0.3**2)))**0.5 ;
mess ' ' ;
mess ' ' ;
mess ' ENERGY RELEASE RATE G : ' G ;
mess 'COMPUTED SOLUTION Kcal : ' K ;
mess 'ANALYTICAL SOLUTION Kthe : ' Kth ;
* ----GOOD WORKING MESSAGE -----
RESI = abs (( k - Kth ) / Kth ) ;
SI (RESI < 5E-2) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
fin;

```



6.4 Test rupt4 (Description sheet)

Test name: rupt4

Calculation type: MECHANICS ELASTIC

Finite element type: QUA8 PLANE STRAIN MODE

Topic: THE USE OF G-INTEGRAL FOR A SIMPLE PLATE WITH A CENTRAL CRACK SUBJECTED TO INTERNAL PRESSURE. The structure is a solid plate containing a central crack. This structure is subjected to a uniform internal pressure. Due to symmetry only 1/4 of the structure is analysed. The CASTEM solution for the stress intensity factor K is given by the relationship:

$$K = (\sqrt{EG/1 - \nu^2}) \quad G \text{ is the energy release rate}$$


Finally this CASTEM K value for the stress intensity factors is compared with the result obtained from the analytical formula defined by the relationship:

$$K_0 = \sigma \sqrt{\pi a}$$

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Compendium of Intensity Factors HMSO (1976) : rupt4 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



GEOMETRY :

Width of the edge : $2b = 1200 \text{ mm}$

Length of the edge : $2h = 2400 \text{ mm}$

Length of the crack : $2a = 200 \text{ mm}$

MATERIAL CHARACTERISTICS :

$E = 20000 \text{ daN mm}^{-2}$

$\nu = 0.3$

LOADING CONDITIONS :


1- Uniform internal pressure

$P = 1 \text{ daN mm}^{-2}$


Solid-Plate with a central crack subject to uniform internal pressure

Test rupt4 (Results)
RESULTS

	$K1(daNmm^{-3/2})$
Reference	17.725
Castem	17.616

CASTEM FIGURES


GIBI FRCIT



FIELD OF PRESSURE AT THE CRACK TIP

 COMPOSANTES
 VECTEURS

FX FY



6.4. TEST RUPT4 (DESCRIPTION SHEET)

Test rupt4 (Listing CASTEM)

```

*****
*
* THE USE OF G-INTEGRAL FOR A SIMPLE PLATE *
* WITH A CENTRALE CRACK STRIP SUBJECTED *
* TO UNIFORM INTERNAL PRESSURE *
*
*****
*
OPTION ECHO 1 ;
OPTION DIME 2 ELEM QUA8 MODE PLAN DEFO ;
*
*---GEOMERTY AND MESHING ----*
*
a = 100. ; b = 200. ;
h = 1200. ; m = 600. ;
t = a / 100 ;
densite t ; pf = (a 0.) ;
c1 = (c ( pf moin (t 0.))
      pf ( pf plus (0. t)))
      c pf ( pf plus (t 0.));
sf = cout pf c1;
r1 = t ; rrl = t ;
reperer bhomo 7;
ri = r1 + ( 0.3 * r1 ) ;
rri = rrl + ri ;
dens ri ;
ci = ( c ( pf moin (rri 0.))
      pf ( pf plus (0. rri)))
      c pf ( pf plus (rri 0.)) ;
sf = sf et ( cout c1 ci ) ;
c1 = ci ; r1 = ri ; rrl = rri ;
fin bhomo ;
dens (a / 3.) ;
p0 = (0. 0.) ; p1 = (b 0.) ;
p2 = p0 plus (0. a) ;
p3 = p1 plus (0. a) ;
pi1 = ci poin 1 ;
l1 = pi1 d p0 ;
n = (nbel l1) * -1 ;
pi2 = ci poin 4 ; l2 = pi2 d n p2 ;
pi3 = ci poin 10 ; l3 = pi3 d n p3 ;
pi4 = ci poin 13 ; l4 = pi4 d p1 ;
ci = inve ci ; ligh = p2 d p3 ;
sc1 = dall l1 (p0 d p2)
      (inve l2) (ci comp pi2 pi1) ;
sc2 = dall l2 ligh (inve l3)
      (ci comp pi3 pi2) ;
sc3 = dall l3 (p3 d p1)
      (inve l4) (ci comp pi4 pi3) ;
sc = sc1 et sc2 et sc3 ;
dens (a / 2.) ;
ligh1 = p1 d p3 ;
mrest1 = ligh1 tran ((m-b) 0.)
        dini (a/2.) ;
ala = mrest1 cote 3 ;
al = mrest1 cote 2 ;
mrest = ligh tran (0. (h - a))
        dini (a / 2.) ;
mrest2 = al tran (0. (h - a))
        dini (a / 2.) ;
lihaut1= mrest cote 3 ;
lihaut = mrest cote 3 ;
lihaut2= mrest2 cote 3 ;
lihaut = lihaut1 et lihaut2 ;
ccp = sf et sc et mrest et

mrest1 et mrest2 ;

elim ccp 0.001 ;
cccp = cont ccp ;
pj = ala poin fina ;
lifis = cccp comp pf p0 ;
libas = cccp comp pj pf ;
licot = ccp poin droi (0. 0.)
        (0. 100.) 0.001 ;
licot = cccp elem appuy stric licot ;
TRAC CCP ;
*
*--MODEL AND MATERIAL CHARACTERISTICS ---*
objaf = modl ccp mecanique elastique isotrope ;
mat = matr objaf YOUN 2.e4 NU 0.3 ;
*
*----- MATRIX OF RIGIDITY-----*
*----- BOUNDARY CONDITIONS -----*
*
rig = rigi objaf mat ;
cdl1 = bloq Uy libas ;
cdl2 = bloq Ux licot ;
*
*
Fp = pres mass objaf 1. lifis ;
vect1 = VECT Fp 50. FX FY rouge ;
TRACE VECT1 CCP TITR
      'FIELD OF PRESSURE AT THE CRACK TIP';
*
*--RESOLUTION-COMPONENT OF THE STRESS FIELD--*
*---- G_THETA PROCEDURE -----*
*---- G ENERGY RELEASE RATE VALUE-----*
*
U = reso (rig et cdl1 et cdl2 ) Fp ;
SIG = sigm mat objaf U;
*
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'J';
SUPTAB.'LEVRE_SUPERIEURE' = lifis;
SUPTAB.'FRONT_FISSURE' = PF ;
SUPTAB.'MODELE' = objaf;
SUPTAB.'CARACTERISTIQUES' = mat;
SUPTAB.'SOLUTION_RESO' = u;
SUPTAB.'CHARGEMENTS_MECANIQUES' = Fp;
SUPTAB.'COUCHE' = 4;
*
SAUT PAGE ;
G_THETA suptab ;
*
*RESULTS AND COMPARAISON *
*WITH THEORITICALS VALUES *
*
G = suptab.'RESULTATS' ;
Kth = 17.7245 ;
K = (20000. * G / (1 - (0.3**2)))**0.5 ;
mess ' ' ;
mess ' ' ;
mess 'ENERGY RELEASE RATE G : ' G ;
mess 'COMPUTED SOLUTION Kcal : ' K ;
mess 'ANALYTICAL SOLUTION Kthe : ' Kth ;
* GOOD WORKING MESSAGE ;
RESI = abs (( k - Kth ) / Kth ) ;
SI (RESI < 5E-2) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
fin;

```

6.5 Test rupt5 (Description sheet)

Test name: rupt5

Calculation type: MECHANICS ELASTIC

Finite element type: QUA8 PLANE STRAIN MODE

Topic: THE USE OF J-INTEGRAL FOR A SIMPLE AXISYMETRICAL TUBE WITH A CENTRAL CRACK STRIP SUBJECTED TO UNIFORM INTERNAL PRESSURE. The structure is a solid cylinder tub containing a central crack . This structure is subject to an internal pressure. Due to symmetry only 1/4 of the structure is analysed. The CASTEM solution for the stress intensity factors K is given by the relationship (axisymmetrical model):

$$K = (\sqrt{EG/1 - \nu^2}) \quad G \text{ is the energy release rate}$$

Finally this CASTEM K value for the stress intensity factor is compared with the result obtained from the analytical formula defined by the relationship:


$$\frac{2}{\sqrt{\pi a}} \int_0^a \frac{rp(r)}{\sqrt{a^2 - r^2}} dr$$

$$P(r) = 1 - \frac{r}{a}$$

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Compendium of Intensity Factors HMSO (1976) : rupt5 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



GEOMETRY :


Width of the edge : $2b = 2000$ mm
 Length of the edge : $2h = 1000$ mm
 Length of the crack : $2a = 200$ mm

MATERIAL CHARACTERISTICS :

$E = 20000$ daN mm⁻²
 $\nu = 0.3$

LOADING CONDITIONS :

1- Internal pressure
 $P(r) = 1 - (r/a)$






Test rupt5 (Results)


RESULTS

	$K1(daNmm^{-3/2})$
Reference	2.4215
Castem	2.4098

CASTEM FIGURES



MESH OF THE CYLINDER TUB



MESH OF THE CYLINDER TUB

Test rupt5 (Listing CASTEM)

```

*****
*
* THE USE OF G-INTEGRAL FOR SIMPLE AXISYMETRICAL *
* TUB WITH A CENTRALE CRACK STRIP SUBJECTED TO *
* UNIFORM INTERNAL PRESSURE *
*
*****
OPTION ECHO 0 ;
OPTION DIME 2 ELEM QUA8 MODE AXIS ;
*
*---GEOMERTY AND MESHING ----*
*
a = 100. ; b = 200. ; h = 500. ; m = 1000 ;
t = a / 100 ; densite t ; pf = (a 0.) ;
c1 = ( c ( pf moin ( t 0.))
      pf ( pf plus ( 0. t)))
      c pf ( pf plus ( t 0.)) ;
sf = cout pf c1 ;
r1 = t ; rrl = t ;
reperter bhomo 7 ;
ri = r1 + ( 0.3 * r1 ) ;
rr1 = rrl + ri ;
dens ri ;
ci = ( c ( pf moin ( rri 0.))
      pf ( pf plus ( 0. rri)))
      c pf ( pf plus ( rri 0.)) ;
sf = sf et ( cout c1 ci ) ;
c1 = ci ; r1 = ri ; rrl = rri ;
fin bhomo ;
dens ( a / 3. ) ;
p0 = ( 0. 0. ) ; p1 = ( b 0. ) ;
p2 = p0 plus ( 0. a ) ;
p3 = p1 plus ( 0. a ) ;
pi1 = ci poin 1 ; l1 = pi1 d p0 ;
n = ( nbel l1 ) * -1 ;
pi2 = ci poin 4 ; l2 = pi2 d n p2 ;
pi3 = ci poin 10 ; l3 = pi3 d n p3 ;
pi4 = ci poin 13 ; l4 = pi4 d p1 ;
ci = inve ci ; ligh = p2 d p3 ;
sc1 = dall l1 ( p0 d p2 )
      ( inve l2 ) ( ci comp pi2 pi1 ) ;
sc2 = dall l2 ligh ( inve l3 )
      ( ci comp pi3 pi2 ) ;
sc3 = dall l3 ( p3 d p1 )
      ( inve l4 ) ( ci comp pi4 pi3 ) ;
sc = sc1 et sc2 et sc3 ;
dens ( a / 2. ) ;
ligh1 = p1 d p3 ;
mrest1 = ligh1 tran ((m-b) 0.)
        dini ( a/2. ) ;
ala = mrest1 cote 3 ;
al = mrest1 cote 2 ;
mrest = ligh tran ( 0. ( h - a ) )
        dini ( a / 2. ) ;
mrest2 = al tran ( 0. ( h - a ) )
        dini ( a / 2. ) ;
lihaut1= mrest cote 3 ;
lihaut = mrest cote 3 ;
lihaut2= mrest2 cote 3 ;
lihaut = lihaut1 et lihaut2 ;
ccp = sf et sc et mrest
      et mrest1 et mrest2 ;
elim ccp 0.001 ;
cccp = cont ccp ;
pj = ala poin fina ;
lifis = cccp comp pf p0 ;

libas = cccp comp pj pf ;
TRAC CCP TITR 'MESH OF THE TUBE' ;
*
*--MODEL AND MATERIAL CHARACTERISTICS ---*

objaf = mode ccp mecanique elastique isotrope ;
*
mat = mate objaf YOUN 2.e4 NU 0.3 ;
*
*----- MATRIX OF RIGIDITY-----*
*----- BOUNDARY CONDITIONS -----*
*
rig = rigi objaf mat ;
cdl1 = bloq Uz libas ;
cdl2 = bloq Uz lihaut ;
*
*---MECHANICS LOADING ----*
*
pchx = coor 1 lifis ;
pcha = manu chpo lifis 1 scal a ;
pchx = nomc 'P' ( pcha - pchx ) ;
pcht = pchx * ( 1/a ) ;
pch0 = pcht * 0. ;
Fp = pres mass objaf pcht ;
*
*--RESOLUTION-COMPONENT OF THE STRESS FIELD--*
*---- G_THETA PROCEDURE -----*
*---- G ENERGY RELEASE RATE VALUE-----*
*
U = reso ( rig et cdl1 et cdl2 ) Fp ;
SIG = sigm mat objaf U ;
*
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'J' ;
SUPTAB.'LEVRE_SUPERIEURE' = lifis ;
SUPTAB.'FRONT_FISSURE' = PF ;
SUPTAB.'MODELE' = objaf ;
SUPTAB.'CARACTERISTIQUES' = mat ;
SUPTAB.'SOLUTION_RESO' = u ;
SUPTAB.'CHARGEMENTS_MECANIQUES' = Fp ;
SUPTAB.'COUCHE' = 4 ;
*
SAUT PAGE ;
G_THETA suptab ;
*
*RESULTS AND COMPARAISON *
*WITH THEORITICALS VALUES *
*
G = suptab.'RESULTATS' ;
Kth = 2.4215 ;
K = ( 20000. * G / ( 1 - ( 0.3**2 ) ) ) ** 0.5 ;
mess ' ' ;
mess ' ' ;
mess 'ENERGY RELEASE RATE G : ' G ;
mess 'COMPUTED SOLUTION Kcal : ' K ;
mess 'ANALYTICAL SOLUTION Kthe : ' Kth ;
*
* GOOD WORKING MESSAGE*
*
RESI = abs ( ( k - Kth ) / Kth ) ;
SI ( RESI < 5E-2 ) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
fin ;

```



6.6 Test rupt6 (Description sheet)

Test name: rupt6

Calculation type: MECHANICS ELASTIC

Finite element type: CU20 MODE TRID

Topic: STRESS INTENSITY FACTOR FOR A CIRCULAR PLANE CRACK IN AN INFINITE SOLID CYLINDER SUBJECTED TO TENSILE STRESS. The structure is an infinite solid cylinder tube enclosing a central circular crack. This structure is subject to tensile stress. Due to symmetry only $1/15^\circ$ of half of the cylinder is analysed. The CASTEM solution for the stress intensity factors K is given with the displacement method by the relationship (threedimensional analysis):

$$K = \frac{E'}{4} \left(\sqrt{\frac{2\Pi}{r-\beta}} \right) v \quad E' = \sqrt{E/1-\nu^2} \quad (\text{planestrain}) \quad \beta = -\frac{d}{c}$$


c and d are the coefficients of the line : $v^2 = cr + d$

Finally this CASTEM K value for the stress intensity factor is compared with the result obtained with the analytical formula defined by the relationship : $K_0 = 2\sigma\sqrt{a\Pi}$

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Copendum of Intensity Factors HMSO (1976) : rupt6 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



Cylinder with circumference crack
subject to Tensile Stress test

GEOMETRY :

Radius of the Cylinder : $R = 100$ mm
Height of the Cylinder : $2h = 200$ mm
Radius of the Crack : $a = 18$ mm

MATERIAL CHARACTERISTICS :



$E = 20000$ daN mm⁽⁻²⁾
 $\nu = 0.3$

LOADING CONDITIONS :

1- Uniform load (Tensile stress)
 $\text{Sigma} = 200$ daN mm⁽⁻²⁾

Test rupt6 (Results)
RESULTS

	$K1(topnode)(daNmm^{-3/2})$	$K1(middelenode)(daNmm^{-3/2})$
Reference	607.41	580.48
Castem	638.47	638.47

CASTEM FIGURES




6.6. TEST RUPT6 (DESCRIPTION SHEET)

Test rupt6 (Listing CASTEM)

```

*****
* STRESS INTENSITY FACTOR FOR A CIRCULAR *
* PLANE CRACK IN AN INFINITE SOLID CYLINDER *
* SUBJECTED TO TENSILE STRESS *
*****
*
option echo 1 dime 3
      elem cu20 mode trid ;
*
*----GEOMETRY AND MESHING ----*
*
dens 0.5 ;
oeilz = -500 500 -500 ;
oeilx = -500 00 00 ;
a0 = 8 ; c0 = 16. ;
b0 = 100. ; p0 = 0 0 0 ;
pa = a0 0 0 ; pb = 0 a0 0 ;
pa1 =(1.5*a0) 0 0 ; pa0 = (0.5*a0) 0 0 ;
pb0 = 0 (0.5*a0) 0 ; pb1 = 0 (1.5*a0) 0 ;
* nbrz Number of finite element
* mesh round the crack tip
nbrz = 6 ;

*--CRACK STRIP MESHING*--*
*
pbz= (0. a0 (0.5*a0)) ;
pbz1=(0. (a0 *(1.- (0.5 / nbrz))))0.);
pbz2=(0. a0 (0.5*a0/nbrz)) ;
pbz3=(0. (a0 *(1.+ (0.5 / nbrz))))0.);
aa = a0*0.5 * (2**-.0.5) ;
pmi1= 0. (a0 + aa) aa ;
pmi2= 0. (a0 - aa) aa ;
*
cc11a = c (2) pb1 pb pmi1 ;
cc11b = c (2) pmi1 pb pbz ;
cc11 = cc11a et cc11b ;
cc12a = c (2) pbz pb pmi2 ;
cc12b = c (2) pmi2 pb pb0 ;
cc12 = cc12a et cc12b ;
cc1 = cc11 et cc12 ;
cc21 = c (4) pbz1 pb pbz2 ;
cc22 = c (4) pbz2 pb pbz3 ;
cc2 = cc21 et cc22 ;

ligz = d (nbrz - 1) pb0 pbz1 ;
ligz1= d (nbrz - 1) pbz3 pb1 ;
ligz2= d (nbrz - 1) pbz pbz2 ;
scz1 = dall cc11 ligz2 cc22 ligz1 ;
scz2 = dall cc12 ligz cc21(inve ligz2) ;

scz1=scz1 et (cout cc22 pb) ;
scz2=scz2 et (cout cc21 pb) ;
scz =scz1 et scz2 ;
elim (scz1 et scz2) 0.001 ;
*
*----- YZ SURFACE-----*
*
pfinay = 0. b0 0. ;
pfinaz = 0. 2.5 b0 ;
pfinayz= 0. b0 b0 ;
pinter = 0. (2*c0) b0 ;
pp = 0. a0 a0 ;
pz = 0. 2.5 a0 ;
p00= 0. 2.5 0. ;
ppp= 0. (2*a0) 0. ;
ppyz= 0. (2*a0) a0 ;

l0a= d (1) pb1 ppp ;
l0b= d (6) ppp pfinay ;
l0 = l0a et l0b ;
l1 = d (2) pfinay pfinayz ;
llbisa = d (6) pfinayz ppyz ;
llbisb = d (1) ppyz pmi1 ;
llbis = llbisa et llbisb ;
l2 = d (2) pfinayz pinter ;
l3a = d (6) pinter pp ;
l3b = d (1) pp pbz ;
l3 = l3a et l3b ;
zaa = dall (inve cc11a) l0 l1 llbis ;
zab = dall llbis cc11b
      (inve l3) (inve l2) ;
za = zaa et zab ;

g2= d (1) pb0 p00 ;
g3= d (2) p00 pz ;
g4= d (2) pz pp ;
zb =dall (g4 et g3)g2 cc12 l3b ;

h1 = d (2) pinter pfinaz ;
h2 = d (6) pfinaz pz ;
zc = dall h2 g4 l3a h1 ;

scr1 = h2 et (inve g3) ;

*---CREATION OF THE VOLUME-----*
*
scr1=scr1 rota 1 (15) (0. 0 -500)
      (0 0 500.) coul roug ;

geo1=za volu 1 'ROTA' (15)
      (0. 0. -500) (0. 0. 500.) ;

geo2=(zb et zc) volu 1 'ROTA' (15)
      (0. 0. -500) (0. 0. 500.) ;

geo3 = scz1 volu 1 'ROTA' (15)
      (0. 0. -500) (0. 0. 500.) ;
geo4 = scz2 volu 1 'ROTA' (15)
      (0. 0. -500) (0. 0. 500.) ;
i = face 3 geo3 ;
i = i et (face 3 geo4) ;

pppp = i poin cylin (0 0 -100) (0 0 100) pb ;
aa0= elem i appu larg pppp ;
aa1= poin aa0 plan p0 (0. 100 0.) (100 0 0) ;
cfis = elem aa0 appu stric aal ;
a1 = enve geo3 ;
a2 = enve geo4 ;
aal= poin a1 plan p0 (0. 100 0.) (100 0 0) ;
aa2= poin a2 plan p0 (0. 100 0.) (100 0 0) ;
aal1 = elem a1 appu stric aal ;
aa21 = elem a2 appu stric aa2 ;
aal2 = cont aal1 ;
aa22 = cont aa21 ;
elim (aal2 et aa22) 0.001 ;
cfissure = elem aal2 appu stric aa22 ;

geo = geo1 et geo2 et geo3 et geo4 ;
elim (geo et scr1 et cfissure et cfis) 0.001 ;
*
*-----FINE MESH-----*
* NEAR THE AXIS Z
c1 = cote (2) scr1 ;

pfi = c1 poin final ;
c1x = droi (1) pfi p0 ;

```

```

b1 = droi (1) p00 p0 ;
su0 = surf ( clx et c1 et b1) plane ;
sul = su0 volu (2) tran (0. 0. a0) ;
ss = su0 plus (0. 0. b0) ;
su2= ss volu (6) tran (0. 0. (a0 - b0)) ;
su = sul et su2 coul vert ;
*
*THE FULL-CYLINDER
cub = geo et su ;
elim cub 0.001 ;
cub = rege cub ;
tot = enve cub ;
trace tot ;
*surface yz ;
poyz = poin tot plan p0 (0. 100 100) (0. 0. 100);
suryz = elem tot appu stric poyz ;
*
*surface z=100 ;
poz100= poin tot plan (0. 0. 100)
(100 0. 100) (0. 100 100) ;
surzz = elem tot appu stric poz100 ;
*
*surfac xy ;
poxy = poin tot plan p0 (0. 100 0.) (100 0 0) ;
surxy= elem tot appu stric poxy ;
az = geo2 et geo4 et su ;
az = enve az ;
aze= az poin plan p0 (0. 100 0.) (100 0 0) ;
aze=elem az appu stric aze ;
co = cont aze;
aze = surxy incl aze;
facxy = surxy diff aze;
TRAC (geo1 et geo2 et geo3 et geo4 et su ) ;

*-----THE SYMMETRY CONDITIONS-----
p = ((-1 * b0 * (sin 15)) (b0 * (cos 15)) 0.);
cond1 =symt depl p0 (0. 0. b0) p cub 0.01;
condi2 =symt depl p0 pfinay pfinayz suryz;
condi3 =symt depl p0 (b0 b0 0.) (b0 0. 0.) facxy;
condit = cond1 et condi2 et condi3 ;

*----- MECHANICS MODEL -----*
affe1 = mode geo1 mecanique elastique isotrope;
affe2 = mode geo2 mecanique elastique isotrope;
affe3 = mode geo3 mecanique elastique isotrope;
affe4 = mode geo4 mecanique elastique isotrope;
affe5 = mode su mecanique elastique isotrope;
affetot = affe1 et affe2 et
affe3 et affe4 et affe5 ;

*-----MATERIAL CHARACTERISTIC-----;

matel = mate affe1 young 20000. nu 0.3 ;
mate2 = mate affe2 young 20000. nu 0.3 ;
mate3 = mate affe3 young 20000. nu 0.3 ;
mate4 = mate affe4 young 20000. nu 0.3 ;
mate5 = mate affe5 young 20000. nu 0.3 ;
matot = matel et mate2 et mate3 et mate4 et mate5 ;
*-----RIGIDITY-----*
rig1 = (rigidite matot affetot) et condit ;
rig2 = bloque uz surzz ;
fol = depi rig2 1. ;
*-----RESOLUTION AND STRESSES----- *
dep = resou (rig1 et rig2) fol ;
sig = sigma matot affetot dep ;

SUPTAB = TABLE;
SUPTAB.'MAILLAGE' = cub;
SUPTAB.'PSF1' = P0;
SUPTAB.'FRFFISS' = elem aal2 appu stri aa22;

*-----SIF PROCEDURE -----*
SIF SUPTAB MATOT dep;
*
*----- G_THETA PROCEDURE-----*
TABG1 = TABLE;
TABG1.'OBJECTIF' = MOT 'J';
TABG1.'COUCHE' = 3;
TABG1.'FRONT_FISSURE' = cfissure;
TABG1.'LEVRE_SUPERIEURE' = aa21 ;
TABG1.'MODELE' = affetot;
TABG1.'SOLUTION_RESO' = dep;
TABG1.'CHARGEMENTS_MECANIQUES' = fol;
TABG1.'CARACTERISTIQUES' = matot;
G_THETA TABG1;
SAUT 1 LIGNE;
***** GOING THROUGH THE SIF-PROCEDURE *****
*****
T = INDEX (SUPTAB.K1);
MESS 'K1 NODE 1 FOR THE CRACK TIP (TOP NODE) : '
SUPTAB.K1.(T.1);
MESS 'K1 NODE 2 FOR THE CRACK TIP (MIDDEL NODE) : '
SUPTAB.K1.(T.2);
*-----ERROR TEST-----*
EC1 = ((SUPTAB.K1.(T.1) - 638.47) /
(SUPTAB.K1.(T.1)) * 100;
EC2 = ((SUPTAB.K1.(T.2) - 638.47) /
(SUPTAB.K1.(T.2)) * 100;
SI ((EC1 < 5.5) ET (EC2 < 11)) ;
MESS 'PROCEDURE <SIF> ERR 0'; ERRE 0;
SINON;
MESS 'PROCEDURE <SIF> ERR 5'; ERRE 5;
FINSI;
***** GOING THROUGH THE G_THETA *****
*****
TBG = TABG1.'RESULTATS';
IND1 = INDE TBG;
GSOM = TBG.(IND1.1);
GMIL = TBG.(IND1.2);
EPRI = 20000. / 0.91;
KSOM = (GSOM * EPRI) ** 0.5;
KMIL = (GMIL * EPRI) ** 0.5;
SAUT 1 LIGNE;
MESS 'K1 NODE 1 FOR THE CRACK TIP(TOP NODE)
: ' KSOM;
MESS 'K1 NODE 2 FOR THE CRACK TIP(MIDDELE NODE)
: ' KMIL;
*-----ERROR TEST-----*
EC1 = ABS (((KSOM - 638.47) / KSOM) * 100);
EC2 = ABS (((KMIL - 638.47) / KMIL) * 100);
SI ((EC1 < 0.9) ET (EC2 < 0.8)) ;
MESS 'PROCEDURE <G_THETA> ERR 0';
ERRE 0;
SINON;
MESS 'PROCEDURE <G_THETA> ERR 5';
ERRE 5;
FINSI;
FIN;

```




Test rupt6 (Comments)

1. DESCRIPTION OF THE SIF PROCEDURE

- SIF SUPTAB MATOT DEP;
- This procedure calculates the stress intensity factor in mode I (possibly in mode II) from the displacements of the sides of the crack. The calculated value is an average of the three points which are closest to the crack tip. This procedure works in two and three dimensions. IN 2D, the problem relative to loading in mixed modes can be solved. In 3D massive elements (for example: 3D massiv elements CUB20), for each node of the crack front, the stress intensity factor is calculated from the displacements of the points located in the plane normal to the crack front at the considered node. (while generating the mesh, make sure that you will have these planes normal to the crack front). If the crack is opening, the plane stress formula is applied to the points located at the surface. In 3D with thin shells, the membrane and bending terms of the stress intensity factor can be computed. The bending term is deduced from the knot rotations inducing a strain at the external boundary of the shells. Mixed mode loadings are allowed.

Input data :

SUPTAB : TABLE type object, indexed by words, used to define the calculation options and parameters :

MATOT : field of material properties

DEP : displacement field

SUPTAB : TABLE type object, indexed by words, used to define the calculation options and parameters :

- for this three-dimensional problem with massive elements the parameters of SUPTAB are

```
SUPTAB = TABLE;
SUPTAB.'MAILLAGE' = cub;
SUPTAB.'PSF1' = P0;
SUPTAB.'FRTFISS' = elem aa12 appu stri aa22;
```

'PSF1'=P0; : point of the crack surface that does not pertain to the front

'FRTFISS'= elem aa12 appu stri aa22; : line describing the front crack

- output data

GOING THROUGH THE SIF-PROCEDURE

```
T = INDEX (SUPTAB.K1);
MESS 'K1 NODE 1 FOR THE CRACK TIP (TOP NODE) : '
      SUPTAB.K1.(T.1);
MESS 'K1 NODE 2 FOR THE CRACK TIP (MIDDEL NODE) : '
      SUPTAB.K1.(T.2);
```

- As output, SUTAB makes it possible to recover the values of the stress intensity factor

K1 : TABLE containing the values of K1 at each node of the front (here at 2 nodes T.1 and T.2)



6.7 Test rupt7 (Description sheet)

Test name: rupt7

Calculation type: MECHANICS ELASTIC

Finite element type: QUA8 DIME 2

Topic: STRESS INTENSITY FACTOR FOR AN OBLIQUE EDGE CRACK IN A PLATE SUBJECTED TO UNIFORM TENSILE STRESS. The structure is a finite plate enclosing an oblique edge crack. This structure is subject to uniform tensile stress. The CASTEM solution for the stress intensity factors K is given with the displacements method in 3 nodes at the crack tip (including the mid-node) from the local displacements which are computed in the local basis $(X'Y')$ for the crack by the relationship:

$$K_I = \frac{E'}{4} \left(\sqrt{\frac{2\Pi}{r-\beta}} \right) v \quad K_{II} = \frac{E'}{4} \left(\sqrt{\frac{2\Pi}{r-\beta'}} \right) u \quad \text{where } E' = \sqrt{E/1-\nu^2} \text{ (planestrain)}$$

$\beta = -\frac{d}{c}$ c and d are the coefficients of the line : $v^2 = cr + d$

$\beta' = -\frac{b}{a}$ a and b are the coefficients of the line : $u^2 = ar + b$

u : is the displacement following the ox axis.

v : is the displacement following the oy axis .


r : is the distance between the point, where K is computed, and the local origin.

This CASTEM K value for the stress intensity factor is compared to the result obtained with the analytical method.

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Compendium of Intensity Factors HMSO (1976): rupt7 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



Tension for a plate with the 45° angled kinked crack

GEOMETRY :

Width of the plate : $b = 100$ mm
 Length of the plate : $2.5b = 250$ mm
 Length of the crack : $a = 50$ mm
 $\beta = 45^\circ$

MATERIAL CHARACTERISTICS :



$E = 20000$ daN mm⁻²
 $\nu = 0.3$

LOADING CONDITIONS :

Tensile Stress :
 $\sigma = 10$ daN mm⁻²

Test rupt7 (Results)
RESULTS

	$K_I(\text{daNmm}^{-3/2})$	$K_{II}(\text{daNmm}^{-3/2})$
Reference	150.37	-71.4
Castem	151.2	-71.56

CASTEM FIGURES




6.7. TEST RUPT7 (DESCRIPTION SHEET)

Test rupt7 (Listing CASTEM)

```
*****
* STRESS INTENSITY FACTOR FOR A OBLIQUE EDGE CRACK*
* IN A PLATE SUBJECTED TO UNIFORM TENSILE STRESS *
*****
opti dime 2 elem qua8 mode plan defo echo 0;
*
*-----GEOMETRY AND MESHING-----*
*
p0=0. 0.;p11=100. 0.;
p12=p11 plus (0. 0.);p2=250. 0.;
p3=0. 100.;
a=50. * (cos 45);
pf=p11 plus (a a);
p4=(p11 plus (a 0.)) plus (0. 100.);p5=250. 100.;
i=0; pd=prog 1. 2.5 5. 9. 13. 19. 25. ;
cim1=pf;cim2=pf;cim3=pf;cim41=pf;cim42=pf;
plfim1=pf;plfim2=pf;pim2=pf;
repete bbbb 7;
  i= i + 1;v=extr pd i;
  pi1=pf moin (v 0.);pi2=pf plus (0. v);
  pi3=pf plus (v 0.);pi4=pf moin (0. v);
  ci4=cerc 2 pi4 pf pi1;
  plfil=ci4 poin 2;plfi2=plfil plus (0. 0.);
  n=2;si ((i ega 6 ) ou (i ega 7));n=4;finsi;
  m = n / 2;
  cil=cerc n pi1 pf pi2;
  ci2=cerc n pi2 pf pi3;
  ci3=cerc n pi3 pf pi4;
  ci41=cerc m pi4 pf plfil;
  ci42=cerc m plfi2 pf pi1;
  sui1=cout cim1 cil;
  sui2=cout cim2 ci2;
  sui3=cout cim3 ci3;
  sui41=cout cim41 ci41;
  sui42=cout cim42 ci42;
  elim 0.001 (sui42 et sui1);
  elim 0.001 (sui1 et sui2 et sui3 et sui41);
  lf11=(cont sui41) comp plfim1 plfil;
  lf12=(cont sui42) comp plfim2 plfi2;
  lhi=pim2 d 1 pi2;
  cim1=cil;cim2=ci2;cim3=ci3;cim41=ci41;cim42=ci42;
  plfim1=plfil;plfim2=plfi2;pim2=pi2;
  sut1=sui42 et sui1;
  sut2=sui2 et sui3 et sui41;
  si (i ega 1);
  sutt1=sut1;sutt2=sut2;lf1=lf11;lf2=lf12;lh=lhi;
  sinon;
  sutt1=sutt1 et sut1;sutt2=sutt2 et sut2;
  lf1=lf1 et lf11;lf2=lf2 et lf12;lh=lh et lhi;
  finsi;
fin bbbb;
lg=p0 d 5 p3;ld=p2 d 5 p5;
lhc= p4 d 2 pi2;
lig1=lg et (p3 d 8 p4) et lhc;
lig1=lig1 et (inve cil) et ci42;
lig1=lig1 et (plfi2 d 2 p12 d 5 p0);
sul=surf lig1 plan;obj1=sul et sutt1;elim 0.001 obj1;
lig2=ld et (p5 d 8 p4) et lhc;
lig2=lig2 et ci2 et ci3 et ci41;
lig2=lig2 et (plfil d 2 p11 d 8 p2);
su2=surf lig2 plan;obj2=su2 et sutt2;elim 0.001 obj2;
perobj1 = inve ( cont sutt1);
perobj2 = inve ( cont sutt2);
lifis1 = perobj1 comp pf plfi2;
lifis2 = perobj2 comp pf plfil;
*
obj=obj1 et obj2;
titre 'OBLIQUE EDGE CRACK IN
PLATE UNDER UNIFORM TENSION';

trace obj ;
*
*----MODEL AND MATERAIL CHARACTERISTICS-----*
*
objaf = mode obj MECANIQUE ELASTIQUE ISOTROPE;
mat=mate objaf youn 20000 nu 0.3;
*
*-----MATRIX OF RIGIDITY-----*
*
rig=rigi mat objaf;
*-----MECHANICS LOADING-----*
*
cdl=(bloq ux uy p0) et (bloq uy p2);
f1=pres mass objaf lg -10.;
f2=pres mass objaf ld -10.;
*
*-----RESOLUTION AND STRESSES-----*
*
u=reso (rig et cdl) (f1 et f2);
sig=sigm mat objaf u;
*----STRESS INTENSITY FACTOR ----*
*---- PROCEDURE SIF -----*
SUPTAB = TABLE;
SUPTAB.'MODMIXTE' = VRAI;
SUPTAB.'LIFIS1' = lifis2;
SUPTAB.'LIFIS2' = lifis1;
SUPTAB.'FRTRFISS' = pf;
SIF SUPTAB MAT U;
K1MOY = SUPTAB.K1;
K2MOY = SUPTAB.K2;
mess '*****';
mess '* THEORITECALS VALUES *';
mess '*****';
*
mess 'K1th = 150.37 ';
EC1 = (K1MOY - 150.37) / K1MOY * 100;
mess 'ECART entre K1MOY et K1TH : 'EC1'%';
mess 'K2th = -71.4 ';
EC2 = (K2MOY + 71.4) / K2MOY * 100;
mess 'ECART entre K2MOY et K2TH : 'EC2'%';
*
*---GOOD WORKING MESSAGE-----*
SI ((EC1 < 0.6) ET (EC2 < 0.25));
  MESS 'ERR 0';
  ERRE 0;
SINON;
  MESS 'ERR 5';
  ERRE 5;
FINSI;
fin;
```

6.8 Test rupt8 (Description sheet)

Test name: rupt8

Calculation type: MECHANICS ELASTICITY

Finite element type: TRI3 MODE TRID

Topic: STRESS INTENSITY FACTOR FOR A CENTRAL CRACK IN A FINITE WIDTH PLATE SUBJECTED TO UNIFORM TENSILE STRESS. The structure is a solid plate enclosing a central crack . This structure is subject to a uniform load by a simple tensile stress test. Due to symmetry only 1/4 the structure is analysed. The CASTEM solution for the stress intensity factors K is given by the relationship:

$$K_I = (\sqrt{EG/1-\nu^2}) \quad G \text{ is the energy release rate}$$


Finally this CASTEM K value for the stress intensity factor is compared to the result obtained with the analytical formula of ROOKE and CARTWRIGHT using the relationship:

$$\frac{K_I}{K_O} = \frac{1 - 0.5\frac{a}{b} + 0.326(\frac{a}{b})^2}{(1 - \frac{a}{b})^{1/2}}$$

$$K_0 = \sigma\sqrt{\pi a}$$

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Compendium of Intensity Factors HMSO (1976): rupt8 Modélisation des structures élastiques dans CASTEM 2000.

Model description:



GEOMETRY :

Width of the plate : $2b = 2 \text{ m}$
 Length of the plate : $2h = 10.5 \text{ m}$
 Length of the crack : $2a = 0.6 \text{ m}$
 Thickness of the plate : $e = 0.001 \text{ m}$

MATERIAL CHARACTERISTICS :

$E = 20000 \text{ daN mm}^{-2}$
 $\nu = 0.3$

LOADING CONDITIONS :

Uniform tensile stress
 $P = 5E6 \text{ N m}^{-2}$

**Test rupt8 (Results)****RESULTS**

	$K_I(Nm^{-3/2})$
Reference	50.1E5
Castem	50.8E5


CASTEM FIGURES

PLATE WITH CENTRAL CRACK SUBJECTED TO TENSILE STRESS




PLATE WITH CENTRAL CRACK SUBJECTED TO TENSILE STRESS

Test rupt8 (Listing CASTEM)

```

*****
* CENTRAL CRACK IN A FINITE WIDTH PLATE *
* SUBJECTED TO UNIFORM BENDING MOMENT *
*****
opti echo 0; OPTI DIME 3 ELEM TRI3 DENS 1.;
N4 = 6; N8 = 13; N12 = (N4 + N8)*(-1);
O1 = 0. 0. 1000.; ADIM = 0.3; BDIM = 1.;
TAFIS = ADIM / 33.; HH1 = BDIM / 2.0;
P1 = 0. 0. 0.; P2 = BDIM 0. 0.;
P3 = ADIM 0. 0.; P3G = (ADIM - TAFIS) 0. 0.;
P3D = (ADIM + TAFIS) 0. 0.; ALF1 = ATG HH1 ADIM;
ALF2 = ATG HH1 (BDIM - ADIM);
X1 = TAFIS * (COS ALF1); Y1 = TAFIS * (SIN ALF1);
P31 = (ADIM - X1) Y1 0.; X1 = TAFIS * (COS ALF2);
Y1 = TAFIS * (SIN ALF2); P32 = (ADIM + X1) Y1 0.;
C1 = C N4 P3G P3 P31; C2 = C N8 P31 P3 P32;
C3 = C N4 P32 P3 P3D; SU1 = COUT (C1 ET C2 ET C3) P3;
P4 = 0. HH1 0.; P5 = BDIM HH1 0.;
L1 = D N4 P1 P4 DINI (2.0*TAFIS) DFIN (8.0*TAFIS);
L2 = D N12 P4 P31 DINI (9.0*TAFIS) DFIN (0.3*TAFIS);
L3 = D N12 P1 P3 P3D DINI (3.0*TAFIS) DFIN (1.5*TAFIS);
L4 = D N8 P4 P5 ;
L5 = D N12 P5 P32 DINI (14.0*TAFIS) DFIN (0.3*TAFIS);
L6 = D N4 P2 P5 DINI (2.0*TAFIS) DFIN (10.0*TAFIS);
L7 = D N12 P2 P3D DINI (10.0*TAFIS) DFIN (1.5*TAFIS);
SU2 = DALL L1 L3 C1 L2; SU3 = DALL L2 L4 L5 C2;
SU4 = DALL L6 L5 C3 L7;
SU5 = L4 TRAN 7 (0. (1.0*BDIM) 0.);
L8 = COTE 3 SU5; Y1 = COOR 2 (POINT L8 FINA);
P6 = 0. (Y1 + (BDIM/4.)) 0.;
P7 = BDIM (Y1 + (BDIM/4.)) 0.;
L9 = D N4 P7 P6; SU6 = COUT L8 L9;
SU7 = L9 TRAN 8 (0. (3.5*BDIM) 0.);
L10 = COTE 3 SU7; Y1 = COOR 2 (POINT L10 FINA);
L11 = D 1 P3 P3D;
SUT = COUL TURQ (SU1 ET SU2 ET SU3
ET SU4 ET SU5 ET SU6 ET SU7);
ELIM 1.E-8 SUT;
L1 = (CONT SUT) ELEM APPU (SUT POIN
DROI (0. 0. 0.) (BDIM 0. 0.) 1.E-8);
L2 = (CONT SUT) ELEM APPU (SUT POIN
DROI (BDIM 0. 0.) (BDIM Y1 0.) 1.E-8);
L3 = (CONT SUT) ELEM APPU (SUT POIN
DROI (BDIM Y1 0.) (0. Y1 0.) 1.E-8);
L4 = (CONT SUT) ELEM APPU (SUT POIN
DROI (0. Y1 0.) (0. 0. 0.) 1.E-8);
PINI3 = POIN L3 INIT; PFINI3 = POIN L3 FINA;
PMIL3 = (CHAN POI1 L3) DIFF (PINI3 ET PFINI3);
MOEXT = 1. / ((NBNO L3) - 1);
MOMIL = MOEXT * (NBNO PMIL3);
*
*---- RESOLUTION-STRESSES AND DISPLACEMENTS ----*
*
EPAI_T = 0.001; NB_MUL = 21;
EPAI_C = EPAI_T * 1.E-6;
EPAI_I = (EPAI_T - (EPAI_C*NB_MUL))/(NB_MUL - 1);
ELEFIS = ELEM 1 (SUT ELEM APPU LARG P3);
TBEPAI = TABLE; TBEXCE = TABLE;
REPETER BC_MUL1 ((2*NB_MUL) - 1);
III = (&BC_MUL1 - NB_MUL) / 2.;
TBEXCE.&BC_MUL1 = (EPAI_I + EPAI_C)*III;
SI (EGA 0. ((&BC_MUL1/2.) - (&BC_MUL1/2)) 1.E-10);
TBEPAI.&BC_MUL1 = EPAI_I;
SINON;
TBEPAI.&BC_MUL1 = EPAI_C;
FINSI;
FIN BC_MUL1;

REPETER BC_MUL2 ((2*NB_MUL) - 1);
SI (EGA &BC_MUL2 1);
MOD_MUL = MODE ELEFIS MECANIQUE ELASTIQUE
DKT CONS (CHAI &BC_MUL2);
MAT_MUL = MATE MOD_MUL YOUN 2.E11 NU 0.3
EPAI TBEPAI.&BC_MUL2 EXCE TBEXCE.&BC_MUL2;
SINON;
MOD_MU1 = MODE ELEFIS MECANIQUE ELASTIQUE
DKT CONS (CHAI &BC_MUL2);
MAT_MU1 = MATE MOD_MU1 YOUN 2.E11 NU 0.3
EPAI TBEPAI.&BC_MUL2 EXCE TBEXCE.&BC_MUL2;
MOD_MUL = MOD_MUL ET MOD_MU1;
MAT_MUL = MAT_MUL ET MAT_MU1;
FINSI;
FIN BC_MUL2;
MOD_R = MODE (SUT DIFF ELEFIS) MECANIQUE ELASTIQUE DKT;
MAT_R = MATE MOD_R YOUN 2.E11 NU 0.3 EPAI EPAI_T;
*
TRACE SUT TITRE 'PLATE WITH CENTRAL
CRACK SUBJECTED TO TENSILE STRESS';
*
MO1 = MOD_MUL ET MOD_R; MA1 = MAT_MUL ET MAT_R;
*
BL1 = BLOQ UY RX (L7 ET L11);
BL2 = BLOQ UX RY L4; BL3 = BLOQ UZ RZ L3;
RI = BL1 ET BL2 ET BL3 ET (RIGI MA1 MO1);
FO2 = ((FORC FY MOEXT (PINI3 ET PFINI3))
ET (FORC FY MOMIL PMIL3))*5.e3; FO3 = FO2;
*
DEPTO = RESO RI FO3 ; REATO = REAC DEPTO RI;
SIGTO = SIGMA MA1 MO1 DEPTO ;
*****
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'J';
SUPTAB.'LEVRE_SUPERIEURE' = comp (cont sut) p3 pl;
SUPTAB.'FRONT_FISSURE' = P3 ;
SUPTAB.'MODELE' = MO1;
SUPTAB.'CARACTERISTIQUES' = MA1;
SUPTAB.'SOLUTION_RESO' = DEPTO;
SUPTAB.'CHARGEMENTS_MECANIQUES' = FO3;
SUPTAB.'COUCHE' = 7;
G_THETA SUPTAB;
*
*RESULTS AND COMPARAISON WITH THEORITECAL VALUE *
*
G = SUPTAB.'RESULTATS'. 'GLOBAL' ;
Kth = 50.1E5 ;
K = (2.E11 * G)**0.5 ;
mess ' ' ;
mess ' ' ;
mess 'ENERGY RELEASE RATE G : ' G ;
mess 'COMPUTED VALUE Kcal : ' K ;
mess 'ANALYTICAL VALUE Kthe : ' Kth ;
*
* ----- GOOD WORKING MESSAGE----- *
*
RESI = abs (( k - Kth ) / Kth ) ;
SI (RESI < 5E-2) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
FIN;

```




6.9 Test rupt9 (Description sheet)

Test name: rupt9

Calculation type: MECHANICS ELASTICITY

Finite element type: QUA8 DIME 2

Topic: VALIDATION OF THE G_THETA AND T_PITETA PROCEDURES FOR AN OBLIQUE EDGE CRACK IN A PLATE SUBJECTED TO UNIFORM TENSILE STRESS. The structure is a finite plate enclosing an edge crack. This structure is subject to uniform tensile stress. The CASTEM solution for the stress intensity factor K is given with the displacement method in 3 nodes at the crack tip (including the mid-node) from the local displacements which are computed in the local basis ($X'Y'$) for the crack by the relationship:


$$K_I = \sqrt{\frac{EG_I}{1-\nu^2}} \quad \text{and} \quad K_{II} = \sqrt{\frac{EG_{II}}{1-\nu^2}}$$

Finally this CASTEM K values for the stress intensity factor is compared to the results obtained with the analytical method.

Reference CASTEM: Test NAFEMS : D.P.Rooke and D. J. Cartwright in Compendium of Intensity Factors HMSO (1976): rupt7 Modélisation des structures élastiques dans CASTEM 2000.

Version: 97' customer version

Model description:



Tension for a plate with the 45° angled kinked crack

GEOMETRY :

Width of the plate : $b = 100$ mm
 Length of the plate : $2.5b = 250$ mm
 Length of the crack : $a = 50$ mm
 $\beta = 45^\circ$

MATERIAL CHARACTERISTICS :


$E = 20000$ daN mm⁻²
 $\nu = 0.3$

LOADING CONDITIONS :


Tensile Stress :
 $\sigma = 10$ daN mm⁻²

Test rupt9 (Results)
RESULTS

	$K_I(\text{daNmm}^{-3/2})$	$K_{II}(\text{daNmm}^{-3/2})$
Reference	149.77	-71.56
Castem	145.98	-72.28

CASTEM FIGURES


GIBI FRCIT



GIBI FRCIT



6.9. TEST RUPT9 (DESCRIPTION SHEET)

Test rupt9 (Listing CASTEM)

```
*****
* VALIDATION OF THE G_THETA AND T_PITETA *
* PROCEDURES FOR A OBLIQUE EDGE CRACK IN A *
* PLATE SUBJECTED TO UNIFORM TENSILE STRESS *
*****
opti dime 2 elem qua8 mode plan defo echo 1;
** DD : INITIAL CRACK OPENING DISPLACEMENT
DD = 0.001;
p11=(125. - (DD/2.)) 0.; p12=(125. + (DD/2.)) 0.;
p0=25. 0.; p2=275. 0.; p3=25. 100.; ppp = 125. 0.;
L=50.; tetal=45; a=L*(cos tetal); b=L*(sin tetal);
pfl=ppp plus (a b); p5=275. 100.;
tgteta = (sin tetal)/(cos tetal);
p4=((100./tgteta) + 125.) 100.;
*****
* MESH OF THE CRACK TIP *
*****
i=0;
pd=prog 1. 2.5 5. 9. 13. 19. 25.;
cim1=pfl;cim2=pfl;cim3=pfl;cim4=pfl;cim12=pfl;
plfim1=pfl;plfim2=pfl;pim3=pfl;
*****
reperer bbbb (14 / 2);
i= i + 1;V=extr pd i; ddl = (v / L)*(DD*0.5);
XX1 = v*(cos tetal); YY1 = v*(sin tetal);
XX2 = ddl*(sin tetal)*(sin tetal);
YY2 = ddl*(sin tetal)*(cos tetal);
pil=pfl moin (XX1 YY1);
pi2=pfl plus ((0. - XX1) YY1);
pi3=pfl plus (XX1 YY1);
pi4=pfl moin ((0. - XX1) YY1);
plfil=pil moin (XX2 (0. - YY2));
plfi2=pil plus (XX2 (0. - YY2));
n=2;si ((i ega 6) ou (i ega 7));
n=4; finsi;
*****
cil1=cerc n plfil pfl pi2;
ci2=cerc n pi2 pfl pi3;
ci3=cerc n pi3 pfl pi4;
cil2=cerc n pi4 pfl plfi2;
suil1=cout cim1 cil1;
suil2=cout cim2 cil2;
sui2=cout cim2 ci2; sui3=cout cim3 ci3;
elim 0.001 (suil1 et sui2);
elim 0.001 (suil2 et sui3);
*****
lfil=(cont suil1) comp plfim1 plfil;
lfi2=(cont suil2) comp plfim2 plfi2;
lhi=pim3 d 1 pi3;
cim1=cil1;cim2=ci2;cim3=ci3;cim12=cil2;
plfim1=plfil;plfim2=plfi2;pim3=pi3;
sut1=suil1 et sui2; sut2=sui3 et suil2;
si (i ega 1);
sutt1=sut1;sutt2=sut2;lf1=lfil;
lf2=lfi2;lh=lhi;
sinon;
sutt1=sutt1 et sut1;sutt2=sutt2 et sut2;
lf1=lf1 et lf1;lf2=lf2 et lfi2;lh=lh et lhi;
finsi;
fin bbbb;
lig1 = (cont sutt1) comp pfl pim3;
lig2 = (cont sutt2) comp pfl pim3;
elim 0.001 lig1 lig2;
*****
* MESH OF THE PLATE *
*****
lg=p0 d 6 p3; ld=p2 d 6 p5;

lig1=lg et (p3 d 13 p4) et lhc;
lig1=lig1 et (inve ci2) et cil1;
lig1=lig1 et (plfil d 3 p11 d 6 p0);
sul=surf lig1 plan; obj1=sul et sutt1;
elim 0.001 obj1;
lig2=ld et (p5 d 3 p4) et lhc;
lig2=lig2 et ci3 et cil2;
lig2=lig2 et (plfi2 d 3 p12 d -9 p2
dini 40 dfin 50);
su2=surf lig2 plan; obj2=su2 et sutt2;
elim 0.001 obj2;
lsutt1 = (cont sutt1) comp pfl pi3;
lsutt2 = (cont sutt2) comp pi3 pfl;
elim 1.e-6 lsutt1 lsutt2;
*
perobj1 = inve (cont sutt1);
perobj2 = inve (cont sutt2);
lifis1 = perobj1 comp pfl plfil;
lifis2 = perobj2 comp pfl plfi2;
obj=obj1 et obj2; surf1=sutt1 et sutt2;
*
ls = (cont obj) comp p11 pfl ;
li = (cont obj) comp pfl p12 ;
TRACE OBJ ;
*---MODEL AND MATERIAL CHARACTERISTICS-----*
objaf = mode obj MECANIQUE ELASTIQUE ISOTROPE;
mat=mate objaf youn 20000 nu 0.3;
*-----MATRIX OF RIGIDITY-----*
rig=rigi mat objaf;
*-----BOUNDARY CONDITIONS-----*
cdl=(bloq ux uy p0 ) et (bloq uy p2);
rigtot = rig et cdl ;
*-----MECHANICS LOADING-----*
f1=pres mass objaf lg -10.;
f2=pres mass objaf ld -10.;
f0 = f1 et f2;
*-----RESOLUTION AND STRESSES-----*
u0=reso rigtot f0; sig0=sigm mat objaf u0;
*****
* G-THETA PROCEDURE AND RESULTS *
*****
SUPTAB = TABLE ;
SUPTAB.'OBJECTIF' = MOT 'DECOUPLAGE';
SUPTAB.'LEVRE_SUPERIEURE' = ls;
SUPTAB.'LEVRE_INFERIEURE' = li;
SUPTAB.'FRONT_FISSURE' = Pfl ;
SUPTAB.'MODELE' = objaf;
SUPTAB.'CARACTERISTIQUES' = mat;
SUPTAB.'SOLUTION_RESO' = u0;
SUPTAB.'CHARGEMENTS_MECANIQUES' = f0;
I=0; repeter bbbb 1; I = I + 1;
SUPTAB.'COUCHE'=6; G_THETA SUPTAB ;
fin bbbb;
K1 = SUPTAB.'RESULTATS'. 'I';
K2 = SUPTAB.'RESULTATS'. 'II';
MESS 'K1 = ' K1 'K2 = ' K2;
ER1 = ABS (((K1 - 149.77) / K1) * 100);
ER2 = ABS (((K2 + 71.56) / K2) * 100);
SI ((ER1 > 0.5) OU (ER2 > 2.0));
ERRE 5;
SINO;
ERRE 0;
FINSI; FIN;
```




Chapter 7

COMPOSITE

7.1 Test comp1 (Description sheet)

Test name: comp1

Calculation type: MECHANICS COMPOSITE 3D

Finite element type: DKT


Topic: A two layer composite cylinder subjected to internal pressure. The cylinder fibers are wound around the Z-axis ($\pm 45^\circ$). The cylinder is subjected to an internal pressure. The displacements of its bottom end are constrained along the Z axis. In order to reduce the number of freedom degrees, one of the top nodes is forced to move along X and Y-axis and in rotation around Z-axis (node PB). The pressure is perpendicular to the internal surface of the cylinder.

Goal: Find the radial displacement of the point PA (1.05, 0, 0) belonging to the bottom of the cylinder.

Reference CASTEM: Report CEA-DMT 89/184 : Modélisation des composites de type multi-couches dans CASTEM 2000.

Version: 97' customer version

Model description:



H	$= 1m$	E_1	$= 7.10^6 Pa$
R	$= 1.05m$	E_2	$= 1.3.10^6 Pa$
th_{tot}	$= 0.1m$	G_{12}	$= 5.10^5$
P	$= 100Pa$	ν_{12}	$= 0.28$




Test comp1 (Results)


Results

	U_r
Theory	1.3776E-4 m
Castem	1.3777E-4 m

CASTEM figures



A TWO COMPOSITE LAYERS CYLINDER SUBJECTED TO INTERNAL PRESSURE



A TWO COMPOSITE LAYERS CYLINDER SUBJECTED TO INTERNAL PRESSURE

Test comp1 (Listing CASTEM)

```

*****
TITRE 'A TWO COMPOSITE LAYERS CYLINDER
      SUBJECTED TO INTERNAL PRESSURE';
OPTION DIME 3 ELEM TRI3 MODE TRIDIM ;
TEMPS ;
DENS 0.1 ;
*
*   GEOMETRY
*
R = 1.05 ; H = 1. ;
PA = R 0. 0. ; PB = R 0. H ;
O1 = 0. 0. 0. ; O2 = 0. 0. H ;
NR1 = 1. 1. 0. ;
LI1 = PA D 4 PB ;
CYLP = ROTA LI1 8 90 O1 O2 ;
CYL1 = ORIENTER CYLP NR1 ;
CYL2 LIB = TOURNER CYL1 LI1 90 O1 O2 ;
CYL3 LIC = TOURNER CYL2 LIB 90 O1 O2 ;
CYL4 LID = TOURNER CYL3 LIC 90 O1 O2 ;
CYL = CYL1 ET CYL2 ET CYL3 ET CYL4 ;
ELIM CYL ;
*
OEIL = 10. 10. 5. ;
TRACE OEIL CACH CYL ;
*
*   DESCRIPTION OF MODEL AND ORTHOTROPIC
*   MATERIAL OF LAYER 1
*
MODORT1 = MODEL CYL MECANIQUE ELASTIQUE
          ORTHOTROPE DKT CONST 'COUCHE 1';
MAT1 = MATER MODORT1 DIRE O2 INCL 45 NR1
       YG1 7E6 YG2 1.3E6 NUI2 0.28 G12 5E5 ;
CAR1 = CARAC MODORT1 EPAI 0.05 EXCENTR 0.025 ;
MAT1=MAT1 ET CAR1;
*
*   DESCRIPTION OF MODEL AND ORTHOTROPIC
*   MATERIAL OF LAYER 2
*
MODORT2 = MODEL CYL MECANIQUE ELASTIQUE
          ORTHOTROPE DKT CONST 'COUCHE 2';
MAT2 = MATER MODORT2 DIRE O2 INCL -45 NR1
       YG1 7E6 YG2 1.3E6 NUI2 0.28 G12 5E5 ;
CAR2 = CARAC MODORT2 EPAI 0.05 EXCENTR -0.025 ;
MAT2=MAT2 ET CAR2;
*
*   RIGIDITY MATRIX
*
MODORT = MODORT1 ET MODORT2 ;
MATTOT = MAT1 ET MAT2 ;
RIGI = RIGI MODORT MATTOT ;
*
*   BOUNDARY CONDITIONS
*
CO1 = COTE 2 CYL1 ; CO2 = COTE 2 CYL2 ;
CO3 = COTE 2 CYL3 ; CO4 = COTE 2 CYL4 ;
COB = CO1 ET CO2 ET CO3 ET CO4 ;
CDL1 = BLOQ UZ COB ;
CDL2 = BLOQ UX UY RZ PB ;
CDL = CDL1 ET CDL2 ;
*
*   CALCULATION
*
RIGITOT = RIGI ET CDL ;
FP = PRES COQUE MODORT1 100. NORM ;
    
```

```

DEPL = RESO RIGITOT FP ;
*
*   OUTPUT
*
DEF01=DEFO CYL DEPL;
DX=EXCO DEPL UX;
TRAC OEIL CACH DEF01 DX;
*
D = EXTR DEPL UX PA ;
SAUT PAGE ;
MESS ' RADIAL REFERENCE DISPLACEMENT
      (NODE PA)      :   1.37762E-4 ' ;
SAUT 1 LIGN ;
MESS ' RADIAL CALCULATED DISPLACEMENT
      (NODE PA)      :   ' D ;
SAUT 2 LIGN ;
TEMPS ;
*
*           GOOD WORKING MESSAGE
DEPREF=1.37762E-4;
RESI=ABS((D-DEPREF)/DEPREF);
SI (RESI <EG 1E-2);
  ERRE 0 ;
SINO;
  ERRE 5 ;
FINSI ;
FIN;
    
```




Test comp1 (Comments)

1. Description of orthotropic material

MAT1 = MATER MODORT1 DIRE O2 INCL 45 NR1 YG1 7E6 YG2 1.3E6 NU12 0.28 G12 5E5 ;

The orthotropy is defined by the first orthotropic direction. First, a reference vector VEC1 is defined, thus giving the first orthotropic direction.

- Reference vector VEC1

DIRE O2 : projecting the direction O2 on the tangential plane to the shell.

- First orthotropic direction

INCL 45 NR1 : The first orthotropic direction makes an angle of 45° with the reference direction VEC1. The direction NR1 gives the direction perpendicular to the shell, orientated outwards.

2. Definition of the offsetting with the median line


CAR1 = CARAC MODORT1 EPAI 0.05 EXCENTR 0.025 ;

The median line of the layer 1 is 0.025 meters (=EPAI/2) away from the median line of the cylinder ($R = 1.05$).

3. Drawing of displacements with deformed shape

TRAC OEIL CACH DEFO1 DX;

It is possible to plot the isodisplacements or isostrains on the deformed shape. For this, the object DEFO1 (DEFORME type) must be added in the TRAC operator.



7.2 Test comp2 (Description sheet)

Test name: comp2

Calculation type: MECHANICS COMPOSITE 3D

Finite element type: DKT


Topic: A two layer composite shell subjected to a distributed pressure. The shell fibers make an angle of $\pm 45^\circ$ with the median line. The square shell is supported on its circumference (displacements and rotation along Z-axis are constrained). The middle node only can move along the vertical Z-axis. The pressure is distributed on the entire shell (upper face).

Goal: Find the axial displacement of the middle point PO (0, 0, 0).

Reference CASTEM: Report CEA-DMT 89/186 :

Version: 97' customer version

Model description:



$$\begin{aligned}
 A &= 127\text{mm} \\
 Th_{tot} &= 5.08\text{mm} \\
 P &= 0.6894\text{MPa}
 \end{aligned}$$

$$\begin{aligned}
 E_1 &= 276.10^3\text{MPa} \\
 E_2 &= 6.3.10^3\text{MPa} \\
 G_{12} &= 3.4.10^3 \\
 \nu_{12} &= 0.25
 \end{aligned}$$




Test comp2 (Results)


Results

	$U_z(PO)$
Reference	23.25
Castem	23.126

CASTEM figures




COMPOSITE SUPPORTED SHELL - DISTRIBUTED PRESSURE



COMPOSITE SUPPORTED SHELL - DISTRIBUTED PRESSURE

VAL - ISO
>= -5.74E-16
< 2.31E+01



AMPLITUDE
DEFORMEE
1.1



7.2. TEST COMP2 (DESCRIPTION SHEET)

Test comp2 (Listing CASTEM)

```
*****
TITRE 'COMPOSITE SUPPORTED SHELL -
      DISTRIBUTED PRESSURE';
OPTI DIME 3 ELEM TRI3 ECHO 0 ;
*
* _____
*   GEOMETRY
* _____

TEMPS ;

A = 127. ; B = A ; AM = -127. ; BM = AM ;
H = 5.08 ; ES = H/4 ; EI = H/-4 ; S = H/2 ;
P = 0.6894 ;
N = 12 ;
P0 = 0. 0. 0. ;
P1 = AM BM 0. ; P2 = A BM 0. ;
P3 = A B 0. ; P4 = AM B 0. ;
V1 = 0. (2*B) 0. ; V2 = 0. 0. 1. ;
L1 = DROI N P1 P2 ;

S11=L1 TRAN V1 N;
S1=ORIE S11 V2;
CONF (S1 POIN PROC P1) P1 ;
CONF (S1 POIN PROC P2) P2 ;
CONF (S1 POIN PROC P3) P3 ;
CONF (S1 POIN PROC P4) P4 ;
CONF (S1 POIN PROC P0) P0 ;
*
TRACE S1 V2 ;
*
* _____
*   MATERIAL
* _____
MOD1S = MODEL S1 MECANIQUE ELASTIQUE
      ORTHOTROPE DKT CONST 'UPPER LAYER';
MOD1I = MODEL S1 MECANIQUE ELASTIQUE
      ORTHOTROPE DKT CONST 'LOWER LAYER';

CARS = CARAC MOD1S EPAI S EXCE ES ;
CARI = CARAC MOD1I EPAI S EXCE EI ;
E11 = 276E3 ; E22 = 6.9E3 ;
N12 = 0.25 ; GX = 3.4E3 ;
MATS = MATER MOD1S DIRE V1 INCLINE 45. V2
      YG1 E11 YG2 E22 NU12 N12 G12 GX ;
MATI = MATER MOD1I DIRE V1 INCLINE -45. V2
      YG1 E11 YG2 E22 NU12 N12 G12 GX ;
MATS = MATS ET CARS;
MATI = MATI ET CARI;
*
* _____
*   BOUNDARY CONDITIONS AND RIGIDITY
* _____
RIS = RIGI MOD1S MATS ;
RII = RIGI MOD1I MATI ;
CL=(BLOQ UZ (CONT S1)) ET (BLOQ RZ (CONT S1))
  ET (BLOQ UX UY RX RY P0);
RIT = RIS ET RII ET CL ;
*
* _____
*   CALCULATION AND OUPUT
* _____
FP = PRES COQU MOD1S P V2 ;
DEP = RESO RIT FP ;
FP1 = EXTR DEP UZ P0 ;
SAUT PAGE ;
SU1=DEFO S1 DEP;
DZ=EXCO DEP UZ;
OEIL= 0 1000 0;
TRAC OEIL CACH SU1 DZ;
*

MESS 'THEORICAL DEFLECTION      23.25' ;
```




Chapter 8

CONCRETE

8.1 Test beton (Description sheet)

Test name: beton

Calculation type: ELASTIC PLASTIC MECHANICS - CONCRETE

Finite element type: CUB8

Topic: Tensile test on a cubic piece made of concrete. The structure is an embedded cube subjected to a tensile stress (imposed displacements). The structure follows a concrete elastic-plastic model. The limite in tension in the first direction for concrete (LRT1) is 4 MPa.


Goal: Test the model of concrete material for two new structures (MMODEL and MCHAMELEM).

Reference: NAH (93/01/19)

Version: 97' customer version

Model description:

$E = 40000 \text{ MPa}$
 $\nu = 0.2$
 $\alpha = 1.10^{-5}$
 $LTR1 = 4 \text{ MPa}$







8.1. TEST BETON (DESCRIPTION SHEET)

Test beton (Results)

CASTEM FIGURES



CONCRETE CUBE SUBJECTED TO TENSILE STRENGTH



CONCRETE CUBE SUBJECTED TO TENSILE STRENGTH

X1.E-4

Test beton (Listing CASTEM)

```

*****
TITR'CONCRETE CUBE SUBJECTED TO TENSILE STRENGTH';
*****
OPTI DIME 3 ELEM CUB8 MODEL TRIDIM ;
OPTI ECHO 0 ;
N = 1 ;
*
*----- GEOMETRY -----*
DENS N ;
A1 = 0 0 0 ; A2 = N 0 0 ;
A3 = N N 0 ; A4 = 0 N 0 ;
B1 = 0 0 N ; B2 = N 0 N ;
B3 = N N N ; B4 = 0 N N ;

A1A2 = A1 DROI N A2 ; A2A3 = A2 DROI N A3 ;
A3A4 = A3 DROI N A4 ; A4A1 = A4 DROI N A1 ;
B1B2 = B1 DROI N B2 ; B2B3 = B2 DROI N B3 ;
B3B4 = B3 DROI N B4 ; B4B1 = B4 DROI N B1 ;
A1B1 = A1 DROI N B1 ; A2B2 = A2 DROI N B2 ;
A3B3 = A3 DROI N B3 ; A4B4 = A4 DROI N B4 ;

A1A2A3A4 = A1A2 A2A3 A3A4 A4A1 DALL PLAN ;
B1B2B3B4 = B1B2 B2B3 B3B4 B4B1 DALL PLAN ;

A1A2B2B1 = A1A2 A2B2 (INVE B1B2) (INVE A1B1)
DALL PLAN ;
A4A3B3B4 = (INVE A3A4) A3B3 B3B4 (INVE A4B4)
DALL PLAN ;
A2A3B3B2 = A2A3 A3B3 B2B3 (INVE A2B2)
DALL PLAN ;
A1A4B4B1 = (INVE A4A1) A4B4 B4B1 (INVE A1B1)
DALL PLAN ;
TOTAL = PAVE A1A2A3A4 B1B2B3B4 A1A2B2B1 A4A3B3B4
A2A3B3B2 A1A4B4B1 ;
TRAC TOTAL CACH (1000 -2000 1000) ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO = MODE TOTAL MECANIQUE ELASTIQUE PLASTIQUE
BETON CUB8 ;
MA = MATE MO YOUN 40000 NU 0.20 ALPH 1E-5 LTR1 4 ;
*
*----- BOUNDARY CONDITIONS -----*
ENC1 = BLOQ UX A1A4B4B1 ;
ENC2 = BLOQ UY A1A2B2B1 ;
ENC3 = BLOQ UZ A1A2A3A4 ;
ENC = ENC1 ET ENC2 ET ENC3 ;

* Locking for imposed displacements
*END1 = BLOQ UX A2A3B3B2 ;
*END2 = BLOQ UY A4A3B3B4 ;
END3 = BLOQ UZ B1B2B3B4 ;
*END = END1 ET END2 ET END3 ;
END = END3 ;
EN = ENC ET END ;

* Values of imposed displacements
*FEXT1 = DEPI END1 1E-4 ;
*FEXT2 = DEPI END2 1E-4 ;
FEXT3 = DEPI END3 1E-4 ;
*FEXT = FEXT1 ET FEXT2 ET FEXT3 ;
FEXT = FEXT3 ;
*
* Strength as a function of time
LT1 = PROG 0. 10. ;
LT2 = PROG 0. 10. ;
EV = EVOL ROUGE MANU TEMPS LT1 FORCE LT2 ;

CHAT = CHAR DIMP FEXT EV ;

```

```

*
*----- SETUP OF THE PASAPAS PROCEDURE -----*
LTT = PROG 0 1 2 ;
TAB1 = TABLE ;
TAB1.BLOCAGES_MECANIQUES = EN ;
TAB1.CARACTERISTIQUES = MA ;
TAB1.MODELE = MO ;
TAB1.CHARGEMENT = CHAT ;
TAB1.TEMPS_CALCULES = LTT ;
PASAPAS TAB1 ;

* TREATMENT OF THE RESULTS
PTT = PROG 0 ;
PSZ = PROG 0 ;
IC = 0 ;
REPETER MABOU 2 ;
IC = IC + 1 ;
DD = TAB1 . DEPLACEMENTS . IC ;
PTT = PTT ET ( PROG ( EXTR DD UZ B3 ) ) ;
SS = TAB1.CONTRAINTE.IC ;
SZ = EXTR ( CHAN CHPO MO SS ) SMZZ B3 ;
PSZ = PSZ ET ( PROG SZ ) ;
FIN MABOU ;

EVSZ = EVOL VERT MANU DEPLACEMENT PTT
CONTRAINTE PSZ ;
DESS EVSZ ;
*
*----- GOOD WORKING MESSAGE -----*
SI ( (ABS ( SZ - 2. )) < 1.E-6 ) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;

FIN ;

```



Test beton (Comments)

- The BETON model in plane strain, axisymmetrical and 3D

```
MO = MODE TOTAL MECANIQUE ELASTIQUE PLASTIQUE
      BETON CUB8 ;
MA = MATE MO YOUN 40000 NU 0.20 ALPH 1E-5 LTR1 4 ;
```

In this model, the concrete behavior is non-linear in tension, and linear everywhere else.

- ('LTR1') : limit in tension in the 1st direction (YOUN*1.2E-4 by default)
('ETR1') : failure strain in tension in the 1st direction (3*LTR1/YOUN by default)
('LTT1') : transitional limit in tension in the 1st direction (0. by default)
('ETT1') : deformation corresponding to LTT1 (ETR1 by default)
('ERS1') : residual deformation in tension in the 1st direction (0. by default)
('VF1i') : three components of the VF1 vector specifying the direction i connected with LTR1
- ('LTR2') : limit in tension in the 2nd direction (by default LTR1)
('ETR2') : failure strain in tension in the 2nd direction (3*LTR2/YOUN by default)
('LTT2') : transitional limit in tension in the 2nd direction (0. by default)
('ETT2') : deformation corresponding to LTT2 (ETR2 by default)
('ERS2') : residual deformation in tension in the 2nd direction (0. by default)
('VF2i') : three components of the VF2 vector specifying the direction i connected with LTR2
- ('LTR3') : limit in tension in the 3rd direction (LTR1 by default)
('ETR3') : failure strain in tension in the 3rd direction (3*LTR3/YOUN by default)
('LTT3') : transitional limit in tension in the 3rd direction (0. by default)
('ETT3') : deformation corresponding to LTT3 (ETR3 by default)
('ERS3') : residual deformation in tension in the 3rd direction (ERS1 by default)
('VF3i') : three components of the VF3 vector specifying the direction i connected with LTR3,
required only in 3D when necessary.
- ('BETR') : residual reduction coefficient for the shear modulus in case of cracking (between 0.
and 1., 0.1 by default)

Caution : The VF1, VF2 and VF3 vectors must be orthogonal.

For computation with a tension limit different from the two others, it is required that the vector corresponding to this limit VF1, VF2 or VF3 be defined.

When LTR1, LTR2 and LTR3 are given, both the VF1 and VF2 vectors are required to define the directions 1, 2 and 3.

Non-zero values corresponding to initial cracks openings in the directions 1, 2 and 3 can be introduced by means of the TAB1 table used in the NONLIN procedure through TAB1.'VARI'. 'OUV1', TAB1.'VARI'. 'OUV2', TAB1.'VARI'. 'OUV3' may be input.

8.2 Test precont1 (Description sheet)

Test name: precont1

Calculation type: REINFORCED ELASTIC MECHANICS - BARR - 2D

Finite element type: QUA4

Topic: Calculation with passive reinforcement. The structure is a reinforced concrete beam. The reinforcement is made of tendon. The beam is subjected to pressure on upper surface, and embedded on its lower side.


Goal: Test the KABL, EQUI RIGI and PROI ARMA operators with passive reinforcement.

Reference: J.S. Fleuret 08/96

Version: 97' customer version

Model description:


<i>Concrete</i>	
E	= $0.4 \cdot 10^{11}$
NU	= 0.0
<i>Tendons</i>	
E	= $2.0 \cdot 10^{11}$
SECT	= 0.02






Test precont1 (Results)

CASTEM FIGURES



AMPLITUDE
0.002e+00
2.0
2.0

MESHING OF THE REINFORCEMENT AFTER KARL



AMPLITUDE
0.002e+00
2.0
2.0

Test precont1 (Listing CASTEM)

```

*****
TITR 'CALCULATION WITH PASSIVE REINFORCEMENT' ;
*****
OPTI DIME 2 ELEM QUA4 MODE PLAN ;
OPTI ECHO 1 ;
*
*----- GEOMETRY -----*
*--- Concrete
PA = 0.0 0.0 ; PB = 10.0 0.0 ;
PC = 10.0 1.0 ; PD = 0.0 1.0 ;
NB1 = 40 ; NB2 = 6 ;
C1 = PA DROI NB1 PB ; C2 = PB DROI NB2 PC ;
C3 = PC DROI NB1 PD ; C4 = PD DROI NB2 PA ;
SUR1 = DALL C1 C2 C3 C4 ;

*--- Tendons
PM = 0.0 0.1 ;
PN = 10.0 0.1 ;
D1 = PM DROI 10 PN ;
*--- Total geometry
D5 = SUR1 ET D1 ;
TITR 'MESHING OF THE BEAM' ;
TRAC ELEM D5 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MODB = MODE SUR1 MECANIQUE ELASTIQUE ;
MOD1 = MODE D1 MECANIQUE ELASTIQUE ARMATURE BARR ;
MOD2 = KABL MODB MOD1 ;

D2 = EXTR MOD2 MAIL ;
D5 = SUR1 ET D2 ;
TITR 'MESHING OF THE REINFORCEMENT AFTER KABL' ;
TRAC ELEM D2 ;

MAT1 = MATE MODB YOUN 0.4E11 NU 0.0 ;
MAT2 = MATE MOD2 YOUN 2.0E11 SECT 0.02 ;
MT = MAT1 ET MAT2 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ PA UX ; CL2 = BLOQ PA UY ;
CL3 = BLOQ PB UY ;
CLTOT = CL1 ET CL2 ET CL3 ;
*
*----- LOAD DEFINITION: PRESSURE -----*
CHAO = PRES MASS MODB C3 2.E6 ;
* Test: PREC and EQUI FORCE should not stop the
* program in the case of insufficiency of values.
* We don't need the prestress load in the end of
* program because of the passive reinforcement
PRE1 = PREC MOD2 MAT2 0.4E6 PM ;
CHA1 = EQUI FORCES MODB MOD2 PRE1 ;

*--- RESOLUTION WITH CONCRETE UNDER PRESSURE ---*
*--- Resolution
RIO = RIGI MODB MAT1 ;
RIG0CL = RIO ET CLTOT ;
DEP0 = RESO RIG0CL CHAO ;
DEP0A = PROI ARMA DEP0 MODB MOD2 ;
*--- Deformed shape
DE0 = DEFO (SUR1 ET D2) DEP0 0. BLAN ;
DE1 = DEFO SUR1 DEP0 2.0 ROUG ;
DE1A = DEFO D2 DEP0A 2.0 ROUG ;
TITR 'Deformed shape of concrete under pressure' ;
TRAC (DE0 ET DE1 ET DE1A) ;
*--- Display of the deflection values
PF = SUR1 POIN PROC (5. 0.) ;

UF = EXTR DEP0 UY PF ;

ERR0 = ABS((ABS(UF)-0.07794)/0.07794)*100 ;
MESS ERR0 ;
*--- Good working message
SI ( ERR0 < 6.E-3 ) ;
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
*
*--- RESOLUTION WITH REINFORCED CONCRETE UNDER
* PRESSURE -----*
*--- Resolution
RIO = RIGI MODB MAT1 ;
RI1 = EQUI RIGI MODB MOD2 MT ;
RIG1CL = RIO ET RI1 ET CLTOT ;
DEP1 = RESO RIG1CL CHAO ;
DEP1A = PROI ARMA DEP1 MODB MOD2 ;
*--- Deformed shape
DE0 = DEFORM (SUR1 ET D2) DEP1 0. BLAN ;
DE1 = DEFORM SUR1 DEP1 2.0 ROUG ;
DE1A = DEFORM D2 DEP1A 2.0 ROUG ;
TITR 'Deformed shape of reinforced concrete
under pressure' ;
TRAC (DE0 ET DE1 ET DE1A) ;
*--- Display of the deflection values
PF = SUR1 POIN PROC (5. 0.) ;
UF = EXTR DEP1 UY PF ;
ERR1 = ABS((ABS(UF)-0.06695)/0.06695)*100 ;
MESS ERR1 ;
*--- Good working message
SI ( ERR1 < 6.E-3 ) ;
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
FIN ;
    
```



Test precont1 (Comments)

1. The ELASTIC ARMATURE material

```
MOD1 = MODE D1 MECANIQUE ELASTIQUE ARMATURE BARR ;  
MAT2 = MATE MOD2 YOUN 2.0E11 SECT 0.02 ;
```

This model concerns the tendons of reinforced concrete (BARR on SEG2). The parameter names for an ELASTIQUE ARMATURE material are as follow:

In the case of passive tendons, the parameters are:

'YOUN' : Young's modulus

'SECT' : section of the tendon

In the case of active tendons (surrounding concrete), the additionnal parameters to be defined are:

For the loss of prestress by friction:

'FF' : ratio of angular friction (0.18 rd-1)

'PHIF' : ratio of linear friction (0.002 m-1)

For the loss of prestress at the anchoring point:

'GANC' : sliding at the anchoring point (0.0)

For the loss of tensile stress of the tendon:

'RMU0' : ratio of the loss of tensile stress of the tendon (0.43)

'FPRG' : stress of warranted rupture (1700.e6 Pa)

'RH10' : loss of tensile stress after 1000 hours (2.5)

2. The KABL operator

```
MODB = MODE SUR1 MECANIQUE ELASTIQUE ;  
MOD1 = MODE D1 MECANIQUE ELASTIQUE ARMATURE BARR ;  
MOD2 = KABL MODB MOD1 ;
```

The KABL operator calculates the intersection points between a concrete mesh and cables and creates new uniaxial tendon elements accordingly. The parameters to input are:

MODB : model object associated to concrete (MMODEL type)

MOD1 : model object associated to tendons (MMODEL type)

MOD2 : resulting model object (MMODEL type)

The KABL operator is usually used with the EQUI, PREC and PROI operators.

3. The EQUI operator

```
RI1 = EQUI RIGI MODB MOD2 MT ;  
CHA1 = EQUI FORCES MODB MOD2 PRE1 ;
```

The operator EQUI calculates the stiffness of tendons equivalent to prestresses expressed at the nodes of the concrete.

'RIGI' : key word to indicate that the calculation of the tendon to the stiffness is required.

'FORCES' : key word to indicate that the calculation of the equivalent forces is required.

MODB : model object associated to concrete (MMODEL type)

MOD2 : model object associated to tendons (MMODEL type)

MT : field of material properties of concrete and tendons (MCHAML type,
CHARACTERISTIQUES subtype)

RI1 : stiffness matrices (RIGIDITE type , RIGIDITE subtype)

PRE1 : stress field (MCHAML type, CONTRAINTES subtype)

CHA1 : force field (CHPOINT type)

4. figures/The PREC operator

```
PRE1 = PREC MOD2 MAT2 0.4E6 PM ;
```

The PREC operator calculates the loss of prestress and the actions of a tendon on the surrounding concrete.

- MOD2 : model object associated to the tendon (MMODEL type)
- MAT2 : field of characteristics associated to the tendon (MCHAML type, CARACTERISTIQUES subtype)
- 0.4E6 : value of the prestressing force applied to the end of the tendon (FLOTTANT type)
- PM : anchoring node (POINT type)
- PRE1 : resulting stress field (MCHAML type, CONTRAINTES subtype)

5. figures/The PROI operator

```
DEP1A = PROI ARMA DEP1 MODB MOD2 ;
```

With the "ARMA" option, the PROI operator projects onto a reinforcement (model MOD2), the displacements of a concrete structure (model MOD1).

The interpolation is done using shape functions of the concrete elements.

- DEP1 : displacements of the concrete structure
- MODB : model of the concrete structure
- MOD2 : model of the reinforcement



8.3 Test precont2 (Description sheet)

Test name: precont2

Calculation type: REINFORCED ELASTIC MECHANICS - BARR - 2D

Finite element type: QUA4

Topic: Calculation with active reinforcement. The structure is a reinforced concrete beam. The reinforcement is made of tendon. The beam is subjected to pressure on its upper surface, and embedded from its lower side.

Goal: Test the KABL, EQUI RIGI and PROI ARMA operators with passive reinforcement.


Reference: J.S. Fleuret 08/96

Version: 97' customer version

Model description:


<i>Concrete</i>	
E	= $0.4 \cdot 10^{11}$
NU	= 0.0

<i>Tendons</i>	
E	= $2.0 \cdot 10^{11}$
SECT	= 0.02
FF	= 0.18 <i>or</i> 0.0
PHIF	= 0.002 <i>or</i> 0.0
GANC	= 0.0
RMU0	= 0.43
FPRG	= $1700 \cdot 10^9$
RH10	= 2.5




Test precont2 (Results)

CASTEM FIGURES



AMPLITUDE
0.00E+00
2.0
2.0





8.3. TEST PRECONT2 (DESCRIPTION SHEET)

Test precont2 (Listing CASTEM)

```
*****
TITR 'CALCULATION WITH ACTIVE REINFORCEMENT' ;
*****
OPTI DIME 2 ELEM QUA4 MODE PLAN ;
OPTI ECHO 1 ;
*
*----- GEOMETRY -----*
LONG = 10.0 ;
*--- Concrete
PA = 0.0 0.0 ; PB = LONG 0.0 ;
PC = LONG 1.0 ; PD = 0.0 1.0 ;
NB1=40 ; NB2=6 ;
C1 = PA DROI NB1 PB ; C2 = PB DROI NB2 PC ;
C3 = PC DROI NB1 PD ; C4 = PD DROI NB2 PA ;
SUR1 = DALL C1 C2 C3 C4 ;

*--- Reinforcement
PM = 0.0 0.5 ; PN = LONG 0.5 ;
D1 = PM DROI 10 PN ;
*--- Total geometry
D5 = SUR1 ET D1 ;
TITR 'MESHING OF THE BEAM' ;
TRAC ELEM D5 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MODB = MODE SUR1 MECANIQUE ELASTIQUE ;
MOD1 = MODE D1 MECANIQUE ELASTIQUE ARMATURE BARR ;
MOD2 = KABL MODB MOD1 ;

D2 = EXTR MOD2 MAIL ;
D5= SUR1 ET D2 ;
TITR 'MESHING OF THE REINFORCEMENT AFTER KABL' ;
TRAC ELEM D2 ;

MAT1 = MATE MODB YOUN 0.4E11 NU 0.0 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ PA UX ;
CL2 = BLOQ PA UY ;
CL3 = BLOQ PB UY ;
CL4 = RELA ENSE UX C2 ;
CL5 = RELA ENSE UX C4 ;
CLTOT = CL1 ET CL2 ET CL3 ET CL4 ET CL5 ;
*
*----- CALCULATION OF THE PRESTRESS -----*
* The loss of prestress results from rubbing
*  $S = SO * (1 - EXP(-FF*ALPHA-PHIF*X))$ 
*--- Calculation of the loss of prestress
*--- at the middle and at the end
*--- with FF = 0.18 ET PHIF = 0.002
MAT2 = MATE MOD2 YOUN 2.0E11 SECT 0.02 FF 0.18
      PHIF 0.002 GANC 0.0 RMU0 0.43 FPRG 1700.E9
      RH10 2.5 ;
PRE1 = PREC MOD2 MAT2 8.0E6 PM ;
PREC101 = EXTR PRE1 EFFX 1 10 2 ;
PREC051 = EXTR PRE1 EFFX 1 5 2 ;
MESS PREC101 ;
MESS PREC051 ;
*--- Calculation of the loss of prestress
*--- at the middle and at the end
*--- with FF = 0.0 ET PHIF = 0.0
MAT2 = MATE MOD2 YOUN 2.0E11 SECT 0.02 FF 0.0
      PHIF 0.0 GANC 0.0 RMU0 0.43 FPRG 1700.E9
      RH10 2.5 ;
PRE1 = PREC MOD2 MAT2 8.0E6 PM ;
PREC102 = EXTR PRE1 EFFX 1 10 2 ;

PREC052 = EXTR PRE1 EFFX 1 5 2 ;

MESS PREC102 ;
MESS PREC052 ;
*--- Comparison of results
*--- The loss should be inferior to 1 % for
*--- the middle and 2 % for the end
ERR10 = ABS(((PREC101-PREC102)/PREC101)*100) ;
ERR05 = ABS(((PREC051-PREC052)/PREC051)*100) ;
MESS ERR10 ;
MESS ERR05 ;
*--- Good working message
SI ( ( ERR10 - 2 < 0.1 ) ET ( ERR05 - 1 < 0.1 ) ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;
*
*----- LOAD DEFINITION -----*
CHA1 = EQUI FORCES MODB MOD2 PRE1 ;
*
*--- RESOLUTION WITH CONCRETE UNDER PRESTRESS ---*
*--- Resolution
RI0 = RIGI MODB MAT1 ;
RIG0CL = RI0 ET CLTOT ;
DEP0 = RESO RIG0CL CHA1 ;
DEP0A = PROI ARMA DEP0 MODB MOD2 ;
*--- Deformed shape
DE0 = DEFO (SUR1 ET D2) DEP0 0. BLAN ;
DE1 = DEFO SUR1 DEP0 2.0 ROUG ;
DE1A = DEFO D2 DEP0A 2.0 ROUG ;
TITR 'Deformed shape of concrete under prestress' ;
TRAC (DE0 ET DE1 ET DE1A) ;
*--- Display of the compressive values
PPM = SUR1 POIN PROC (LONG 0.5) ;
PPN = SUR1 POIN PROC ( 0. 0.5 ) ;
UM = EXTR DEP0 UX PPM ;
UN = EXTR DEP0 UX PPN ;
MESS UM ;
MESS UN ;
MESS ( ABS ( UM - UN ) ) ;
ERR1 = ( ( ABS(UM-UN) - 0.1 ) / 0.1 ) * 100 ;
MESS ERR1 '%' ;
*--- Good working message
SI ( ERR1 < 1.E-3 ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;

FIN ;
```

Test precont2 (Comments)

- The RELA operator

```
CL4 = RELA ENSE UX C2 ;
```

The ENSE option of the RELA operator enables the user to construct the stiffness CL4 (RIGIDITE type) associated with a global motion : the value on the degree of freedom UX (MOT type) is imposed as being identical for all the nodes of C2 (MAILLAGE type).



8.4 Test precont3 (Description sheet)

Test name: precont3

Calculation type: REINFORCED ELASTIC MECHANICS - BARR - 2D

Finite element type: QUA4

Topic: Calculation with active reinforcement and different loss of prestress. The structure is a reinforced concrete beam. The reinforcement is made of tendon. The beam is subjected to pressure on its upper surface, and embedded from its lower side.

Goal: Test the PREC operators with passive reinforcement.

Reference: J.S. Fleuret 07/96


Version: 97' customer version

Model description:

Concrete

E	=	$0.4 \cdot 10^{11}$
NU	=	0.0

<i>Tendons</i>	without loss	without loss	with loss for friction	with loss for friction and slipping	with loss for friction slipping and relaxation	
E	=	$2.1 \cdot 10^{11}$	=	$2.1 \cdot 10^{11}$	=	$2.1 \cdot 10^{11}$
SECT	=	0.02	=	0.02	=	0.02
FF	=	0.0	=	0.18	=	0.18
PHIF	=	0.0	=	0.002	=	0.002
GANC	=	0.0	=	0.0	=	0.005
RMU0	=	0.43	=	0.43	=	0.43
FPRG	=	$1700 \cdot 10^9$	=	$1700 \cdot 10^9$	=	$1700 \cdot 10^9$
RH10	=	2.5	=	2.5	=	2.5



Test precont3 (Results)**CASTEM FIGURES**

MESHING OF THE BEAM

MESHING OF THE REINFORCEMENT AFTER KABL



8.4. TEST PRECONT3 (DESCRIPTION SHEET)

Test precont3 (Listing CASTEM)

```
*****
TITR 'CALCULATION WITH ACTIVE REINFORCEMENT' ;
*****
OPTI DIME 2 ELEM QUA4 MODE PLAN ;
OPTI ECHO 1 ;
*
*----- GEOMETRY -----*
LONG = 10.0 ;
*--- Concrete
PA = 0.0 0.0 ; PB = LONG 0.0 ;
PC = LONG 1.0 ; PD = 0.0 1.0 ;
NB1=40 ; NB2=6 ;
C1 = PA DROI NB1 PB ; C2 = PB DROI NB2 PC ;
C3 = PC DROI NB1 PD ; C4 = PD DROI NB2 PA ;
SUR1 = DALL C1 C2 C3 C4 ;

*--- Reinforcement
PM = 0.0 0.5 ; PN = LONG 0.5 ;
D1 = PM DROI 10 PN ;
*--- Total geometry
D5 = SUR1 ET D1 ;
TITR 'MESHING OF THE BEAM' ;
TRAC ELEM D5 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MODB = MODE SUR1 MECANIQUE ELASTIQUE ;
MOD1 = MODE D1 MECANIQUE ELASTIQUE ARMATURE BARR ;
MOD2 = KABL MODB MOD1 ;

D2 = EXTR MOD2 MAIL ;
D5 = SUR1 ET D2 ;
TITR 'MESHING OF THE REINFORCEMENT AFTER KABL' ;
TRAC ELEM D2 ;
MAT1=MATE MODB YOUN 0.4E11 NU 0.0 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQ PA UX ; CL2 = BLOQ PA UY ;
CL3 = BLOQ PB UY ;
CL4 = RELA ENSE UX C2 ; CL5 = RELA ENSE UX C4 ;
CLTOT = CL1 ET CL2 ET CL3 ET CL4 ET CL5 ;
*
*----- CALCULATION OF THE PRESTRESS -----*
* Calculation without loss
MAT2 = MATE MOD2 YOUN 2.1E11 SECT 0.02
      FF 0.0 PHIF 0.0 GANC 0.0 RMUO 0.43
      FPRG 1700.E9 RH10 2.5 ;
PRE1 = PREC MOD2 MAT2 12.0E6 PM ;
PREC100 = EXTR PRE1 EFFX 1 10 2 ;
PREC050 = EXTR PRE1 EFFX 1 5 2 ;
MESS PREC100 ; MESS PREC050 ;
ERR0 = ( (ABS(PREC100-600.0E6)/600.0E6)
        + (ABS(PREC050-600.0E6)/600.0E6) )/2 ;
MESS ERR0 '%' ;
*--- Good working message
SI ( ERR0 < 1. ) ;
      ERRE 0 ;
SINON ;
      ERRE 5 ;
FINSI ;

* Second calculation without loss
MAT2 = MATE MOD2 YOUN 2.1E11 SECT 0.02
      FF 0.0 PHIF 0.0 GANC 0.0 RMUO 0.43
      FPRG 1700.E9 RH10 2.5 ;
PRE2=PREC MOD2 MAT2 12.0E6 PN PRE1 ;
PREC100 = EXTR PRE2 EFFX 1 10 2 ;

PREC050 = EXTR PRE2 EFFX 1 5 2 ;
MESS PREC100 ; MESS PREC050 ;

ERR0 = ( (ABS(PREC100-1200.0E6)/1200.0E6)
        + (ABS(PREC050-1200.0E6)/1200.0E6) )/2 ;
MESS ERR0 '%' ;
*--- Good working message
SI ( ERR0 < 1. ) ;
      ERRE 0 ;
SINON ;
      ERRE 5 ;
FINSI ;

* Calculation with loss by rubbing
MAT2 = MATE MOD2 YOUN 2.1E11 SECT 0.02
      FF 0.18 PHIF 0.002 GANC 0.0
      RMUO 0.43 FPRG 1700.E9 RH10 2.5 ;
PRE1 = PREC MOD2 MAT2 24.0E6 PM ;
PREC101 = EXTR PRE1 EFFX 1 10 2 ;
PREC051 = EXTR PRE1 EFFX 1 5 2 ;
MESS PREC101 ; MESS PREC051 ;
ERR1 = ( (ABS(PREC101-1176.0E6)/1176.0E6)
        + (ABS(PREC051-1188.0E6)/1188.0E6) )/2 ;
MESS ERR1 '%' ;
*--- Good working message
SI ( ERR1 < 1. ) ;
      ERRE 0 ;
SINON ;
      ERRE 5 ;
FINSI ;

* Calculation with loss by rubbing and slipping
MAT2 = MATE MOD2 YOUN 2.1E11 SECT 0.02
      FF 0.18 PHIF 0.002 GANC 0.005
      RMUO 0.43 FPRG 1700.E9 RH10 2.5 ;
PRE1 = PREC MOD2 MAT2 24.0E6 PM ;
PREC102 = EXTR PRE1 EFFX 1 10 2 ;
PREC052 = EXTR PRE1 EFFX 1 5 2 ;
MESS PREC102 ; MESS PREC052 ;
ERR2 = ( (ABS(PREC102-1176.0E6)/1176.0E6)
        + (ABS(PREC052-1183.6E6)/1183.6E6) )/2 ;
MESS ERR2 '%' ;
*--- Good working message
SI ( ERR2 < 1. ) ;
      ERRE 0 ;
SINON ;
      ERRE 5 ;
FINSI ;

* Calculation with loss by rubbing, slipping
* and relaxation
MAT2 = MATE MOD2 YOUN 2.1E11 SECT 0.02
      FF 0.18 PHIF 0.002 GANC 0.005
      RMUO 0.43 FPRG 1700.E6 RH10 2.5 ;
PRE1 = PREC MOD2 MAT2 24.0E6 PM ;
PREC103 = EXTR PRE1 EFFX 1 10 2 ;
PREC053 = EXTR PRE1 EFFX 1 5 2 ;
MESS PREC103 ; MESS PREC053 ;
ERR3 = ( (ABS(PREC103-1129.8E6)/1129.8E6)
        + (ABS(PREC053-1049.5E6)/1049.5E6) )/2 ;
MESS ERR3 ' EN %' ;
*--- Good working message
SI ( ERR3 < 1. ) ;
      ERRE 0 ;
SINON ;
      ERRE 5 ;
FINSI ;
FIN ;
```




Chapter 9

POROSITY

9.1 Test pore1 (Description sheet)

Test name: pore1

Calculation type: ELASTOPLASTIC MECHANICS - DRUCKER-PRAGER MODEL IN POROSITY
- MODE AXIS

Finite element type: QUAP

Topic: Porous thick cylinder. The structure is an elastoplastic thick cylinder. It follows a Drucker-Prager law with non-associated yield. We release the stress within the hole.


Goal: Calculate the convergence of the hall and the stress along the circumference

Reference: Benchmark INTERCLAY 1.1 Variante 2

Version: 97' customer version

Model description:

E	= 200.13E06	Young's modulus
NU	= 0.25	Poisson's ratio
RHO	= 0.	density
ALPH	= 0.	thermal expansion coefficient
MOB	= 7500E6	Biot's modulus
COB	= 1.	Biot's coefficient
PERM	= 4.E-12	intrinsic permeability
VISC	= 4.	fluid dynamic viscosity
ALPM	= 0.	coefficient of the pressure-temperature coupling






Test pore1 (Results)

RESULTS

	Hall convergence	Stress along the circumference
Reference (*)	-0.266 m	-3.47E6 Pa
Castem	-0.26586 m	-3.46667E6 PA
Error	5.21515E-4 %	9.59243E-4 %

(*) The reference values are calculated with castem2000 which agree with INTERCLAY's results.

CASTEM FIGURES



Test pore1 (Listing CASTEM)

```

*****
TITR 'A POROUS THICK CYLINDER' ;
*****
OPTI DIME 2 MODE AXIS ELEM QUA8 ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
DENS 0.5 ;
P1 = 2.5 0. ; P2 = 25. 0. ;
Q1 = 2.5 0.5 ; Q2 = 25. 0.5 ;
PP1 = 5. 0. ; QQ1 = 5. 0.5 ;
LI1 = P1 DROI P2 DINI 0.25 DFIN 2.5 ;
LI2 = P2 DROI Q2 ;
LI3 = INVE ( Q1 DROI Q2 DINI 0.25 DFIN 2.5 ) ;
LI4 = Q1 DROI P1 ;
LI = LI1 ET LI2 ET LI3 ET LI4 ;
CADRE = DALL LI1 LI2 LI3 LI4 PLAN ;
TRAC CADRE ;
*
*----- BOUNDARY CONDITIONS -----*
BL1 = BLOQ UZ LI1 ;
BL2 = BLOQ UZ LI3 ;
CL = BL1 ET BL2 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO=MODE CADRE POREUX ELASTIQUE PLASTIQUE
      DRUCKER_PRAGER ;
MAT=MATE MO YOUNG 200.13E6 NU 0.25
      MOB 7500E6 COB 1.
      PERM 4.E-12 VISCO 4.
      ALPH 0. KF 0. RHOF 0. ALPF 0. ALPM 0. RHO 0.
      ALFA 0.1270509 BETA 0.57735027 K 0.
      GAMM 0. DELT 0.57735027
      ETA 0.1270509 MU 0.57735027 KL 0. H 0. ;
*
*----- LOAD DEFINITION -----*
TITR 'INITIAL STRESSES' ;
ZOZO = MANU CHML MO SMRZ 0.
      SMRR -5.E6 SMTT -5.E6 SMZZ -5.E6 P 2.5E6
      TYPE CONTRAINTES STRESSES ;
FF = BSIGMA MO ZOZO ;
EVT1 = EVOL MANU TEMPS (PROG 0 14 )
      CHARGE (PROG 1 1 ) ;
DESS EVT1 ;
CHA1 = CHAR MECA FF EVT1 ;

TITR 'PRESSURE' ;
FPI2 = PRESS MASS MO -1.E6 LI4 ;
EVT2 = EVOL MANU TEMPS (PROG 0 PAS 1 14 )
      CHARGE (PROG 0 PAS 0.25 3.5) ;
DESS EVT2 ;
CHA2 = CHAR MECA FPI2 EVT2 ;
CHATT = CHA1 ET CHA2 ;
*
*----- SETUP OF THE PASAPAS PROCEDURE -----*
TAB1 = TABLE ;
TAB1.'CONTRAINTES' = TABLE ;
TAB1.'DEPLACEMENTS' = TABLE ;
TAB1.'CONTRAINTES'.0 = ZOZO ;
TAB1.'DEPLACEMENTS'.0 = MANU CHPO CADRE 1 P 2.5E6 ;
LREE = PROG 0 PAS 1 14 ;
TAB1.'CARACTERISTIQUES' = MAT ;
TAB1.'CHARGEMENT' = CHATT ;
TAB1.'MODELE' = MO ;
TAB1.'TEMPS_CALCULES' = LREE ;
TAB1.'BLOCAGES_MECANIQUES' = CL ;

PASAPAS TAB1 ;
*
*----- RESULTS AND OUTPUT -----*
MESS ' RESULTS OF TEST PORE1 ' ;
MESS ' ----- ' ;
SAUT 2 LIGN ;
MESS 'We search the convergence of the hole and' ;
MESS 'the stress along the circumference' ;
SAUT 1 LIGN ;
SAUT 1 LIGN ;
CONVTH = -0.266 ;
SIGTTH = -3.47E6 ;
LEDEP = PECHE TAB1 DEPLACEMENTS 14 ;
SIGAA = PECHE TAB1 CONTRAINTES 14 ;
SIGAB = CHAN 'CHPO' MO SIGAA ;
CONVXX = EXTR LEDEP UR P1 ;
SIGTXX = EXTR SIGAB SMTT P1 ;
ERGXXA = ( CONVXX - CONVTH ) / CONVTH ;
ERGXXA = ABS ERGXXA ;
ERGXXB = ( SIGTXX - SIGTTH ) / SIGTTH ;
ERGXXB = ABS ERGXXB ;

MESS ' Theoretical convergence:' CONVTH 'm' ;
MESS ' Calculated convergence:' CONVXX 'm' ;
MESS ' Error:' ERGXXA '%' ;
SAUT 1 LIGN ;
MESS 'Theoretical Stress along the circumference:'
      SIGTTH 'Pa' ;
MESS 'Calculated Stress along the circumference:'
      SIGTXX 'Pa' ;
MESS ' Error :' ERGXXB '%' ;
*
*----- GOOD WORKING MESSAGE -----*
SAUT 2 LIGN ;
SI (ERGXXA <EG 5) ;
SI (ERGXXB <EG 5) ;
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
SINON ;
ERRE 5 ;
FINSI ;

FIN ;

```

**Test pore1 (Comments)**

1. Parameter name for an ELASTIQUE ISOTROPE material in the case of a POREUX formulation
The parameters which should be input in the POREUX formulation are as follow:

E	Young's modulus
NU	Poisson's ratio
RHO	density
ALPH	thermal expansion coefficient
MOB	Biot's modulus
COB	Biot's coefficient
PERM	intrinsic permeability
VISC	fluid dynamic viscosity
ALPM	coefficient of the pressure-temperature coupling

2. The Drucker-Prager's model with non-associated yield and stress hardening

```
MO=MODE CADRE POREUX ELASTIQUE PLASTIQUE
      DRUCKER_PRAGER ;
MAT=MATE MO YOUNG 200.13E6 NU 0.25
MOB 7500E6      COB 1.
PERM 4.E-12     VISCO 4.
ALPH 0. KF 0.   RHOF 0. ALPF 0. ALPM 0. RHO 0.
ALFA 0.1270509 BETA 0.57735027 K 0.
GAMM 0.        DELT 0.57735027
ETA 0.1270509 MU 0.57735027 KL 0. H 0. ;
```

The equations of the plastic DRUCKER-PRAGER model are as follow:

-> Notations: S stress tensor

Seq Von Mises equivalent stress

p cumulated equivalent plastic strain

-> initial criterion: $ALFA * Tr(S) + BETA * Seq = K$

-> ultimate criterion: $ETA * Tr(S) + MU * Seq = KL$

-> Hardening: $dK = H * dp$ (H in algebraic value)

-> Flow potential: $GAMM * Tr(S) + DELT * Seq$

The parameters to be defined are: ALFA, BETA, K, ETA, MU, KL, H, GAMM, DE

3. Application of the initial stresses

```
ZOZO = MANU CHML MO      SMRZ 0.
      SMRR -5.E6 SMTT -5.E6 SMZZ -5.E6 P 2.5E6
      TYPE CONTRAINTE STRESSES ;
FF    = BSIGMA MO ZOZO;
EVT1  = EVOL MANU TEMPS (PROG 0 14 )
      CHARGE (PROG 1 1 ) ;
CHA1  = CHAR MECA FF EVT1 ;
```

The MANU CHML operator creates a field by element (MCHAML type) connected with the MO object (MMODEL type). The key word 'stresses' means that the field by element will be expressed with respect to element stress calculation points.

The BSIGMA operator calculates the nodal force field resulting from the stress field integration.

9.2 Test pore2 (Description sheet)

Test name: pore2

Calculation type: ELASTOPLASTIC MECHANICS - DRUCKER-PRAGER MODEL IN POROSITY
- MODE AXIS

Finite element type: QUAP

Topic: Porous thick cylinder with thermal effects. The structure is an elastoplastic thick cylinder. It follows the Von-Mises model with associated yield. We release the stress within the hole, and we set a temperature yield.


Goal: Calculate the convergence of the hole and the stress along the circumference

Reference: Benchmark INTERCLAY 1.4 Basic Model

Version: 97' customer version

Model description:

E	= 200.13E06	Young's modulus
NU	= 0.25	Poisson's ratio
RHO	= 0.	density
ALPH	= (5.E-5 / 3.)	thermal expansion coefficient
MOB	= 7500E6	Biot's modulus
COB	= 1.	Biot's coefficient
PERM	= 4.E-12	intrinsic permeability
VISC	= 4.	fluid dynamic viscosity
ALPM	= (19.E-5 / 3.)	coefficient of the pressure-temperature coupling







9.2. TEST PORE2 (DESCRIPTION SHEET)

Test pore2 (Results)


CASTEM FIGURES




A TESTE CALIBRADO DE PORE2 REALIZADO COM O TERMOCO APARELHO






CRITICAL PORE2



PRELIMINAR



PRELIMINAR



Test pore2 (Listing CASTEM)

```

*****
TITR'A POROUS THICK CYLINDER WITH THERMIC EFFECTS';
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
DENS 0.5 ;
P1 = 2.5 0. ; P2 = 25. 0. ;
Q1 = 2.5 0.5 ; Q2 = 25. 0.5 ;
PP1 = 5 0. ; QQ1 = 5 0.5 ;
LI1 = P1 DROI P2 DINI 0.25 DFIN 2.5 ;
LI2 = P2 DROI Q2 ;
LI3 = INVE ( Q1 DROI Q2 DINI 0.25 DFIN 2.5 ) ;
LI4 = Q1 D P1 ;
LI = LI1 ET LI2 ET LI3 ET LI4 ;
CADRE = DALL LI1 LI2 LI3 LI4 PLAN ;
TRAC CADRE ;
*
*----- BOUNDARY CONDITIONS -----*
BL1 = BLOQ UZ LI1 ;
BL2 = BLOQ UZ LI3 ;
BL = BL1 ET BL2 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO=MODE CADRE POREUX ELASTIQUE PLASTIQUE
DRUCKER_PRAGER ;
MAT=MATE MO YOUNG 200.13E6 NU 0.25 MOB 7500E6
COB 1. PERM 4.E-12 VISCO 4.
ALPH (5.E-5 / 3. ) KF 3000E6 RHOF 0.
ALPF 40E-5 ALPM (19.E-5 / 3. ) RHO 0.
ALFA 0. BETA 0.57735027 K 1.E6
GAMM 0. DELT 0.57735027
ETA 0. MU 0.57735027 KL 1.E6 H 0. ;
*
*----- LOAD DEFINITION -----*
* FIRST STEP: mechanic load
* SETUP OF THE TABLE FOR CALL TO PASAPAS
TITR 'INITIAL STRESSES' ;
ZOZO = MANU CHML MO SMRZ 0.
SMRR -5.E6 SMTT -5.E6 SMZZ -5.E6 P 2.5E6
TYPE CONTRAINTE STRESSES ;
FF = BSIGMA MO ZOZO ;
EVT1= EVOL MANU TEMPS (PROG 0 14 )
CHARGE (PROG 1 1 ) ;
DESS EVT1 ;
CHA1 = CHAR MECA FF EVT1 ;

TITR 'PRESSURE' ;
FPI2 = PRESS MASS MO -1.E6 LI4 ;
EVT2= EVOL MANU TEMPS (PROG 0 PAS 1 14 )
CHARGE (PROG 0 PAS 0.25 3.5) ;
DESS EVT2 ;
CHA2 = CHAR MECA FPI2 EVT2 ;
CHATT = CHA1 ET CHA2 ;

TAB1 = TABLE ;
TAB1 . 'CONTRAINTE' = TABLE ;
TAB1 . 'DEPLACEMENTS' = TABLE ;
TAB1.'CONTRAINTE' . 0 = ZOZO ;
TAB1.'DEPLACEMENTS' . 0 =
MANU CHPO CADRE 1 P 2.5E6 ;
TAB1.'TEMPS_CALCULES' = PROG 0 6 10 14 ;
TAB1.'MODELE' = MO ;
TAB1.'CARACTERISTIQUES' = MAT ;

TAB1.'BLOCAGES_MECANIQUES' = BL ;

TAB1.'CHARGEMENT' = CHATT ;

PASAPAS TAB1 ;

* SECOND STEP: temperature
* SETUP OF THE TABLE FOR CALL TO PASAPAS
TAB2 = TABLE ;
TAB2 . 'VARIABLES_INTERNES' = TABLE ;
TAB2 . 'DEPLACEMENTS' = TABLE ;
TAB2 . 'CONTRAINTE' = TABLE ;
SS0 = PECHE TAB1 CONTRAINTE ;
TAB2.'CONTRAINTE' . 0 = SS0 ;
TAB2.'PRECISION'=5.E-7 ;
TAB2.'VARIABLES_INTERNES' . 0 =
PECHE TAB1 VARIABLES_INTERNES ;
TAB2.'DEPLACEMENTS' . 0 =
PECHE TAB1 DEPLACEMENTS ;

FF = BSIGMA MO SS0 ;
EVT1= EVOL MANU TEMPS (PROG 0 100)
CHARGE (PROG 1 1 ) ;
TITR 'STRESSES' ;
DESS EVT1 ;
CHA1 = CHAR MECA FF EVT1 ;

LEFAC = 1. / ( LOG ( 2.5/25. ) ) ;
RAD = COOR 1 CADRE ;
DT = NOMC ( LEFAC *( LOG ( RAD / 25. ) ) ) T ;

TABTER = TABLE ; TEMP = TABLE ;
TEMP . 0 = 0. ;
TEMP . 1 = 100. ;
TABTER . 0 = MANU CHPO CADRE 1 T 0. ;
TABTER . 1 = (DT * 100.) ;
CHA2 = CHAR T TEMP TABTER ;

CHATT = CHA1 ET CHA2;

SI COMPLET ;
TFIN = 70. ;
SINON ;
TFIN = 15. ;
FINSI ;
TAB2.'TEMPS_CALCULES' = PROG 0 PAS 5. TFIN ;
TAB2.'MODELE' = MO ;
TAB2.'CARACTERISTIQUES' = MAT ;
TAB2.'BLOCAGES_MECANIQUES' = BL ;
TAB2.'CHARGEMENT' = CHATT ;

PASAPAS TAB2 ;
*
*----- OUTPUT -----*
MESS ' RESULTS TEST PORE2 ' ;
MESS 'We search the stress and the pressure.' ;
MESS 'they are compare with reference values.' ;
SAUT 1 LIGN ;
* RESULTS
* DISTRIBUTION OF STRESSES AND PRESSURE FOR T=70.

SI COMPLET ;
TT = 70.;
SINON ;
TT = 15.;
FINSI ;

SIGAA = PECHE TAB2 CONTRAINTE TT ;
SIGAB = CHAN 'CHPO' MO SIGAA ;
SIGARR = EVOL CHPO SIGAB SMRR LI1 ;
SIGATT = EVOL CHPO SIGAB SMTT LI1 ;

```




9.2. TEST PORE2 (DESCRIPTION SHEET)

```

LAPP = EVOL CHPO SIGAB P LI1 ;
DESS SIGARR ; DESS SIGATT ; DESS LAPP ;

EVSIGAR = EXTR SIGARR ORDO 1 ;
EVSIGAT = EXTR SIGATT ORDO 1 ;
EVPP = EXTR LAPP ORDO 1 ;
*
* THEORETICAL VALUES (1/2/93)
SI COMPLET ;
* RADIAL STRESS
SOLRR = PROG
-1.59459E+06 -1.56051E+06 -1.78383E+06
-1.76850E+06 -1.96931E+06 -1.97473E+06
-2.15711E+06 -2.17942E+06 -2.34668E+06
-2.38270E+06 -2.53753E+06 -2.58467E+06
-2.72927E+06 -2.78543E+06 -2.92159E+06
-2.98507E+06 -3.11427E+06 -3.18371E+06
-3.30718E+06 -3.38150E+06 -3.50030E+06
-3.57863E+06 -3.69361E+06 -3.77540E+06
-3.88744E+06 ;
SOLRR = SOLRR ET ( PROG
-3.97178E+06 -4.08114E+06 -4.16891E+06
-4.29345E+06 -4.34768E+06 -4.48547E+06
-4.54414E+06 -4.62704E+06 -4.68421E+06
-4.74775E+06 -4.79065E+06 -4.83863E+06
-4.86963E+06 -4.90481E+06 -4.92612E+06
-4.95084E+06 -4.96421E+06 -4.98036E+06
-4.98723E+06 -4.99637E+06 -4.99792E+06
-5.00142E+06 ) ;
* TANGENTIAL STRESS
SOLTT = PROG
-3.59051E+06 -3.55538E+06 -3.77813E+06
-3.76143E+06 -3.96142E+06 -3.96508E+06
-4.14632E+06 -4.16639E+06 -4.33216E+06
-4.36546E+06 -4.51847E+06 -4.56247E+06
-4.70501E+06 -4.75775E+06 -4.89181E+06
-4.95180E+06 -5.07914E+06 -5.14532E+06
-5.26743E+06 -5.33906E+06 -5.45722E+06
-5.53370E+06 -5.64905E+06 -5.72971E+06
-5.84300E+06 ;
SOLTT = SOLTT ET ( PROG
-5.92761E+06 -6.03950E+06 -6.12667E+06
-6.24105E+06 -6.30245E+06 -6.28413E+06
-6.13481E+06 -6.01293E+06 -5.89978E+06
-5.80418E+06 -5.70545E+06 -5.62098E+06
-5.53670E+06 -5.46384E+06 -5.39075E+06
-5.32680E+06 -5.26302E+06 -5.20651E+06
-5.15031E+06 -5.09985E+06 -5.04987E+06
-5.00444E+06 ) ;
* PRESSURE
SOLPP = PROG
1.36747E+06 1.44034E+06 1.51321E+06
1.58674E+06 1.66027E+06 1.73387E+06
1.80747E+06 1.88113E+06 1.95479E+06
2.02829E+06 2.10179E+06 2.17490E+06
2.24800E+06 2.32049E+06 2.39297E+06
2.46467E+06 2.53636E+06 2.60709E+06
2.67783E+06 2.74773E+06 2.81762E+06
2.88621E+06 2.95480E+06 3.02386E+06
3.09291E+06 ;
SOLPP = SOLPP ET ( PROG
3.15682E+06 3.22073E+06 3.29927E+06
3.37782E+06 3.40824E+06 3.43867E+06
3.36939E+06 3.30011E+06 3.24558E+06
3.19105E+06 3.13296E+06 3.07487E+06
3.01765E+06 2.96042E+06 2.90299E+06
2.84556E+06 2.78819E+06 2.73081E+06
2.67343E+06 2.61604E+06 2.55866E+06

2.50128E+06 ) ;
SINON ;
SOLRR = PROG
-1.56503E+06 -1.57213E+06 -1.75778E+06
-1.77784E+06 -1.94803E+06 -1.98244E+06
-2.14000E+06 -2.18611E+06 -2.33334E+06
-2.38903E+06 -2.52782E+06 -2.59133E+06
-2.72321E+06 -2.79313E+06 -2.91938E+06
-2.99454E+06 -3.11618E+06 -3.19563E+06
-3.31352E+06 -3.39647E+06 -3.51129E+06
-3.59714E+06 -3.70949E+06 -3.79760E+06
-3.90787E+06 ;
SOLRR = SOLRR ET ( PROG
-3.99816E+06 -4.12859E+06 -4.17682E+06
-4.32275E+06 -4.37916E+06 -4.46316E+06
-4.52542E+06 -4.59310E+06 -4.64335E+06
-4.69793E+06 -4.73815E+06 -4.78194E+06
-4.81401E+06 -4.84898E+06 -4.87436E+06
-4.90211E+06 -4.92201E+06 -4.94386E+06
-4.95928E+06 -4.97629E+06 -4.98804E+06
-5.00116E+06 ) ;
SOLTT = PROG
-3.56189E+06 -3.56846E+06 -3.75410E+06
-3.77364E+06 -3.94386E+06 -3.97775E+06
-4.13540E+06 -4.18103E+06 -4.32842E+06
-4.38367E+06 -4.52270E+06 -4.58583E+06
-4.71802E+06 -4.78765E+06 -4.91426E+06
-4.98918E+06 -5.11126E+06 -5.19054E+06
-5.30890E+06 -5.39174E+06 -5.50708E+06
-5.59285E+06 -5.70571E+06 -5.79390E+06
-5.90481E+06 ;
SOLTT = SOLTT ET ( PROG
-5.99480E+06 -6.11303E+06 -6.16665E+06
-6.10728E+06 -6.00175E+06 -5.91312E+06
-5.83171E+06 -5.76581E+06 -5.69639E+06
-5.63972E+06 -5.58251E+06 -5.53566E+06
-5.48786E+06 -5.44850E+06 -5.40854E+06
-5.37545E+06 -5.34188E+06 -5.31389E+06
-5.28556E+06 -5.26174E+06 -5.23769E+06
-5.21735E+06 ) ;
SOLPP = PROG
3.52566E+05 4.45884E+05 5.39202E+05
6.32610E+05 7.26018E+05 8.19101E+05
9.12184E+05 1.00502E+06 1.09786E+06
1.19042E+06 1.28298E+06 1.37529E+06
1.46759E+06 1.55963E+06 1.65167E+06
1.74345E+06 1.83523E+06 1.92692E+06
2.01860E+06 2.10969E+06 2.20078E+06
2.29305E+06 2.38533E+06 2.47274E+06
2.56016E+06 ;
SOLPP = SOLPP ET ( PROG
2.66334E+06 2.76651E+06 2.80560E+06
2.84469E+06 2.82168E+06 2.79866E+06
2.78895E+06 2.77924E+06 2.76631E+06

2.75339E+06 2.74124E+06 2.72909E+06
2.71676E+06 2.70442E+06 2.69213E+06
2.67984E+06 2.66755E+06 2.65525E+06
2.64295E+06 2.63066E+06 2.61836E+06
2.60606E+06 ) ;
FINSI ;
*
*----- GOOD WORKING MESSAGE -----*
ERSIGAR = (MAXI ABS ( EVSIGAR - SOLRR )) /
(MAXI ABS SOLRR) ;
ERSIGAT = (MAXI ABS ( EVSIGAT - SOLTT )) /
(MAXI ABS SOLTT) ;
ERPP = (MAXI ABS ( EVPP - SOLPP )) /

```

```
(MAXI ABS SOLPP) ;
SAUT 1 LIGN ;
MESS ' Difference between the radial stresses:'
      ERSIGAR '%' ;
SAUT 1 LIGN ;
MESS ' Difference between the stresses along the
      circumference:' ERSIGAT '%' ;
SAUT 1 LIGN ;
MESS ' Difference between the pressures:'
      ERPP '%' ;
*The maximum difference between the theoretical
*and the calculated values should be below 0.001
SI (ERSIGAR <EG 0.001) ;
  SI (ERSIGAT <EG 0.001) ;
    SI (ERPP <EG 0.001) ;
      ERRE 0 ;
    SINON ;
      ERRE 5 ;
  FINSI ;
SINON ;
  ERRE 5 ;
FINSI ;
SINON ;
  ERRE 5 ;
FINSI ;
FIN ;
```



Test pore2 (Comments)

- The PECHE operator

SS0 = PECHE TAB1 CONTRAINTES ;

This procedure enables the user to retrieve for a given time the results of a calculation carried out using the PASAPAS procedure. CONTRAINTES is the requested result index. If no time input (like here), the requested results for the last calculated time will be retrieved.

9.3 Test pore3 (Description sheet)

Test name: pore3

Calculation type: POROUS-ELASTIC MECHANIC - MODE PLAN CONT

Finite element type: QUAP

Topic: One-dimensionnal consolidation. The structure is a porous elastic piece subjected to a pressure at its upper surface.


Goal: Compare pressures and curves of the vertical displacements with analytical solution.

Reference: Problem of Terzaghi

Version: 97' customer version

Model description:

E	= 7000D6	Young's modulus
NU	= 0.4	Poisson's ratio
RHO	= 0.	density
ALPH	= 0.	thermal expansion coefficient
MOB	= 1.D18	Biot's modulus
COB	= 1.	Biot's coefficient
PERM	= 1.D-9	intrinsic permeability
VISC	= 1.	fluid dynamic viscosity
ALPM	= 0.	coefficient of the pressure-temperature coupling






Test pore3 (Results)


CASTEM FIGURES




A THICK CYLINDER IN POROUS MIDDLE WITH THERMIC EFFECTS



COMPARISON BETWEEN THE PRESSURE PROFILES



COMPARISON BETWEEN THE PROFILES OF VERTICAL DISPLACEMENTS



COMPARISON BETWEEN THE PROFILES OF VERTICAL DISPLACEMENTS

Test pore3 (Listing CASTEM)

```

*****
TITR 'POROUS ELASTIC PIECE SUBJECTED TO PRESSURE';
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
OPTI DIME 2 ELEM QUA8 MODE PLAN DEFO ;
OPTI ECHO 1 ;

MOTP = MOTS P ;
LREP = PROG 1.E6 ;
OPTI NORM MOTP LREP ;
*
*----- GEOMETRY -----*
DENS 1. ; H = 10. ; H2 = H * H ;
P1 = 0. 0. ; P2 = 1. 0. ;
P3 = 1. 10. ; P4 = 0. 10. ;
LI1= P1 DROI 1 P2 ;
LI2= P2 DROI P3 DINI 1. DFIN 0.1 ;
LI3= P3 DROI 1 P4 ;
LI4= P4 DROI P1 DINI 0.1 DFIN 1. ;
LI = LI1 ET LI2 ET LI3 ET LI4 ;
CADRE = DALL LI1 LI2 LI3 LI4 PLAN ;
TRAC CADRE ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MO=MODE CADRE POREUX ;

LAM = 1.D10 ;
MU = 2.5D9 ;
PK = 1.D-9 ;
MAT=MATE MO YOUNG 7000D6 NU 0.4 MOB 1.D18
COB 1. PERM PK VISCO 1. ALPH 0. KF 1.
RHO 0. ALPF 0. ALPM 0. RHO 0. ;
*
*----- BOUNDARY CONDITIONS -----*
BL1 = BLOQ UY LI1 ;
BL2 = BLOQ UX (LI2 ET LI4) ;
BLP = BLOQ P LI3 ;
*
*----- CONSOLIDATION -----*
* imposed pressure on the upper surface
P0 = 4.D3 ;
FPP = PRESSION MASS MO P0 LI3 ;
PER = PERM MO MAT ;
EV = EVOL MANU T ( PROG 0 1.D-5 1000)
F ( PROG 0 1 1 ) ;
CHA = CHAR MECA FPP EV ;
SI COMPLET ;
LETEMP = PROG 0 PAS 1.D-2 1.D-1 PAS 1.D-1 1.D0
PAS 5.D-1 1.D1 ;
SINON ;
LETEMP = PROG 0 PAS 1.D-2 1.D-1 PAS 1.D-1 5.D-1 ;
FINSI ;

NBT = DIME LETEMP ;
TFINAL = EXTR LETEMP NBT ;
*
*----- SETUP OF THE PASAPAS PROCEDURE -----*
TABLE = TABLE ;
TABL.'MODELE' = MO ;
TABL.'CARACTERISTIQUES' = MAT ;
TABL.'BLOCAGES_MECAIQUES' = BL1 ET BL2 ET BLP ;
TABL.'TEMPS_CALCULES' = LETEMP ;
TABL.'CHARGEMENT' = CHA ;
TABL.'CONSOLIDATION' = VRAI ;
TABL.'PERMEABILITE' = PER ;

PASAPAS TAB1 ;
*
*----- OUTPUT -----*
MESS ' RESULTATS TEST PORE3 ' ;
MESS 'We search the vertical displacement and the
pressure' ;
MESS 'They are compared with reference values.' ;

TTA = INDEX ( TAB1.DEPLACEMENTS ) ;
NT = DIME TTA ;
LIST NT ;
IT = 0 ;
REPETER BOPI NT ;
IT = IT + 1 ;
LET = TTA . IT ;
DD= TAB1.DEPLACEMENTS.LET ;
SI ( EGA IT 1 ) ;
PRT = PROG LET ;
PRD = PROG ( EXTR DD UY P3 ) ;
EV = EVOL CHPO DD P LI2 ;
SINON ;
PRT = PRT ET (PROG LET) ;
PRD = PRD ET (PROG ( EXTR DD UY P3)) ;
EV = EV ET (EVOL CHPO DD P LI2) ;
FINSI ;
FIN BOPI ;
FV = EVOL ROSE MANU T PRT UY PRD ;

DD = PECHE TAB1 DEPLACEMENTS ;
EV1 = EVOL ROSE CHPO DD P LI2 ;
EVD1 = EVOL ROSE CHPO DD UY LI2 ;
*
*----- DETERMINATION OF THE ANALYTIC SOLUTION -----*
X = COOR 2 LI2 ;
XX = ( MANU CHPO LI2 1 SCAL 1 ) - ( X / H ) ;
XX = NOMC XX P ;
FACT = (LAM + ( 2. * MU )) * ( PK / H2 ) ;
FACU = -1. * ( (LAM + (2. * MU)) / (P0 * H) ) ;
PI = 3.141592654D0 ;
PI2 = PI / 2.D0 ;
DEGPI = 180. / PI ;

ITT = 0 ;
REPE BOUT NBT ;
ITT = ITT + 1 ;
T= EXTR LETEMP ITT ;
PP = MANU CHPO LI2 1 P 0. ;
UU = X / H ;
UU = NOMC UU P ;
TT = FACT * T ;
N = 0 ;
SI COMPLET ;
NFOI = 50 ;
SINON ;
NFOI = 10 ;
FINSI ;
REPE BOU NFOI ;
M = PI2 * ( ( 2. * N ) + 1. ) ;
M2 = M*M ;
EX = EXP ( ( -1. * M2 ) * TT ) ;
SSI = SIN ( ( M * DEGPI ) * XX ) ;
CCO = COS ( ( M * DEGPI ) * XX ) ;
PP = PP + ( ( ( 2. / M ) * EX ) * SSI ) ;
UU = UU - ( ( ( 2. / M2 ) * EX ) * CCO ) ;
N=N + 1 ;

FIN BOU ;
PP = PP * P0 ;
UU = UU / FACU ;
SI ( EGA ITT 1 ) ;

```



9.3. TEST PORE3 (DESCRIPTION SHEET)

```
PGT = PROG T ;
PGU = PROG ( EXTR UU P P3) ;
SINON ;
PGT = PGT ET (PROG T) ;
PGU = PGU ET (PROG ( EXTR UU P P3)) ;
FINSI ;

SI ( EGA ITT NBT ) ;
EV2 = EVOL TURQ CHPO PP P LI2 ;
TITRE ' COMPARISON BETWEEN THE PRESSURE
      PROFILES' ;
DESS (EV1 ET EV2 ) ;
EVD2 = EVOL TURQ CHPO UU P LI2 ;
TITRE ' COMPARISON BETWEEN THE PROFILES
      OF VERTICAL DISPLACEMENTS' ;
DESS (EVD1 ET EVD2 ) ;
FINSI ;

FIN BOUT ;
FVT= EVOL TURQ MANU 'T' PGT 'UY' PGU ;
DESS ( FV ET FVT) ;

* Calculation of the differencies
EVY1 = EXTR EV1 ORDO ;
EVY2 = EXTR EV2 ORDO ;
EVDY1 = EXTR EVD1 ORDO ;
EVDY2 = EXTR EVD2 ORDO ;
ERPP = ( MAXI ABS ( EVY1 - EVY2 ) ) /
        ( MAXI ABS EVY2 ) ;
ERDUY = ( MAXI ABS ( EVDY1 - EVDY2 ) ) /
         ( MAXI ABS EVDY2 ) ;
MESS ' Error on the pressure:' ERPP  '%' ;
MESS ' Error on the profile of vertical
      displacement:' ERDUY  '%' ;
*
*----- GOOD WORKING MESSAGE -----*
SI (ERDUY <EG 0.1) ;
SI (ERPP <EG 0.1) ;
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
SINON ;
ERRE 5 ;
FINSI ;

FIN ;
```

Test pore3 (Comments)

1. The POREUX formulation

```

MO=MODE CADRE POREUX ;
LAM = 1.D10 ;
MU = 2.5D9 ;
PK = 1.D-9 ;
MAT=MATE MO YOUNG 7000D6 NU 0.4 MOB 1.D18
      COB 1. PERM PK VISCO 1. ALPH 0. KF 1.
      RHOF 0. ALPF 0. ALPM 0. RHO 0. ;
  
```

This procedure is to create a porous model and to give the material data.

2. The MOTS operator

```
MOTP = MOTS P ;
```

The MOTS operator creates a list of 4-character words.

3. The OPTI NORM operator

```
OPTI NORM MOTP LREP ;
```

The NORM option in the OPTI operator allows to normalize variables in view of factorization. It is required that a LISTMOT containing the initial unknowns to be normalized be given as well as the LISTREEL providing the standardization factor. We may also ask for automatic standardization ('AUTO') but this may take some time.

4. The INDEX operator

```
TTA = INDEX ( TAB1.DEPLACEMENTS) ;
```

The INDEX operator enables the user to get all the indices of a table. 'TAB1.DEPLACEMENTS' and 'TTA' are respectively the table which indices are requested and the generated object containing the indice (TABLE type).



Chapter 10

METALLURGY OF POWDERS

10.1 Test poudre1 (Description sheet)

Test name: poudre1

Calculation type: POWDER COMPACTION - 2D MODE AXIS

Finite element type: QUA8

Topic: Viscoplastic model of Abouaf for consolidation of powders. The structure is a cylinder consolidated by pressure load. The initial density is assumed uniform, whereas the final density is obtained with analytical calculation. The cylinder is also subjected to a field of temperature. It follows a viscoplastic law according to the model of Abouaf.




Goal: Test the model of Abouaf: poudre_A.

Version: 97' customer version

Model description:

$$E = 118.0 \cdot 10^3 \text{ Pa}$$

$$\text{NU} = 0.33$$








Test poudre1 (Results)

RESULTS: Displacements for an isotropic densification

	<i>UR</i>	<i>UZ</i>
<i>Analytical results</i>	$-1.25297E - 03$	$-1.88229E - 03$
<i>Castem results</i>	$-1.25292E - 03$	$-1.88222E - 03$

CASTEM FIGURES



Test poudrel (Listing CASTEM)

```

*****
TITR 'MODEL OF ABOUAF FOR CONSOLIDATION OF POWDERS' ;
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;

OPTI DIME 2 ELEM QUA8 MODE AXIS ;
OPTI ECHO 0 ;
*----- GEOMETRY -----*
RAYON_1 = 10.0 ; HAUTE_1 = 15.0 ;
*
* DEFINITION OF POINTS AND DENSITY
*      A0      *      A1      *
XA0 = 0.0      ; XA1 = RAYON_1 ;
YA0 = 0.0      ; YA1=YA0      ;
DENA0=10.0     ; DENA1 = 10.0  ;
*
*      A2      *      A3      *
XA2 = XA1      ; XA3 = 0.0    ;
YA2 = YA1 + HAUTE_1 ; YA3 = YA2 ;
DENA2 = 10.0   ; DENA3=10.0   ;

DENS DENA0 ; A0 = XA0 YA0 ;
DENS DENA1 ; A1 = XA1 YA1 ;
DENS DENA2 ; A2 = XA2 YA2 ;
DENS DENA3 ; A3 = XA3 YA3 ;

* DEFINITION OF LINES
LA0A1 = A0 DROI 2 A1 ; LA1A2 = A1 DROI 2 A2 ;
LA2A3 = A2 DROI 2 A3 ; LA3A0 = A3 DROI 2 A0 ;
LA3A2 = INVE LA2A3 ; LA2A1 = INVE LA1A2 ;

* LINES FOR BOUNDARY CONDITIONS
L_SYMZ = LA0A1 ; L_SYMX = LA3A0 ;
L_EXT1 = LA1A2 ; L_EXT2 = LA2A3 ;

* DEFINITION OF SURFACE
ELMAT1 = SURF (LA0A1 ET LA1A2 ET LA2A3 ET LA3A0) PLAN;
PIECE_1 = COUL ELMAT1 BLEU ;
TOT = PIECE_1 ;
ELIM (TOT ET A2) 0.001 ;
TRAC TOT ;
*
*----- CALCULATION -----*
T0TEMPE = 925.0 ; T1TEMPE = 925.0 ;
T0TEMPS = 0.0 ; T1TEMPS = 1200.0 ;
T2TEMPS = 2400.0 ; T3TEMPS = 3600.0 ;

*----- CHAMP-POINT OF TEMPERATURE -----*
CLTH0001 = BLOQ T (L_EXT1 ET L_EXT2) ;
P_BLOTH1 = DEPI CLTH0001 1. ;
P_TEMPS1 = PROG T0TEMPS T1TEMPS T2TEMPS T3TEMPS ;
P_THETA1 = PROG T0TEMPE T1TEMPE T1TEMPE T0TEMPE ;
EV_THER1 = EVOL MANU 'TEMPS' P_TEMPS1 'T' P_THETA1 ;
DESS EV_THER1 ;
CH_THER1 = CHAR TIMP P_BLOTH1 EV_THER1 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
P1MODEL = MODE PIECE_1 MECANIQUE ELASTIQUE
          VISCOPLASTIQUE POWDRE_A CONS DEUX ;

V1YOUNG = 118.0E+03 ;
VINU = 0.33 ;
V1ALPHA = 10.0E-10 ;
V1A = 7.76E-15 ;
VIN = 4.55 ;

V1QSR1 = 0.0 ;
V1RHOR = 0.68 ;

V1F0 = 0.1098 ; V1F1 = 1.0168 ; V1F2 = -0.2591 ;
V1F3 = 0.0 ; V1F4 = 0.0 ; V1F5 = 0.66 ;
V1C0 = 2.10851 ; V1C1 = 1.21767 ; V1C2 = -0.43081 ;
V1C3 = 0.0 ; V1C4 = 0.0 ; V1C5 = 50.0 ;

P1MATER = MATE P1MODEL YOUNG V1YOUNG NU VINU
          ALPHA V1ALPHA A V1A N VIN QSR1 V1QSR1 F0 V1F0 F1 V1F1
          F2 V1F2 F3 V1F3 F4 V1F4 F5 V1F5 C0 V1C0 C1 V1C1
          C2 V1C2 C3 V1C3 C4 V1C4 C5 V1C5 RHOR V1RHOR ;

P2MODEL = MODE PIECE_1 THERMIQUE ISOTROPE CONS UN ;

P2MATER = MATE P2MODEL 'C' 460 'K' 14.6 'RHO' 4.54 ;
*
*----- BOUNDARY CONDITIONS -----*
P0BLOCA = BLOQ UZ L_SYMZ ; P1BLOCA = BLOQ UR L_SYMX ;
P_BLOCA = P0BLOCA ET P1BLOCA ;
*
*----- LOAD DEFINITION -----*
T0PRES = 0.0 ;
T1PRES = 100.0 ;

LI_TPS = PROG T0TEMPS T1TEMPS T2TEMPS T3TEMPS ;
LI_PRES = PROG T0PRES T1PRES T1PRES T0PRES ;
EV_PRES = EVOL MANU TEMPS LI_TPS PRESSION LI_PRES ;
DESS EV_PRES TITR 'EVOLUTION OF THE PRESSURE LOAD' ;

P2FORCE = PRES MASS P1MODEL 1.0 L_EXT1 ;
P3FORCE = PRES MASS P1MODEL 1.0 L_EXT2 ;
TITR 'APPLIED STRENGTH' ;
VECT1 = VECT (P2FORCE ET P3FORCE) 'FR' 'FZ' 0.1 ;
TRAC VECT1 PIECE_1 ;

P2CHARG = CHAR P2FORCE EV_PRES 'MECA' ;
P3CHARG = CHAR P3FORCE EV_PRES 'MECA' ;
P_CHARG = P2CHARG ET P3CHARG ;

*----- CALCULATION AND OUTPUT -----*
P_MODEL = P1MODEL ET P2MODEL ;
P_MATER = P1MATER ET P2MATER ;

* SETUP OF THE TABLE FOR THE PASAPAS PROCEDURE
TAB1 = TABLE ;
TAB1 . 'CHARGEMENT' = (P_CHARG ET CH_THER1) ;
TAB1 . 'MODELE' = P_MODEL ;
TAB1 . 'CARACTERISTIQUES' = P_MATER ;
TAB1 . 'BLOCAGES_MECANIQUES' = P_BLOCA ;
TAB1 . 'BLOCAGES_THERMIQUES' = CLTH0001 ;

SI COMPLET; TFIN = 2400. ;
SINON ; TFIN = 400. ;
FINSI;
TAB1 . 'TEMPS_CALCULES' = PROG 0.0
          PAS 50.0 TFIN ;
TAB1 . 'TEMPS_SAUVES' = PROG TFIN ;
TAB1 . 'MAXITERATION' = 49 ;
TAB1 . 'MAXISOUSPAS' = 500 ;
TAB1 . 'MOVA' = 'RHOR' ;
TAB1 . 'PRECISION' = 1.0E-4 ;
TAB1 . 'TALPHA_REFERENCE' = 20.0 ;
TAB1 . 'ACCELERATION' = 2 ;
TAB1 . 'PROCEDURE_THERMIQUE' = LINEAIRE ;

PASAPAS TAB1 ;

*
*----- GOOD WORKING TEST -----*
IND1 = INDEX (TAB1 . 'VARIABLES_INTERNES') ;
NB1 = DIME IND1 ;

```



10.1. TEST POUFRE1 (DESCRIPTION SHEET)

```
* FINAL DENSITY
VM1 = TAB1 . 'VARIABLES_INTERNES' .(IND1.NB1) ;
VM2 = EXCO RHOR VM1 ;
VM3 = REDU VM2 PIECE_1 ;
RHOF1 = MAXI VM3 ;

* THEORETICAL CALCULATION OF THE FINAL DENSITY
SI COMPLET ; RHOF2 = 0.7418622 ;
SINON      ; RHOF2 = 0.68006 ;
FINSI;

* DISPLACEMENTS AT THE P2 POINT
IND2 = INDEX (TAB1 . 'DEPLACEMENTS' ) ;
NB2 = DIME IND2 ;
UR1 = (EXTR TAB1 . 'DEPLACEMENTS' .(IND2.NB2) UR A2) ;
UZ1 = (EXTR TAB1 . 'DEPLACEMENTS' .(IND2.NB2) UZ A2) ;

* ANALYTIC CALCULATION OF DISPLACEMENTS
FOR AN ISOTROPIC CONSOLIDATION
SI COMPLET ;
      UR2 = RAYON_1 * ((V1RHOR/RHOF1)**(1./3.) -1.) ;
      UZ2 = HAUTE_1 * ((V1RHOR/RHOF1)**(1./3.) -1.) ;
SINON ; UR2 = -1.25297E-03 ;
      UZ2 = -1.88229E-03 ;
FINSI;
LIST UR2 ; LIST UR1 ;
LIST UZ2 ; LIST UZ1 ;

* CALCULATION OF ERRORS ON DISPLACEMENTS AND DENSITIES
ERH01 = ABS ((RHOF1 - RHOF2 )/RHOF2);
EUR1 = ABS ((UR1 - UR2) / UR2) ;
EUZ1 = ABS ((UZ1 - UZ2) / UZ2) ;

OPTI ECHO 0 ;
SAUTER 3 LIGNES ;
MESS 'CALCULATION (ISOTROPIC CASE):' ;
MESS 'ERROR ON THE RADIAL DISPLACEMENT: ' EUR1 ;
MESS 'ERROR ON THE VERTICAL DISPLACEMENT: ' EUZ1 ;
MESS 'ERROR ON THE FINAL DENSITY: ' ERH01 ;
SAUT 3 LIGNES ;

*
SI ( (EUR1 + EUZ1 + ERH01 ) >EG 0.05 ) ;
  MESS '----- WRONG RESULTS -----' ;
  SAUT 3 LIGNES ;
  ERRE 5;
SINON ;
  MESS '----- GOOD RESULTS -----' ;
  SAUT 3 LIGNES ;
FINSI ;
*
FIN ;
```

Test poudre1 (Comments)

1. ABOUAF model for consolidation of powders: POUFRE_A

```
P1MODEL = MODE PIECE_1 MECANIQUE ELASTIQUE
~~~VISCOPLASTIQUE POUFRE_A CONS DEUX ;
P1MATER = MATE P1MODEL YOUNG V1YOUNG NU V1NU ALPHA V1ALPHA
~~~A V1A N V1N QSRT V1QSRT F0 V1F0 F1 V1F1 F2 V1F2 F3 V1F3
~~~F4 V1F4 F5 V1F5 C0 V1C0 C1 V1C1 C2 V1C2 C3 V1C3 C4 V1C4
~~~C5 V1C5 RHOR V1RHOR ;
```

The ABOUAF model is a viscoplastic model for consolidation of powders.

The constitutive equations of Abouaf model are:

$$\begin{aligned} \text{Seq} &= (f \cdot I_1^{2+3/2 \cdot c} \cdot J_2^{2})^{0.5} \\ \text{devp/dt} &= \rho \cdot A \cdot \exp(-Q/RT) \cdot \text{Seq}^{(n-1)} \cdot (f \cdot I_1^{d+3/2 \cdot c} \cdot s) \\ \text{tr}(\text{devp/dt}) &= -(\text{drho/dt})/\rho \end{aligned}$$

With the following notations:

S stress tensor
 s deviatoric stress tensor
 d unit tensor
 Seq equivalent stress
 I1 first invariant of stress tensor
 J2 second invariant of deviatoric stress tensor
 rho relative density of porous material
 f function of relative density
 c function of relative density
 evp viscoplastic deformation
 t time
 T temperature

We should put several data in the MATE operator:

A coefficient of creep law
 N power creep law exponent
 QSRT activation energy
 F0 to F5 parameters defining the f function
 C0 to C5 parameters defining the c function
 RHOR initial relative density of the material

2. Isotropic thermal model

```
P2MODEL = MODE PIECE_1 THERMIQUE ISOTROPE CONS UN ;
P2MATER = MATE P2MODEL 'C' 460 'K' 14.6 'RHO' 4.54 ;
```

With the isotropic thermal model, we have to specify the specific heat (C), the thermal conductivity (K) and the density (RHO). The word CONS indicates the number of component. It allows to associate in the same calculation several models to a same mesh object.



Chapter 11

JOINT

11.1 Test joi22 (Description sheet)

Test name: joi22

Calculation type: ISOTROPIC ELASTIC MECHANIC


Finite element type: SEG2, JOI2

Topic: Tensile test on a 2D-joint. The structure is a 2D-joint subjected to a tensile strength on its upper surface, in the direction perpendicular to its plan. It is embedded from its lower surface.

Goal: Test the JOI2 element.

Version: 97' customer version

Model description:





Test joi22 (Results)

RESULTS Analytical solution:

From the principle of virtual work, we can show for a one-axis tensile test that the delta displacement at each node is equal to:

$$\Delta = \frac{F}{k * L * l}$$

where:

F is the strength

k is the stiffness

L is the length of the joint element

l is the width of the joint element

Numerical application:

$$F = 100000.0 \text{ N}$$

$$k = 4.2E10 \text{ N/m}^3$$

$$L = 2.0 \text{ m}$$

$$l = 1.0 \text{ m}$$

$$\Delta = 100000 / (4.2E10 * 2.0 * 1.0)$$

$$= 1.190476E-6$$

Castem solution:

Point	Unknown	Point	Unknown	Point	Unknown	Point	Unknown
5	0.00000E0	6	0.00000E0	7	2.50000E4	8	2.50000E-4
9	0.000E0	10	2.50000E4	11	2.50000E4	12	2.50000E-4

Point	UX	UY
1	0.00000E0	1.19048E - 6
2	0.00000E0	1.19048E - 6
3	0.00000E0	-2.11615E - 22
4	0.000000E0	-1.05736E - 22

Strain calculation	
<i>First component</i>	
point 1 :	0.000E0
point 2 :	0.0000E0
<i>Second component</i>	
point 1 :	-1.190E - 6
point 2 :	-1.190E - 6

Stress calculation	
<i>First component</i>	
point 1 :	0.000E0
point 2 :	0.0000E0
<i>Second component</i>	
point 1 :	-5.000E4
point 2 :	-5.000E4

Test joi22 (Listing CASTEM)

```

*****
TITR 'TENSILE TEST ON A 2D-JOINT' ;
*****
OPTI DIME 2 ELEM SEG2 MODE PLAN CONT ;
OPTI ECHO 0 ;
*
SOLANA = 1.190476E-6 ;
SOLNUL = 0.0 ;
*
*----- GEOMETRY -----*
* THE UPPER SURFACE OF THE JOINT
A1 = 0.00 0.00 ;
B1 = 2.00 0.00 ;
H1 = A1 DROI 1 B1 ; L1 = H1 ;
*
* THE LOWER SURFACE OF THE JOINT
IA1 = 0.00 0.00 ;
IB1 = 2.00 0.00 ;
IH1 = IA1 DROI 1 IB1 ; IL1 = IH1 ;
*
* JOINT CREATION : JOI2
OPTI ELEM RAC2 ;
VOL = RACCORD 0.00001 L1 IL1 ;
*
*----- BOUNDARY CONDITIONS -----*
CL11 = BLOQ IA1 UX ; CL12 = BLOQ IA1 UY ;
CL1 = CL11 ET CL12 ;
*
CL21 = BLOQ IB1 UX ; CL22 = BLOQ IB1 UY ;
CL2 = CL21 ET CL22 ;
*
CL = CL1 ET CL2 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD1 = MODE VOL MECANIQUE ELASTIQUE ISOTROPE JOI2 ;
MA1 = MATE MOD1 KS 4.2E08 KN 4.2E10 ;
*
*----- LOAD DEFINITION: TENSILE STRENGTH -----*
FO1 = FORCE ( 0. 100000.0 ) L1 ;

*-- CALCULATION OF THE HOOKE AND RIGIDITY MATRIX -*
HOO1 = HOOKE MOD1 MA1 ;

RI1 = RIGI MOD1 HOO1 ;
RI2 = RI1 ET CL ;
*
*----- CALCULATION AND OUTPUT -----*
RE = RESO RI2 FO1 ;
MESS ' ' ;
MESS ' ' ;
MESS ' Analytic solution: ' ;
MESS ' ' ;
MESS ' UX =' SOLN ;
MESS ' UY =' SOLANA ;
MESS ' ' ;
MESS ' ' ;
MESS ' ' ;
MESS ' Calculated solution: ' ;
MESS ' ' ;
LIST RE ;
*
*----- GOOD WORKING MESSAGE -----*
DEPA1 = EXTR RE UY A1 ;
RESI = ABS( (DEPA1-SOLANA)/SOLANA ) ;
*
SI (RESI <EG 1E-4 ) ;

ERRE 0 ;
    
```

```

SINON;
    ERRE 5 ;
FINSI ;
*
*----- STRAIN CALCULATION -----*
EPS1 = EPSI MOD1 RE ;
LIST EPS1 ;
*
*----- STRESS CALCULATION -----*
SIG1 = SIGMA MOD1 MA1 RE ;
LIST SIG1 ;
*
FIN ;
    
```



Test joi22 (Comments)

1. JOINT operator

```
MOD1 = MODE VOL MECANIQUE ELASTIQUE ISOTROPE JOI2 ;  
MA1 = MATE MOD1 KS 4.2E08 KN 4.2E10 ;
```

The JOI2 element is used in the modelling of a 2D-joint. This joint is an element with four nodes associated with the geometrical support RAC2. Within the MATE operator we should specify the shear and the normal stiffnesses, respectively KS and KN.

2. RACCORD operator

```
OPTI ELEM RAC2 ;  
VOL = RACCORD 0.00001 L1 IL1 ;
```

The RACCORD operator enables the user to create an ordinary joint element. It generates a line VOL (MAILLAGE type) composed of double points that connects the objects L1 and IL1 (MAILLAGE type). The objects L1 and IL1 may be segments of 2 or 3 points. An element is created between an element of the L1 object and an element of the IL1 object; the distance between their points is smaller than a distance criterion (FLOTTANT type: 0.00001).

For the creation of a 2D joint element JOI2 (or JOI3) L1 and IL1 represent respectively lines 1 and 2 of the element. These lines should be read in the same direction. This direction is that of line 1. It is defined by considering a vector n normal to this line. When it is orientated positively in the direction of the opening of the joint element, n must necessarily lie on the right handside of the segment going from node 1 to node 2 of line 1. (We define the opening of the element as the separation movement of line 1 from line 2 when line 2 is held fixed).

3. HOOKE operator

```
HOO1 = HOOKE MOD1 MA1 ;
```

The HOOKE operator constructs the field of HOOKE's matrix from a field of geometrical and material properties MA1 (MCHAML type, CARACTERISTIQUES subtype). Here HOO1 is a field by element of Hooke's matrices (MCHAML type, MATRICE DE HOOKE subtype).

MA1 is required only for the types of elements the geometry of which cannot be deduced from the mesh, for instance, beam, pipe, or shell elements.

4. EPSI and SIGMA operators

```
EPS1 = EPSI MOD1 RE ;          SIG1 = SIGMA MOD1 MA1 RE ;
```

The EPSI operator enables to calculate a strain field either from a displacement field, according to usual expressions with or without second order terms (1-st syntax), or by taking the natural logarithm of a symmetric gradient field (2-nd syntax). This second possibility is now available only for the massive formulation.

For some elements, (beams, pipes, thin shells, with or without transverse shear) it describes generalized strains, i.e. membrane-type strains and curvature variations. For the joint elements, it describes relative displacements. The strains for the solid elements are computed in the general basis, those for the shell, plate, and beam elements are computed in the local axes.

Here MOD1 is a model object (MMODEL type), GRAD1 is a symmetrical gradient field (MCHAML type, GRADIENT subtype) and EPS1 is the generated displacement field (MCHAML type, DEFORMATIONS subtype).

The SIGMA operator calculates a stress field from a displacement field. It is assumed that the material has an elastic linear behaviour and that there is no initial strain. For some elements, it describes forces, (bars, beams, pipes), for others, it describes generalized stresses (thin shells). The stresses are calculated in the general basis for the solid elements, and in the local axes for the shell, plate and beam elements. Here SIG1 is the generated stress field (MCHAML type, CONTRAINTES subtype).



11.2 Test joi41 (Description sheet)

Test name: joi41

Calculation type: ISOTROPIC ELASTIC MECHANICS

Finite element type: QUA4, CUB8,JOI4


Topic: Two superposed cubes with a JOI4 joint in their intersection. We study the effects of joints with the help of two examples: cubes with joint and cubes without joint.

Goal: Calculate the effect of a JOI4 joint on displacements, strains and stresses.

Version: 97' customer version

Model description:

$$\begin{aligned} E &= 2.1 \cdot 10^{11} \text{ Pa} \\ \nu &= 0.3 \end{aligned}$$



Test joi41 (Listing CASTEM)

```

*****
TITR ' 2 SUPERPOSED CUBES WITH A JOINT IN THE MIDDLE';
*****
MESS -----!';
MESS '! CASE OF TWO SUPERPOSED CUBES WITHOUT JOINT !';
MESS '!-----!';
OPTI DIME 3 ELEM QUA4 MODE TRID ;
*
*----- GEOMETRY -----*
* THE UPPER CUBE
SA1 = 0.00 0.00 0.00 ; SB1 = 2.00 0.00 0.00 ;
SC1 = 2.00 2.00 0.00 ; SD1 = 0.00 2.00 0.00 ;

SH1 = SA1 DROI 1 SB1 ; SH2 = SB1 DROI 1 SC1 ;
SH3 = SC1 DROI 1 SD1 ; SH4 = SD1 DROI 1 SA1 ;
*
SL1 = SH1 ET SH2 ET SH3 ET SH4 ;
SS1 = SURF SL1 PLAN ;

* THE LOWER CUBE
IA1 = 0.00 0.00 -4.00 ; IB1 = 2.00 0.00 -4.00 ;
IC1 = 2.00 2.00 -4.00 ; ID1 = 0.00 2.00 -4.00 ;

IH1 = IA1 DROI 1 IB1 ; IH2 = IB1 DROI 1 IC1 ;
IH3 = IC1 DROI 1 ID1 ; IH4 = ID1 DROI 1 IA1 ;

IL1 = IH1 ET IH2 ET IH3 ET IH4 ;
IS1 = SURF IL1 PLAN ;
*
*
OPTI ELEM CUB8 ;
*
VOLS = SS1 VOLU 1 TRAN ( 0.00 0.00 -2.00 ) ;
FACS = FACE 2 VOLS ;
VOLI = IS1 VOLU 1 TRAN ( 0.00 0.00 2.00 ) ;
FACI = FACE 2 VOLI ;
VT1 = VOLS ET VOLI ;
VT1 = ELIM 0.00001 VT1 ;
VTSAN = VT1 ;
*
*----- BOUNDARY CONDITIONS -----*
CL11 = BLOQ IC1 UX ; CL12 = BLOQ IC1 UY ;
CL13 = BLOQ IC1 UZ ;
CL1 = CL11 ET CL12 ET CL13 ;
*
CL21 = BLOQ IH2 UX ; CL22 = BLOQ IH2 UZ ;
CL2 = CL21 ET CL22 ;
*
CL31 = BLOQ IH3 UY ; CL32 = BLOQ IH3 UZ ;
CL3 = CL31 ET CL32 ;
*
CL4 = BLOQ IA1 UZ ;
*
CL = CL1 ET CL2 ET CL3 ET CL4 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD1 = MODE VT1 MECANIQUE ELASTIQUE ISOTROPE ;
MA1 = MATE MOD1 YOUN 2.1E11 NU 0.3 ;
*
*----- LOAD DEFINITION: UNIFORM PRESSURE -----*
CHPO1 = PRES MASS MOD1 -100000.00 SS1 ;
*
*----- CALCULATION OF THE RIGIDITY MATRIX -----*
RI1 = RIGI MOD1 MA1 ;
RI2 = RI1 ET CL ;
*
*----- CALCULATION AND OUTPUT -----*
RE = RESO RI2 CHPO1 ;
*
*-- RESTRICTION OF THE DISPLACEMENTS -----*
*----- ON THE DIFFERENT ELEMENTS-----*
DEPS = REDU RE VOLS ;
MESS ' DISPLACEMENTS OF THE UPPER CUBE NODES ' ;
LIST DEPS ;
*
DESUSA = DEPS ;
*
DEPI = REDU RE VOLI ;
MESS ' DISPLACEMENTS OF THE LOWER CUBE NODES ' ;
LIST DEPI ;
*
DEINSA = DEPI ;
*
*----- CALCULATION OF THE DEFORMATIONS -----*
EPS1 = EPSI MOD1 RE ;
*
*--- RESTRICTION OF THE DEFORMATIONS -----*
*----- ON THE DIFFERENT ELEMENTS -----*
EPSS = REDU EPS1 VOLS ;
MESS ' DEFORMATIONS OF THE UPPER CUBE NODES ' ;
LIST EPSS ;
*
EPSB = REDU EPS1 VOLI ;
MESS ' DEFORMATIONS OF THE LOWER CUBE NODES ' ;
LIST EPSB ;
*
*----- CALCULATION OF THE STRESS -----*
SIG1 = SIGMA MOD1 MA1 RE ;
*
* RESTRICTION OF THE STRESS ON THE DIFFERENT ELEMENTS*
SIGS = REDU SIG1 VOLS ;
MESS ' STRESS OF THE UPPER CUBE NODES ' ;
LIST SIGS ;
*
SIGI = REDU SIG1 VOLI ;
MESS ' STRESS OF THE LOWER CUBE NODES ' ;
LIST SIGI ;
*
SAUT PAGE;

MESS '!-----!';
MESS '! CASE OF TWO SUPERPOSED CUBES WITH A JOINT !';
MESS '!-----!';

OPTI DIME 3 ELEM QUA4 MODE TRID ;
*
*----- GEOMETRY -----*
* THE UPPER CUBE
SA1A = 0.00 0.00 0.00 ; SB1A = 2.00 0.00 0.00 ;
SC1A = 2.00 2.00 0.00 ; SD1A = 0.00 2.00 0.00 ;

SH1A = SA1A DROI 1 SB1A ; SH2A = SB1A DROI 1 SC1A ;
SH3A = SC1A DROI 1 SD1A ; SH4A = SD1A DROI 1 SA1A ;
*
SL1A = SH1A ET SH2A ET SH3A ET SH4A ;
SS1A = SURF SL1A PLAN ;

* THE LOWER CUBE
IA1A = 0.00 0.00 -4.00 ;
IB1A = 2.00 0.00 -4.00 ;

IC1A = 2.00 2.00 -4.00 ;
ID1A = 0.00 2.00 -4.00 ;
IH1A = IA1A DROI 1 IB1A ; IH2A = IB1A DROI 1 IC1A ;
IH3A = IC1A DROI 1 ID1A ; IH4A = ID1A DROI 1 IA1A ;
    
```



11.2. TEST JOI41 (DESCRIPTION SHEET)

```
IL1A = IH1A ET IH2A ET IH3A ET IH4A ;
IS1A = SURF IL1A PLAN ;
*
*
OPTI ELEM CUB8 ;
*
VOLSA = SS1A VOLU 1 TRAN ( 0.00 0.00 -2.00 ) ;
FACSA = FACE 2 VOLSA ;
VOLIA = IS1A VOLU 1 TRAN ( 0.00 0.00 2.00 ) ;
FACIA = FACE 2 VOLIA ;
VT1A = VOLSA ET VOLIA ;
*
*----- BOUNDARY CONDITIONS -----*
CL11A = BLOQ IC1A UX ;
CL12A = BLOQ IC1A UY ;
CL13A = BLOQ IC1A UZ ;
CL1A = CL11A ET CL12A ET CL13A ;
*
CL21A = BLOQ IH2A UX ;
CL22A = BLOQ IH2A UZ ;
CL2A = CL21A ET CL22A ;
*
CL31A = BLOQ IH3A UY ;
CL32A = BLOQ IH3A UZ ;
CL3A = CL31A ET CL32A ;
*
CL4A = BLOQ IA1A UZ ;
*
CLA = CL1A ET CL2A ET CL3A ET CL4A ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD1A = MODE VT1A MECANIQUE ELASTIQUE ISOTROPE ;
MA1A = MATE MOD1A YOUN 2.1E11 NU 0.3 ;
*
*-- CREATION OF THE JOINT ELEMENT BETWEEN THE CUBES -*
*
OPTI ELEM LIA4 ;
VOLJA = LIAISON 0.0001 FACSA FACIA ;
*
VTOTAL=VT1A ET VOLJA ;
*
*----- DEFINITION OF THE JOINT MODEL -----*
MODJA = MODE VOLJA MECANIQUE ELASTIQUE ISOTROPE JOI4 ;
MAJA = MATE MODJA KS 4.2E15 KN 4.2E15 ;
*
*----- MODEL AND MATERIAL OF THE WHOLE STRUCTURE ----*
MODTA = MOD1A ET MODJA ;
MATTA = MA1A ET MAJA ;
*
*----- LOAD DEFINITION: UNIFORM PRESSURE -----*
CHPO1A= PRES MASS MODTA -100000.00 SS1A ;
*
*----- CALCULATION OF THE RIGIDITY MATRIX -----*
RITA = RIGI MODTA MATTA ;
RITOA = RITA ET CLA ;
*----- CALCULATION AND OUTPUT -----*
RETA = RESO RITOA CHPO1A ;
*
*- RESTRICTION OF THE DEFORMATIONS -----*
*----- ON THE DIFFERENT ELEMENTS -----*
DEPSA = REDU RETA VOLSA ;

MESS ' DISPLACEMENTS OF THE UPPER CUBE NODES ' ;
LIST DEPSA ;
*
DEPIA = REDU RETA VOLIA ;
MESS ' DISPLACEMENTS OF THE LOWER CUBE NODES ' ;
LIST DEPIA ;
*
DEPJ = REDU RETA VOLJA ;

MESS ' DISPLACEMENTS OF THE JOINT NODES ' ;
LIST DEPJ ;
*----- CALCULATION OF THE DEFORMATIONS -----*
EPSTA = EPSI MODTA RETA ;
*
*- RESTRICTION OF THE DEFORMATIONS -----*
*----- ON THE DIFFERENT ELEMENTS -----*
EPSSA = REDU EPSTA VOLSA ;
MESS ' DEFORMATIONS OF THE UPPER CUBE ' ;
LIST EPSSA ;
*
EPSBA = REDU EPSTA VOLIA ;
MESS ' DEFORMATIONS OF THE LOWER CUBE ' ;
LIST EPSBA ;
*
EPSJA = REDU EPSTA VOLJA ;
MESS ' DEFORMATIONS OF THE JOINT ' ;
LIST EPSJA ;
*
*----- CALCULATION OF THE STRESS -----*
SIGTA = SIGMA MODTA MATTA RETA ;
*
* RESTRICTION OF THE STRESS ON THE DIFFERENT ELEMENTS*
SIGSA = REDU SIGTA VOLSA ;
MESS ' STRESS OF THE UPPER CUBE ' ;
LIST SIGSA ;
*
SIGIA = REDU SIGTA VOLIA ;
MESS ' STRESS OF THE LOWER CUBE ' ;
LIST SIGIA ;
*
SIGJA = REDU SIGTA VOLJA ;
MESS ' STRESS OF THE JOINT ' ;
LIST SIGJA ;
*
*----- GOOD WORKING MESSAGE -----*
DESUAV = DEPS ;
* DIFFERENCE BETWEEN THE DISPLACEMENTS OF
* THE UPPER CUBE WITH AND WITHOUT JOINT ;
DESDIF = DESUSA - DESUAV ;

DEINAV = DEPI ;
* DIFFERENCE BETWEEN THE DISPLACEMENTS OF
* THE LOWER CUBE WITH AND WITHOUT JOINT ;
DEIDIF = DEINSA - DEINAV ;
*
MAXDIS = MAXI DESDIF ABS ; MAXDII = MAXI DEIDIF ABS ;
**
SI ((MAXDIS <EG 1E-9) ET (MAXDII <EG 1E-9));
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
*
*
FIN;
```

Test joi41 (Comments)

1. LIAISON OPERATOR: creation of linkage elements

```
OPTI ELEM LIA4 ;
VOLJA = LIAISON 0.0001 FACSA FACIA ;
```

The LIAISON operator generates the set of linkage elements between two surfacic objects (FACSA and FACIA: MAILLAGE type). VOLJA is the generated object (MAILLAGE type), and 0.0001 is the proximity criterion between the two surfacic objects.

For the creation of a joint element JOI4 (3D):

-FACSA and FACIA represent respectively surfaces 1 and 2 of the element. These surfaces should be read in the same direction. This direction is that of the boundary of surface 1. The numbering of the nodes of surface 1 must be such that the axes (1,2,N) make up a direct trihedron, with N orientated positively in the direction of the opening of the joint element,

We define the following quantities :

- . axis 1 = vector connecting node 1 to node 2 of surface 1
- . axis 2 = vector connecting node 1 to node 4 of surface 1
- . axis N = vector normal to the plane defined by vectors 1 and 2
- . opening of the joint element = separator movement of surface 1 from surface 2 when surface 2 is held fixed

-To take into account initial free space x in a joint element, FACSA and FACIA should have a distance x between them. In addition, x should be input as an initial normal inelastic strain before the use of the NONLIN procedure. (See report DMT/93.655)

LIA4 is a connection element with $2*4$ nodes.

2. REDU OPERATOR: restriction of a field or an object

```
DEPS = REDU RE VOLS ; (OBJET3 = REDU OBJET1 OBJET2 ;)
```

The REDU operator restricts :

- a field by points to the points of a given mesh or to the non-zero values of a CHPOINT
- a field by elements to the elements of a given mesh or to the given MMODEL object
- a MMODEL object to a given mesh
- a NUAGE object to given component

We have different possible types:

<i>OBJET1</i>	<i>OBJET2</i>	<i>OBJET3</i>
<i>CHPOINT</i>	<i>MAILLAGE</i>	<i>CHPOINT</i>
<i>CHPOINT</i>	<i>POINT</i>	<i>CHPOINT</i>
<i>CHPOINT</i>	<i>CHPOINT</i>	<i>CHPOINT</i>
<i>MCHAML</i>	<i>MAILLAGE</i>	<i>MCHAML</i>
<i>MCHAML</i>	<i>MMODEL</i>	<i>MCHAML</i>
<i>MMODEL</i>	<i>MAILLAGE</i>	<i>MMODEL</i>
<i>NUAGE</i>	<i>MOT1,MOT2...</i>	<i>NUAGE</i>

In case of restriction of a MCHAML or a MMODEL on a mesh :

- this mesh must be composed of elements of the same type as the mesh that supports the MCHAML or the MMODEL
- all the elements of the mesh must be included in the mesh that supports the MCHAML or the MMODEL



11.3 Test joi44 (Description sheet)

Test name: joi44

Calculation type: ORTHOTROPIC ELASTIC MECHANICS

Finite element type: QUA4, JOI4


Topic: Calculation of the thermal stresses on a 3D joint. The structure is a parallelepipedic joint embedded at the lower surface. The upper surface is free. The joint is subjected to a field of temperatures of CHAMELEM type, and of subtype TEMPERATURES. The MCHAML is pertaining to a MMODEL object at first, and to a mesh object secondly.

Goal: Calculate the field of thermal stresses induced by a field of temperatures.

Version: 97' customer version

Model description:

KS1	=	$4.2 \cdot 10^{10}$
KS2	=	$3.0 \cdot 10^5$
KN	=	$4.2 \cdot 10^{20}$
ALPN	=	2.5
T	=	1000.0



Test joi44 (Results)
RESULTS Analytical solution:

$$\text{SIGMA} = K * \text{ALPN} * \text{DELTA}(T)$$

where

K is the stiffness in the direction perpendicular to the joint plan

ALPN is the thermal expansion coefficient in the direction n, perpendicular to the joint plan (ALPN is homogeneous to length per degrees; the length is the thickness of the joint)

DELTA(T) is the CHAMALEM of temperature

$$\begin{aligned} \text{Then SIGMA} &= 4.2E20 * 2.5 * 1000.0 \\ &= 1.05E24 \end{aligned}$$

Castem solution:

	<i>CHAMELEM ON A MODEL OBJECT</i>		
	<i>SMS1</i>	<i>SMS2</i>	<i>SMSN</i>
<i>Analytical solution</i>	0.00000E00	0.00000E00	1.05000E24
<i>Castem Solution</i>			
<i>Point 1</i>	0.000E00	0.000E00	1.050E24
<i>Point 2</i>	0.000E00	0.000E00	1.050E24
<i>Point 3</i>	0.000E00	0.000E00	1.050E24
<i>Point 4</i>	0.000E00	0.000E00	1.050E24

	<i>CHAMELEM ON A MESH OBJECT</i>		
	<i>SMS1</i>	<i>SMS2</i>	<i>SMSN</i>
<i>Analytical solution</i>	0.00000E00	0.00000E00	1.05000E24
<i>Castem Solution</i>			
<i>Point 1</i>	0.000E00	0.000E00	1.050E24
<i>Point 2</i>	0.000E00	0.000E00	1.050E24
<i>Point 3</i>	0.000E00	0.000E00	1.050E24
<i>Point 4</i>	0.000E00	0.000E00	1.050E24



11.3. TEST JOI44 (DESCRIPTION SHEET)

Test joi44 (Listing CASTEM)

```

*****
TITR 'CALCULATION OF THERMAL STRESSES ON A 3D JOINT' ;
*****
OPTI DIME 3 ELEM QUA4 MODE TRID ;
*
SOLNUL = 0.0 ;
SOLANA = 1.05E24 ;
*
*----- GEOMETRY -----*
* THE UPPER SURFACE OF THE JOINT
A1 = 0.00 0.00 0.00 ; B1 = 2.00 0.00 0.00 ;
C1 = 2.00 2.00 0.00 ; D1 = 0.00 2.00 0.00 ;

H1 = A1 DROI 1 B1 ; H2 = B1 DROI 1 C1 ;
H3 = C1 DROI 1 D1 ; H4 = D1 DROI 1 A1 ;
L1 = H1 ET H2 ET H3 ET H4 ;
S1 = SURF L1 PLAN ;

* THE LOWER SURFACE OF THE JOINT
IA1 = 0.00 0.00 0.00 ; IB1 = 2.00 0.00 0.00 ;
IC1 = 2.00 2.00 0.00 ; ID1 = 0.00 2.00 0.00 ;

IH1 = IA1 DROI 1 IB1 ; IH2 = IB1 DROI 1 IC1 ;
IH3 = IC1 DROI 1 ID1 ; IH4 = ID1 DROI 1 IA1 ;
IL1 = IH1 ET IH2 ET IH3 ET IH4 ;
IS1 = SURF IL1 PLAN ;

* JOINT CREATION: JOI4
OPTI ELEM LIA4 ;
VOL = LIAISON 0.00001 S1 IS1 ;
*
*----- BOUNDARY CONDITIONS -----*
CL11 = BLOQ IA1 UX ; CL12 = BLOQ IA1 UY ;
CL13 = BLOQ IA1 UZ ;
CL1 = CL11 ET CL12 ET CL13 ;
*
CL21 = BLOQ IB1 UX ; CL22 = BLOQ IB1 UY ;
CL23 = BLOQ IB1 UZ ;
CL2 = CL21 ET CL22 ET CL23 ;
*
CL31 = BLOQ IC1 UX ; CL32 = BLOQ IC1 UY ;
CL33 = BLOQ IC1 UZ ;
CL3 = CL31 ET CL32 ET CL33 ;
*
CL41 = BLOQ ID1 UX ; CL42 = BLOQ ID1 UY ;
CL43 = BLOQ ID1 UZ ;
CL4 = CL41 ET CL42 ET CL43 ;
*
CL = CL1 ET CL2 ET CL3 ET CL4 ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
P1 = 0.707 0.707 0.707 ;
MOD1 = MODE VOL MECANIQUE ELASTIQUE ORTHOTROPE JOI4 ;
MA1 = MATE MOD1 DIRECTION P1 PERPENDICULAIRE
      KS1 4.2E10 KS2 3.0E5 KN 4.2E20 ALPN 2.5 ;
*
*----- CALCULATION OF THE RIGIDITY MATRIX -----*
RI1 = RIGI MOD1 MA1 ;
RI2 = RI1 ET CL ;
*
*----- THERMIC STRESSES -----*
* MCHAML TYPE OBJECT PERTAINING TO A MMODEL OBJECT
*
C1T = MANU CHML MOD1 T 1000.0 TYPE TEMPERATURES ;
SI1 = THET MOD1 MA1 C1T ;
*

MMODEL OBJECT ' ;
MESS ' Analytic solution :' ;
MESS ' ;
MESS ' SMS1 =' SOLNUL ;
MESS ' SMS2 =' SOLNUL ;
MESS ' SMN =' SOLANA ;
MESS ' ;
MESS ' Calculated solution :' ;
LIST SI1 ;

*-----*
* MCHAML TYPE OBJECT PERTAINING TO A MESH OBJECT
*
C2T = MANU CHML VOL T 1000.0 TYPE TEMPERATURES ;
SI2 = THET MOD1 MA1 C2T ;
MESS ' ;
MESS 'MCHAML TYPE OBJECT PERTAINING TO A MESH OBJECT':
MESS ' Analytic solution :' ;
MESS ' ;
MESS ' SMS1 =' SOLNUL ;
MESS ' SMS2 =' SOLNUL ;
MESS ' SMN =' SOLANA ;
MESS ' ;
MESS ' Calculated solution :' ;
LIST SI2 ;
*
*----- GOOD WORKING MESSAGE -----*
MAXCO1 = MAXI SI1 ABS ;
RESI1 = ABS( (MAXCO1-SOLANA)/SOLANA ) ;

MAXCO2 = MAXI SI2 ABS ;
RESI2 = ABS( (MAXCO2-SOLANA)/SOLANA ) ;
*
SI ( (RESI1 <EG 1E-4) ET (RESI2 <EG 1E-4) ) ;
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
*
FIN ;

MESS 'MCHAML TYPE OBJECT PERTAINING TO A

```

Test joi44 (Comments)

- MANU OPERATOR: creation of a MCHAML object type

```
C1T = MANU CHML MOD1 T 1000.0 TYPE TEMPERATURES ;  
SI1 = THET MOD1 MA1 C1T ;  
C2T = MANU CHML VOL T 1000.0 TYPE TEMPERATURES ;  
SI2 = THET MOD1 MA1 C2T ;
```

The MANU operator enables the user to simply create objects of the following types: MAILLAGE, CH-POINT, SOLUTION, RIGIDITE. Here we have the creation of a MCHAML type object. The MANU CHML operator creates a new field by element (MCHAML type) connected with the VOL object (MAILLAGE type) or with the MOD1 object (MMODEL type). In the first case the MCHAML type object is pertaining to a MMODEL object whereas in the second case, it is pertaining to a mesh object.

The temperature T is the name of a component (MOT type) and 1000.0 is its constant value. The field subtype (MOT type) is TEMPERATURES.

The generated field by element is always expressed at the nodes.



11.4 Test joi45 (Description sheet)

Test name: joi45

Calculation type: ORTHOTROPIC ELASTIC MECHANICS

Finite element type: QUA4, JOI4

Topic: Calculation of the thermal stresses on a 3D joint. The structure is a parallelepipedic joint embedded from the lower surface. The upper surface is free. The joint is subjected to a shear load applied on the whole circumference of the upper surface in the direction O-P1.

Goal: Calculate the displacements in the direction of the load, the deformations, the stresses and the nodal force field resulting from a stress field integration.


Version: 97' customer version

Model description:

$$\begin{aligned} \text{KS1} &= 4.2 \cdot 10^{10} \\ \text{KS2} &= 4.2 \cdot 10^8 \\ \text{KN} &= 4.2 \cdot 10^{20} \end{aligned}$$

Angle (OX,OP1) = 55 degrees

O-P1 axis : first orthotropic direction



Test joi45 (Results)

RESULTS Analytical solution:

F

$\delta = \frac{F}{K \cdot A}$

K * A

where

F is the total load on the upper surface in the direction OP1

K is the stiffness in the first orthotropic direction

A is the surface of the joint

Projection of the solution in the axis X and Y:

$\delta_x = \delta \cdot \cos(55)$

$\delta_y = \delta \cdot \sin(55)$

Numerical application:

F = 100000

K = 4.2 E20

A = 4.0

$\delta_x = 3.416 \text{ E-7}$

$\delta_y = 4.875 \text{ E-7}$



Test joi45 (Listing CASTEM)

```

*****
TITR ' TEST OF SHEAR ON A ORTHOTROPIC 3D-JOINT' ;
*****
OPTI DIME 3 ELEM QUA4 MODE TRID ;
OPTI ECHO 0 ;
*
*----- GEOMETRY -----*
OO = 00.00 0.00 0.00 ; ON = 00.00 0.00 1.00 ;
AN = 10.00 ;
*
* DEF. OF P1 POINT FOR THE ORTHOTROPIC DIRECTIONS
P1 = 0.707106781 0.707106781 0.00 ;
P1 = P1 TOUR AN OO ON ;
*
* DEFINITION OF THE LOWER SURFACE OF THE JOINT
A1 = 1.414213562 0.000000000 0.00 ;
B1 = 0.000000000 1.414213562 0.00 ;
C1 = -1.414213562 0.000000000 0.00 ;
D1 = 0.000000000 -1.414213562 0.00 ;
*
A1 = A1 TOUR AN OO ON ; B1 = B1 TOUR AN OO ON ;
C1 = C1 TOUR AN OO ON ; D1 = D1 TOUR AN OO ON ;
*
* MESH
H1 = A1 DROI 1 B1 ; H2 = B1 DROI 1 C1 ;
H3 = C1 DROI 1 D1 ; H4 = D1 DROI 1 A1 ;
L1 = H1 ET H2 ET H3 ET H4 ;
S1 = SURF L1 PLAN ;
*
* DEFINITION OF THE UPPER SURFACE OF THE JOINT
IA1 = 1.414213562 0.000000000 0.00 ;
IB1 = 0.000000000 1.414213562 0.00 ;
IC1 = -1.414213562 0.000000000 0.00 ;
ID1 = 0.000000000 -1.414213562 0.00 ;
*
IA1 = IA1 TOUR AN OO ON ; IB1 = IB1 TOUR AN OO ON ;
IC1 = IC1 TOUR AN OO ON ; ID1 = ID1 TOUR AN OO ON ;
*
* MESH
IH1 = IA1 DROI 1 IB1 ; IH2 = IB1 DROI 1 IC1 ;
IH3 = IC1 DROI 1 ID1 ; IH4 = ID1 DROI 1 IA1 ;
IL1 = IH1 ET IH2 ET IH3 ET IH4 ;
IS1 = SURF IL1 PLAN ;
*
* CREATION OF THE JOI4 JOINT
OPTION ELEM LIA4 ;
VOL = LIAISON 0.00001 S1 IS1 ;
*
*----- BOUNDARY CONDITIONS -----*
CL11 = BLOQ A1 UX ; CL12 = BLOQ A1 UY ;
CL13 = BLOQ A1 UZ ;
CL1 = CL11 ET CL12 ET CL13 ;
*
CL21 = BLOQ B1 UX ; CL22 = BLOQ B1 UY ;
CL23 = BLOQ B1 UZ ;
CL2 = CL21 ET CL22 ET CL23 ;
*
CL31 = BLOQ C1 UX ; CL32 = BLOQ C1 UY ;
CL33 = BLOQ C1 UZ ;
CL3 = CL31 ET CL32 ET CL33 ;
*
CL41 = BLOQ D1 UX ; CL42 = BLOQ D1 UY ;
CL43 = BLOQ D1 UZ ;
CL4 = CL41 ET CL42 ET CL43 ;
*
CL = CL1 ET CL2 ET CL3 ET CL4 ;
*
*----- DESCRIPTION OF THE MODEL MATERIAL -----*
MOD1 = MODE VOL MECANIQUE ELASTIQUE ORTHOTROPE JOI4 ;
MA1 = MATE MOD1 DIRECTION P1 PARALLELE
      KS1 4.2E10 KS2 4.2E08 KN 4.2E20 ;
*****
*** OTHERS MODEL OF MATERIALS ***
*** FLOT1 = 0.0 ; ***
*** P3 = 3.0 1.0 1.0 ; ***
*** ***
*** MA1 = MATE MOD1 DIRECTION P1 INCLINE FLOT1 P3 ***
*** KS1 4.2E10 KS2 4.2E08 KN 4.2E20 ; ***
*****
*** MA1 = MATE MOD1 RADIAL P1 ***
*** KS1 4.2E10 KS2 4.2E08 KN 4.2E20 ; ***
*****
*----- LOAD DEFINITION : TENSILE LOAD -----*
FO1 = FORCE ( 57358.00 81915.00 0.00 ) IL1 ;
*
*----- CALCULATION OF THE RIGIDITY MATRIX -----*
RI1 = RIGI MOD1 MA1 ;
RI2 = RI1 ET CL ;
*
*----- CALCULATION AND OUTPUT -----*
RE = RESO RI2 FO1 ;
*****
*****
*****
SOLAX=3.416E-7 ;
SOLAY=4.875E-7 ;
MESS ' Analytic solution : ' ;
MESS ' deltax = ' SOLAX ;
MESS ' deltay = ' SOLAY ;
MESS ' ' ;
MESS ' Calculated solution : ' ;
*
LIST RE ;
*
*----- GOOD WORKING MESSAGE -----*
DEPIAX = EXTR RE UX IAL ;
RESI1 = ABS( (DEPIAX-SOLAX)/SOLAX ) ;
*
DEPIAY = EXTR RE UY IAL ;
RESI2 = ABS( (DEPIAY-SOLAY)/SOLAY ) ;
*
SI ( (RESI1 <EG 1E-3 ) ET (RESI2 <EG 1E-3 ) ) ;
ERRE 0 ;
SINO ;
ERRE 5 ;
FINSI ;
*
*----- STRAIN CALCULATION -----*
EPS1 = EPSI MOD1 MA1 RE ;
LIST EPS1 ;
*
*----- STRESS CALCULATION -----*
SIG1 = SIGMA MOD1 MA1 RE ;
LIST SIG1 ;
*
*----- NODAL FORCE CALCULATION -----*
BSI1 = BSIGMA MOD1 SIG1 ;
LIST BSI1 ;
*
FIN ;

```

Test joi45 (Comments)

- BSI1 = BSI1: creation of a MCHAML object type

```
BSI1 = BSI1 MOD1 SIG1 ;
```

The BSI1 operator calculates the nodal force field (BSI1) resulting from a stress field (SIG1) integration.



Chapter 12

DYNAMICS

12.1 Test dyna6 (Description sheet)

Test name: dyna6

Calculation type: MECHANICS DYNAMICS 3D

Finite element type: POUT

Topic: Bending mode shape of a clamped-free beam. The beam is clamped at one end and free on the other end. The problem consists in finding frequencies of the first and second mode shapes of the beam. The three options of the VIBR operator are tested in this example :


- within intervals (keyword INTERVALLE) : separation of mode shapes by dichotomy on a given interval
- by proximity (keyword PROCHE) : computation of mode shapes which are close to the given values
- by Lanczos method (keyword SIMULANE) : projection on a subspace

Goal: Find the frequencies of the first and second mode shapes of the beam.

Reference: Analytical solution : Timoshenko.

Version: 97' customer version

Model description:






Test dyna6 (Results)

Results

	Analytic	INTERVALLE	PROCHE	SIMULTANE
Freq (Hz) Shape1	17.71	17.704	17.704	17.704
Freq (Hz) Shape2	110.98	110.71	110.71	110.71
$U_z(Q2)(m)$ Shape2	0.0	$1.2480 \cdot 10^{-3}$	$1.2483 \cdot 10^{-3}$	$1.2483 \cdot 10^{-3}$

CASTEM figures



Test dyna6 (Listing CASTEM)

```

*****
TITRE 'BENDING OF A CLAMPED-FREE BEAM';
OPTIO DIME 3 ELEM SEG2 MODE TRID IMPI 0;
OPTI ECHO 0;
TEMPS ;
DENSITE 0.1;
*
*----- MESHING -----
*---- Q2 IS A NODE FOR THE SECOND SHAPE --
*
Q1 = 0. 0. 0.; Q2 = (2.*0.783) 0. 0. ;
Q3 = 2. 0. 0. ;
LIG= Q1 D Q2 D Q3 ;
OEIL1 = 5 5 5 ;
TRAC OEIL1 LIG ;
*
STAN= MODEL LIG MECANIQUE POUT ;
*
*----- MATERIAL AND CHARACTERISTICS -----
*
MATPL1 = MATER STAN YOUNG 2.E11 NU 0.3
        RHO 7800;
CARPL1 = CARA STAN SECT 0.01 INRY 6.25E-6
        INRZ 1.E-3 TORS 1.E-2;
MATPL1 = MATPL1 ET CARPL1;
*
*----- RIGIDITY AND MASS -----
*
RIGPL1 =RIGI STAN MATPL1;
MASPLA1=MASS STAN MATPL1 ;
*
*----- CLAMPING ON Q1 -----
*
BLOQ1 = BLOQ DEPL ROTA Q1 ;
RIGPLA1 = RIGPL1 ET BLOQ1 ;
*
*-----
*          FIRST OPTION (INTERVALLE)
*          SEPARATION OF SHAPES BY DICHOTOMY
*          THEN USING OF INVERSE ITERATIONS
*-----
MODPLA1 = VIBR INTE 10. 120. RIGPLA1
        MASPLA1 ;
*
*-----
*          SECOND OPTION (PROCHE)
*          USING OF INVERSE ITERATIONS FROM
*          FREQUENCIES GIVEN IN THE LISTREEL
*-----
PR = PROG 17. 110.;
MODPLA2 = VIBR PROC PR RIGPLA1 MASPLA1 ;
*
*-----
*          THIRD OPTION (SIMULTANE)
*          USING OF LANCZOS METHOD
*          (PROJECTION ON A SUBSPACE)
*-----
MODPLA3 = VIBR SIMU 1. 2 RIGPLA1 MASPLA1 ;
*
*----- RESULTS -----
*
OPTI ECHO 0;
SAUTER PAGE;
F1=TIRE MODPLA1 FREQ RANG 1;

G1=TIRE MODPLA1 FREQ RANG 2;

F2=TIRE MODPLA2 FREQ RANG 1;
G2=TIRE MODPLA2 FREQ RANG 2;
F3=TIRE MODPLA3 FREQ RANG 1;
G3=TIRE MODPLA3 FREQ RANG 2;
*
MESS '*';
MESS '***** CHECKING OF FREQUENCIES';
MESS '*';
MESS '          ANALYTIC          * INTERVALLE
          * PROCHE          * SIMULTANE';
MESS '          -----*-----';
MESS ' SHAPE 1      17.71          * ' F1 '*' F2
          ' * ' F3 ;
MESS ' SHAPE 2     110.98          * ' G1 '*' G2
          ' * ' G3 ;
MESS ' ';
MESS ' ';
MESS '*';
MESS '**** CHECKING OF THE SECOND SHAPE ';
MESS '*';
MESS ' ';
X1=EXTR (TIRE MODPLA1 DEPL RANG 2) UZ Q2;
X2=EXTR (TIRE MODPLA2 DEPL RANG 2) UZ Q2;
X3=EXTR (TIRE MODPLA3 DEPL RANG 2) UZ Q2;
MESS '          ANALYTIC          * INTERVALLE
          * PROCHE          * SIMULTANE';
MESS '          -----*-----';
MESS ' SHAPE 2      0.00          * ' X1 '*' X2
          ' * ' X3 ;
TEMPS ;
*          GOOD WORKING MESSAGE
FREF=17.71; GREF=110.98;
RESI1=ABS((F1-FREF)/FREF);
RESI2=ABS((G1-GREF)/GREF);
SI((RESI1 <EG 1E-2) ET (RESI2 <EG 1E-2));
ERRE 0;
SINO;
ERRE 5;
FINSI;
FIN ;
    
```



Test dyna6 (Comments)

1. Computation of rigidity and mass matrices

RIGPL1 = RIGI STAN MATPL1;
MASPLA1= MASS STAN MATPL1 ;

The operators RIGI and MASS require the model (here STAN) and the material and geometrical characteristics (here MATPL1).

The type of these two objects is RIGIDITE. The subtype of RIGPL1 is RIGIDITE and the subtype of MASPLA1 is MASSE.

2. Computation of eigen shapes : option INTERVALLE

MODPLA1 = VIBR INTE 10. 120. RIGPLA1 MASPLA1 ;

The operator VIBR requires the rigidity and mass matrices of the system. VIBR with keyword INTE computes the natural eigen shapes which frequencies are contained between 10 Hz and 120 Hz.

3. Computation of eigen shapes : option PROCHE

PR = PROG 17. 110. ;
MODPLA2 = VIBR PROC PR RIGPLA1 MASPLA1 ;

VIBR with keyword PROC computes the natural eigen shapes which frequencies are the nearest to the given list (here PR).

4. Computation of eigen shapes : option SIMULTANE

MODPLA3 = VIBR SIMU 1. 2 RIGPLA1 MASPLA1 ;

VIBR with keyword SIMU computes the two natural eigen shapes which frequencies are the nearest to the given real (here 1.).

12.2 Test dyna7 (Description sheet)

Test name: dyna7

Calculation type: MECHANICS DYNAMICS 2D FOURIER

Finite element type: LQUA4, LSU2

Topic: Vibrating mode shapes of a thin fluid web. The cavity is supposed undeformable. The gravity effect is taken into consideration. The shaking mode shapes are computed in an annular space with Fourier modes number 1 and 3. The three options of the VIBR operator are tested in this example.


Goal: Find the frequencies of the first and third Fourier shape.

Reference: Analytical solution.

Version: 97' customer version

Model description:

L	$= 2m$	ρ	$= 1000kg/m^3$	ρ_{ref}	$= 1kg/m^3$
$L2$	$= 1.2m$	S_s	$= 1200m/s$	S_{ref}	$= 1m/s$
Th	$= 0.01m$	G	$= 9.81m/s^2$	L_{car}	$= 1m$



**Test dyna7 (Results)****Results**

	Analytical	INTERVALLE	PROCHE	SIMULTANE
Freq (Hz) Shape1	0.489	0.4895	0.4895	0.4895
Freq (Hz) Shape3	0.863	0.8638	0.8638	0.8638
P(A1) Shape1	1.	1.	1.	1.
P(A2) Shape1	1.81	1.8108	1.8108	1.8108
P(A3) Shape1	3.76	3.763	3.763	3.763
P(A1) Shape3	1.	1.	1.	1.
P(A2) Shape3	18.31	18.375	18.375	18.375
P(A3) Shape3	201.72	202.86	202.86	202.86

CASTEM figures



SHAKING SHAPES OF A THIN FLUID WEB



12.2. TEST DYNA7 (DESCRIPTION SHEET)

Test dyna7 (Listing CASTEM)

```

*****
TITRE 'SHAKING SHAPES OF A THIN FLUID WEB' ;
OPTI DIME 2 ELEM QUA4;
OPTI ECHO 0;
TEMPS ;

*----- MESHING -----
*----- ONLY THE FLUID IS MESHEDED -----
*
A1=0.995 0. ; A2=1.005 0. ;
A3=1.005 2. ; A4=0.995 2. ;
A5=1.005 1.2; A6=0.995 1.2;
L1=D 1 A1 A2;
L2=A2 D 24 A5 D 16 A3;
L3=D 1 A3 A4;
L4=A4 D 16 A6 D 24 A1;
FLUID=DALL L1 L2 L3 L4;
*
TRAC FLUID ;
*
OPTI ELEM SEG2;
SURFA=D 1 A3 A4;
*
TRAC SURFA ;
*
MAIL=FLUID ET SURFA;
*
TRAC MAIL;
*
*----- DEFINITION OF SOME CONSTANTS -----
*
ROREF=1. ; CEREF=1. ; LCARA=1. ;
RO0=1000. ; CS=1200. ; GRA=9.81 ;
*
*----- MODEL ASSIGNED TO FLUID -----
*-----AND FREE SURFACE -----
*----- FOURIER HARMONIC 1 -----
*
OPTI MODE FOUR 1;
MODLIQ1 = MODE FLUID LIQUIDE LQU4;
MODLIQ2 = MODE SURFA LIQUIDE LSU2;
MODLIQ = MODLIQ1 ET MODLIQ2;
*
*----- LIQUID MATERIAL -----
*
MAT=MATE MODLIQ RHO RO0 RORF ROREF CSON CS
      CREF CEREF LCAR LCARA G GRA;
*
*----- RIGIDITY AND MASS -----
*
RAI=RIGI MODLIQ MAT;
MAS=MASS MODLIQ MAT;
*
* FIRST SHAPE IS COMPUTED WITH 'INTERVALLE'
* YIELD OF DISPLACEMENTS (AND OF PRESSURE)
* IS NORMALIZED IN ORDER TO HAVE A UNITARY
* PRESSURE IN ALTITUDE 0

* PRESSURE IS EXTRACTED IN POINTS A2, A5
* AND A3 .
*
MOD1 = VIBR INTE 0. 1. RAI MAS;
F1 = TIRE MOD1 FREQ RANG 1;
PRE = TIRE MOD1 DEPL RANG 1;

PBAS = EXTR PRE P A2;

```

```

PP = PRE/PBAS;
X1 = EXTR PP P A2;
Y1 = EXTR PP P A5;
Z1 = EXTR PP P A3;
*
*
* FIRST SHAPE IS COMPUTED WITH 'PROCHE' OPTION
* OF 'VIBR'
*
MOD2 = VIBR PROC (PROG 0.5) RAI MAS;
F2 =TIRE MOD2 FREQ RANG 1;
PRE=TIRE MOD2 DEPL RANG 1;
PBAS=EXTR PRE P A2;
PP=PRE/PBAS;
X2=EXTR PP P A2;Y2=EXTR PP P A5;Z2=EXTR PP P A3;
*
*
* FIRST SHAPE IS COMPUTED WITH 'SIMULTANE'
* OPTION OF 'VIBR'
*
MOD3 = VIBR SIMU 0.5 1 RAI MAS;
F3 =TIRE MOD3 FREQ RANG 1;
PRE=TIRE MOD2 DEPL RANG 1;
PBAS=EXTR PRE P A2;
PP=PRE/PBAS;
X3=EXTR PP P A2;Y3=EXTR PP P A5;Z3=EXTR PP P A3;
*
*
* SAME OPERATIONS WITH FOURIER HARMONIC 3
*
OPTI MODE FOUR 3;
MODLIQ1=MODE FLUID LIQUIDE LQU4;
MODLIQ2=MODE SURFA LIQUIDE LSU2;
*
MODLIQ=MODLIQ1 ET MODLIQ2;
*
MAT=MATE MODLIQ RHO RO0 RORF ROREF CSON CS
      CREF CEREF LCAR LCARA G GRA;
*
RAI=RIGI MODLIQ MAT;
MAS=MASS MODLIQ MAT;
*
MOD1=VIBR INTE 0. 1. RAI MAS;
*
G1=TIRE MOD1 FREQ RANG 1;
PRE=TIRE MOD1 DEPL RANG 1;
PBAS=EXTR PRE P A2;
PP=PRE/PBAS;
T1=EXTR PP P A2;U1=EXTR PP P A5;
V1=EXTR PP P A3;
*
MOD2=VIBR PROC (PROG 0.9) RAI MAS;
*
G2=TIRE MOD2 FREQ RANG 1;
PRE=TIRE MOD2 DEPL RANG 1;
PBAS=EXTR PRE P A2;
PP=PRE/PBAS;
T2=EXTR PP P A2;U2=EXTR PP P A5;
V2=EXTR PP P A3;
*
MOD3=VIBR SIMU 0.9 1 RAI MAS;
G3=TIRE MOD3 FREQ RANG 1;
PRE=TIRE MOD3 DEPL RANG 1;

PBAS=EXTR PRE P A2;
PP=PRE/PBAS;

```

```

T3=EXTR PP P A2;U3=EXTR PP P A5;
V3=EXTR PP P A3;
*-----
SAUT PAGE; SAUT 2 LIGNE ;
MESS '*';
MESS '***** CHECKING OF FREQUENCIES';
MESS '*';
MESS ' ';
MESS '          ANALYTIC * INTERVALLE *
          PROCHE * SIMULTANE';
MESS ' -----*-----*
          -----*-----*';
MESS ' HARM 1    0.489    * ' F1 '**
          F2 '* ' F3 ;
MESS ' HARM 3    0.863    * ' G1 '**
          G2 '* ' G3 ;

MESS ' ';
MESS '*';
MESS '***** CHECKING OF PRESSURE YIELD
          AT POINTS A1, A5 AND A3';
MESS '*';
MESS ' ';
MESS '          ANALYTIC * INTERVALLE *
          PROCHE * SIMULTANE';
MESS ' -----*-----*
          -----*-----*';
MESS 'HARM 1    1.00    * ' X1 '**
          X2 '* ' X3 ;
MESS 'HARM 1    1.81    * ' Y1 '**
          Y2 '* ' Y3 ;
MESS 'HARM 1    3.76    * ' Z1 '**
          Z2 '* ' Z3 ;
MESS 'HARM 3    1.00    * ' T1 '**
          T2 '* ' T3 ;
MESS 'HARM 3    18.31   * ' U1 '**
          U2 '* ' U3 ;
MESS 'HARM 3    201.72  * ' V1 '**
          V2 '* ' V3 ;

TEMPS ;

*----- GOOD WORKING MESSAGE -----*
SAUT 1 PAGE;
ERR=TABLE;
ERR.1=100*(ABS(0.489 -F1)/0.489);
ERR.2=100*(ABS(0.489-F2)/0.489);
ERR.3=100*(ABS(0.489 -F3)/0.489);
ERR.4=100*(ABS(0.863 -G1)/0.863);
ERR.5=100*(ABS(0.863 -G2)/0.863);
ERR.6=100*(ABS(0.863 -G3)/0.863);
ERR.7=100*(ABS(1-X1)/1);
ERR.8=100*(ABS(1-X2)/1);
ERR.9=100*(ABS(1-X3)/1);
ERR.10=100*(ABS(1.81 -Y1)/1.81);
ERR.11=100*(ABS(1.81 -Y2)/1.81);
ERR.12=100*(ABS(1.81 -Y3)/1.81);
ERR.13=100*(ABS(3.76 -Z1)/3.76);
ERR.14=100*(ABS(3.76 -Z2)/3.76);
ERR.15=100*(ABS(3.76 -Z3)/3.76);
ERR.16=100*(ABS(1-T1));
ERR.17=100*(ABS(1-T2));
ERR.18=100*(ABS(1-T3));
ERR.19=100*(ABS(18.31 -U1)/18.31);
ERR.20=100*(ABS(18.31 -U2)/18.31);
ERR.21=100*(ABS(18.31 -U3)/18.31);
ERR.22=100*(ABS(201.72-V1)/201.72);

ERR.23=100*(ABS(201.72-V2)/201.72);
ERR.24=100*(ABS(201.72-V3)/201.72);

I=0;
REPETER BOUC1 24;
I=I+1;

```



Test dyna7 (Comments)

1. Liquid formulation

MODLIQ1 = MODE FLUID LIQUIDE LQU4;
MODLIQ2 = MODE SURFA LIQUIDE LSU2;

The operator MODEL with keyword LIQUIDE defines a liquid formulation. The type of finite elements depends of the type of formulation. LQU4 is a four node massive quadrangle and LSU2 is a two node element of free surface.

2. Liquid material

MAT=MATE MODLIQ RHO RO0 RORF ROREF CSON CS CREF CEREF LCAR LCARA G GRA;

The material characteristics are defined with MATER operator and use the following parameters :

- RHO : volumic mass
- RORF : referential volumic mass
- CSON : sonic speed
- CREF : referential speed
- LCAR : characteristic length
- G : gravitational acceleration

3. Results extraction

PRE = TIRE MOD1 DEPL RANG 1;
PBAS = EXTR PRE P A2;

The object PRE is a ponctual field and one of its components is the pressure P. The data PBAS is the pressure at point A2.

12.3 Test fsi4 (Description sheet)

Test name: fsi4

Calculation type: MECHANICS DYNAMICS 2D FOURIER, FLUID-STRUCTURE INTERACTION

Finite element type: COQ2, LQUA4, RACO

Topic: Frequencies of concentric shells coupled by fluid. Both the concentric shells are separated by a volume of water. Springs are on the bottom of the shells. The inner radius is 0.5m and the outer radius is 1m.


Goal: Find the frequencies of the first and second mode shapes of the shells ($M=1$).

Reference CASTEM: Report CEA 91/479 : Validation of CASTEM 2000 for fluid structure interaction problems.

Version: 97' customer version

Model description:

R_i	= 500mm	E_{shell}	= $2.10^5 MPa$	G	= $0mm/s^2$
R_o	= 1000mm	ν_{shell}	= 0.3	ρ_{ref}	= $1000kg/m^3$
H	= 1000mm	ρ_{water}	= $1000kg/m^3$	S_{ref}	= $1435.10^3 mm/s$
Th_{shell}	= 5mm	S_s	= $1435.10^3 mm/s$	L_{car}	= 1000mm
ρ_{shell}	= $8000kg/m^3$				






Test fsi4 (Results)

Results

	Theoretical freq (Hz)	CASTEM freq (Hz)	Margin
Shape1	18.88	18.647	1.24 %
Shape2	55.278	56.083	1.46 %

CASTEM figures



#1
FSI4 : MESHING

Test fsi4 (Listing CASTEM)

```

*****
TEMPS;
*
OPTI DIME 2;
OPTI MODE FOUR 1 ;
OPTI ELEM QUA4;
*
*=====
*           geometry           *
*=====
*
* unit of length : millimeter
*
* Points
*
N1 = 1           ; N2 = 1           ;
P5 = 500.0 1000.0 ; P6 = 500.0 0.0 ;
P7 = 1000. 1000.0 ; P8 = 1000. 0.0 ;
*
S5 =P5 D N1 P6 ; S6 = P8 D N1 P7 ;
*
STEEL = S5 ET S6           ;
WATER = (S5 PLUS (2.5 0.)) TRAN N2 (495.0 0. ) ;
RAC1 = RACC 3.0 WATER STEEL ;
*
TITR ' FSI4 : MESHING';
TRAC QUAL (WATER ET (0 0));
*
*=====
*           model - material - rigidity - mass           *
*=====
*
MODCOQ=MODEL STEEL MECANIQUE COQ2 ;
MODLIQ=MODEL WATER LIQUIDE LQU4;
MODRAC=MODEL RAC1 MECANIQUE LIQUIDE RACO;
*
MATCOQ=MATER MODCOQ RHO 8.E-9 YOUN 2.E5 NU 0.3 ;
MATLIQ = MATER (MODLIQ ET MODRAC) RHO 1.E-9
          RORF 1.E-9 CSON 1435.0E3 CREF 1435.0E3
          LCAR 1000. G 0.;
CARCOQ=CARAC MODCOQ EPAI 5.0 ;
CARRAC=CARAC MODRAC LIQU WATER;
MATCOQ=MATCOQ ET CARCOQ;
MATLIQ=MATLIQ ET CARRAC;
*
K =3.7458E4/2.0;
PP =PROG K (-1*K) K ;
PPNEW = PROG K ( -1*K) ( -1*K) K;
RIG1 = RIGI (MODCOQ ET MODLIQ ET MODRAC)
        (MATCOQ ET MATLIQ) ;
MAS4 = MASS (MODRAC ET MODLIQ) (MATLIQ) ;
*
*=====
*           boundary conditions           *
*=====
*
RES1 = BLOQ UZ RT (S5 ET S6)           ;
RIG2 = (APPU UR UT K P8 ) ET (APPU UR UT K P6 ) ;
RIG3 = ( MANU 'RIGI' type 'RIGIDITE' (P6 D 1 P8)
        (MOTS UR) PPNEW )
        ET ( MANU 'RIGI' type 'RIGIDITE' (P6 D 1 P8)
        (MOTS UT) PPNEW );
RIG4 = RIG1 ET RIG2 ET RIG3 ;
*
*=====
* theoretical value of frequency
*
FREL = TABLE ;
FREL.1 = 18.88 ;

FREL.2 = 55.278;
LIST1 = PROG FREL.1 FREL.2 ;
*
*=====
*           computation of the frequencies           *
*           and           *
*           extraction of some results           *
*=====
* Use of the operator VIBR. (option PROC)
*
RESUL = VIBR PROC LIST1 (RES1 ET RIG4 ) MAS4 ;
*
*=====
*           results           *
*=====
*
MESS ' RESULTS ';
MESS ' ----- ';
SAUT 1 LIGN;
*
FREQ2 = TABLE; MOD = TABLE; DEF = TABLE; ERG = TABLE;
*
I = 0;
REPETER BLOC1 2;
  I = I + 1;
  FREQ2.I = TIRE RESUL FREQ RANG I;
  ERG.I = 100 * (ABS ((FREL.I - FREQ2.I) / FREL.I));
  MESS ' MODE SHAPE NUMBER' I ;
  MESS ' -----';
  MESS 'Theoretical frequency :' FREL.I 'Hz';
  MESS 'Computed frequency :' FREQ2.I 'Hz';
  MESS ' Margin : ' ERG.I '%';
  SAUT 1 LIGN;
FIN BLOC1;
*
*=====
*           code validation           *
*=====
*
ERGMAX = MAXI (PROG ERG.1 ERG.2 );
*
SI (ERGMAX <EG 5.);
  ERRE 0;
SINON; ERRE 5;
FINSI;
*
SAUT 1 LIGN;
TEMPS;
SAUT 1 LIGN;
*
FIN;
    
```



Test fsi4 (Comments)

1. Union of fluid and structure

WATER = (S5 PLUS (2.5 0.)) TRAN N2 (495.0 0.) ;
 RAC1 = RACC 3.0 WATER STEEL ;

If the distance between points of STEEL and WATER is smaller than 3.0 mm, the operator RACCORD creates a line of double points by joining the STEEL structure with the WATER fluid. This corresponds to the creation of union elements between STEEL and WATER.

Note that there is a clearance between WATER boundaries and STEEL median line, corresponding here with the thickness of the shell.

2. Declaration of models

MODCOQ=MODEL STEEL MECANIQUE COQ2 ;
 MODLIQ=MODEL WATER LIQUIDE LQU4;
 MODRAC=MODEL RAC1 MECANIQUE LIQUIDE RACO;

Fluid-structure interaction requires three different models : the liquid, the solid and the model of union elements.

3. Declaration of material and geometrical characteristics

MATLIQ = MATER (MODLIQ ET MODRAC) RHO 1.E-9
 RORF 1.E-9 CSON 1435.0E3 CREF 1435.0E3
 LCAR 1000. G 0. ;
 ...
 CARRAC=CARAC MODRAC LIQU WATER;

The definition of material properties is the same for the liquid model and the union model.

The liquid model has no geometrical characteristics but the position of the liquid has to be known : hence the operator CARAC is used with the keyword LIQU to indicate that the liquid concerning the union elements of MODRAC is represented by the object WATER.

4. Modelling of springs

RIG2 = (APPU UR UT K P8) ET (APPU UR UT K P6) ;
 RIG3 = (MANU 'RIGI' type 'RIGIDITE' (P6 D 1 P8) (MOTS UR) PPNEW)
 ET (MANU 'RIGI' type 'RIGIDITE' (P6 D 1 P8) (MOTS UT) PPNEW);

Springs can be modelled by two methods :

- with operator APPUI if the spring is connected to a fixed point,
- with operator MANUEL RIGI if the spring joins two points of the structure. In this case, a stiffness is added to each point.

12.4 Test dyna5 (Description sheet)

Test name: dyna5

Calculation type: MECHANICS DYNAMICS 3D, SUB-STRUCTURATION

Finite element type: POUT

Topic: Methods of sub-structuration. The problem consists in using sub-structuration methods to find the natural frequency of a clamped-clamped beam. The beam is decomposed in two pieces of same length. The three methods correspond to the three different ways to describe the constraints between the two common tips


- Q2 free - Q3 free
- Q2 and Q3: - Q2 clamped - Q3 clamped
- Q2 clamped - Q3 free

Goal: Find the first natural frequency of the beam.

Reference: Analytical solution.

Version: 97' customer version

Model description:



$$l = 2m$$

$$A = 0.1m^2$$

$$I_y = 2.08 \cdot 10^{-7} m^4$$

$$I_z = 3.33 \cdot 10^{-8} m^4$$

$$I_O = 2.41 \cdot 10^{-7} m^4$$

$$E = 2 \cdot 10^{11} Pa$$

$$\nu = 0.3$$

$$\rho = 7800 kg/m^3$$




Test dyna5 (Results)

Results

	Analytic	1st method	2nd method	3rd method
Freq (Hz)	0.65	0.65072	0.65083	0.64926

CASTEM figures



Test dyna5 (Listing CASTEM)

```

*****
TITRE 'DYNA5 - METHODS OF SUB-STRUCTURATION' ;
OPTIO DIME 3 ELEM SEG2 MODE TRID IMPI 0;
SOL=TABLE;
TEMPS ;
*
*-----
*   FIRST METHOD FREE-FREE
*-----
*
*----- Definition of the 1st beam geometry -----
*
Q1 = 0. 0. 0.; Q2 =2. 0. 0.;
LIG1= D 12 Q1 Q2;
OEIL1 = 5 5 5 ;
TRACE OEIL1 LIG1 ;
*
*- model, material and geometrical characteristics -
*----- of the 1st beam -----
*
M01= MODEL LIG1 MECANIQUE ELASTIQUE POUT ;

MATPL1=MATER M01  YOUNG 2.E11 NU 0.3 RHO 7800;
CARPL1=CARAC M01 SECT 0.1 INRY 2.08E-7 INRZ 3.33E-8
      TORS 2.41E-7;
MATPL1=MATPL1 ET CARPL1;
*
*----- rigidity, mass, boundary conditions -----
*----- of the 1st beam -----
*
RIGPL1 =RIGI MATPL1 M01 ;
MASPLA1=MASS MATPL1 M01 ;
BLOQ1=BLOQ DEPL ROTA Q1 ;
RIGPLA1=RIGPL1 ET BLOQ1 ;
*
*-- Computation of the eigen mode of the 1st beam --
*
PR=PROG 0.4 ;
MODPLA1=VIBR PROC PR RIGPLA1 MASPLA1;
*
*----- Definition of the 2nd beam geometry -----
*
Q3 =2. 0. 0.; Q4 =4. 0. 0.;
LIG2= D 12 Q3 Q4;
TRAC OEIL1 LIG2 ;
*
POUTRE=LIG1 ET LIG2 ;
TRAC OEIL1 POUTRE;
*
*-- model, material and geometrical characteristics -
*----- of the 2nd beam -----
*
M02= MODEL LIG2 MECANIQUE ELASTIQUE POUT ;
MATPL2=MATER M02  YOUNG 2.E11 NU 0.3 RHO 7800;
CARPL2=CARAC M02 SECT 0.1 INRY 2.08E-7 INRZ 3.33E-8
      TORS 2.41E-7;
MATPL2=MATPL2 ET CARPL2;
*
*----- rigidity, mass, boundary conditions -----
*----- of the 2nd beam -----
*
RIGPL2 =RIGI MATPL2 M02 ;
MASPLA2=MASS MATPL2 M02 ;
BLOQ4=BLOQ DEPL ROTA Q4;
RIGPLA2=RIGPL2 ET BLOQ4 ;
*
*-- Computation of the eigen mode of the 2nd beam ---
*
MODPLA2=VIBR PROC PR RIGPLA2 MASPLA2;

*
*----- Creation of the sub-structures -----
*----- related to the two beams -----
*
SPLA1=STRU RIGPLA1 MASPLA1 ;
SPLA2=STRU RIGPLA2 MASPLA2 ;
ELM2=ELST Q2 SPLA1;
ELM3=ELST Q3 SPLA2;
PR2=PROG 1. ;PR3=PROG -1. ;
*
*----- Description of the constraints -----
*----- between the sub-structures -----
*
LIUX=RELA ELM2 UX PR2 ELM3 UX PR3;
LIUY=RELA ELM2 UY PR2 ELM3 UY PR3;
LIUZ=RELA ELM2 UZ PR2 ELM3 UZ PR3;
LIRX=RELA ELM2 RX PR2 ELM3 RX PR3;
LIRY=RELA ELM2 RY PR2 ELM3 RY PR3;
LIRZ=RELA ELM2 RZ PR2 ELM3 RZ PR3;
ENCL=LIUY ET LIRX ET LIRY ET LIRZ ET LIUX ET LIUZ;
*
*- Creation of modal basis related to the two beams -
*
BASE1=BASE SPLA1 MODPLA1 ENCL ;
BASE2=BASE SPLA2 MODPLA2 ENCL ;
*
*----- Complete modal basis -----
*----- related to the reunion of the -----
*----- two elementary basis -----
*
BASET= BASE1 ET BASE2 ;
RIGMO  = RIGI BASET ;
MASMO  = MASS BASET ;
*
*----- Searching of the 1st eigen frequency -----
*----- of the complete structure -----
*
SOL.1= VIBR PROC (PROG 0.65) MASMO RIGMO ;
FREQ1=TIRE SOL.1 FREQ RANG 1;
*
*-----
*   SECOND METHOD CLAMPED-CLAMPED
*-----
*
*----- Boundary conditions -----
*----- of the 1st beam -----
*
BLOQUX2=BLOQ UX Q2;
BLOQUY2=BLOQ UY Q2;
BLOQUZ2=BLOQ UZ Q2;
BLOQRX2=BLOQ RX Q2;
BLOQRY2=BLOQ RY Q2;
BLOQRZ2=BLOQ RZ Q2;
*
*----- Rigidity of 1st beam -----
*
RIGPLA1=RIGPL1 ET BLOQ1
ET BLOQUX2 ET BLOQUY2 ET BLOQUZ2 ET BLOQRX2
ET BLOQRY2 ET BLOQRZ2;
*
*---- Computation of the eigen mode of the 1st beam --
*
PR1=PROG 2.6 ;
MODPLA1=VIBR PROC PR1 RIGPLA1 MASPLA1;
*
*----- Boundary conditions -----
    
```



12.4. TEST DYNA5 (DESCRIPTION SHEET)

```

*----- of the 2nd beam -----
*
*-- Q2 point of 1st beam and Q3 point of 2nd beam ---
*-- have the same co-ordinates -----
*
BLOQUX3=BLOQ UX Q3;
BLOQUY3=BLOQ UY Q3;
BLOQUZ3=BLOQ UZ Q3;
BLOQRX3=BLOQ RX Q3;
BLOQRY3=BLOQ RY Q3;
BLOQRZ3=BLOQ RZ Q3;
*
*----- Rigidity of 2nd beam -----
*
RIGPLA2=RIGPL2 ET BLOQ4
ET BLOQUX3 ET BLOQUY3 ET BLOQUZ3 ET BLOQRX3
ET BLOQRY3 ET BLOQRZ3;
*
*--- Computation of the eigen mode of the 2nd beam ---
*
MODPLA2=VIBR PROC PR1 RIGPLA2 MASPLA2;
*
*----- Creation of the sub-structures -----
*----- related to the two beams -----
*
SPLA1=STRU RIGPLA1 MASPLA1 ;
SPLA2=STRU RIGPLA2 MASPLA2 ;
CLUX2=CLST SPLA1 BLOQUX2;
CLUY2=CLST SPLA1 BLOQUY2;
CLUZ2=CLST SPLA1 BLOQUZ2;
CLR2=CLST SPLA1 BLOQRX2;
CLRY2=CLST SPLA1 BLOQRY2;
CLRZ2=CLST SPLA1 BLOQRZ2;
CLUX3=CLST SPLA2 BLOQUX3;
CLUY3=CLST SPLA2 BLOQUY3;
CLUZ3=CLST SPLA2 BLOQUZ3;
CLR3=CLST SPLA2 BLOQRX3;
CLRY3=CLST SPLA2 BLOQRY3;
CLRZ3=CLST SPLA2 BLOQRZ3;
*
*----- Description of the constraints -----
*----- between the sub-structures -----
*
PR2=PROG 1.1;PR3=PROG 1.1;
LIUX=RELA CLUX2 LX PR2 CLUX3 LX PR3;
LIUY=RELA CLUY2 LX PR2 CLUY3 LX PR3;
LIUZ=RELA CLUZ2 LX PR2 CLUZ3 LX PR3;
LIRX=RELA CLR2 LX PR2 CLR3 LX PR3;
LIRY=RELA CLRY2 LX PR2 CLRY3 LX PR3;
LIRZ=RELA CLRZ2 LX PR2 CLRZ3 LX PR3;
ENCL=LIUY ET LIRX ET LIRY ET LIRZ ET LIUX ET LIUZ;
*
*-- Creation of modal basis related to the two beams --
*
BASE1=BASE SPLA1 MODPLA1 ENCL ;
BASE2=BASE SPLA2 MODPLA2 ENCL ;
*
*----- Complete modal basis -----
*----- related to the reunion of the -----
*----- two elementary basis -----
*
BASET= BASE1 ET BASE2 ;
RIGMO = RIGI BASET ;
MASMO = MASS BASET ;
*
*----- Searching of the 1st eigen frequency -----
*----- of the complete structure -----
*
SOL.2= VIBR PROC (PROG 0.65) MASMO RIGMO ;
FREQ2=TIRE SOL.2 FREQ RANG 1;
*
*----- Searching of the 1st eigen frequency -----
*----- of the complete structure -----
*
SOL.3= VIBR PROC (PROG 0.65) MASMO RIGMO ;
FREQ3=TIRE SOL.3 FREQ RANG 1;
*
*----- RESULTS -----
*
SAUT PAGE ; SAUT 4 LIGNE ;
MESS 'THEORICAL FREQUENCY 0.65 HZ';
SAUT 2 LIGNE ;

```

```
MESS 'WITH 1ST METHOD : ' FREQ1;
MESS 'WITH 2ND METHOD : ' FREQ2;
MESS 'WITH 3RD METHOD : ' FREQ3;
TEMPS ;
*           code validation
REF=0.65;
RESI1=ABS((FREQ1-REF)/REF);
RESI2=ABS((FREQ2-REF)/REF);
RESI3=ABS((FREQ3-REF)/REF);
SI((RESI1 <EG 5E-2) ET (RESI2 <EG 5E-2)
    ET (RESI3 <EG 5E-2));
    ERRE 0;
SINO;
    ERRE 5;
FINSI;
FIN ;
```



Test dyna5 (Comments)

1. Creation of sub-structures related to the two beams

```
SPLA1=STRU RIGPLA1 MASPLA1 ;
SPLA2=STRU RIGPLA2 MASPLA2 ;
ELM2=ELST Q2 SPLA1;
ELM3=ELST Q3 SPLA2;
...
CLUX2=CLST SPLA1 BLOQUX2;
CLUY2=CLST SPLA1 BLOQUY2;
CLUZ2=CLST SPLA1 BLOQUZ2;
CLR2=CLST SPLA1 BLOQRX2;
CLRY2=CLST SPLA1 BLOQRY2;
CLRZ2=CLST SPLA1 BLOQRZ2;
```

The operator STRU creates an object of STRUCTURE type by indicating the mass and the rigidity.

The operator ELST creates an object of ELEMSTRU type by indicating the structure and the point of the interface. This type of object is used to describe the constraints between sub-structures when the point of the interface is free.

The operator CLST creates an object of BLOQSTRU type by indicating the structure and a rigidity of fixing. This type of object is used to describe the constraints between sub-structures when the point of the interface is constrained.

2. Description of elementary constraints

- Interface with free points

```
PR2=PROG 1. ; PR3=PROG -1. ;
```

```
LIUX=RELA ELM2 UX PR2 ELM3 UX PR3;
LIUY=RELA ELM2 UY PR2 ELM3 UY PR3;
LIUZ=RELA ELM2 UZ PR2 ELM3 UZ PR3;
LIRX=RELA ELM2 RX PR2 ELM3 RX PR3;
LIRY=RELA ELM2 RY PR2 ELM3 RY PR3;
LIRZ=RELA ELM2 RZ PR2 ELM3 RZ PR3;
```

The operator RELA describes elementary constraints between the sub-structures, it creates an object of ATTACHE type.

The element of structure ELM2 (that is to say Q2 point) has the same displacements as the element ELM3 (Q3 point).

- Interface with fixed point

```
PR2=PROG 1. 1.; PR3=PROG 1. 1.;
```

```
LIUX=RELA CLUX2 LX PR2 CLUX3 LX PR3;
LIUY=RELA CLUY2 LX PR2 CLUY3 LX PR3;
LIUZ=RELA CLUZ2 LX PR2 CLUZ3 LX PR3;
LIRX=RELA CLR2 LX PR2 CLR3 LX PR3;
LIRY=RELA CLRY2 LX PR2 CLRY3 LX PR3;
LIRZ=RELA CLRZ2 LX PR2 CLRZ3 LX PR3;
```

The fixing of structure CLUX2 (related to Q2 X-displacement) has the same component as CLUX3. Note that 'LX' is the compulsory component for a fixing.

- Interface with fixed and free points

CL3=ELST SPLA2 Q3;

...

PR=PROG 1 1;

REL1=RELA CL3 UX 1 CLUX2 LX PR ;

REL2=RELA CL3 UY 1 CLUY2 LX PR ;

REL3=RELA CL3 UZ 1 CLUZ2 LX PR ;

REL4=RELA CL3 RX 1 CLRX2 LX PR ;

REL5=RELA CL3 RY 1 CLRY2 LX PR ;

REL6=RELA CL3 RZ 1 CLRZ2 LX PR ;

It is possible to describe a constraint between a fixed point and a free point, in this case the operator

RELA is used with an element of structure (CL3) and a fixing of structure (CLUX2, CLUY2, ...).

3. Creation of modal basis

BASE1=BASE SPLA1 MODPLA1 ENCL ;

BASE2=BASE SPLA2 MODPLA2 ENCL ;

The operator BASE creates an object of BASEMODA type. This object contains :

- the structure (SPLA1),
- the mode shapes related to this structure (MODPLA1),
- the constraints (ENCL).

4. Reunion of the elementary bases

BASET = BASE1 ET BASE2 ;

RIGMO = RIGI BASET ;

MASMO = MASS BASET ;

The two modal bases are joined together.

The operators RIGI and MASS calculate rigidity and mass matrices by using a modal basis (BASET).



12.5 Test vibr6 (Description sheet)

Test name: vibr6

Calculation type: MECHANICS DYNAMICS 3D, COMPOUND EIGEN SHAPES

Finite element type: TUYA


Topic: Compound eigen modes of a rod. The rod is subjected to an external axial permanent flow. Both the tips of the rod are articulated. Computations are made for different flow velocities (0 up to 120 m/s).

Goal: Find the real and imaginary parts of frequencies for the different velocities.

Reference: Model of Paidoussis & Ortoja-Starzewski.

Version: 97' customer version

Model description:



L	$= 2m$	ρ_{fl}	$= 1000kg/m^3$
R	$= 10^{-3}m$	V_{min}	$= 0m/s$
Th	$= 10^{-3}m$	V_{max}	$= 120m/s$
ρ	$= 4000kg/m^3$	$step_{velo}$	$= 1m/s$
E	$= 10^{10}Pa$		
ν	$= 0$		

Test vibr6 (Results)
Results
Velocity $V = 0$

Mode	Frequency	Damping	real part	imag. part	stability
1	2.78	$-7.9 \cdot 10^{-16}$	$1.4 \cdot 10^{-14}$	17.5	STAB
2	11.11	$-7.0 \cdot 10^{-16}$	$4.9 \cdot 10^{-14}$	69.8	STAB
3	24.98	$1.0 \cdot 10^{-16}$	$-1.6 \cdot 10^{-14}$	157.0	STAB
4	44.41	$3.6 \cdot 10^{-15}$	$-1.0 \cdot 10^{-12}$	279.0	STAB
5	69.37	$-7.7 \cdot 10^{-15}$	$3.4 \cdot 10^{-12}$	435.8	STAB


Velocity $V = 120$ m/s

Mode	Frequency	Damping	real part	imag. part	stability
1	2.9	-5.2	95.4	18.3	INST
2	2.9	5.2	-95.4	18.3	STAB
3	13.3	1.7	-141.6	83.5	STAB
4	13.3	-1.7	141.6	83.5	INST
5	33.2	$1.2 \cdot 10^{-12}$	$-2.5 \cdot 10^{-10}$	208.4	STAB




12.5. TEST VIBR6 (DESCRIPTION SHEET)


CASTEM figures




Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)
Frequencies (real)




Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)
Frequencies (imagin.)



Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)
Lambda (real)



Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)
Lambda (imagin.)



Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)
Lambda (real/imagin.)

Test vibr6 (Listing CASTEM)

```

*****
Opti Echo 0;
*
*****
* Characteristics of the system *
*****
*--- Material
Rhostru = 4000.;
Youstru = 1.E10;
Nustru = 0.;
*--- Length of the rod
Lstru = 2.;
*--- Radius of the rod
R = 10.E-3;
*--- Section of the rod
Sstru = PI*R*R;
*--- Mass per unit length of the rod
Mstru = Rhostru*Sstru;
*--- Density of the fluid
Rhof = 1000.;
*--- Added mass of the fluid (non enclosed fluid)
Mfaj = Rhof*PI*R*R;
*--- Equivalent density
Rhoe = (Mstru+Mfaj)/Sstru;
*--- Axial velocity of fluid
Vmin = 0.;
Vmax = 120.;
PVa = 1.;
*
LVa = Prog Vmin PAS PVa Vmax;
*
*****
* Meshing of the system *
*****
*
NbN = 100;
Mess 'Number of nodes : ' NbN;
Mess ' ';
Opti Dime 3 Elem Seg2;
PA = (-.5*Lstru) 0. 0.;
PB = (.5*Lstru) 0. 0.;
L1 = Droi NbN PA PB;
*
Modl1 = Model L1 Mecanique Elastique Isotrope Tuyau;
Matr1 = Mater Modl1 Youn Youstru Nu Nustru Rho Rhoe
      Epai R Rayo R;
*
*****
* Masses, Rigidities ... of system *
*****
*--- Mass
Mstru = Masse Modl1 Matr1;
*--- Structural rigidity
Kstru = Rigi Modl1 Matr1;
*--- Boundary Conditions : plane problem
Cl1 = Bloq Ux L1;
Cl2 = Bloq Uy L1;
Cl3 = Bloq Rx L1;
Cl4 = Bloq Rz L1;
Cl5 = Bloq Uz (PA et PB);
*
Kstru = Kstru et Cl1 et Cl2 et Cl3 et Cl4 et CL5;
*
*****
* Added rigidities *
*****
*
Lpout = Lstru/NbN;
*
*--- Damping due to the flow
PAfu = (Prog -.5 0. 0. (Lpout/10.) .5 0. 0.
        (-1.*Lpout/10.) );
PAfu = PAfu et (Prog 0. -.5 (-1.*Lpout/10.) 0. 0. .5
               (Lpout/10.) 0.);
PAfu = PAfu et (Prog 0. (Lpout/10.) 0. 0. 0.
               (-1.*Lpout/10.) (-1.*Lpout*Lpout/60.) 0.);
PAfu = PAfu et (Prog (-1.*Lpout/10.) 0. 0. 0. (Lpout/10.)
               0. 0. (-1.*Lpout*Lpout/60.));
PAfu = PAfu et (Prog -.5 0. 0. (-1.*Lpout/10.) .5 0. 0.
               (Lpout/10.) );
PAfu = PAfu et (Prog 0. -.5 (Lpout/10.) 0. 0. .5
               (-1.*Lpout/10.) 0. );
PAfu = PAfu et (Prog 0. (-1.*Lpout/10.) (Lpout*Lpout/60.)
               0. 0. (Lpout/10.) 0. 0.);
PAfu = PAfu et (Prog (Lpout/10.) 0. 0. (Lpout*Lpout/60.)
               (-1.*Lpout/10.) 0. 0. 0. );
Afu = Manu Rigi type AMORTISSEMENT L1 (MOTS UY UZ RY RZ)
      PAfu;
*
*--- Rigidity due to the flow
PKfu = (Prog (-6./(5.*Lpout)) 0. 0. -1.1
        (6./(5.*Lpout)) 0. 0. -1.);
PKfu = PKfu et (Prog 0. (-6./(5.*Lpout)) 1.1 0. 0.
               (6./(5.*Lpout)) .1 0.);
PKfu = PKfu et (Prog 0. .1 (-2.*Lpout/15.) 0. 0. -1
               (Lpout/30.) 0. );
PKfu = PKfu et (Prog -.1 0. 0. (-2.*Lpout/15.) .1 0. 0.
               (Lpout/30.) );
PKfu = PKfu et (Prog (6./(5.*Lpout)) 0. 0. .1
               (-6./(5.*Lpout)) 0. 0. 1.1);
PKfu = PKfu et (Prog 0. (6./(5.*Lpout)) -1.1 0. 0.
               (-6./(5.*Lpout)) -1.1 0. );
PKfu = PKfu et (Prog 0. .1 (Lpout/30.) 0. 0. -.1
               (-2.*Lpout/15.) 0. );
PKfu = PKfu et (Prog -.1 0. 0. (Lpout/30.) .1 0. 0.
               (-2.*Lpout/15.) );
Kfu = Manu Rigi type RIGIDITE L1 (MOTS UY UZ RY RZ) PKfu;
*
*****
* Modal analysis *
*****
*
PMod = Prog 2.77 11.1 24.99 44.43 69.42;
Basr = VIBR PROCH PMod Kstru Mstru TBAS;
nnor (Basr.'MODES') (mots 'UZ');
*
Basm = Basr.'MODES';
Nbmod = DIME Basm-2;
Mess ' ';
Mess 'Stopped fluid';
Mess ' Mode Frequency ';
Mess '-----';
Repeter LMod Nbmod;
      i=&LMod;
      Mess ' ' i ' ' Basm.i.'FREQUENCE';
      Fin LMod;
      Mess ' ';
      Mess ' ';
      Mess ' ';
Finsi;
*
*****
* Table of results *
*****
*

```



12.5. TEST VIBR6 (DESCRIPTION SHEET)

```

Mod = Table;
Mod.Reel = Table;
Mod.Imag = Table;
Repeter Bloc6 (2*Nbmod);
  i=&Bloc6;
  Mod.Reel.i=Prog;
  Mod.Imag.i=Prog;
Fin Bloc6;
*
*****
*      Projecting matrices      *
*****
*
*--- Mass matrix
Mt = (PJBA MStru Basr);
*--- Damping matrices
Afp = (PJBA Afu Basr);
*--- Rigidity matrices
Ksp = (PJBA Kstru Basr);
Kfp = (PJBA Kfu Basr);
*
*--- Locating the critical velocity
VSauv = 'N';
* ===== *
*      Loop on the axial velocity      *
* ===== *
*
Repeter Bloc10 (DIME LVa);
  k=&Bloc10;
  Va = Extr LVa k;
  Mess 'Iteration' k;
  Mess 'Axial velocity : ' Va ;
  Mess ' ';
  Mess '-----'
-----';
  Mess ' Mode      Frequency      Damping
Real part      Imag. part
  Stabil.  ';
*
*****
*      Masses, rigidities ...      *
*****
*
*--- Damping
At = (2*Mfaj*Va*Afp);
*--- Rigidity
Kt = (Ksp) et ((Mfaj*Va*Va)*Kfp);
*
*****
*      Computation of compound eigen shapes *
*****
*
  Basc = VIBC Mt Kt At;
  Basm = Basc.'MODES';
*
*--- Extracting frequencies
Prf = Prog;
Pif = Prog;
Repeter Bloc2 (2*Nbmod);
  i=&Bloc2;
  Modn = Basm.i;
  f = Modn.'FREQUENCE_REELLE';

  Prf = Prf et (Prog Modn.'FREQUENCE_REELLE');
  Pif = Pif et (Prog Modn.'FREQUENCE_IMAGINAIRE');
  Fin bloc2;
*
*--- Sorting of frequencies (rising row)
  n1 = DIME Prf;
  Prfc = Prog (extr Prf 1);

Pifc = Prog (extr Pif 1);
Repeter Bloc3 (n1-1);
  i=&Bloc3+1;
  f = extr Prf i;
  n2 = DIME Prfc;
  Pos = 1;
  Repeter Bloc4 n2;
    j = &Bloc4;
    fl = extr Prfc j;
    Si (> f fl);
      Pos = Pos + 1;
    Finsi;
  Fin Bloc4;
  Prfc = INSE Prfc Pos f;
  Pifc = INSE Pifc Pos (extr Pif i);
  Fin Bloc3;
*--- Saving
Repeter Bloc5 n1;
  i=&Bloc5;
  f = (Extr Prfc i);
  if = (Extr Pifc i);
  Msg = 'STAB.';
  Si ((< if -1.D-10) et (>EG f 0.));
    Msg = 'INST.';
    Si (EGA VSauv 'N');
      Vcrit = Va - (PVa/2.);
      VSauv = 'O';
    Finsi;
  Finsi;
  Si (> f 0.);
* Eigen values related to lambda = i omega = ix2PIxf
  Mess i f (if/f) (-2*PI*if) (2*PI*f) Msg;
  Sinon;
  Si (EGA f 0.);
    Mess i f ' ----- ' (-2*PI*if) (2*PI*f)
    Msg;
  Finsi;
  Finsi;
  Mod.Reel.i = Mod.Reel.i et (Prog f);
  Mod.Imag.i = Mod.Imag.i et (Prog if);
  Fin Bloc5;
  Mess '-----'
-----';
  Mess ' ';
*
Fin Bloc10;
*
*****
*      Graphes      *
*****
*
*
*
Nbcour=2*Nbmod;
Tabl = Table;
Repeter Ident1 Nbcour;
  k=&Ident1;
  Tabl.k = 'NOLI';
  Fin Ident1;
*
  k=1;
*--- Graphes related to frequency
  Titr 'Frequencies (real)';
  Evol1 = Evol manu Lva (Mod.Reel.k);
  Titr 'Frequencies (imagin)';
  Evol2 = Evol manu Lva (Mod.Imag.k);
*--- Graphes related to lambda = i 2 Pi f
  Titr 'Lambda (real)';
  Evol3 = Evol manu Lva (-2.*PI*(Mod.Imag.k));
  Titr 'Lambda (imagin.)';
  Evol4 = Evol manu Lva (2*PI*(Mod.Reel.k));

```

```

Titre 'Lambda (real/imagin)';
Evol5 = Evol manu (2.*PI*(Mod.Reel.k)
                (-2.*PI*(Mod.Imag.k)));
Repete Tracl (Nbcour-1);
k=&Tracl+1;
Si (> (DIME Mod.Reel.k) 0);
Titre 'Frequencies (real)';
Evol1 = Evol1 et (Evol manu Lva (Mod.Reel.k));
Titre 'Frequencies (imagin)';
Evol2 = Evol2 et (Evol manu Lva (Mod.Imag.k));
Titre 'Lambda (real)';
Evol3 = Evol3 et (Evol manu Lva (-2.*PI*(Mod.Imag.k)));
Titre 'Lambda (imagin.)';
Evol4 = Evol4 et (Evol manu Lva (2.*PI*(Mod.Reel.k)));
Titre 'Lambda (real/imagin)';
Evol5 = Evol5 et (Evol manu
                (2.*PI*(Mod.Reel.k) (-2.*PI*(Mod.Imag.k)));
Finsi;
Fin Tracl;
Dess Evol1 XBOR 0. 150. YBOR 0. 70. Lege
Titx 'Vax (m/s)'
Tity 'Re(F) (Hz)' Tabl;
Dess Evol2 Lege
Titx 'Vax (m/s)'
Tity 'Im(F) (Hz)' Tabl;
Dess Evol3 XBOR 0. 150. YBOR -150. 150. Lege
Titx 'Vax (m/s)'
Tity 'Re(i2.PI.F)' Tabl;
Dess Evol4 XBOR 0. 150. YBOR 0. 450. Lege
Titx 'Vax (m/s)'
Tity 'Im(i2.PI.F)' Tabl;
Dess Evol5 XBOR 0. 500. YBOR -150. 150. Lege
Titx 'Re(i2.PI.F)'
Tity 'Im(i2.PI.F)' Tabl;
*
*****
* Code validation *
*****
*
Vth = (Pi/Lstru)*((Youstru*Pi*(R**4)/(4*mfaj))**0.5);
Si (Abs((Vcrit-25.)/25.) < 0.05);
Erre 0;
Sinon;
Erre 5;
Finsi;
*
FIN ;
    
```



Test vibr6 (Comments)

1. Creation of added stiffnesses

Afu = Manu Rigi type AMORTISSEMENT L1 (MOTS UY UZ RY RZ) PAfu;
Kfu = Manu Rigi type RIGIDITE L1 (MOTS UY UZ RY RZ) PKfu;

The operator MANU followed by RIGI keyword creates a RIGIDITE subtype object in which all elementary stiffness matrices are similar :

- the elementary matrix is given by a list of reals (PAfu and PKfu),
- the geometrical object is given by L1,
- the type of stiffness is specified (AMORTISSEMENT and RIGIDITE),
- the name of the components for each node of L1 is given by a list of words (MOTS UY UZ RY RZ).

In this case, each elementary matrix is defined by a 8X8 matrix : 2 x (UY UZ RY RZ).

2. Projection on modal basis

* Mass matrix

Mt = (PJBA MStru Basr);

* Damping matrices

Afp = (PJBA Afu Basr);

* Rigidity matrices

Ksp = (PJBA Kstru Basr);

Kfp = (PJBA Kfu Basr);

The PJBA operator calculates the projection of stiffness objects (MStru, Afu, Kstru and Kfu) on an eigen mode basis (Basr).

3. Computation of compound eigen shapes

Basc = VIBC Mt Kt At;

Basm = Basc.'MODES';

...

Modn = Basm.i;

Prf = Prf et (Prog Modn.'FREQUENCE_REELLE');

Pif = Pif et (Prog Modn.'FREQUENCE_IMAGINAIRE');

VIBC computes the complex eigen values and eigen vectors of the general equation : $M\ddot{x} + C\dot{x} + Kx = 0$. The mass (Mt), stiffness (Kt) and damping (At) are given in matrices previously expressed relatively to a real eigen mode basis. If these matrices have not their subtypes corresponding with MASSE, RIGIDITE and AMORTISSEMENT, the input order is very important because they can't be sorted: they then are processed in order of their input.

The generated object is a BASE_MODALÉ subtype table (Basc).

Basc.'MODES' is a BASE_DE_MODALÉ subtype table (Basm).

Basm.i is a MODE_COMPLEXE subtype table (Modn) :

Modn.'FREQUENCE_REELLE' : real part of the eigen frequency.

Modn.'FREQUENCE_IMAGINAIRE' : imaginary part of the eigen frequency.

Modn.'NUMERO_MODE' : number of the mode.

Modn.'DEFORMEE_MODALÉ_REELLE' : real part of the eigen vector.

Modn.'DEFORMEE_MODAL_E_IMAGINAIRE' : imaginary part of the eigen vector.



4. Options for evolution display

Dess Evol1 XBOR 0. 150. YBOR 0. 70. Lege
Titx 'Vax (m/s)' Tity 'Re(F) (Hz)' Tab1;

The XBOR and YBOR keywords impose boundaries on X-axis (mini 0. and maxi 150.) and on Y-axis (mini 0. and maxi 70.).

The LEGE keyword imposes a caption for the different plotted curves. It imposes a square frame and the maximal number of curves is nine.

The TITX and TITY keywords allow to modify the names of abscissas and ordinates.

12.6 Test vibr10 (Description sheet)

Test name: vibr10

Calculation type: MECHANICS DYNAMICS, MASS UNBALANCE RESPONSE

Finite element type:


Topic: Mass unbalance response. The system is represented by a point (mass M1, stiffness K1). A mass unbalance is added in the equation under the shape of a force. The system to be solved is : $M\ddot{x} + C\dot{x} + Kx = F$ with F defined by $F = F_s \cdot \sin(\Omega \cdot t) + F_c \cdot \cos(\Omega \cdot t)$. The response is computed with VIBC operator and its option BALOUR.

Goal: Find the small and big axes of the ellipse described by the system.

Reference: Analytical solution.

Version: 97' customer version

Model description:



Test vibr10 (Results)

X_{theor} (m)	$9.46402 \cdot 10^{-8}$
X_s (m)	$9.46402 \cdot 10^{-8}$
X_c (m)	$9.46402 \cdot 10^{-8}$



Test vibr10 (Listing CASTEM)

```

*****
OPTI 'DIME' 3 'ELEM' SEG2 echo 0 ;
*
*-Data
P1 = 1. 0. 0. ;
m1 = 40. ;
k1 = 4.e5 ;
f1 = ((k1 / m1) ** 0.5) / (2. * pi) ;
MESS 'f1 =' f1 ;
*
TR1 = MANU 'POI1' P1 ;
*
*-Matrices of mass and rigidity :
MMass = MANU 'RIGI' TR1 (MOTS 'UY' 'UZ')
        (PROG m1 0. 0. m1) ;
MRigi = MANU 'RIGI' TR1 (MOTS 'UY' 'UZ')
        (PROG k1 0. 0. k1) ;
*
*-Mode shapes :
Phi1 = MANU 'CHPO' TR1 2 'UY' 1. 'UZ' 0. ;
Phi2 = MANU 'CHPO' TR1 2 'UY' 0. 'UZ' 1. ;
*
*-Points identifying the modes
PR1 = 0. 0. 0. ;
PR2 = 0. 0. 0. ;
*
*-Basis of real modes :
TBasR = TABLE 'BASE_MODAL' ;
TBasR.MODES = TABLE 'BASE_DE_MODES' ;
TBasR.MODES . 'MAILLAGE' = TR1 ;
TBasR.MODES . 1 = TABLE 'MODE' ;
TBasR.MODES . 1 . 'NUMERO_MODE' = 1 ;
TBasR.MODES . 1 . 'POINT_REPERE' = PR1 ;
TBasR.MODES . 1 . 'FREQUENCE' = f1 ;
TBasR.MODES . 1 . 'MASSE_GENERALISEE' = m1 ;
TBasR.MODES . 1 . 'DEPLACEMENTS_GENERALISES' =
    k1 - (2 * pi * f1) ;
TBasR.MODES . 1 . 'DEFORMEE_MODAL' = Phi1 ;
TBasR.MODES . 2 = TABLE 'MODE' ;
TBasR.MODES . 2 . 'NUMERO_MODE' = 2 ;
TBasR.MODES . 2 . 'POINT_REPERE' = PR2 ;
TBasR.MODES . 2 . 'FREQUENCE' = f1 ;
TBasR.MODES . 2 . 'MASSE_GENERALISEE' = m1 ;
TBasR.MODES . 2 . 'DEPLACEMENTS_GENERALISES' =
    k1 - (2 * pi * f1) ;
TBasR.MODES . 2 . 'DEFORMEE_MODAL' = Phi2 ;
*
*-Whirl speed (rad/s)
omegl = 19.0985 ;
*
*-Projection of matrices on the real basis :
MMasP = PJBA TBasR MMass ;
MRigP = PJBA TBasR MRigi ;
*
*-Mass unbalance excitation
MxD = 1e-4 ;
Fyls = MxD * (omegl ** 2) ;
Fzls = (-1) * MxD * (omegl ** 2) ;
Fylc = MxD * (omegl ** 2) ;
Fzlc = MxD * (omegl ** 2) ;
*
Fls = MANU 'CHPO' P1 2 'FY' Fyls 'FZ' Fzls ;
Flc = MANU 'CHPO' P1 2 'FY' Fylc 'FZ' Fzlc ;
PjFls = PJBA Fls TBasR ;
PjFlc = PJBA Flc TBasR ;
*
GEOx1 = EXTR PjFls 'MAIL' ;
GEOx2 = GEOx1 ;
*
XVal1 = PROG ;
XVal2 = PROG ;
REPE bbb2 (NBEL GEOx1) ;
i2 = &bbb2 ;
XP1 = 0. 0. 0. ;
GEOx2 = GEOx2 ET XP1 ;
Px1 = POIN GEOx1 i2 ;
XVal1 = XVal1 ET (PROG (EXTR PjFls 'FALF' Px1)) ;
XVal2 = XVal2 ET (PROG (EXTR PjFlc 'FALF' Px1)) ;
FIN bbb2 ;
XVal2 = XVal1 ET XVal2 ;
*
PjFBal1 = MANU CHPO GEOx2 1 'FALF' XVal2 ;
*
*-Computation of the response :
RepRel RepIm1 = VIBC 'BALOU' MMasP MRigP TBasR
                PjFBal1 omegl ;
*
*-Computation of big and small axes of the ellipse :
YR1 = EXTR RepRel 'UY' P1 ;
ZR1 = EXTR RepRel 'UZ' P1 ;
MESS YR1 ZR1 ;
*
opti donn 5;
*-Wanted results :
An1 = (MxD * (omegl ** 2)) / (k1 - ((omegl ** 2) * m1)) ;
MESS 'Wanted results : ' An1 ;
*
FIN ;

```

Test vibr10 (Comments)

1. Mass unbalance excitation

```

F1s = MANU 'CHPO' P1 2 'FY' Fy1s 'FZ' Fz1s ;
F1c = MANU 'CHPO' P1 2 'FY' Fy1c 'FZ' Fz1c ;
PjF1s = PJBA F1s TBasR ;
PjF1c = PJBA F1c TBasR ;
GEOx1 = EXTR PjF1s 'MAIL' ;
GEOx2 = GEOx1 ;
*
XVal1 = PROG ;
XVal2 = PROG ;
REPE bbb2 (NBEL GEOx1) ;
i2 = &bbb2 ;
XP1 = 0. 0. 0. ; GEOx2 = GEOx2 ET XP1 ;
Px1 = POIN GEOx1 i2 ;
XVal1 = XVal1 ET (PROG (EXTR PjF1s 'FALF' Px1)) ;
XVal2 = XVal2 ET (PROG (EXTR PjF1c 'FALF' Px1)) ;
FIN bbb2 ;
XVal2 = XVal1 ET XVal2 ;
*
PjFBal1 = MANU CHPO GEOx2 1 'FALF' XVal2 ;
    
```

The mass unbalance force (PjFBal1) is expressed relatively to a real eigen mode basis. It is a CHAMPOIN object with one dual component FALF.

It must be obtained by the following way :

- Both components F_s and F_c are separated into two CHAMPOIN objects (here F1s and F1c).
- Those two CHAMPOIN objects are projected on the n real eigen mode basis (here TBasR basis is composed of 2 eigen modes). The projected objects are PjF1s and PjF1c.
- Then PjF1s and PjF1c are combined in order to obtain a 2 x n vector that is to say a CHAMPOIN object composed by one component FALF and supported by 4 points (GEOx2)

2. Computation of the response

```
RepRe1 RepIm1 = VIBC 'BALOU' MMasP MRigP TBasR PjFBal1 omeg1 ;
```

The operator VIBC followed by BALOU keyword computes the complex solutions in the case of an unbalanced excitation. It creates two CHAMPOIN objects (RepRe1 and RepIm1) which contain the displacement vectors :

- RepRe1 deals with $Cos(\Omega.t)$ components, and the real components.
- RepIm1 deals with $Sin(\Omega.t)$ components, and the imaginary components.



12.7 Test dyna12 (Description sheet)

Test name: dyna12

Calculation type: MECHANICS DYNAMICS 3D, TRANSIENT RESPONSE

Finite element type: BARR


Topic: Transient response of a launcher. The launcher is submitted to an axial load during a limited period. Considering rigid body movements and vibration of the structure, the displacement history is searched for the bottom point.

Goal: Find the displacement history of the bottom point (P1).

Reference: J. S. Przemieniecki : Theory of matrix structural analyses (New York : McGraw-Hill Book Co., 1968) p340.

Version: 97' customer version

Model description:



Test dyna12 (Results)

	Analytic	CASTEM	Margin
UY(t=1.9125) Point P1	10.9	10.660	-2.20%
UY(t=2.0625) Point P1	11.732	11.289	-3.78%

Test dyna12 (Listing CASTEM)

```

*****
OPTION ECHO 0 ;
OPTIO DIME 3 ELEM SEG2 ;
*
* MESHING
*
P1 = 0. 0. 0. ; P2 = 0. 70. 0. ; P3 = 0. 140. 0. ;
L1 = D 1 P1 P2 ; L2 = D 1 P2 P3 ; LTOT = L1 ET L2 ;
*
* MODEL
*
MOD1 = MODELE LTOT MECANIQUE ELASTIQUE BARR ;
*
* MATERIAL AND GEOMETRICAL CHARACTERISTICS
*
MAT1 = MATER MOD1 YOUN 1.E4 NU 0.3 RHO 0.1 ;
CAR1 = CARAC MOD1 SECT 1. ;
*
* RIGIDITY MATRIX
*
RIG1 = RIGIDITE MOD1 (MAT1 ET CAR1) ;
*
* MASS MATRIX
*
MAS1 = MASSE MOD1 (MAT1 ET CAR1) ;
*
* LOAD
*
LI1 = PROG 0. 0.0125 1.0125 1.025 3. ;
LI2 = PROG 0. 1. 1. 0. 0. ;
F1 = FORCE (0. 100. 0.) P1 ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR 'FORC' F1 EV ;
*
* INPUT DATA FOR DYNAMIC SUBROUTINE
*
LIS1 = PROG 1.9125 2.0625 ;
TAB1 = TABLE ;
TAB1.'DEPL' = MANU CHPO LTOT 3 UZ 0. UY 0. UX 0. ;
TAB1.'VITE' = MANU CHPO LTOT 3 UZ 0. UY 0. UX 0. ;
TAB1.'CHAR' = CHA1 ;
TAB1.'RIGI' = RIG1 ;
TAB1.'MASS' = MAS1 ;
TAB1.'FREQ' = 20. ;
TAB1.'INST' = LIS1 ;
TAB2 = DYNAMIC TAB1 ;
*
* COMPARISON WITH ANALYTIC RESULTS
*
DEP1 = TAB2 . 1 . DEPL ;
U1 = EXTR DEP1 P1 UY ; U1TH = 10.89972 ;
ERR1 = ((U1 - U1TH) / (ABS U1TH)) * 100. ;
DEP2 = TAB2 . 2 . DEPL ;
U2 = EXTR DEP2 P1 UY ; U2TH = 11.73228 ;
ERR2 = ((U2 - U2TH) / (ABS U2TH)) * 100. ;
MESS ' VERTICAL DISPLACEMENT OF THE BASIS T= 1.9125' ;
MESS 'THEORY 'U1TH'CASTEM 'U1 '%MARGIN 'ERR1 ;
MESS ' VERTICAL DISPLACEMENT OF THE BASIS T= 2.0625' ;
MESS 'THEORY 'U2TH'CASTEM 'U2 '%MARGIN 'ERR2 ;
*
SI ((MAXI ABS (PROG ERR1 ERR2 )) < 5);
ERRE 0;
SINON;
ERRE 5;
FINSI;
FIN ;

```



Test dyna12 (Comments)

1. Load as a function of time

```
LI1 = PROG 0. 0.0125 1.0125 1.025 3. ;
LI2 = PROG 0. 1. 1. 0. 0. ;
F1 = FORCE (0. 100. 0.) P1 ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR 'FORC' F1 EV ;
```

The operator CHAR builds an object of type CHARGEMENT, which contains spatial and temporal descriptions of the load :

- the spatial description is given by the object F1,
- the temporal description is given by the object EV which is called evolution (built by the operator EVOL, EV describes the load (object LI2) as a function of the time (object LI1)).

2. Use of DYNAMIC subroutine

```
* INPUT DATA FOR DYNAMIC SUBROUTINE
*
LIS1 = PROG 1.9125 2.0625 ;
TAB1 = TABLE ;
TAB1.'DEPL' = MANU CHPO LTOT 3 UZ 0. UY 0. UX 0. ;
TAB1.'VITE' = MANU CHPO LTOT 3 UZ 0. UY 0. UX 0. ;
TAB1.'CHAR' = CHA1 ;
TAB1.'RIGI' = RIG1 ;
TAB1.'MASS' = MAS1 ;
TAB1.'FREQ' = 20. ;
TAB1.'INST' = LIS1 ;
TAB2 = DYNAMIC TAB1 ;
```

The operator DYNAMIC carries out step by step dynamic computations, using the Newmark centered algorithm. All required data are given in the object TAB1 defined by the following indexes :

- 'DEPL' : initial displacements,
- 'VITE' : initial velocity which components are the same as components of initial displacements,
- 'CHAR' : the load (as a function of time),
- 'RIGI' : the rigidity matrix,
- 'MASS' : the mass matrix,
- 'FREQ' : the cut off frequency which determines the time step,
- 'INST' : the list of moments to save.

The value of time step will be equal to 0.25 divided by the cut off frequency (in this case 20 Hz) which gives a step of 0.0125 sec.

3. Description of DYNAMIC result

```
DEP1 = TAB2 . 1 . DEPL ; U1 = EXTR DEP1 P1 UY ;
DEP2 = TAB2 . 2 . DEPL ; U2 = EXTR DEP2 P1 UY ;
```

The object created by DYNAMIC subroutine is a table which contains as many tables as saved moments. Here TAB2 is composed of two tables because the list TAB1.INST contains two moments : TAB2 . 1 and TAB2 . 2.

Each table has three indexes :

- 'TEMP' : the value of the moment
- 'DEPL' : the displacement
- 'VITE' : the velocity

TAB2 . 1 . DEPL is a displacement field. A value can be extracted by using EXTR operator and specifying the desired point and the required component.



12.8 Test dyna11 (Description sheet)

Test name: dyna11

Calculation type: MECHANICS DYNAMICS 2D

Finite element type: QUA4


Topic: Transient response of an internal pressure. A thick-walled cylinder is suddenly subjected to an internal pressure. The problem consists in finding displacement maxima of some points.

Goal: Find displacement maxima of three points located on the cylinder half-length.


Reference: S. H.Crandall, N. C. Dahl and T. J. Lardner : An introduction to the mechanics of solids, 2nd ed. (New York : McGraw-Hill Book Company, 1972) pp. 293-297.

Version: 97' customer version

Model description:



- $L = 8in$
- $R_i = 6in$
- $R_o = 12in$
- $ur_1 = R_i$
- $ur_2 = 9in$
- $ur_3 = 11in$
- $\rho = 0.28lb/in^3$
- $1/g = 1/(386.4in/sec^2)$
- $= 0.002588$
- $E = 30.10^6lb/in^2$
- $\nu = 0.0$
- $P = 10lb/in^2$




Test dyna11 (Results)

Results : Ratio transient/static

	Analytical	CASTEM	%Margin
M1	2	1.93	-3.46
M2	2	2.06	2.79
M3	2	2.05	2.74

CASTEM figures



Test dyna11 (Listing CASTEM)

```

*****
OPTION ECHO 0 ;
OPTIO DIME 2 MODE AXIS ELEM QUA4 ;
*
* MESHING
*
P1 = 6. 0. ; P2 = 12. 0. ; P3 = 12. 8. ; P4 = 6. 8. ;
L1 = D 6 P1 P2 ; L2 = D 8 P2 P3 ; L3 = D 6 P3 P4 ;
L4 = D 8 P4 P1 ; LTOT = L1 ET L2 ET L3 ET L4 ;
S1 = SURF LTOT PLAN ;
*
* MODEL
*
MOD1 = MODELE S1 MECANIQUE ELASTIQUE ;
*
* MATERIAL PROPERTIES
*
RO = 0.28 * 0.002588 ;
MAT1 = MATER MOD1 YOUN 30.E6 NU 0.0 RHO RO ;
*
* BOUNDARY CONDITIONS
*
CL1 = BLOQ L1 UZ ; CL2 = BLOQ L3 UZ ;
*
* RIGIDITY MATRIX
*
RIG1 = RIGIFITE MOD1 MAT1 ;
*
* MASS MATRIX
*
MAS1 = MASSE MOD1 MAT1 ;
*
* LOAD
*
LI1 = (PROG 0. 1.E-6 2.E-6 3.E-6 4.E-6 ) ET
      (PROG 5e-6 pas 1e-6 204E-6 ) ;
LI2 = (PROG 0. 0. 0. 0. 0. ) ET (PROG 200 * 1.) ;
F1 = PRES MOD1 'MASS' 10. L4 ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR 'PRES' F1 EV ;
*
* INPUT DATA FOR DYNAMIC SUBROUTINE
*
LIS1 = PROG 0. PAS 1.E-6 200.E-6 ;
TAB2 = TABLE ;
TAB2.'DEPL' = MANU CHPO S1 2 UZ 0. UR 0. ;
TAB2.'VITE' = MANU CHPO S1 2 UZ 0. UR 0. ;
TAB2.'CHAR' = CHA1 ;
TAB2.'RIGI' = RIG1 ;
TAB2.'MASS' = MAS1 ;
TAB2.'FREQ' = 250000. ;
TAB2.'INST' = LIS1 ;
TAB1 = DYNAMIC TAB2 ;
*
* EXTRACTING OF DISPLACEMENT EVOLUTIONS
*
LT = PROG ; LUR1 = PROG ;
LUR2 = PROG ;
LUR3 = PROG ;
I = 0 ;
M1 = NOEUD 25 ; M2 = NOEUD 62 ; M3 = NOEUD 36 ;
NBB = (DIME LIS1) - 1 ;
REPETER Bb1 NBB ;
I = I + 1 ;
LT = LT ET( PROG(TAB1 . I . 'TEMP' )) ;
DEP1 = TAB1 . I . 'DEPL' ;
LUR1 = LUR1 ET( PROG (EXTR DEP1 UR M1) ) ;
LUR2 = LUR2 ET( PROG (EXTR DEP1 UR M2) ) ;
LUR3 = LUR3 ET( PROG (EXTR DEP1 UR M3) ) ;
FIN Bb1 ;
*
EV1 = EVOL MANU 'TIME' LT 'UR ' LUR1 ;
EV2 = EVOL MANU 'TIME' LT 'UR ' LUR2 ;
EV3 = EVOL MANU 'TIME' LT 'UR ' LUR3 ;
*
* --- DISPLAY OF GRAPH
*
TABDD = TABLE ;
TABDD.1 = 'MARQ ETOI NOLI TITR Node25' ;
TABDD.2 = 'MARQ CARR NOLI TITR Node62' ;
TABDD.3 = 'MARQ TRIA NOLI TITR Node36' ;
TITR 'RADIAL DISPLACEMENT THICK-WALLED CYLINDER ' ;
DESS (EV1 ET EV2 ET EV3) TABDD 'LEGE' ;
*
* COMPARISON WITH THEORY
*
UR1 = (MAXI LUR1) / 3.326E-6 ; UR1TH = 2. ;
ERR1 = ((UR1 - UR1TH) / (ABS UR1TH)) * 100. ;
UR2 = (MAXI LUR2) / 2.773E-6 ; UR2TH = 2. ;
ERR2 = ((UR2 - UR2TH) / (ABS UR2TH)) * 100. ;
UR3 = (MAXI LUR3) / 2.673E-6 ; UR3TH = 2. ;
ERR3 = ((UR3 - UR3TH) / (ABS UR3TH)) * 100. ;
MESS 'RATIO TRANSIENT / STATIC ' ;
MESS 'NODE 25 THEORY 'UR1TH 'CASTEM ' UR1
      '%MARGIN 'ERR1 ;
MESS 'NODE 62 THEORY 'UR2TH 'CASTEM ' UR2
      '%MARGIN 'ERR2 ;
MESS 'NODE 36 THEORY 'UR3TH 'CASTEM ' UR3
      '%MARGIN 'ERR3 ;
*
SI ((MAXI ABS (PROG ERR1 ERR2 ERR3 )) < 5) ;
ERRE 0 ;
SINON ;
ERRE 5 ;
FINSI ;
FIN ;

```



Test dyna11 (Comments)

1. Unit system

RO = 0.28 * 0.002588 ;
MAT1 = MATER MOD1 YOUN 30.E6 NU 0.0 RHO RO ;

There is no unit system on CASTEM. Therefore data must be given in a right system according to the basic law of mechanics : $F = M \cdot \gamma$.

It is necessary here to give the density in mass per volume unit that is to say to divide the density (given in lb/in^3) by the acceleration constant g (given in in/sec^2).

2. Extracting of DYNAMIC results

```
LT = PROG ; LUR1 = PROG ;
...
I = 0 ;
M1 = NOEUD 25 ;
...
NBB = (DIME LIS1) - 1 ;
REPETER BB1 NBB ;
I = I + 1 ;
LT = LT ET( PROG(TAB1 . I . 'TEMP' ) );
DEP1 = TAB1 . I . 'DEPL' ;
LUR1 = LUR1 ET( PROG (EXTR DEP1 UR M1) );
...
FIN BB1 ;
EV1 = EVOL MANU 'TIME' LT 'UR ' LUR1 ;
```

To get the displacement evolution as a function of time, it is necessary to extract each saved instant and store it in a list of reals (LT). In the same way, each saved UR-displacement of the required point is stored in the LUR1 list of reals. Then an object of EVOLUTION type (EV1) can be obtained which represents the UR-displacement of M1 point as a function of time.

3. Display of graph with a caption

```
TABDD = TABLE ;
TABDD.1 = 'MARQ ETOI NOLI TITR Node25' ;
TABDD.2 = 'MARQ CARR NOLI TITR Node62' ;
TABDD.3 = 'MARQ TRIA NOLI TITR Node36' ;
TITR 'RADIAL DISPLACEMENT THICK-WALLED CYLINDER ' ;
DESS (EV1 ET EV2 ET EV3) TABDD 'LEGE' ;
```

To get a caption the LEGE keyword is added to the DESS operator, but it is necessary to indicate some specific options. These options are defined in the TABDD table. Each index of the table correspond to one curve : TABDD.1 is related to the first curve indicated in the DESS call.

The MARQ keyword followed by ETOI indicates that a symbol (a star) will be plotted as a marker at different points of the curve (CARR correspond to squares, TRIA correspond to triangles).

NOLI indicates that the segments connecting the points won't be plotted.

The TITR keyword followed by a string (maximum 20 letters and no gap between letters) defines the caption related to the required curve.

12.9 Test dy_devo5 (Description sheet)

Test name: dy_devo5

Calculation type: MECHANICS DYNAMICS 3D, TRANSIENT RESPONSE

Finite element type: POUT


Topic: Transient response of a supported beam. The beam is simply supported by elastic supports. The triangular distributed load has its amplitude as a function of time. The problem consists in finding the maximum deflection for different moments.

Goal: Find the maximum deflection of the beam.


Reference: Walter C. Hury and Moshe F. Rubinstein : Dynamics of structures (Englewood Cliffs : Prentice-Hall, Inc., 1964) p.294.

Version: 97' customer version

Model description:



$$\begin{aligned}
 L &= 50m \\
 K_y &= (974.1/2)N/m \\
 A &= 1m^2 \\
 I_y &= 10m^4 \\
 I_z &= 10m^4 \\
 G &= 1.m^4 \\
 E &= 10^6 Pa \\
 \nu &= 0.3 \\
 \rho &= 0.1kg/m^3 \\
 F(x,t) &= \int_0^{2L} -\left(\frac{2x}{L}\right) \cdot F(t) \cdot S \cdot \rho dx
 \end{aligned}$$







Test dy_devo5 (Results)

Results

	Theoretical	CASTEM	%margin
ω (shape 1)	7.155 rad/s	7.1498 rad/s	0.007
ω (shape 2)	28.23 rad/s	29.679 rad/s	5.13
$deflection_{max}(t = 1.45s)$	0.23382 m	0.23361 m	0.009
$deflection_{max}(t = 1.9s)$	-0.24283 m	-0.2385 m	1.78

CASTEM figures



Test dy_devo5 (Listing CASTEM)

```

*****
OPTION ECHO 0 ;
OPTION DIME 3 ELEM SEG2 ;
*
* MESHING
*
P1 = 0. 0. 0. ; P2 = 50. 0. 0. ;
L1 = D 4 P1 P2 ;
*
* MODEL
*
MOD1 = MODELE L1 MECANIQUE ELASTIQUE POUT ;
*
* MATERIAL AND GEOMETRICAL CHARACTERISTICS
*
MAT1 = MATER MOD1 YOUN 1.E6 NU 0.3 RHO 0.1 ;
CAR1 = CARAC MOD1 SECT 1 INRY 10. INRZ 10. TORS 1.
      VECT (0. 1. 0.) ;
*
* BOUNDARY CONDITIONS : ELASTIC SUPPORT AND
* SYMETRICAL CONDITIONS
*
K = 974.1 / 2 ;
CL1 = APPUI UY K P1 ;
CL2 = BLOQ L1 RX RY UX UZ ;
CL3 = BLOQ P2 RZ ;
CLTOT = CL1 ET CL2 ET CL3 ;
*
* MATRIX OF RIGIDITY
*
RIG1 = RIGIDITE MOD1 (MAT1 ET CAR1) ;
*
* MATRIX OF MASS
*
MAS1 = MASSE MOD1 (MAT1 ET CAR1) ;
*
* COMPUTATION OF MODE SHAPES
*
TAB1 = VIBR 'INTERVALE' 0. 5.
      (RIG1 ET CLTOT) MAS1 'TBAS' ;
TAB2 = TAB1.MODES ;
NNOR TAB2 AVEC (MOTS UY) ;
NMOD = (DIME TAB2) - 2 ;
IB = 0 ;
REPETE BOUC NMOD ;
  IB = IB + 1 ;
  F1 = (TAB2.IB). 'FREQUENCE' ;
  OMEG1 = (F1 * 2.) * PI ;
  SI (IB 'EGA' 1) ; OMEG1TH = 7.155 ;
  SINON ; OMEG1TH = 28.23 ; FINSI ;
  ERR1 = ((OMEG1 - OMEG1TH) / (ABS OMEG1TH)) * 100. ;
  MESS 'MODE SHAPE 'IB ;
  MESS 'THEORETICAL ANGULAR FREQUENCY 'OMEG1TH
      'CASTEM 'OMEG1 '%MARGIN 'ERR1 ;
  DEP1 = (TAB2.IB).DEFORMEE_MODAL ;
  DEF1 = DEFORM DEP1 L1 1. ROUG ;
  DEF0 = DEFORM DEP1 L1 0. ;
  TRAC (0 0 10000) (DEF0 ET DEF1) ;
FIN BOUC ;
*
TABM = TAB1 ;
*
* -- INPUT DATA FOR USING OF DYNE
*
*
*
*
* DISTRIBUTED LOAD
*
FC1 = COOR 1 L1 ;
FC2 = (FC1 / -50.) * 2. ;
FC3 = NOMC FC2 UY ;
MODB = MODEL L1 MECANIQUE ELASTIQUE BARR ;
MATB = MATER MODB SECT 1. RHO 1. ;
MASB = MASSE MODB MATB ;
FC4 = MASB * FC3 ;
FC5 = EXCO FC4 FY FY ;
LI1 = PROG 0. 1. 1.025 3.0 ;
LI2 = PROG 0. 1. 0. 0. ;
EV = EVOL MANU T LI1 F(T) LI2 ;
CHA1 = CHAR 'FORC' FC5 EV ;
TABCHAR = TABLE 'CHARGEMENT' ;
*
* PROJECTING ON THE MODAL BASIS
*
CHA2 = PJBA CHA1 TAB1 ;
TABCHAR.'BASE_A' = CHA2 ;
*
* OUTPUT TABLE
*
TABSOR = TABLE 'SORTIE' ;
TABVAR = TABLE 'VARIABLE' ;
TABSOR.'VARIABLE' = TABVAR ;
TABVAR.'DEPLACEMENT' = VRAI ;
TABVAR.'VITESSE' = FAUX ;
TABVAR.'ACCELERATION' = FAUX ;
*
* CALLING DYNE DE_VOGELAERE
*
TABRES = DYNE DE_VOGELAERE TABM TABCHAR
      TABSOR 80 0.025 1 ;
*
* RECOMBINATION FOR MOMENTS 1.45 AND 1.9
*
DEP1 = RECO TABRES TABM 1.45 'DEPL' ;
DEP2 = RECO TABRES TABM 1.9 'DEPL' ;
*
* COMPARISON WITH ANALYTIC RESULTS
*
FLECH1 = EXTR DEP1 P2 UY ; FLECH1TH = 0.23382 ;
ERR1 = (( FLECH1 - FLECH1TH) / (ABS FLECH1TH)) * 100. ;
MESS ' MAXIMUM DEFLECTION T = 1.45 ' ;
MESS 'THEORETICAL' FLECH1TH'CASTEM 'FLECH1
      '%MARGIN 'ERR1 ;
*
FLECH2 = EXTR DEP2 P2 UY ; FLECH2TH = -0.24283 ;
ERR2 = (( FLECH2 - FLECH2TH) / (ABS FLECH2TH)) * 100. ;
MESS ' MAXIMUM DEFLECTION T = 1.9 ' ;
MESS 'THEORETICAL' FLECH2TH'CASTEM 'FLECH2
      '%MARGIN 'ERR2 ;
*
SI ((MAXI ABS (PROG ERR1 ERR2)) < 5);
ERRE 0;
SINON;
ERRE 5;
FINSI;
FIN ;

```



Test dy_devo5 (Comments)

1. Norming of mode shapes

NNOR TAB2 AVEC (MOTS UY) ;

The NNOR instruction norms the object TAB2 so that the greatest value equal 1. The keyword AVEC limits the research only to UY component.

2. Distributed load

FC1 = COOR 1 L1 ;
 FC2 = (FC1 / -50.) * 2. ;
 FC3 = NOMC FC2 UY ;
 MODB = MODEL L1 MECANIQUE ELASTIQUE BARR ;
 MATB = MATER MODB SECT 1. RHO 1. ;
 MASB = MASSE MODB MATB ;
 FC4 = MASB * FC3 ;

The distributed load is created by the multiplication of the load FC3 (which is a CHPOINT) by the mass matrix MASB. This method allows to balance the load in the meaning of finite elements.

3. Projection of the load onto the modal basis

TABCHAR = TABLE 'CHARGEMENT' ;
 CHA2 = PJBA CHA1 TAB1 ;
 TABCHAR.'BASE_A' = CHA2 ;

The PJBA operator projects the load CHA1 onto the modal basis TAB1. The projected load CHA2 is put in the table TABCHAR of CHARGEMENT subtype and indexed by BASE_A. The A-basis is the modal basis in which the equations are uncoupled.

4. Declaration of the output table

```
* OUTPUT TABLE
*
TABSOR = TABLE 'SORTIE' ;
TABVAR = TABLE 'VARIABLE' ;
TABSOR.'VARIABLE' = TABVAR ;
TABVAR.'DEPLACEMENT' = VRAI ;
TABVAR.'VITESSE' = FAUX ;
TABVAR.'ACCELERATION' = FAUX ;
```

It is necessary to define the output table and its subtype : the TABSOR table of SORTIE subtype defines the data to be saved during the use of DYNE operator.

TABSOR.VARIABLE is a table too, its subtype is VARIABLE.

In this example, only the displacements will be saved because the content of the boolean variable (TABVAR.'DEPLACEMENT') is true and the content of the other boolean variables (TABVAR.'VITESSE' and TABVAR.'ACCELERATION') is false.

5. Calling of DYNE De_Vogelaere operator

```
TABRES = DYNE DE_VOGELAERE TABM TABCHAR TABSOR 80 0.025 1 ;
```

- The TABM table contains the modal basis.
- The TABCHAR table contains the evolution of applied free forces (which are projected onto the modal base).
- The TABSOR table defines the results requested in the TABRES table.
- The number of steps is 80, the value of time step is 0.025.
- A result is saved every 1 computation step.
- TABRES contains the results of the computation.

The maximum value of time step is about one tenth of the smallest period of the used mode shapes.

6. Content of the result table TABRES

The TABRES subtype is RESULTAT_DYNE.

- TABRES.'TEMPS_DE_SORTIE' contains the list of output times.
- TABRES. i is a table containing the results of the i^{th} output step. The indices of TABRES. i depend on the requested variables given previously in TABVAR. Here we will find only the DEPLACEMENT index.

TABRES. i .DEPLACEMENT is an object of CHPOINT type which component is ALFA because the computation was done in A-basis.

7. figures/Recombination of TABRES table for moments 1.45 and 1.9

```
DEP1 = RECO TABRES TABM 1.45 'DEPL' ;
```

```
DEP2 = RECO TABRES TABM 1.9 'DEPL' ;
```

The RECO operator recombines the modes and the static solutions contained in a modal basis from the modal contributions. The requested time is indicated (here 1.45 then 1.9) with the name of the variable (here DEPLACEMENT). DEP1 and DEP2 are CHPOINT objects which can be used like results of a static computation.



12.10 Test dy_devo6 (Description sheet)

Test name: dy_devo6

Calculation type: MECHANICS DYNAMICS 3D, IMPACT, LINKAGE POINT-PLANE


Topic: Impact of a great mass against a plane. A great mass with a sinusoidal movement (composed by two different amplitudes) impacts a plane. The computation is done with a linkage point-plane and with taking into consideration the plastification of the impact spring. The computation restart is also checked.

Goal: Get the evolution of the impact force according to the displacement and get the evolution of the plastic displacement according to the time.

Reference:

Version: 97' customer version

Model description:



$$M1 = 100000kg$$

$$Gap = 0$$

$$E = 50Pa$$

$$K_e = 100N/m$$



Test dy_devo6 (Results)

Results

Maxima of plastic displacements :

	Analytic	CASTEM	Margin
1st computation	0.5 m	0.4993 m	0.14 %
2nd computation	1.5 m	1.4947 m	0.35 %

CASTEM figures





Test dy_devo6 (Listing CASTEM)

```

*****
opti echo 0 ;
*****
m1 = 100000 ;   comm 'impacted mass';
ke = 100. ;     comm 'impact stiffness';
sel = 50. ;     comm 'elastic limit';
dt = .01 ;     comm 'time step';
t1 = 12 ;      comm 'computation duration';
*****
opti dime 3 ;
ppl = 0 0 0 ;
pl = 0 0 0 ;
*
tabl = table 'BASE_MODAL' ;
TAB2 = TABLE 'BASE_DE_MODES' ;
TAB1 . MODES = TAB2 ;
TAB2 . 1 = TABLE 'MODE' ;
TAB2 . 1 . 'POINT_REPERE' = PPL ;
TAB2 . 1 . 'FREQUENCE' = 0 ;
TAB2 . 1 . 'MASSE_GENERALISEE' = M1 ;
TAB2 . 1 . 'DEFORMEE_MODAL' = manu chpo pl 1 ux 1. ;
*
** definition of load
w0 = .125*2*pi ;
list1 = prog 0. pas dt (t1/2-dt) ;
list2 = prog (t1/2) pas dt t1 ;
lisyl = prog sinu .125 ampl (w0**2) list1 ;
lisyl2 = prog sinu .125 ampl (1.5*(w0**2)) list2 ;
evl = evol manu 'time' (list1 et list2)
      'accel' (lisyl et lisyl2) ;
fx1 = force fx (-1*m1) pl ;
chal = char fx1 evl ;
*
** load and initial conditions in A basis
cham1 = pjba tabl chal ;
TCHAL = TABLE 'CHARGEMENT' ;
TCHAL.'BASE_A' = CHAM1 ;
TINIT = TABLE 'INITIAL' ;
TINIT.'VITESSE' = manu chpo ppl 1 alfa w0 ;
*
** definition of the linkage
TLIA = TABLE 'LIAISON' ;
TLB = TABLE 'LIAISON_B' ;
TLIA.'LIAISON_B' = TLB ;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'POINT_PLAN' ;
TL1.'SUPPORT' = P1 ;
TL1.'NORMALE' = (1 0 0) ;
TL1.'RAIDEUR' = KE ;
TL1.'JEU' = 0. ;
t11.'SEUIL_PLASTIQUE' = sel ;
TL1.'AMORTISSEMENT' = 0. ;
TLB . 1 = TL1 ;
*
** definition of output
TSORT = TABLE 'SORTIE' ;
TSORL = TABLE 'LIAISON_B' ;
TSORT.'LIAISON_B' = TSORL ;
TSORL.TL1 = VRAI ;
*
** number of steps
npas = (enti (t1/dt)) ;
npas = npas / 2 ;
nins = 1 ;
*
** 1st computation block
tres1 = dyne de_vogelaere tlia tabl tchal tsort
      tinit npas dt nins ;
TTL1 = TRES1.TL1 ;
LIS_TEM0 = TRES1.'TEMPS_DE_SORTIE' ;
lisxpl = TTL1.'UX' ;
lisyp1 = t11 . 'FORCE_DE_CHOC' ;
titre 'force - displacement' ;
evdf1 = evol manu 'displ' lisxpl 'force' lisyp1 ;
titre 'plastic displacement - time' ;
evdp1 = evol manu 'time' lis_tem0 'disp_pl'
      ttl1.deplacement_plastique ;
*
** restart of the computation
tinit = table initial ;
tinit.reprise = tres1.reprise ;
tres2 = dyne de_vogelaere tlia tabl tchal tsort tinit
      npas dt nins ;
TTL2 = TRES2.TL1 ;
LIS_TEM2 = TRES2.'TEMPS_DE_SORTIE' ;
lisxp2 = TTL2.'UX' ;
lisyp2 = t11 . 'FORCE_DE_CHOC' ;
titre 'force - displacement' ;
evdf2 = evol rouge manu 'displ' lisxp2 'force' lisyp2 ;
titre 'plastic displacement - time' ;
evdp2 = evol rouge manu 'time' lis_tem2 'disp_pl'
      ttl2.deplacement_plastique ;
*
titre 'force - displacement' ;
dess (evdf1 et evdf2) ;
titre 'plastic displacement - time' ;
dess (evdp1 et evdp2) ;
*
** maxima of theoretical plastic displacements
d11 = .5 ; d22 = 1.5 ;
mess 'max of theoretical plastic displ.' d11 and' d22 ;
d1 = maxi (extr evdp1 ordo) ;
d2 = maxi (extr evdp2 ordo) ;
mess 'max of CASTEM2000 plastic displ.' d1 and' d2 ;
er1 = abs(d1/d11 - 1) ; er2 = abs (d2/d22 - 1) ;
mess 'margin 'er1%' and 'er2%';
err = maxi (prog er1 er2) ;
si (err > .005) ;
erre 5 ;
sinon ;
erre 0 ;
finsi ;
fin ;

```

Test dy_devo6 (Comments)

1. Definition of initial conditions

```
TINIT = TABLE 'INITIAL' ;
TINIT.'VITESSE' = manu chpo pp1 1 alfa w0 ;
```

For this first block of the computation initial values must be given in TINIT table of INITIAL subtype. The load is defined on A basis, then the component of the initial values is alfa (A basis is related to uncoupled equations).

2. Definition of the linkage

```
TLIA = TABLE 'LIAISON' ;
TLB = TABLE 'LIAISON_B' ;
TLIA.'LIAISON_B' = TLB ;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'POINT_PLAN' ;
TL1.'SUPPORT' = P1 ;
TL1.'NORMALE' = (1 0 0);
TL1.'RAIDEUR' = KE ;
TL1.'JEU' = 0.;
tl1.'SEUIL_PLASTIQUE' = se1 ;
TL1.'AMORTISSEMENT'=0. ;
TLB. 1 = TL1 ;
```

TLIA is a table of subtype LIAISON.

TLIA.'LIAISON_B' (or TLB) is a table which gathers all linkages on B basis. This table has as many indexes as the number of linkages.

TLB.1 (or TL1) defines an elementary linkage. It is a table of 'LIAISON_ELEMENTAIRE' subtype. The first index to be specified is the type of the linkage (here 'POINT_PLAN').

The POINT-PLAN linkage allows to model the impact of a point onto a semi-infinite medium which is delimited by a fixed plane. The linkage is given in form of an imaginary spring which is defined by a stiffness (TL1.'RAIDEUR'), a yield limit (tl1.'SEUIL_PLASTIQUE') in case of plastification and a damping (TL1.'AMORTISSEMENT') in case of damping.

Other data is related to the impacting point (TL1.'SUPPORT'), the initial gap between the point and the fixed plane (TL1.'JEU'), the normal perpendicular to the plane and oriented towards the inside of the solid (TL1.'NORMALE').

3. Declaration of the output table

```
** definition of output
TSORT = TABLE 'SORTIE' ;
TSORL = TABLE 'LIAISON_B' ;
TSORT.'LIAISON_B' = TSORL ;
TSORL.TL1 = VRAI ;
```

TSORT is the table of SORTIE subtype.

TSORT.'LIAISON_B' (or TSORL) is a table of LIAISON_B subtype and defines all linkages on B basis to be saved.



12.10. TEST DY_DEVO6 (DESCRIPTION SHEET)

Every index of TSORL is the table defining the linkage to be saved (here TL1). The value of TSORL.TL1 is VRAI (true); thus all internal data of the linkage TL1 will be saved.

4. Output of results

```

TTL1 = TRES1.TL1 ;
LIS_TEM0 = TRES1.'TEMPS_DE_SORTIE';
lisxp1 = TTL1.'UX' ;
lisyp1 = ttl1 .'FORCE_DE_CHOC' ;
titre 'force - displacement';
evdf1 = evol manu 'displ' lisxp1 'force' lisyp1 ;
titre 'plastic displacement - time' ;
evdpl = evol manu 'time' lis_tem0 'disp_pl'
ttl1.deplacement_plastique ;
    
```

TRES.TL1 (or TTL1) is a table which contains the results related to the linkage defined in TL1 table.

- TTL1.'UX' is the list of UX displacements.
- ttl1 .'FORCE_DE_CHOC' is the list of the force of impact.
- ttl1.deplacement_plastique is the list of the plastic displacements.

5. Restart of the computation

```

tinit = table initial ;
tinit.reprise = tres1.reprise ;
tres2 = dyne de_vogelaere tlia tab1 tcha1 tsort tinit npas dt nins ;
    
```

For this second block of the computation initial values are defined in tinit.reprise. All data are given in the table tres1.reprise.



12.11 Test dy_devo3 (Description sheet)

Test name: dy_devo3

Calculation type: MECHANICS DYNAMICS 3D, IMPACT, LINKAGE CIRCLE-PLANE-FRICTION


Topic: A wheel subjected to a torsional moment. A wheel of 1 meter diameter is clamped by 2 planes. The wheel is subjected to a torsional moment. The linkages used are a circle-point with taking into consideration the friction.

Goal: Find the rotation around Z-axis.

Reference:

Version: 97' customer version


Model description:




R	$= 1m$
Gap	$= -0.00025m$
M_z	$= 1000N.m$
K_e	$= 4.10^6 N/m$
K_t	$= (10 * K_e) / R^2$
C	$= 0N.s/m$
C_t	$= \sqrt{K_t * M}$
$Coeff_{sliding}$	$= 0.4$
$Coeff_{adherence}$	$= 0.4$

Test dy_devo3 (Results)


Results	Time	0.4
	Rotation RZ	16.013

CASTEM figures



***** TEL- NORMAL IMPACT FORCE *****




***** TEL- TANGENTIAL IMPACT FORCE *****




***** TEL- WEAR POWER *****



***** TEL- VE CONTACT POINT *****



***** TEL- VE POINT OF MEDIAN LINE *****



***** TEL- RZ POINT OF MEDIAN LINE *****



Test dy_devo3 (Listing CASTEM)

```

*****
OPTION ECHO 0 ;
TEMPS ;
OPTION ELEM SEG2 DIME 3 ;
*
*   Definition of geometry
*
P1 = 0. 0. 0. ;
*
*   Modal basis
*
CHP_M1 = MANU CHPO 6 P1 'UX' 1.0 'UY' 0.0
        'UZ' 0.0 'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHP_M2 = MANU CHPO 6 P1 'UX' 0.0 'UY' 1.0
        'UZ' 0.0 'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHP_M3 = MANU CHPO 6 P1 'UX' 0.0 'UY' 0.0
        'UZ' 0.0 'RX' 0.0 'RY' 0.0 'RZ' 1.0 ;
*
TBAS = TABLE 'BASE_MODAL' ;
TBAS . 'MODES' = TABLE 'BASE_DE_MODES' ;
TBAS . 'MODES' . 'MAILLAGE' = P1 ;
TBAS . 'MODES' . 1 = TABLE 'MODE' ;
TBAS . 'MODES' . 1 . 'POINT_REPERE' = ( 0 0 0 ) ;
TBAS . 'MODES' . 1 . 'FREQUENCE' = 0.0 ;
TBAS . 'MODES' . 1 . 'MASSE_GENERALISEE' = 1.0 ;
TBAS . 'MODES' . 1 . 'DEFORMEE_MODAL' = CHP_M1 ;
TBAS . 'MODES' . 2 = TABLE 'MODE' ;
TBAS . 'MODES' . 2 . 'POINT_REPERE' = ( 0 0 0 ) ;
TBAS . 'MODES' . 2 . 'FREQUENCE' = 0.0 ;
TBAS . 'MODES' . 2 . 'MASSE_GENERALISEE' = 1.0 ;
TBAS . 'MODES' . 2 . 'DEFORMEE_MODAL' = CHP_M2 ;
TBAS . 'MODES' . 3 = TABLE 'MODE' ;
TBAS . 'MODES' . 3 . 'POINT_REPERE' = ( 0 0 0 ) ;
TBAS . 'MODES' . 3 . 'FREQUENCE' = 0.0 ;
TBAS . 'MODES' . 3 . 'MASSE_GENERALISEE' = 1.0 ;
TBAS . 'MODES' . 3 . 'DEFORMEE_MODAL' = CHP_M3 ;
*
*   Load field
*
XFOR = 1000.0 ;
CHP_F1 = MANU CHPO 3 P1 'MX' 0.0
                    'MY' 0.0
                    'MZ' XFOR ;
EVO_F1 = EVOL MANU ( PROG -1. 10. ) ( PROG 1. 1. ) ;
CHA_F1 = CHARGE 'FORC' CHP_F1 EVO_F1 ;
TCHARG = TABLE 'CHARGEMENT' ;
TCHARG . 'BASE_A' = PJBA TBAS CHA_F1 ;
*
*--- Definition of linkages
*
TLIA1 = TABLE 'LIAISON' ;
TLB1 = TABLE 'LIAISON_B' ;
TLIA1 . 'LIAISON_B' = TLB1 ;
*
*   Linkage CERCLE_PLAN_FROTTEMENT with damping
*
RAY_SUP = 1.0 ;
XNU = 0.4 ;
XRAID = 4.0E6 ;
RAID_T = ( 10.0 * XRAID ) / ( RAY_SUP ** 2 ) ;
AMOR_T = ( RAID_T * 1.0 ) ** 0.5 ;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1 . 'TYPE_LIAISON' = MOT 'CERCLE_PLAN_FROTTEMENT' ;
TL1 . 'SUPPORT' = P1 ;
TL1 . 'JEU' = -0.00025 ;
TL1 . 'NORMALE' = ( 0 -1 0 ) ;
TL1 . 'RAIDEUR' = XRAID ;
TL1 . 'RAYON_SUPPORT' = RAY_SUP ;
TL1 . 'COEFFICIENT_GLISSEMENT' = XNU ;
TL1 . 'COEFFICIENT_ADHERENCE' = XNU ;
TL1 . 'RAIDEUR_TANGENTIELLE' = RAID_T ;
TL1 . 'AMORTISSEMENT_TANGENTIEL' = AMOR_T ;
TL1 . 'AMORTISSEMENT' = 0.0 ;
TL2 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL2 . 'TYPE_LIAISON' = MOT 'CERCLE_PLAN_FROTTEMENT' ;
TL2 . 'SUPPORT' = P1 ;
TL2 . 'JEU' = -0.00025 ;
TL2 . 'NORMALE' = ( 0 1 0 ) ;
TL2 . 'RAIDEUR' = XRAID ;
TL2 . 'RAYON_SUPPORT' = RAY_SUP ;
TL2 . 'COEFFICIENT_GLISSEMENT' = XNU ;
TL2 . 'COEFFICIENT_ADHERENCE' = XNU ;
TL2 . 'RAIDEUR_TANGENTIELLE' = RAID_T ;
TL2 . 'AMORTISSEMENT_TANGENTIEL' = AMOR_T ;
TL2 . 'AMORTISSEMENT' = 0.0 ;
*
TLB1 . 1 = TL1 ;
TLB1 . 2 = TL2 ;
*
*--- Definition of output results
*
TSORT = TABLE 'SORTIE' ;
TSOR1 = TABLE 'SORTIE' ;
TSORT . 'VARIABLE' = TSOR1 ;
TSOR1 . 'DEPLACEMENT' = FAUX ;
TSOR1 . 'VITESSE' = FAUX ;
*
TSOR2 = TABLE 'SORTIE' ;
TSORT . 'LIAISON_B' = TSOR2 ;
TSOR2 . TL1 = VRAI ;
*
*   Time
*
PDT = 4.0E-5 ;
NP1 = 5000 ;
NP2 = 5000 ;
NINS = 20 ;
*
*----- Operator DYNE -----
*
TRESU1 = DYNE 'DE_VOGELAERE' TBAS TCHARG TSORT TLIA1
          NP1 PDT NINS ;
*
TINIT = TABLE 'INITIAL' ;
TINIT . 'REPRISE' = TRESU1 . 'REPRISE' ;
TRESU2 = DYNE 'DE_VOGELAERE' TBAS TCHARG TSORT TLIA1
          TINIT NP2 PDT NINS ;
*
*   Display of graph
*
LISTE1 = TRESU1 . 'TEMPS_DE_SORTIE' ;
LISTE2 = TRESU2 . 'TEMPS_DE_SORTIE' ;
*
TITRE '***** - TL1- NORMAL IMPACT FORCE *****' ;
EVO_FN1 = EVOL MANU LISTE1
          TRESU1 . TL1 . 'FORCE_DE_CHOC_NORMALE' ;
EVO_FN2 = EVOL MANU LISTE2
          TRESU2 . TL1 . 'FORCE_DE_CHOC_NORMALE' ;
EVO_FN = CONCAT EVO_FN1 EVO_FN2 ;
*
TITRE '***** - TL1- TANGENTIAL IMPACT FORCE *****' ;
EVO_FT1 = EVOL MANU LISTE1
          TRESU1 . TL1 . 'FORCE_DE_CHOC_TANGENTIELLE' ;

```

```

EVO_FT2 = EVOL MANU LISTE2
      TRESU2.TL1.'FORCE_DE_CHOC_TANGENTIELLE' ;
EVO_FT = CONCAT EVO_FT1 EVO_FT2 ;
*
TITRE '***** - TL1- WEAR POWER *****' ;
EVO_PU1 = EVOL MANU LISTE1
      TRESU1.TL1.'PUISSANCE_USURE_INSTANTANEE' ;
EVO_PU2 = EVOL MANU LISTE2
      TRESU2.TL1.'PUISSANCE_USURE_INSTANTANEE' ;
EVO_PU = CONCAT EVO_PU1 EVO_PU2 ;
*
TITRE '***** - TL1- VX CONTACT POINT *****' ;
EVO_VX1 = EVOL MANU LISTE1
      TRESU1.TL1.'VX_POINT_CONTACT' ;
EVO_VX2 = EVOL MANU LISTE2
      TRESU2.TL1.'VX_POINT_CONTACT' ;
EVO_VX = CONCAT EVO_VX1 EVO_VX2 ;
*
TITRE '***** - TL1- MZ POINT OF MEDIAN LINE *****' ;
EVO_MZ1 = EVOL MANU LISTE1
      TRESU1.TL1.'MZ_POINT_FIBRE_NEUTRE' ;
EVO_MZ2 = EVOL MANU LISTE2
      TRESU2.TL1.'MZ_POINT_FIBRE_NEUTRE' ;
EVO_MZ = CONCAT EVO_MZ1 EVO_MZ2 ;
*
TITRE '***** - TL1- RZ POINT OF MEDIAN LINE *****' ;
EVO_RZ1 = EVOL MANU LISTE1
      TRESU1.TL1.'RZ_POINT_FIBRE_NEUTRE' ;
EVO_RZ2 = EVOL MANU LISTE2
      TRESU2.TL1.'RZ_POINT_FIBRE_NEUTRE' ;
EVO_RZ = CONCAT EVO_RZ1 EVO_RZ2 ;
*
DESS EVO_FN 'MIMA' ;
DESS EVO_FT 'MIMA' ;
DESS EVO_PU 'MIMA' ;
DESS EVO_VX 'MIMA' ;
DESS EVO_MZ 'MIMA' ;
DESS EVO_RZ 'MIMA' ;
*
LIS_RZ = EXTRAIRE EVO_RZ 'ORDO' ;
LISTEM = EXTRAIRE EVO_RZ 'ABSC' ;
D = EXTRAIRE LIS_RZ 501 ;
T = EXTRAIRE LISTEM 501 ;
SAUT PAGE ;
MESS          'Value of time step :' T ;
MESS 'value of rotation around Z-axis :' D ;
SAUT 2 LIGNES ;
*
TEMPS ;
*
*   Code validation
*
DEPREF = 100. * T * T ;
RESI = ABS ( ( D - DEPREF ) / DEPREF ) ;
SI ( RESI <EG 1E-2 ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;
*
FIN ;

```



Test dy_devo3 (Comments)

1. Linkage CERCLE_PLAN_FROTTEMENT

```

TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'CERCLE_PLAN_FROTTEMENT' ;
TL1.'SUPPORT' = P1 ;
TL1.'JEU' = -0.00025 ;
TL1.'NORMALE' = ( 0 -1 0 ) ;
TL1.'RAIDEUR' = XRAID ;
TL1.'RAYON_SUPPORT' = RAY_SUP ;
TL1.'COEFFICIENT_GLISSEMENT' = XNU ;
TL1.'COEFFICIENT_ADHERENCE' = XNU ;
TL1.'RAIDEUR_TANGENTIELLE' = RAID_T ;
TL1.'AMORTISSEMENT_TANGENTIEL' = AMOR_T ;
TL1.'AMORTISSEMENT' = 0.0 ;

```

The linkage CERCLE_PLAN_FROTTEMENT is used for impacts of a circle onto a semi-infinite medium. The circle is simply described by its center point (TL1.'SUPPORT') and its radius (TL1.'RAYON_SUPPORT'). TL1.'JEU' is the gap between a point of the circle and the medium. Here the gap is negative : it means the medium penetrates into the circle.

The linkage takes into consideration the friction, so specific data must be given to describe the imaginary spring : the impact stiffness (TL1.'RAIDEUR'), the tangential stiffness (TL1.'RAIDEUR_TANGENTIELLE'), the sliding coefficient (TL1.'COEFFICIENT_GLISSEMENT'), the adherence stress coefficient (TL1.'COEFFICIENT_ADHERENCE'), the impact damping (TL1.'AMORTISSEMENT') and the tangential damping (TL1.'AMORTISSEMENT_TANGENTIEL'). Note that the impact damping is optional.

2. Use of 2 linkages

```

TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'CERCLE_PLAN_FROTTEMENT' ;
TL1.'SUPPORT' = P1 ;
...
TL1.'NORMALE' = ( 0 -1 0 ) ;
...
TL2 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL2.'TYPE_LIAISON' = MOT 'CERCLE_PLAN_FROTTEMENT' ;
TL2.'SUPPORT' = P1 ;
...
TL2.'NORMALE' = ( 0 1 0 ) ;

```

The difference between TL1 and TL2 is only made by the definition of the normals TL1.'NORMALE' and TL2.'NORMALE'.

3. Declaration of the output table

```

TSORT = TABLE 'SORTIE' ;
TSOR1 = TABLE 'SORTIE' ;
TSORT . 'VARIABLE' = TSOR1 ;
TSOR1 . 'DEPLACEMENT' = FAUX ;
TSOR1 . 'VITESSE' = FAUX ;
*
TSOR2 = TABLE 'SORTIE' ;
TSORT . 'LIAISON_B' = TSOR2 ;
TSOR2 . TL1 = VRAI ;

```

The output table will contain only the results related to TL1 linkage.

4. Output of results

```

LISTE1 = TRESU1.'TEMPS_DE_SORTIE' ;
LISTE2 = TRESU2.'TEMPS_DE_SORTIE' ;
...
TITRE '***** - TL1- RZ POINT OF MEDIAN LINE *****' ;
EVO_RZ1 = EVOL MANU LISTE1 TRESU1.TL1.'RZ_POINT_FIBRE_NEUTRE' ;
EVO_RZ2 = EVOL MANU LISTE2 TRESU2.TL1.'RZ_POINT_FIBRE_NEUTRE' ;
EVO_RZ = CONCAT EVO_RZ1 EVO_RZ2 ;
...
LIS_RZ = EXTRAIRE EVO_RZ 'ORDO' ;
LISTEM = EXTRAIRE EVO_RZ 'ABSC' ;
D = EXTRAIRE LIS_RZ 501 ;
T = EXTRAIRE LISTEM 501 ;

```

The computation is done in two parts: thus the results are separate (some in TRESU1 and others in TRESU2). The CONCAT operator performs the joining up of two objects (of EVOLUTION type).

To obtain a specific value from the EVO_RZ object, it is first necessary to extract every list of reals from the EVOLUTION object (EVO_RZ) : the EXTRAIRE operator extracts the lists of reals related to the abscissas ('ABSC') and the ordinates ('ORDO'). Then you can extract the required value from the obtained list of reals : 501 is the row of the required value in the list.



12.12 Test dy_devo4 (Description sheet)

Test name: dy_devo4

Calculation type: MECHANICS DYNAMICS 3D, IMPACT, LINKAGE SECTION-SECTION-OUTSIDE

Finite element type: POUT

Topic: Impact of a beam with concentrated mass against a plane. A concentrated mass is fixed at the extremity of a cantilever beam. A normal force is applied on this tips, this load gives the initial displacements of the beam. Two different linkages are used in this example : the PROFIL_PROFIL_SECTION linkage is tested then compared with the POINT_PLAN linkage.


Goal: Find the impact force and displacement UX at time $T = 0.4$ s.

Reference:

Version: 97' customer version


Model description:

L	$= 0.9m$	E	$= 2.1 \cdot 10^{11} Pa$
M	$= 2kg$	ν	$= 0.3$
A	$= 1.1 \cdot 10^{-6} m^2$	ρ	$= 78kg/m^3$
I_y	$= 8.33 \cdot 10^{-10} m^4$	$F1$	$= 1N$
I_z	$= 8.33 \cdot 10^{-10} m^4$	K_e	$= 1.10^5 N/m$
G	$= 5.10^{-7} m^4$	Gap	$= 10^{-3} m$




Test dy_devo4 (Results)
Results (t=0.4s)


	PROFIL_PROFIL_EXTERIEUR	POINT_PLAN
Impact force	-29.833 N	-29.833 N
Displacement UX	$1.298 \cdot 10^{-3}$ m	$1.298 \cdot 10^{-3}$ m

CASTEM figures



SIZE: PRCY




***** LINEAGE PROFIL_PROFIL_EXTERIEUR *****



***** LINEAGE PROFIL_PROFIL_EXTERIEUR *****



***** LINEAGE POINT_PLAN *****



***** LINEAGE POINT_PLAN *****



Test dy_devo4 (Listing CASTEM)

```

OPTION ECHO 0 ;
TEMPS ;
*
NB_M = 3 ;
NB_F = 1 ;
R_PPL = 1.E+05 ;
R_PP = R_PPL ;
EXPO = 0.5 ;
*
OPTION ELEM SEG2 DIME 3 ;
*
*   Definition of geometry
*
P1 = 0. 0. 0. ;
P2 = 0. 0. 0.9 ;
LIG1 = P1 D 10 P2 ;
*
*   Definition of material
*
MODL1 = MODEL LIG1 MECANIQUE ELASTIQUE POUT;
*
MA1 = MATER MODL1 YOUNG 2.1E11 NU 0.3 RHO 78. ;
CA1 = CARAC MODL1 SECT 1.1E-6 INRY 8.33E-10
      INRZ 8.33E-10 TORS 5.0E-7 VECT (1. 0. 1. ) ;
*
RIG1 = RIGI MODL1 (MA1 ET CA1) ;
MAST = (MASSE MODL1 (MA1 ET CA1)) ET
        ( MASSE DEPL 2.0 P2 ) ;
*
*   Boundary conditions
*
ENC1 = BLOQUE DEPL ROTA P1 ;
ENC2 = BLOQUE 'UY' LIG1 ;
RIGT = RIG1 ET ENC1 ET ENC2 ;
*
*   Mode shapes computation
*
MOD1=VIBRATION PROCHE ( PROG 1.0 ) RIGT MAST ;
*
*   Modal basis
*
TB1 = TRADUIRE MOD1 ;
TBAS = TABLE 'BASE_MODAL' ;
TBAS.'MODES' = TB1 ;
*
RIGI1 = RIGI TB1 ;
*
*   Damping
*
AMOR1 = AMOR TBAS ( PROG 0.0 ) ;
*
*
TA = TABLE 'AMORTISSEMENT' ;
TA.'AMORTISSEMENT' = AMOR1 ;
*
*   Initial displacement
*
FE2 = FORCE FX 1. P2 ;
XE2 = RESOU RIGT FE2 ;
XMAX = MAXI XE2 AVEC (MOTS UX ) ;
FE1 = FE2 * (0.139E-2 / XMAX ) ;
FGE1 = PJBA TBAS FE1 ;
ALO = RESO RIGI1 FGE1 ;
*
TINIT = TABLE 'INITIAL' ;
TINIT.'DEPLACEMENT' = ALO ;
*
*--- Definition of linkages

```

```

*
TLIA = TABLE 'LIAISON' ;
TLB = TABLE 'LIAISON_B' ;
TLIA.'LIAISON_B' = TLB ;
NORM1 = 0 0 1 ;
*
*   Linkage PROFIL_PROFIL_EXTERIEUR
*
*   moving section
P1_1 = -0.5 0.0 0.9 ;
P1_2 = 0.0 -0.5 0.9 ;
P1_3 = 0.5 0.0 0.9 ;
P1_4 = 0.0 0.5 0.9 ;
MAIL_1 = P1_1 D NB_M P1_2 D NB_M P1_3 D NB_M
          P1_4 D NB_M P1_1 ;
*
*   fixed section
P2_1 = 0.501 1.0 0.9 ;
P2_2 = 0.501 -1.0 0.9 ;
P2_3 = 0.601 -1.0 0.9 ;
P2_4 = 0.601 1.0 0.9 ;
MAIL_2 = P2_1 D NB_F P2_2 D NB_F P2_3 D NB_F
          P2_4 D NB_F P2_1 ;
*
TRAC ( 0 0 1000 ) ( MAIL_1 ET MAIL_2 et P2 ) ;
*
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'PROFIL_PROFIL_EXTERIEUR' ;
TL1.'SUPPORT' = P2 ;
TL1.'PROFIL_FIXE' = MAIL_2 ;
TL1.'PROFIL_MOBILE' = MAIL_1 ;
TL1.'NORMALE' = NORM1 ;
TL1.'RAIDEUR' = R_PP ;
TL1.'EXPOSANT_RAIDEUR' = EXPO ;
*
TLB . 1 = TL1 ;
*
*--- Definition of output
*
*
TSORT = TABLE 'SORTIE' ;
TSOR1 = TABLE 'SORTIE' ;
TSORT . 'VARIABLE' = TSOR1 ;
TSOR1 . 'DEPLACEMENT' = FAUX ;
TSOR1 . 'VITESSE' = FAUX ;
*
TSOR2 = TABLE 'SORTIE' ;
TSORT.'LIAISON_B' = TSOR2 ;
TSOR2.TL1 = VRAI ;
*
*   Time
*
PDT = 0.1E-3 ;
NP1 = 5000 ;
NINS = 5 ;
*
*----- Operator DYNE -----
*
TRESU1 = DYNE 'DE_VOGELAERE' TBAS TA TINIT TSORT TLIA
          NP1 PDT NINS ;
*
*   Display of graph
*
TITRE '***** LINKAGE PROFIL_PROFIL_EXTERIEUR *****' ;
DESS (EVOL MANU 'TIME' TRESU1 . 'TEMPS_DE_SORTIE'
      'IMPACT FORCE' TRESU1 . TL1 . 'FORCE_DE_CHOC') 'MIMA';
*
DESS ( EVOL MANU 'TIME' TRESU1 . 'TEMPS_DE_SORTIE'

```

```

        'UX'      TRESU1 . TL1 . 'UX' ) 'MIMA' ;
D_FC_1 = EXTRAIRE ( TRESU1 . TL1 . 'FORCE_DE_CHOC' ) 801 ;
D_DE_1 = EXTRAIRE ( TRESU1 . TL1 . 'UX' ) 801 ;
*
SAUT PAGE ;
MESS '-----> linkage PROFIL_PROFIL_EXTERIEUR' ;
MESS '          value of impact force : ' D_FC_1 ;
MESS '          value of displacement : ' D_DE_1 ;
*
*   Linkage POINT_PLAN
*
NORM1 = 1 0 0 ;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'POINT_PLAN' ;
TL1.'SUPPORT'      = P2 ;
TL1.'NORMALE'      = NORM1 ;
TL1.'RAIDEUR'      = R_PPL ;
TL1.'JEU'          = 0.001 ;
*
TLB . 1 = TL1 ;
TSOR2.TL1 = VRAI ;
*
TRESU2 = DYNE 'DE_VOGELAERE' TBAS TA TINIT TSORT TLIA NP1 PDT NINS ;
*
*   Display of graph
*
TITRE '***** LINKAGE POINT_PLAN *****' ;
DESS ( EVOL MANU 'TIME' TRESU2 . 'TEMPS_DE_SORTIE'
      'IMPACT FORCE' TRESU2 . TL1 . 'FORCE_DE_CHOC' ) 'MIMA' ;
*
DESS ( EVOL MANU 'TIME' TRESU2 . 'TEMPS_DE_SORTIE'
      'UX' TRESU2 . TL1 . 'UX' ) 'MIMA' ;
*
D_FC_2 = EXTRAIRE ( TRESU2 . TL1 . 'FORCE_DE_CHOC' ) 801 ;
D_DE_2 = EXTRAIRE ( TRESU2 . TL1 . 'UX' ) 801 ;
SAUT 2 LIGNES ;
MESS '-----> linkage POINT_PLAN ' ;
MESS '          value of impact force : ' D_FC_2 ;
MESS '          value of displacement : ' D_DE_2 ;
SAUT 3 LIGNES ;
*
TEMPS ;
*
*   Code validation
*
R_FC = ABS ( ( D_FC_1 - D_FC_2 ) / D_FC_2 ) ;
R_DE = ABS ( ( D_DE_1 - D_DE_2 ) / D_DE_2 ) ;
SI ( ( R_FC <EG 1E-2 ) OU ( R_DE <EG 1E-2 ) ) ;
  ERRE 0 ;
SINON ;
  ERRE 5 ;
FINSI ;
*
FIN ;

```




Test dy_devo4 (Comments)

1. Creation of the table of BASE_MODALE subtype

```
TB1 = TRADUIRE MOD1 ;
TBAS = TABLE 'BASE_MODALE' ;
TBAS.'MODES' = TB1 ;
```

The DYNE operator uses a BASE_MODALE subtype table but the VIBR operator creates a SOLUTION object if keyword TBAS is not specified. Also it is necessary to create a BASE_MODALE subtype table from a SOLUTION object of MODE subtype.

First of all, the TRADUIRE operator creates a BASE_DE_MODES subtype table (TB1) from a SOLUTION object (MOD1). After the creation of a BASE_MODALE subtype table (TBAS), TB1 is given in TBAS.'MODES'.

2. Generalized stiffnesses

```
RIGI1 = RIGI TB1 ;
```

RIGI1 is the stiffness due to the modes contained in TB1 (BASE_DE_MODES subtype table).

3. Initial displacement

```
FGE1 = PJBA TBAS FE1 ;
ALO = RESO RIGI1 FGE1 ;
```

ALO is a CHPOINT object which has only one component (ALPHA). Indeed this object was created with FGE1 load (of modal component) and RIGI1 (generalized stiffness).

4. Linkage PROFIL_PROFIL_EXTERIEUR

```
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'PROFIL_PROFIL_EXTERIEUR' ;
TL1.'SUPPORT' = P2 ;
TL1.'PROFIL_FIXE' = MAIL_2 ;
TL1.'PROFIL_MOBILE' = MAIL_1 ;
TL1.'NORMALE' = NORM1 ;
TL1.'RAIDEUR' = R_PP ;
TL1.'EXPOSANT_RAIDEUR' = EXPO ;
TLB . 1 = TL1 ;
```

The linkage PROFIL_PROFIL_EXTERIEUR is used for impacts of a moving section onto a fixed section.

The moving section (TL1.'PROFIL_MOBILE') is connected to the structure by a point (TL1.'SUPPORT') and is outside the fixed section (TL1.'PROFIL_FIXE'). The sections must be oriented anti clockwise and in rest position both the mesh sections must have any intersection point.

The given normal (TL1.'NORMALE', N1 in the model description) must be perpendicular to the plane composed by both sections.

The normal impact force F is given by : $F = -K * A^b$
where K is the impact stiffness (TL1.'RAIDEUR'),

A is the section surface which passed through fixed section,
b is the stiffness exponent (TL1.'EXPOSANT_RAIDEUR').

5. Linkage POINT_PLAN

To obtain the same configuration as the PROFIL_PROFIL_EXTERIEUR linkage, the value of the gap must be equal to 0.001 (the gap along X-axis between moving section and fixed section was of 0.001 too) and the normal must be oriented along X-axis (N2 in the model description).



12.13 Test dy_devo8 (Description sheet)

Test name: dy_devo8

Calculation type: MECHANICS DYNAMICS 3D, IMPACT, LINKAGE POINT-MOVING CIRCLE, LINKAGE LINE-LINE-FRICTION

Finite element type:


Topic: Impact of a thrust against a moving circle. A point impacts a moving circle. Two different linkages are used and compared in this test. The POINT_CERCLE_MOBILE linkage gives the impact of a point onto a mobile circle, taking into consideration the dry friction in the tangent plane of contact. The LIGNE_LIGNE_FROTTEMENT gives the impact of a slave line against a master line.

Goal: Find displacements UX and UY of the thrust. Comparison between both methods.

Reference:

Version: 97' customer version





Model description:



R	$= 0.04m$
U_o	$= 5m/s$
V_o	$= 10m/s$
K_e	$= 1.10^8 N/m$
K_t	$= -100N/m$
C_t	$= 0.01N.s/m$
$Coe_{f_{sliding}}$	$= 20$
$Coe_{f_{adherence}}$	$= 20$

Test dy_devo8 (Results)

CASTEM figures





Test dy_devo8 (Listing CASTEM)

```

*****
OPTION ECHO 0 ;
OPTION ELEM SEG2 DIME 3 ;
*****
* PROBLEM *
* Comparison test between *
* linkages POINT_CERCLE_MOBILE *
* and LIGNE_LIGNE_FROTTEMENT. *
* *
* Impact of a thrust *
* onto a moving circle. *
* *
*****
OC = 0. 0. 0. ;
A = 0.04 0. 0. ;
B = 0. 0.04 0. ;
N = 0. 0. 1. ;
C1 = CERC 100 B OC A ;
C2 = C1 TOUR -90. OC N ;
C3 = C1 ET C2 ;
C4 = C3 TOUR -180. OC N ;
C5 = C3 ET C4 ;
OB = 0.01 0.038 0. ;
CE1 = ELIM C5 0.0001 ;
PCEX = 0 0 0 ;
PCEY = 0 0 0 ;
PCEZ = 0 0 0 ;
PBX = 0 0 0 ;
PBY = 0 0 0 ;
PCX = 0 0 0 ;
PCY = 0 0 0 ;
N1 = 0. 0. 1. ;
OEIL = 1. 1. 50. ;
*****
* MODAL BASIS *
*****
MAI1 = (CE1 ET OB) ;
CHPCE1 = MANU CHPO 6 (CE1 ET OC) 'UX' 1.0 'UY' 0.0
          'UZ' 0.0 'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHPCE2 = MANU CHPO 6 (CE1 ET OC) 'UX' 0.0 'UY' 1.0
          'UZ' 0.0 'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHPCE3 = MANU CHPO 6 (CE1 ET OC) 'UX' 0.0 'UY' 0.0
          'UZ' 0.0 'RX' 0.0 'RY' 0.0 'RZ' 1.0 ;
CHPOB1 = MANU CHPO 6 OB 'UX' 1.0 'UY' 0.0 'UZ' 0.0
          'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHPOB2 = MANU CHPO 6 OB 'UX' 0.0 'UY' 1.0 'UZ' 0.0
          'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHPOC1 = MANU CHPO 6 OC 'UX' 1.0 'UY' 0.0 'UZ' 0.0
          'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;
CHPOC2 = MANU CHPO 6 OC 'UX' 0.0 'UY' 1.0 'UZ' 0.0
          'RX' 0.0 'RY' 0.0 'RZ' 0.0 ;

* POINT LIGNE_FROTTEMENT
Z=0 0 0;
TBAS1 = TABLE 'BASE_MODAL' ;
TBAS1 . 'MODES' = TABLE 'BASE_DE_MODAL' ;
TBAS1 . 'MODES'. 'MAILLAGE' = (CE1 ET OC);
TBAS1 . 'MODES'. 1 = TABLE 'MODE' ;
TBAS1 . 'MODES'. 1 . 'POINT_REPERE' = PCEX ;
TBAS1 . 'MODES'. 1 . 'FREQUENCE' = 0.0 ;
TBAS1 . 'MODES'. 1 . 'MASSE_GENERALISEE' = 40. ;
TBAS1 . 'MODES'. 1 . 'DEFORMEE_MODAL' = CHPCE1 ;
TBAS1 . 'MODES'. 2 = TABLE 'MODE' ;
TBAS1 . 'MODES'. 2 . 'POINT_REPERE' = PCEY ;
TBAS1 . 'MODES'. 2 . 'FREQUENCE' = 0.0 ;
TBAS1 . 'MODES'. 2 . 'MASSE_GENERALISEE' = 40. ;
TBAS1 . 'MODES'. 2 . 'DEFORMEE_MODAL' = CHPCE2 ;
TBAS1 . 'MODES'. 3 = TABLE 'MODE' ;
TBAS1 . 'MODES'. 3 . 'POINT_REPERE' = PCEZ ;

TBAS1 . 'MODES'. 3 . 'FREQUENCE' = 0.0 ;
TBAS1 . 'MODES'. 3 . 'MASSE_GENERALISEE' = 0.4 ;
TBAS1 . 'MODES'. 3 . 'DEFORMEE_MODAL' = CHPCE3 ;
TBAS1 . 'MODES'. 3 . 'CORPS_RIGIDE' = 'VRAI' ;
TBAS1 . 'MODES'. 3 . 'CENTRE_DE_GRAVITE' = OC ;
TBAS2 = TABLE 'BASE_MODAL' ;
TBAS2 . 'MODES' = TABLE 'BASE_DE_MODAL' ;
TBAS2 . 'MODES'. 'MAILLAGE' = (OB et Z);
TBAS2 . 'MODES'. 1 = TABLE 'MODE' ;
TBAS2 . 'MODES'. 1 . 'POINT_REPERE' = PBX ;
TBAS2 . 'MODES'. 1 . 'FREQUENCE' = 0.0 ;
TBAS2 . 'MODES'. 1 . 'MASSE_GENERALISEE' = 40. ;
TBAS2 . 'MODES'. 1 . 'DEFORMEE_MODAL' = CHPOB1 ;
TBAS2 . 'MODES'. 2 = TABLE 'MODE' ;
TBAS2 . 'MODES'. 2 . 'POINT_REPERE' = PBY ;
TBAS2 . 'MODES'. 2 . 'FREQUENCE' = 0.0 ;
TBAS2 . 'MODES'. 2 . 'MASSE_GENERALISEE' = 40. ;
TBAS2 . 'MODES'. 2 . 'DEFORMEE_MODAL' = CHPOB2 ;
TBAST1 = TABLE 'ENSEMBLE_DE_BASIS';
TBAST1.1=TBAS1;
TBAST1.2=TBAS2;
* POINT CERCLE_FROTTEMENT
MAI2 = OC ET OB ;
TBAS3 = TABLE 'BASE_MODAL' ;
TBAS3 . 'MODES' = TABLE 'BASE_DE_MODAL' ;
TBAS3 . 'MODES'. 'MAILLAGE' = (OC et z);
TBAS3 . 'MODES'. 1 = TABLE 'MODE' ;
TBAS3 . 'MODES'. 1 . 'POINT_REPERE' = PCX ;
TBAS3 . 'MODES'. 1 . 'FREQUENCE' = 0.0 ;
TBAS3 . 'MODES'. 1 . 'MASSE_GENERALISEE' = 40. ;
TBAS3 . 'MODES'. 1 . 'DEFORMEE_MODAL' = CHPOC1 ;
TBAS3 . 'MODES'. 2 = TABLE 'MODE' ;
TBAS3 . 'MODES'. 2 . 'POINT_REPERE' = PCY ;
TBAS3 . 'MODES'. 2 . 'FREQUENCE' = 0.0 ;
TBAS3 . 'MODES'. 2 . 'MASSE_GENERALISEE' = 40. ;
TBAS3 . 'MODES'. 2 . 'DEFORMEE_MODAL' = CHPOC2 ;
TBAST2 = TABLE 'ENSEMBLE_DE_BASIS';
TBAST2.1 = TBAS3;
TBAST2.2 =TBAS2;
*****
* LOAD *
*****
* TIME *
*****
PDT = 5.E-5 ;
NPDT = 405 ;
NPDT2 = 200 ;
NINS = 1 ;
*****
* INITIAL CONDITIONS *
*****
U0 = 5. ;
V0 = 10. ;
TINIT = TABLE 'INITIAL' ;
TINIT.'VITESSE'=(MANU CHPO PBX 1 alfa U0 'NATURE'
'DIFFUS') ET (MANU CHPO PBY 1 alfa V0 'NATURE'
'DIFFUS');
*****
* DAMPING *
*****
AMOR1 = AMOR TBAST1 (PROG 0. 0. 0. 0. 0. ) ;
TA1 = TABLE 'AMORTISSEMENT' ;
TA1.'AMORTISSEMENT' = AMOR1 ;
AMOR2 = AMOR TBAST2 (PROG 0. 0. 0. 0. 0. ) ;
TA2 = TABLE 'AMORTISSEMENT' ;
TA2.'AMORTISSEMENT' = AMOR2 ;
*****
* LINKAGE LIGNE_LIGNE_FROTTEMENT*

```

```

*****
TLIA1 = TABLE 'LIAISON' ;
TLB1 = TABLE 'LIAISON_B' ;
TLIA1 .'LIAISON_B' = TLB1 ;
CHRA=MANU 'CHPO' 1 OB 'RAIN' 100000000.;
CHAM=MANU 'CHPO' 1 OB 'AMON' 63245.5;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON'= MOT 'LIGNE_LIGNE_FROTTEMENT';
TL1.'LIGNE_MAITRE' = CE1 ;
TL1.'LIGNE_ESCLAVE' = OB ;
TL1.'COEFFICIENT_GLISSEMENT' = 20.0 ;
TL1.'COEFFICIENT_ADHERENCE' = 20.0 ;
TL1.'RAIDEUR_TANGENTIELLE' = -100. ;
TL1.'RAIDEURS' = CHRA;
TL1.'AMORTISSEMENT_TANGENTIEL' = 0.01 ;
*TL1.'AMORTISSEMENTS' =CHAM;
TL1.'NORMALE' = N1;
TL1.'SYMETRIE' = 'GLOBALE';
TLB1.1 = TL1 ;
*****
* LINKAGE POINT_CERCLE_MOBILE *
*****
TLIA2 = TABLE 'LIAISON' ;
TLB2 = TABLE 'LIAISON_B' ;
TLIA2 .'LIAISON_B' = TLB2 ;
TL2 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL2.'TYPE_LIAISON'= MOT 'POINT_CERCLE_MOBILE';
TL2.'POINT' = OB ;
TL2.'CERCLE' = OC ;
TL2.'COEFFICIENT_GLISSEMENT' = 20.0 ;
TL2.'COEFFICIENT_ADHERENCE' = 20.0 ;
TL2.'RAIDEUR_TANGENTIELLE' = -100. ;
TL2.'RAIDEUR' = 100000000. ;
TL2.'AMORTISSEMENT_TANGENTIEL' = 0.01 ;
*TL2.'AMORTISSEMENT' =63245.5 ;
TL2.'NORMALE' = N1 ;
TL2.'RAYON' = 0.04 ;
TLB2.1 = TL2 ;
*****
* OUTPUT *
*****
TSORT1 = TABLE 'SORTIE' ;
TSORT2 = TABLE 'SORTIE' ;
TSOR1 = TABLE 'SORTIE' ;
TSORT1.'VARIABLE' = TSOR1 ;
TSORT2.'VARIABLE' = TSOR1 ;
TSOR1.'DEPLACEMENT' = VRAI ;
TSOR1.'VITESSE' = FAUX ;
TSOR2 = TABLE 'SORTIE' ;
TSORT1.'LIAISON_B' = TSOR2 ;
TSOR2.TL1 = FAUX ;
TSOR3 = TABLE 'SORTIE' ;
TSORT2.'LIAISON_B' = TSOR3 ;
TSOR3.TL2 = FAUX ;
*****
* OPERATOR DYNE *
*****
TRESU11 = DYNE 'DE_VOGELAERE' TBAST1 TA1 TSORT1
          TLIA1 TINIT NPDT PDT NINS ;
TRESU2 =DYNE 'DE_VOGELAERE' TBAST2 TA2 TSORT2
          TLIA2 TINIT NPDT PDT NINS ;
EV5 = EVOL 'RECO' TRESU11 TBAST1 'DEPL' OB 'UX';
EV6 = EVOL 'RECO' TRESU11 TBAST1 'DEPL' OB 'UY';
EV7 = EVOL 'RECO' TRESU2 TBAST2 'DEPL' OB 'UX';
EV8 = EVOL 'RECO' TRESU2 TBAST2 'DEPL' OB 'UY';
DESS EV5 MIMA TITRE 'UX DISPLACEMENT OF OB :
          LIGNE_LIGNE_FROTTEMENT';
DESS EV6 MIMA TITRE 'UY DISPLACEMENT OF OB :
          LIGNE_LIGNE_FROTTEMENT';
DESS EV7 MIMA TITRE 'UX DISPLACEMENT OF OB :
          POINT_CERCLE_MOBILE';
DESS EV8 MIMA TITRE 'UY DISPLACEMENT OF OB :
          POINT_CERCLE_MOBILE';
          POINT_CERCLE_MOBILE';
TX1 = EXTR EV5 'ORDO';
TX2 = EXTR EV7 'ORDO';
TY1 = EXTR EV6 'ORDO';
TY2 = EXTR EV8 'ORDO';
NFOIS=NPDT+1;
max_x = 0.;
max_y = 0.;
*
* SEARCH OF MARGIN BETWEEN BOTH METHODS
REPETER BLOC1 NFOIS;
x1 = EXTR TX1 (&BLOC1);
x2 = EXTR TX2 (&BLOC1);
y1 = EXTR TY1 (&BLOC1);
y2 = EXTR TY2 (&BLOC1);
ecart_x = ABS(x1 - x2);
ecart_y = ABS(y1 - y2);
SI (ecart_x > max_x);
max_x = ecart_x;
FINSI;
SI (ecart_y > max_y);
max_y = ecart_y;
FINSI;
FIN BLOC1;
SI (max_x > 0.0035);
ERREUR 5;
SINON;
SI (max_y > 0.0025);
ERREUR 5;
SINON;
ERREUR 0;
FINSI;
FINSI;
FIN ;

```



Test dy_devo8 (Comments)

1. Linkage LIGNE_LIGNE_FROTTEMENT

```

TLIA1 = TABLE 'LIAISON' ;
TLB1 = TABLE 'LIAISON_B' ;
TLIA1.'LIAISON_B' = TLB1 ;
CHRA = MANU 'CHPO' 1 OB 'RAIN' 100000000.;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL1.'TYPE_LIAISON' = MOT 'LIGNE_LIGNE_FROTTEMENT';
TL1.'LIGNE_MAITRE' = CE1 ;
TL1.'LIGNE_ESCLAVE' = OB ;
TL1.'COEFFICIENT_GLISSEMENT' = 20.0 ;
TL1.'COEFFICIENT_ADHERENCE' = 20.0 ;
TL1.'RAIDEUR_TANGENTIELLE' = -100. ;
TL1.'RAIDEURS' = CHRA ;
TL1.'AMORTISSEMENT_TANGENTIEL' = 0.01 ;
TL1.'NORMALE' = N1 ;
TL1.'SYMETRIE' = 'GLOBALE';
TLB1.1 = TL1 ;

```

The LIGNE_LIGNE_FROTTEMENT linkage is used for impacts of a slave line (TL1.'LIGNE_ESCLAVE') against a master line (TL1.'LIGNE_MAITRE'). The slave line can be restricted to one point like here. The vector N1 (TL1.'NORMALE') represents the normal perpendicular to the contact plane. The computation tests the penetration of the slave nodes in the master surface (TL1.'SYMETRIE').

The linkage takes into consideration the friction, also specific data must be given : the sliding coefficient (TL1.'COEFFICIENT_GLISSEMENT'), the adherence stress coefficient (TL1.'COEFFICIENT_ADHERENCE'), the tangential stiffness (TL1.'RAIDEUR_TANGENTIELLE'), the tangential damping (TL1.'AMORTISSEMENT_TANGENTIEL'). The impact stiffness (TL1.'RAIDEURS') is given in CHPOINT form (if the slave line is not restricted to one point, it is necessary to define the impact stiffness for each slave node).

2. Linkage POINT_CERCLE_MOBILE

```

TLIA2 = TABLE 'LIAISON' ;
TLB2 = TABLE 'LIAISON_B' ;
TLIA2.'LIAISON_B' = TLB2 ;
TL2 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL2.'TYPE_LIAISON' = MOT 'POINT_CERCLE_MOBILE';
TL2.'POINT' = OB ;
TL2.'CERCLE' = OC ;
TL2.'COEFFICIENT_GLISSEMENT' = 20.0 ;
TL2.'COEFFICIENT_ADHERENCE' = 20.0 ;
TL2.'RAIDEUR_TANGENTIELLE' = -100. ;
TL2.'RAIDEUR' = 100000000. ;
TL2.'AMORTISSEMENT_TANGENTIEL' = 0.01 ;
TL2.'NORMALE' = N1 ;
TL2.'RAYON' = 0.04 ;
TLB2.1 = TL2 ;

```

The POINT_CERCLE_MOBILE linkage is used for impacts of a point (TL2.'POINT') against a mobile

circle. The circle is simply described by its center (TL2.'CERCLE') and its radius (TL2.'RAYON'). The given normal (TL2.'NORMALE') is perpendicular to the circle plane.

The linkage takes into consideration the friction. Specific data are the same as the previous linkage except for the impact stiffness which is here a real (because it is related to one point).

3. Loop counter

```
REPETER BLOC1 NFOIS;  
x1 = EXTR TX1 (&BLOC1);  
x2 = EXTR TX2 (&BLOC1);  
y1 = EXTR TY1 (&BLOC1);  
y2 = EXTR TY2 (&BLOC1);  
...  
FIN BLOC1 ;
```

&BLOC1 is an integer which contains the number of times the loop BLOC1 has been in execution, it can therefore be used as a loop counter.



12.14 Test dyne01 (Description sheet)

Test name: dyne01

Calculation type: MECHANICS DYNAMICS 3D, POLYNOMIAL LINKAGE

Finite element type:


Topic: Using a polynomial linkage and test of restart computation. The polynomial linkage calculates a force applied to a mode i and composed of a set of contributions stemming from other modes. There is no geometry in this test, the computation is done in a modal basis, the used points are only modal points.

Goal: Comparison of displacement, velocity and acceleration between both computations.

Reference:

Version: 97' customer version







Model description:



$$\begin{aligned}
 Gap &= 1 \\
 K_e &= 0 \\
 C &= 0 \\
 U_o &= 0.033 \\
 V_o &= 0.54 \\
 \dot{U}_o &= 5 \\
 \dot{V}_o &= -0.9
 \end{aligned}$$

Test dyne01 (Results)

CASTEM figures





Test dyne01 (Listing CASTEM)

```

*****
TITRE 'DYNE with POLYNOMIALE linkage : test
      of computation restart' ;
OPTION DIME 3 MODE TRID ECHO 0 ;
*
* Creation of modal points
*
P0 = 0. 0. 0. ;
PX = 0. 0. 0. ;
PY = 0. 0. 0. ;
*
* Data
*
MG = 1.25 ;
FR = 5.23 ;
T = 1. / FR ;
DEFMX = MANU 'CHPO' P0 3 UX 1. UY 0. UZ 0. ;
DEFMY = MANU 'CHPO' P0 3 UX 0. UY 1. UZ 0. ;
*
* Table of modal basis
*
TAB1 = TABLE 'BASES_DE_MODE' ;
TABM01 = TABLE 'MODE' ;
TABM01.'POINT_REPERE' = PX ;
TABM01.'FREQUENCE' = FR ;
TABM01.'MASSE_GENERALISEE' = MG ;
TABM01.'DEFORMEE_MODAL' = DEFMY ;
TAB1.1 = TABM01 ;
*
TABM02 = TABLE 'MODE' ;
TABM02.'POINT_REPERE' = PY ;
TABM02.'FREQUENCE' = FR ;
TABM02.'MASSE_GENERALISEE' = MG ;
TABM02.'DEFORMEE_MODAL' = DEFMY ;
TAB1.2 = TABM02 ;
*
TAB1.'MAILLAGE' = P0 ;
TBAS = TABLE 'BASE_MODAL' ;
TBAS.'MODES' = TAB1 ;
*
* Table of initial conditions
*
CPID = (MANU 'CHPO' PX 1 ALFA .033 ) +
      (MANU 'CHPO' PY 1 ALFA .54 ) ;
CPIV = (MANU 'CHPO' PX 1 ALFA 5.) +
      (MANU 'CHPO' PY 1 ALFA -0.9) ;
TINIT = TABLE 'INITIAL' ;
TINIT.'DEPLACEMENT' = CPID ;
TINIT.'VITESSE' = CPIV ;
TINIT2 = TINIT ;
*
* Tables of linkages
*
EPS1 = -1.E-9 ;
TLIAI = TABLE 'LIAISON' ;
TLA = TABLE 'LIAISON_A' ;
*
TLB = TABLE 'LIAISON_B' ;
TLIAI.'LIAISON_B' = TLB ;
TLB1 = TABLE 'LIAISON_ELEMENTAIRE' ;
TLB1.'TYPE_LIAISON' = MOT 'POINT_PLAN' ;
TLB1.'SUPPORT' = PX ;
TLB1.'NORMALE' = (1.0 0.0 0.0) ;
TLB1.'RAIDEUR' = 0. ;
TLB1.'JEU' = 1.0 ;
TLB1.'AMORTISSEMENT' = 0. ;
TLB.1 = TLB1 ;
*
TLIAI.'LIAISON_A' = TLA ;
TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
*
TL1.'TYPE_LIAISON' = MOT 'POLYNOMIALE' ;
TL1.'SUPPORT' = PX ;
TL1.'COEFFICIENT' = .67 ;
TL2 = TABLE 'CONTRIBUTION' ;
TL1.PY = TL2 ;
TL2.'EXPOSANT_DEPLACEMENT' = 1.2 ;
TL2.'RETARD_DEPLACEMENT' = 0.01 ;
TL2.'EXPOSANT_VITESSE' = 0. ;
TL2.'RETARD_VITESSE' = 0. ;
TL2.'JEU_DEPLACEMENT' = 0. ;
TL2.'JEU_VITESSE' = 0. ;
TLA.1 = TL1 ;
TL3 = TABLE 'LIAISON_ELEMENTAIRE' ;
TL3.'TYPE_LIAISON' = MOT 'POLYNOMIALE' ;
TL3.'SUPPORT' = PX ;
TL3.'COEFFICIENT' = .67 ;
TL4 = TABLE 'CONTRIBUTION' ;
TL3.PY = TL4 ;
TL4.'EXPOSANT_DEPLACEMENT' = 1.2 ;
TL4.'RETARD_DEPLACEMENT' = 0. ;
TL4.'EXPOSANT_VITESSE' = 0. ;
TL4.'RETARD_VITESSE' = 0. ;
TL4.'JEU_DEPLACEMENT' = EPS1 ;
TL4.'JEU_VITESSE' = 0. ;
TLA.2 = TL3 ;
*
* Output table for generalized data
*
TSORT = TABLE 'SORTIE' ;
TSORV = TABLE 'VARIABLE' ;
TSORV.'DEPLACEMENT' = vrai ;
TSORV.'VITESSE' = vrai ;
TSORV.'ACCELERATION' = vrai ;
TSORT.'VARIABLE' = TSORV ;
*
* Output table for A-basis linkage data
*
TSORL = TABLE 'LIAISON_A' ;
TSORL.TL1 = TABLE ;
TSORL.TL1.'FORCE_POLYNOMIALE' = vrai ;
TSORL.TL3 = TABLE ;
TSORL.TL3.'FORCE_POLYNOMIALE' = vrai ;
TSORT.'LIAISON_A' = TSORL ;
*
* Output table for B-basis linkage data
*
TSORB = TABLE 'LIAISON_B' ;
TSORB.TLB1 = TABLE ;
TSORB.TLB1.'FORCE_DE_CHOC' = VRAI ;
TSORT.'LIAISON_B' = TSORB ;
*
* Time values : NT & DT
*
NT = 1200 ;
DT = 0.0025 ;
TT = NT * DT ;
NPC = 5 ;
LST = PROG 0. PAS (DT * NPC) NPAS (NT / NPC) ;
*
*-----> calling DYNE with NT time steps
*
TRES1 = DYNE 'DE_VOGELAERE'
      TBAS TLIAI TINIT NT DT NPC TSORT ;
*
*-----> calling DYNE with NT/2 time steps
*
      then restart with NT/2 additional steps

```

```

*
NT2 = ENTIER(NT/2) ;
TRES21 = DYNE 'DE_VOGELAERE'
      TBAS TLIAI TINIT NT2 DT NPC TSPORT ;
*
TREP = TRES21.'REPRISE' ;
TINIT.'REPRISE' = TREP ;
*
TRES22 = DYNE 'DE_VOGELAERE'
      TBAS TLIAI TINIT NT2 DT NPC TSPORT ;
*
LIS1 = TRES1.'TEMPS_DE_SORTIE' ;
LIS21 = TRES21.'TEMPS_DE_SORTIE' ;
LIS22 = TRES22.'TEMPS_DE_SORTIE' ;
*
* Displacement / velocity / acceleration
*
I = 0 ;
NF1 = ENTIER ( NT / NPC ) + 1 ;
REPETER BOUC1 NF1 ;
  I = I + 1 ;
  CHD1 = TRES1.I.'DEPLACEMENT' ;
  DEP1 = EXTR CHD1 ALFA PX ;
  CHV1 = TRES1.I.'VITESSE' ;
  VIT1 = EXTR CHV1 ALFA PX ;
  CHA1 = TRES1.I.'ACCELERATION' ;
  ACC1 = EXTR CHA1 ALFA PX ;
  SI ( I EGA 1 ) ;
    LD1 = PROG DEP1 ;
    LV1 = PROG VIT1 ;
    LA1 = PROG ACC1 ;
  SINON ;
    LD1 = LD1 ET (PROG DEP1) ;
    LV1 = LV1 ET (PROG VIT1) ;
    LA1 = LA1 ET (PROG ACC1) ;
  FINSI ;
FIN BOUC1 ;
*
I = 0 ;
NF21 = ENTIER((NT / NPC) / 2) + 1 ;
REPETER BOUC21 NF21 ;
  I = I + 1 ;
  CHD21 = TRES21.I.'DEPLACEMENT' ;
  DEP21 = EXTR CHD21 ALFA PX ;
  CHV21 = TRES21.I.'VITESSE' ;
  VIT21 = EXTR CHV21 ALFA PX ;
  CHA21 = TRES21.I.'ACCELERATION' ;
  ACC21 = EXTR CHA21 ALFA PX ;
  SI ( I EGA 1 ) ;
    LD21 = PROG DEP21 ;
    LV21 = PROG VIT21 ;
    LA21 = PROG ACC21 ;
  SINON ;
    LD21 = LD21 ET (PROG DEP21) ;
    LV21 = LV21 ET (PROG VIT21) ;
    LA21 = LA21 ET (PROG ACC21) ;
  FINSI ;
FIN BOUC21 ;
I = 0 ;
NF22 = ENTIER((NT / NPC) / 2) ;
REPETER BOUC22 NF22 ;
  I = I + 1 ;
  CHD22 = TRES22.I.'DEPLACEMENT' ;
  DEP22 = EXTR CHD22 ALFA PX ;
  CHV22 = TRES22.I.'VITESSE' ;

  VIT22 = EXTR CHV22 ALFA PX ;
  CHA22 = TRES22.I.'ACCELERATION' ;
  ACC22 = EXTR CHA22 ALFA PX ;
  SI ( I EGA 1 ) ;
    LD22 = PROG DEP22 ;
    LV22 = PROG VIT22 ;

    LA22 = PROG ACC22 ;
  SINON ;
    LD22 = LD22 ET (PROG DEP22) ;
    LV22 = LV22 ET (PROG VIT22) ;
    LA22 = LA22 ET (PROG ACC22) ;
  FINSI ;
FIN BOUC22 ;
*
TITRE 'COMPUTATION 1 : DISPLACEMENT' ;
EVD1 = EVOL MANU 'TIME' LST 'DISPLACEMENT' LD1 ;
DESS EVD1 ;
TITRE 'COMPUTATION 2 : DISPLACEMENT' ;
EVD2 = EVOL MANU 'TIME' LST 'DISPLACEMENT' LD2 ;
DESS EVD2 ;
TITRE 'COMPUTATION 1 : VELOCITY' ;
EVD1 = EVOL MANU 'TIME' LST 'VELOCITY' LV1 ;
DESS EVD1 ;
TITRE 'COMPUTATION 2 : VELOCITY' ;
EVD2 = EVOL MANU 'TIME' LST 'VELOCITY' LV2 ;
DESS EVD2 ;
TITRE 'COMPUTATION 1 : ACCELERATION' ;
EVA1 = EVOL MANU 'TIME' LST 'ACCELERATION' LA1 ;
DESS EVA1 ;
TITRE 'COMPUTATION 2 : ACCELERATION' ;
EVA2 = EVOL MANU 'TIME' LST 'ACCELERATION' LA2 ;
DESS EVA2 ;
*
* Comparison of displacements, velocities
* and accelerations at last time step
*
DFIN1 = EXTRAIRE LD1 NF1 ;
DFIN2 = EXTRAIRE LD22 NF22 ;
ERD = ABS ( (DFIN1 - DFIN2) / DFIN1 ) ;
BOLD = (ERD <EG 0.001) ;
*
VFIN1 = EXTRAIRE LV1 NF1 ;
VFIN2 = EXTRAIRE LV22 NF22 ;
ERV = ABS ( (VFIN1 - VFIN2) / VFIN1 ) ;
BOLV = (ERV <EG 0.001) ;
*
AFIN1 = EXTRAIRE LA1 NF1 ;
AFIN2 = EXTRAIRE LA22 NF22 ;
ERA = ABS ( (AFIN1 - AFIN2) / AFIN1 ) ;
BOLA = (ERA <EG 0.001) ;
*
SI ( BOLA et BOLV et BOLD ) ;
  ERREUR 0 ;
SINON ;
  ERREUR 5 ;
FINSI ;
*
FIN ;

```



Test dyne01 (Comments)

1. Linkage POLYNOMIALE

TLIAI.'LIAISON_A' = TLA ;
 TL1 = TABLE 'LIAISON_ELEMENTAIRE' ;
 TL1.'TYPE_LIAISON' = MOT 'POLYNOMIALE' ;
 TL1.'SUPPORT' = PX ;
 TL1.'COEFFICIENT' = .67 ;
 TL2 = TABLE 'CONTRIBUTION' ;
 TL1.PY = TL2 ;
 TL2.'EXPOSANT_DEPLACEMENT' = 1.2 ;
 TL2.'RETARD_DEPLACEMENT' = 0.01 ;
 TL2.'EXPOSANT_VITESSE' = 0. ;
 TL2.'RETARD_VITESSE' = 0. ;
 TL2.'JEU_DEPLACEMENT' = 0. ;
 TL2.'JEU_VITESSE' = 0. ;
 TLA.1 = TL1 ;

The polynomial linkage calculates a force f_i applied to a mode i and composed of a set of contributions stemming from other modes j :

$$f_i = a_i [Q_j^b(t - T_j) \dot{Q}_j^c(t - T_j)] [Q_k^d(t - T_k) \dot{Q}_k^e(t - T_k)]$$

with $Q_j = \langle q_j - J_d \rangle$

$$\dot{Q}_j = \langle \dot{q}_j - J_v \rangle$$

a_i is the modal coefficient (TL1.'COEFFICIENT').

Each modal contribution is defined in a CONTRIBUTION subtype table (TL2) which contains :

- the displacement exponent b (TL2.'EXPOSANT_DEPLACEMENT')
- the displacement term of delay d (TL2.'RETARD_DEPLACEMENT')
- the velocity exponent c (TL2.'EXPOSANT_VITESSE')
- the velocity term of delay e (TL2.'RETARD_VITESSE')
- the gap connected with the displacement J_d (TL2.'JEU_DEPLACEMENT')
- the gap connected with the velocity J_v (TL2.'JEU_VITESSE')

TL2 is given in TL1.PY where PY is the modal origin point.

12.15 Test vibr2 (Description sheet)

Test name: vibr2

Calculation type: MECHANICS DYNAMICS 2D AXISYMMETRICAL

Finite element type: QUA8

Topic: Uniform radial mode shapes of a thick sphere. Eigen frequencies and shapes of a thick sphere are calculated. Only the uniform radial vibration shapes are studied. Only a sector of the sphere is meshed and computation is done with axisymmetrical mode. Boundary conditions corresponding to this meshing are obtained by using the operator RELA : nodes of similar radius have similar displacements.


Goal: Find the eigen frequencies of the sphere.

Reference: Test NAFEMS number fv42.

Version: 97' customer version

Model description:

$$\begin{array}{ll}
 R_1 = 1.8m & \rho = 8000kg/m^3 \\
 R_2 = 6.m & E = 2.10^{11}MPa \\
 \alpha = 5^\circ & \nu = 0.3
 \end{array}$$






Test vibr2 (Results)





Results

	Theoretical freq (Hz)	CASTEM freq (Hz)	margin
Shape 1	369.91	368.62	0.35 %
Shape 2	838.03	834.92	0.37 %
Shape 3	1451.2	1447.5	0.25 %
Shape 4	2117.0	2123.6	0.31 %
Shape 5	2795.8	2841.8	1.64 %

CASTEM figures



VIBR2 - MESHING



Test vibr2 (Listing CASTEM)

```

*****
OPTI ECHO 1;
*
OPTI DIME 2;
OPTI MODE AXIS;
OPTI ELEM QUA8;
*
*-----*
*      geometry - meshing      *
*-----*
*
ALPHA = 5.; COSALPHA = COS ALPHA;
SINALPHA = SIN ALPHA;
*
P0 = 0. 0.;
P1 = 1.8 ( 1.8 * SINALPHA);
P2 = 1.8 (-1.8 * SINALPHA);
P3 = 6.0 (-6.0 * SINALPHA);
P4 = 6.0 ( 6.0 * SINALPHA);
*
D23 = P2 DROI 5 P3;
*
SUR1 = D23 ROTA 1 (2 * ALPHA) P0;
D41 = COTE 3 SUR1;
*
ELIM (SUR1 ET D23 ET D41) 0.001;
*
TITR 'VIBR2 : MESHING';
TRAC QUAL SUR1;
*
*-----*
*      model - material - rigidity - mass      *
*-----*
*
MOD1 = MODEL SUR1 MECANIQUE QUA8;

MAT1 = MATER MOD1 YOUN 2.E11 NU 0.3
      RHO 8E3;
RIG1 = RIGI MOD1 MAT1;
MAS1 = MASS MOD1 MAT1;
*
*-----*
*      boundary conditions      *
*-----*
*
DIR1 = (-1 * SINALPHA) (-1 * COSALPHA);
DIR2 = (-1 * SINALPHA) COSALPHA;
*
CL1 = BLOQ DEPL DIRECTION DIR1 D23;
CL2 = BLOQ DEPL DIRECTION DIR2 D41;
CLI1 = CL1 ET CL2;
*
DIR3 = COSALPHA (-1 * SINALPHA);
DIR4 = COSALPHA SINALPHA;
*
LIST1 = LECT 27 22 17 23 18 24 19
        25 20 26 21;
LIST2 = LECT 1 6 2 7 3 8 4
        9 5 10 11;
*
I = 0;
REPETER BLOC1 11;
  I = I + 1;
  J = EXTR LIST1 I;
  K = EXTR LIST2 I;
  CL3 = RELA 1 DEPL DIRECTION DIR4
        (NOEUD J) - 1 DEPL DIRECTION
        DIR3 (NOEUD K);

  CLI1 = CLI1 ET CL3;

FIN BLOC1;
*
LIST1 = LECT 27 17 18 19 20 21;
LIST2 = LECT 28 12 13 14 15 16;
*
I = 0;
REPETER BLOC2 6;
  I = I + 1;
  J = EXTR LIST1 I;
  K = EXTR LIST2 I;
  CL4 = RELA 1 DEPL DIRECTION DIR4
        (NOEUD J) - 1 UR (NOEUD K);
  CLI1 = CLI1 ET CL4;
FIN BLOC2;
*
RIG1 = RIG1 ET CLI1;
*
*-----*
*      eigen frequencies calculation      *
*-----*
*
* Using of operator VIBR.
*
FRE1 = TABLE;
FRE1.1 = 369.91;
FRE1.2 = 838.03;
FRE1.3 = 1451.2;
FRE1.4 = 2117.0;
FRE1.5 = 2795.8;
*
LIST1 = PROG FRE1.1 FRE1.2 FRE1.3 FRE1.4
        FRE1.5;
*
RESUL = VIBR PROC LIST1 RIG1 MAS1;
*
*-----*
*      extracting results      *
*-----*
*
MESS ' RESULTS FV42.DGIBI';
MESS ' -----';
SAUT 2 LIGN;
MESS ' SPHERE IN RADIAL VIBRATION ';
SAUT 2 LIGN;
*
FRE2 = TABLE; ERG = TABLE;
DEP = TABLE; DEF = TABLE; EVO = TABLE;
*
I = 0;
REPETER BLOC3 5;
*
  I = I + 1;
  FRE2.I = TIRE RESUL FREQ RANG I;
  ERG.I = 100 * (ABS ((FRE1.I - FRE2.I)
                    / FRE1.I));
  MESS ' MODE ' I;
  MESS ' -----';
  MESS ' Theoretical eigen frequency : '
        FRE1.I 'Hz';
  MESS ' Computed eigen frequency : '
        FRE2.I 'Hz';
  MESS '      That is to say a margin of : '
        ERG.I '%';
  SAUT 1 LIGN;
*
* Eigen shapes graphes
*
DEP.I = TIRE RESUL DEPL RANG I;
    
```




12.15. TEST VIBR2 (DESCRIPTION SHEET)

```
DEFO = DEFO SUR1 DEP.I 0. ROUGE;
DEF.I = DEFO SUR1 DEP.I VERT;
TITR ' Eigen shape : Radial displacement';
TRAC (DEFO ET DEF.I);
EVO.I = EVOL VERT CHPO DEP.I UR D23;
DESSIN EVO.I;
*
FIN BLOC3;
*
*-----*
*      good working message      *
*-----*
*
ERGMAX = MAXI (PROG ERG.1 ERG.2 ERG.3
               ERG.4 ERG.5);
*
SAUT 2 LIGN;
SI (ERGMAX <EG 5.);
  ERRE 0;
SINON;
  ERRE 5;
FINSI;
*
SAUT 2 LIGN;
TEMPS;
SAUT 2 LIGN;
*
FIN;
```

Test vibr2 (Comments)

Boundary conditions

$$\text{DIR1} = (-1 * \text{SINALPHA}) (-1 * \text{COSALPHA});$$

$$\text{DIR2} = (-1 * \text{SINALPHA}) \text{COSALPHA};$$


$$\text{CL1} = \text{BLOQ DEPL DIRECTION DIR1 D23};$$

$$\text{CL2} = \text{BLOQ DEPL DIRECTION DIR2 D41};$$

$$\text{CLI1} = \text{CL1 ET CL2};$$

Only the radial vibrations are studied, so the normal displacements are constrained : DIR1 is the normal to D23 and DIR2 to D41.

Nodes of similar radius have similar radial displacements : this condition is given in loops BLOC1 and BLOC2. The different lists of integers correspond to the numbers of nodes (display the number of nodes when drawing the meshing). The loop BLOC1 is relative to nodes on D41 and D23. The loop BLOC2 is relative to nodes on D41 and middle nodes.





Chapter 13

NON LINEAR DYNAMICS

13.1 Test dyna_nl1 (Description sheet)

Test name: dyna_nl1

Calculation type: MECHANICS 2D, NON LINEAR GEOMETRICAL DYNAMICS

Finite element type: BARR


Topic: Free oscillations of an oscillator of Duffing type. The bar is clamped at one end and prestressed. On the free end is imposed a vertical displacement. Then the bar is released and has free oscillations.

Goal: Find the amplitude and period of the free oscillations of the bar.

Reference:

Version: 97' customer version

Model description:



L	$= 1m$	E	$= 10^7 Pa$
δ	$= 0.2m$	ν	$= 0.3$
A	$= 1m^2$	ρ	$= 0kg/m^3$
M_y	$= 250kg$	F_{pre}	$= 500N$




Test dyna_nl1 (Results)

Results

Margin in amplitude : 0.007%

Margin in period :0.3%

CASTEM figures



GIBI FECIT

Test dyna_n1 (Listing CASTEM)

```

opti echo 0 ;

OPTI DIME 2 ELEM SEG2 MODE PLAN CONT ; ;
P1 = 0. 0. ; P2 = 1. 0. ;
ST = P1 D 1 P2 ;

MO = MODE ST MECANIQUE BARR ;
MA1 = MATE MO YOUN 1E7 NU 0.3 RHO 0.;
CAR1 = CARA MO SECT 1. ;
MACA= MAL ET CAR1;
MASS1 = MASS UY 250. P2 ;

CL1 = BLOQ UX UY P1 ;
CL2 = BLOQ UX P2 ;
CL3 = BLOQ UY P2 ;
CL = CL1 ET CL2 ;

* prestress of 500 N

RI1 = RIGI MACA MO ;
f1 = force fx 500 p2 ;
depl = reso (ril et cll et cl3) f1 ;
sig1 = sigm depl mo maca ;
*sig1 = manu 'CHML' mo 'EFFX' 500. 'TYPE'
* 'CONTRAIN' 'STRESSES' ;

* imposed displacement in statics

t1 = 1e-10 ;
FP11 = DEPI CL3 .2;
LIX1 = PROG 0. t1 100 ;
LIY1 = PROG 0. 1 1 ;
EV1 = EVOL MANU T LIX1 F(T) LIY1 ;
CHA1 = CHAR MECA FP11 EV1 ;
tab2 = table ;
tab2.grandes_rotations = vrai ;
TAB2.'BLOCAGES_MECANIQUES' = CL et cl3 ;
TAB2.'MODELE' = MO;
Tab2.'K_SIGMA' = VRAI ;
TAB2.'CARACTERISTIQUES' = MACA;
TAB2.'CHARGEMENT' = CHA1;
TAB2.'CONTRAINTES' = TABLE;
TAB2.'CONTRAINTES' . 0 = SIG1 ;
LIS11 = PROG 0. (.33*t1) (.66*t1) t1 ;
TAB2.'TEMPS_CALCULES' = LIS11;
PASAPAS TAB2 ;

* release and free oscillations in dynamics
t2 = .27 ;
dt = .002 ;
f1 = force fy 0. p2 ;
cha2 = char meca f1 ev1 ;
tab2.dynamique = vrai ;
tab2.masse_constante = mass1 ;
tab2.ftol = 1e-4 ;
TAB2.'BLOCAGES_MECANIQUES' = CL ;
LIS22 = PROG (t1 + dt) pas dt (t1 + t2) ;
TAB2.'TEMPS_CALCULES' = LIS22;
TAB2.'CHARGEMENT' = CHA2;
PASAPAS TAB2 ;
NDIM = (DIME ( TAB2 . DEPLACEMENTS )) ;
PGY = PROG ;

REPETER TBOU2 NDIM;
LEDEP = TAB2 . DEPLACEMENTS. (&TBOU2 - 1);
V = EXTR LEDEP UY P2 ;
PGY = PGY ET ( PROG V ) ;
FIN TBOU2 ;

ev1 = evol manu 'time' (lis11 et lis22) 'displ' pgy ;
dess ev1 ;

*searching of amplitude and period
dim1 = enti ((t2 - .25)/dt) ;
maxuy = 0 ;
tt = .25 ;
repeter boul dim1 ;
tt = tt + dt ;
uyl = ipol tt (lis11 et lis22) pgy ;
si (uyl > maxuy) ;
maxuy = uyl ;
tp = tt ;
finsi ;
fin boul ;
valres = .2;

** theoretical period .264791269s
** theoretical amplitude .2 ;
err1 = abs ((maxuy - valres)/valres * 100) ;
err2 = abs ((tp - .2648)/.2648 *100) ;
mess 'margin in amplitude'err1'%';
mess 'margin in period'err2'%';
si ((err1 > .1) ou (err2 > .5));
erre 5 ;
sinon ;
erre 0 ;
finsi ;
fin ;

```



Test dyna_nl1 (Comments)

1. First step : Computation of prestress

```
RI1 = RIGI MACA MO ;
f1 = force fx 500 p2 ;
dep1 = reso (ri1 et cl1 et cl3) f1 ;
sig1 = sigm dep1 mo maca ;
```

A horizontal force of 500 N is applied on the free end of the bar. To get a prestress we compute the stresses related to this load. Those results will be added in the following step.

2. Second step : Imposed displacement

```
FP11 = DEPI CL3 .2 ;
LIX1 = PROG 0. t1 100 ;
LIY1 = PROG 0. 1 1 ;
EV1 = EVOL MANU T LIX1 F(T) LIY1 ;
CHA1 = CHAR MECA FP11 EV1 ;
```

The operator DEPI imposes a specific value for the degree of freedom described in CL3. It means that UY-displacement of point P2 (related to CL3) will be equal to 0.2 m. The object created by DEPI is of CHPOINT type and must be added to the second member of the equation. To perform a dynamic computation the second member must be a function of time : the operator CHAR is then used to multiply the second member by a factor which is defined as a function of time (EV1).

3. Second step : Computation by using PASAPAS subroutine

```
tab2 = table ;
tab2.grandes_rotations = vraï ;
TAB2.'BLOCAGES_MECANIQUES' = CL et cl3 ;
TAB2.'MODELE' = MO;
Tab2.'K_SIGMA' = VRAI ;
TAB2.'CARACTERISTIQUES' = MACA;
TAB2.'CHARGEMENT' = CHA1;
TAB2.'CONTRAINTES' = TABLE;
TAB2.'CONTRAINTES' . 0 = SIG1 ;
LIS11 = PROG 0. (.33*t1) (.66*t1) t1 ;
TAB2.'TEMPS_CALCULES' = LIS11;
PASAPAS TAB2 ;
```

The non linear problem is solved by using PASAPAS subroutine. The single object necessary for the computation is a table (TAB2). The user must create this table which has indexes defined by keywords :

- TAB2.grandes_rotations : boolean equal to VRAI if the second-order terms must be taken into account for the strain tensor (in the case of large rotations or large deformations)
- TAB2.BLOCAGES_MECANIQUES : contains the boundary conditions
- TAB2.MODELE : contains the model object
- TAB2.K_SIGMA : boolean equal to VRAI if the matrix K_SIGMA is to be added to the iteration operator

- TAB2.CARACTERISTIQUES : contains material and geometrical characteristics
- TAB2.CHARGEMENT : contains the load object (defined as a function of time)
- TAB2.CONTRAINTE : is a table : TAB2.'CONTRAINTE'.0 contains the stresses at the beginning of first step. The given values are the previous computed prestresses.
- TAB2.TEMPS_CALCULES : the given list of reals defines the values of time for which results are requested.

The subroutine is called by the following instruction : PASAPAS TAB2 .

4. Third step : Release and free oscillations

```
t2 = .27 ;
dt = .002 ;
f1 = force fy 0. p2 ;
cha2 = char meca f1 ev1 ;
tab2.dynamique = vrai ;
tab2.masse_constante = mass1 ;
tab2.ftol = 1e-4 ;
TAB2.'BLOCAGES_MECANIQUES' = CL ;
LIS22 = PROG (t1 + dt) pas dt (t1 + t2) ;
TAB2.'TEMPS_CALCULES' = LIS22;
TAB2.'CHARGEMENT' = CHA2;
PASAPAS TAB2 ;
```

The free oscillations are simulated by a zero force.

TAB2.dynamique is a boolean. If its value is true, it means that it is a dynamic analysis.

TAB2.masse_constante defines a constant lumped mass matrix (mass1).

TAB2.ftol indicates that the effort equilibrium will be checked up to about 1e-4.

5. Results extraction

```
NDIM = (DIME ( TAB2 . DEPLACEMENTS )) ;
PGY = PROG ;
REPETER TBOU2 NDIM;
LEDEP = TAB2 . DEPLACEMENTS. (&TBOU2 - 1);
V = EXTR LEDEP UY P2 ;
PGY = PGY ET ( PROG V ) ;
FIN TBOU2 ;
ev1 = evol manu 'time' (lis11 et lis22) 'displ' pgy ;
dess ev1 ;
```

The results of PASAPAS subroutine are stored in TAB2 table : TAB2.DEPLACEMENTS is a table which has as many indexes as the number of requested saving times.

TAB2.DEPLACEMENTS.i contains the displacements computed for the ith step.



13.2 Test dyna_nl3 (Description sheet)

Test name: dyna_nl1

Calculation type: MECHANICS 3D, NON LINEAR DYNAMICS

Finite element type: POUT, COQ3


Topic: Follower pressure and wavering of a beam. The response of a released beam is studied with an axial compression force which is a little higher than the theoretical value of wavering force ($F_{cr} = 20.05 * E * I / l^2$). The beam is clamped-free. The follower force is described by an equivalent pressure which is applied on a small shell located on the end of the beam. The system is disturbed by a small transversal displacement which is imposed at the free end.

Goal: Find the maximum amplitude of the displacement of the free end.

Reference:

Version: 97' customer version

Model description:



L	$= 12$	A_{beam}	$= 0.15$	
l	$= 0.01$	Th_{shell}	$= 0.01$	
δ	$= 0.01$	E_{beam}	$= 2.1 \cdot 10^7$	
I_y	$= 2.8125 \cdot 10^{-4}$	ρ_{beam}	$= 7.8 \cdot 10^{-3}$	$P_f = 1.05 * [20.05 * E_{beam} * I_y / L^2] / (4l^2)$
I_z	$= 2.8125 \cdot 10^{-4}$	E_{shell}	$= 2.1 \cdot 10^{10}$	
I_o	$= 10^{-5}$	ρ_{shell}	$= 7.8 \cdot 10^{-10}$	
		ν	$= 0.3$	



Test dyna_nl3 (Results)

Results

Maximum amplitude (point bb) : 0.97959

Margin : 0.04%

CASTEM figures





Test dyna_nl3 (Listing CASTEM)

```

complet = vrai;
*
*****
* to get a complete computation put the data
* complet with vrai;
*****
*
option dime 3 elem seg2 tri3 echo 0;
*=====
* Definition of geometry
* The follower force is described by an equivalent
* pressure which is applied on a small shell located
* on the end of the beam
*=====
aa = 0 0 0 ;
bb = 12 0 0;
cc = 12 0.01 0.01 ;
dd = 12 0.01 -0.01 ;
ee = 12 -0.01 -0.01 ;
ff = 12 -0.01 0.01 ;
dens1 = 10 ;
lig1 = aa d dens1 bb ;
lig2 = cc d 2 dd d 2 ee d 2 ff d 2 cc ;
surf1 = surf plan lig2 ;
geo = lig1 et surf1 ;
elim 1.e-4 geo ;
*=====
* Definition of boundary conditions
* Basis of the beam clamped and move in xOy plane
*=====
c11 = bloq rota depla aa ;
c12 = bloq uz rx ry surf1 ;
c13 = bloq bb uy ;
*=====
* Definition of model, material and characteristics
*=====
mo1 = mode lig1 mecanique pout ;
mo2 = mode surf1 mecanique coq3 ;
ma1 = mate mo1 youn 2.1e7 nu 0.3 rho 7.8e-3 ;
ma2 = mate mo2 youn 2.1e10 nu 0.3 rho 7.8e-10 ;
ca1 = cara mo1 sect 0.15 inry 2.8125e-4
      inrz 2.8125e-4 tors 1.e-5 ;
ca2 = cara mo2 epai 0.01 ;
mat1 = mal et ca1 ;
mat2 = ma2 et ca2 ;
matttotal = mat1 et mat2 ;
mototal = mol et mo2 ;
*=====
* calcul of rigidity and mass matrices
*=====
ril = rigi mototal matttotal ;
ritotal = ril et c11 et c12 ;
masst = masse mototal matttotal ;
*=====
* Load : pressure equivalent to the critical wavering
* force and balanced by a coefficient of 1.05 in order
* to be above the instability.
*
* A small transversal displacement is imposed to disturb
* the system.
*=====
fo2 = depi c13 .01 ;
fo0 = manu chpo geo 1 p (2050781.25*1.05) nature
      discret ;

*=====
*
* non linear computation
* Static computation under the compression force and
* the disturbing transversal force
*=====
dt = 0.00001 ;
lil = prog 0. dt ;
li2 = prog 0. 1 ;
ev = evol manu t (lil ) f(t) (li2 ) ;
fcons = char dimp fo2 ev ;
fncons = char psui fo0 ev ;
fco = fncons et fcons ;
lisla = prog 0. pas (dt/5) dt ;

tab3 = table ;
tab3.grands_deplacements = vrai ;
tab3.chargement = fco ;
tab3.blocages_mecaniques = c11 et c12 et c13 ;
tab3.modele = mototal ;
tab3.caracteristiques = matttotal ;
tab3.temps_calcules = lisla ;

pasapas tab3 ;

oeil = 0 0 10000 ;
dim1 = dime (tab3.deplacements) ;
lisx = prog ; lisy = prog ;
repetier boul dim1 ;
n = &boul - 1 ;
uy1 = extr tab3.deplacements.n 'UY' bb ;
lisy = lisy et (prog uy1) ;
lisx = lisx et (prog tab3.temps. n);
def1 = defo tab3.deplacements.n geo 100 ;
si (&boul ega 1) ; deft = def1 ; sinon ;
deft = deft et def1 ; finsi ;
fin boul ;
titre 'Static computation' ;
ev1 = evol manu 'time' lisx 'displ' lisy ;
trac deft oeil cach anime ;

*=====
*
* non linear and dynamic computation
*=====
* Definition of the load evolution
* duration of static load : dt

ev2 = evol manu (prog dt 100) (prog 1 1 ) ;
*
fncons = char psui fo0 ev2 ;
si complet;
lislc = prog dt pas .001 .045 ;
sinon;
lislc = prog dt pas .001 0.005;
finsi;
tab3.chargement = fncons ;
tab3.blocages_mecaniques = c11 et c12 ;
tab3.dynamique = vrai ;
tab3.temps_calcules = lislc ;

pasapas tab3 ;

*=====
* Results : displacement of the free end of
* the beam (point bb)
*=====

```

```
dim1 = dime (tab3.deplacements) ;
lisx = prog ; lisy = prog ;
repeter boul dim1 ;
  n = &boul - 1 ;
  uyl = extr tab3.deplacements.n 'UY' bb ;
  lisy = lisy et (prog uyl) ;
  lisx = lisx et (prog tab3.temps. n);
fin boul ;
titre 'Dynamic computation' ;
evl = evol manu 'time' lisx 'displ' lisy ;
list lisy ;
dess evl ;

* The reference of this test is the result of a previous C2000 version
dmax = maxi abs lisy ;
mess 'Maximum value of amplitude :' dmax ;
si complet ;
valtes = .98;
sinon;
valtes = .082;
finsi;
err1 = abs ((dmax - valtes)/valtes);
mess 'margin for the max amplitude :' err1 ;
  si (err1 > .05) ; erreur 5 ;
  sinon ; erre 0 ; finsi ;
fin ;
fin ;
```



Test dyna_nl3 (Comments)

1. First step : Static computation under the compression force and the disturbing transversal force

```
fo2 = depi c13 .01 ;
fo0 = manu chpo geo 1 p (2050781.25*1.05) nature discret ;
*
dt = 0.00001 ;
li1 = prog 0. dt ;
li2 = prog 0. 1 ;
ev = evol manu t (li1 ) f(t) (li2 ) ;
fcons = char dimp fo2 ev ;
fncons = char psui fo0 ev ;
fco = fncons et fcons ;
lis1a = prog 0. pas (dt/5) dt ;
*
tab3 = table ;
tab3.grands_deplacements = vrai ;
tab3.chargement = fco ;
tab3.blocages_mecaniques = c11 et c12 et c13 ;
tab3.modele = mototal ;
tab3.caracteristiques = mattotal ;
tab3.temps_calcules = lis1a ;
pasapas tab3 ;
```

In the static computation the free tip of the beam has an imposed vertical displacement (fo2). The follower pressure is described with the fo0 object. The loads are defined as a function of time with CHAR operator and the evolution ev.

The PASAPAS subroutine is called by using 6 computation steps (defined with lis1a).

2. Second step : Dynamic computation under the compression force and the release of the beam

```
ev2 = evol manu (prog dt 100) (prog 1 1 ) ;
*
fncons = char psui fo0 ev2 ;
si complet;
lis1c = prog dt pas .001 .045 ;
sinon;
lis1c = prog dt pas .001 0.005;
finsi;
tab3.chargement = fncons ;
tab3.blocages_mecaniques = c11 et c12 ;
tab3.dynamique = vrai ;
tab3.temps_calcules = lis1c ;
*
pasapas tab3 ;
```

The beam is released : the boundary condition c13 and imposed displacement fo2 are removed. The applied load is the follower pressure defined as a function of time (fncons).

The computation is a dynamic analysis so tab3.dynamique is true.



Chapter 14

THERMAL ANALYSIS

14.1 Test ther2 (Description sheet)

Test name: ther2

Calculation type: THERMAL STEADY STATE, 2D AXISYMMETRICAL

Finite element type: QUA4

Topic: Test of the boundary conditions of temperature with an axisymmetrical computation. Three different configurations are tested on a thick-walled cylinder :

- an imposed temperature and an imposed flux
- an imposed temperature and a volumetric source of heat
- an imposed temperature and a forced convection


Goal: Find the field of temperatures in the cylinder for each configuration.

Reference: Analytical solution (1st and 2nd parts).

Comparison with results computed by THERMX (3rd part).

Version: 97' customer version

Model description:



1st configuration

$$R_i = 0.0515m$$

$$R_e = 0.2015m$$

$$H = 0.5m$$


$$T_i = 0.57^{\circ}C$$

$$\phi_e = 0.72W/m^2$$

$$K = 130W/m^{\circ}$$



14.1. TEST THER2 (DESCRIPTION SHEET)



2nd configuration

$$\begin{aligned}R_i &= 6m \\R_e &= 16m \\H &= 4m \\T_i &= 100^\circ C \\ \phi_e &= 0.W/m^2 \\K &= 1W/m^\circ \\Q &= 20W\end{aligned}$$

3rd configuration

$$\begin{aligned}R_i &= 6m \\R_e &= 16m \\H &= 4m \\T_i &= 100^\circ C \\T_a &= 125^\circ C \\K &= 8W/m^\circ \\H_c &= 5W/m^{2o}\end{aligned}$$


Test ther2 (Results)

Results


Value of the temperature at the point located on the external radius.

	Reference	Computed
Configuration 1	0.571	0.571
Configuration 2	1510.9	1505.3
Configuration 3	122.68	122.68


CASTEM figures



MESHING - 1ST PROBLEM



MESHING - 2ND PROBLEM



MESHING - 3RD PROBLEM



14.1. TEST THER2 (DESCRIPTION SHEET)

Test ther2 (Listing CASTEM)

```
*****
TITRE 'THER2 - BOUNDARY COND. TEMPERATURE
      - AXISYMMETRICAL MODE' ;
OPTION DIME 2 ELEM QUA4 MODE AXIS ;
OPTION ECHO 0 ;
TEMPS ;
*****
*--- 1st problem
*****
*--- GEOMETRY:
*
P1 = 0.0515 0. ; P20 = 0.0665 0. ;
P19 = 0.0665 0.5 ; P2 = 0.0515 0.5 ;
P18 = 0.0815 0. ; P17 = 0.0815 0.5 ;
P16 = 0.0965 0. ; P15 = 0.0965 0.5 ;
P14 = 0.1115 0. ; P13 = 0.1115 0.5 ;
P12 = 0.1265 0. ; P11 = 0.1265 0.5 ;
P10 = 0.1415 0. ; P9 = 0.1415 0.5 ;
P8 = 0.1565 0. ; P7 = 0.1565 0.5 ;
P6 = 0.1715 0. ; P5 = 0.1715 0.5 ;
P24 = 0.1865 0. ; P23 = 0.1865 0.5 ;
P4 = 0.2015 0.5 ; P3 = 0.2015 0. ;
*
SURF1 = MANUEL QUA4 P1 P20 P19 P2 ;
SURF2 = MANUEL QUA4 P20 P18 P17 P19 ;
SURF3 = MANUEL QUA4 P18 P16 P15 P17 ;
SURF4 = MANUEL QUA4 P16 P14 P13 P15 ;
SURF5 = MANUEL QUA4 P14 P12 P11 P13 ;
SURF6 = MANUEL QUA4 P12 P10 P9 P11 ;
SURF7 = MANUEL QUA4 P10 P8 P7 P9 ;
SURF8 = MANUEL QUA4 P8 P6 P5 P7 ;
SURF9 = MANUEL QUA4 P6 P24 P23 P5 ;
SURF10 = MANUEL QUA4 P24 P3 P4 P23 ;
SURF11 = SURF1 ET SURF2 ET SURF3 ET SURF4
        ET SURF5 ET SURF6 ET SURF7 ET SURF8
        ET SURF9 ET SURF10 ;
*
TRAC SURF11 TITRE 'MESHING - 1ST PROBLEM' ;
*
D1 = DROITE 1 P1 P2 ;
D3 = DROITE 1 P3 P4 ;
*
*--- DATA FOR THE THERMIC PROBLEM:
*
*--- MODEL:
*
MOD1 = MODEL SURF11 THERMIQUE ISOTROPE ;
*
*--- MATERIAL CHARACTERISTICS:
*
MAT1 = MATER MOD1 K 130. ;
*
*--- CONDUCTIVITY MATRIX :
*
CND1 = CONDUCTIVITE MOD1 MAT1 ;
*
*--- IMPOSED TEMPERATURES : USING
      OF BLOQUE AND DEPI
*
B1 = BLOQUE T D1 ;
E1 = DEPI B1 0.57 ;
*
*--- IMPOSED FLUX :
*
F1 = FLUX MOD1 0.72 D3 ;
*
*--- GATHERING OF FIRST AND SECOND MEMBERS:
```

```
*
RIG1 = CND1 ET B1 ;
FF1 = E1 ET F1 ;
*
*--- RESOLUTION:
*
CHTER1 = RESOUDRE RIG1 FF1 ;
TETA1 = EXTR CHTER1 T P3 ;
*
*****
*--- 2nd problem
*****
*--- GEOMETRY:
*
P1 = 6 0 ;
P2 = 16 0 ;
P3 = 16 4 ;
P4 = 6 4 ;
D1 = DROITE 8 P1 P2 ;
D2 = DROITE 6 P2 P3 ;
D3 = DROITE 9 P3 P4 ;
D4 = DROITE 4 P4 P1 ;
LIG1 = D1 ET D2 ET D3 ET D4 ;
SURF1 = SURFACE LIG1 PLANE ;
*
TRAC SURF1 TITRE 'MESHING - 2ND PROBLEM' ;
*
*
*--- DATA FOR THE THERMIC PROBLEM:
*
*--- MODEL:
*
MOD1 = MODEL SURF1 THERMIQUE ISOTROPE ;
*
*--- MATERIAL CHARACTERISTICS:
*
MAT1 = MATER MOD1 K 1. ;
*
*--- CONDUCTIVITY MATRIX:
*
CND1 = CONDUCTIVITE MOD1 MAT1 ;
*
*--- IMPOSED TEMPERATURES : USING
      OF BLOQUE AND DEPI
*
B1 = BLOQUE T D4 ;
E1 = DEPI B1 100. ;
*
*--- IMPOSED FLUX :
*
E2 = FLUX MOD1 0. D2 ;
*
*--- VOLUMETRIC SOURCE:
*
E3 = SOURCE MOD1 20. SURF1 ;
*
*--- GATHERING OF FIRST AND SECOND MEMBERS:
*
RIG1 = CND1 ET B1 ;
FF1 = E1 ET E2 ET E3 ;
*
*--- RESOLUTION:
*
CHTER1 = RESOUDRE RIG1 FF1 ;
TETA2 = EXTR CHTER1 T P3 ;
*
*****
*--- 3rd problem
```

```

*****
*
*--- GEOMETRY:
*
*   - inner radius RI = 6.
*   - outer radius RE = 16.
*   - height      H = 4.
*   - inner surface : D4
*   - outer surface : D2
*
P1 = 6 0 ;
P2 = 16 0 ;
P3 = 16 4 ;
P4 = 6 4 ;
D1 = DROITE 8 P1 P2 ;
D2 = DROITE 5 P2 P3 ;
D3 = DROITE 9 P3 P4 ;
D4 = DROITE 6 P4 P1 ;
LIG1 = D1 ET D2 ET D3 ET D4 ;
SURF1 = SURF LIG1 PLANE ;
*
TRAC SURF1 TITRE 'MESHING - 3RD PROBLEM';
*
*--- DATA FOR THE THERMIC PROBLEM:
*
*--- MODEL:
*
MOD1 = MODEL SURF1 THERMIQUE ISOTROPE ;
MOD2 = MODEL D2 CONVECTION ;
*
*--- MATERIAL CHARACTERISTICS:
*
MAT1 = MATER MOD1 K 8. ;
MAT2 = MATER MOD2 H 5. ;
*
*--- CONDUCTIVITY MATRIX:
*
CND1 = CONDUCTIVITE MOD1 MAT1 ;
CNF1 = CONDUCTIVITE MOD2 MAT2 ;
*
*--- FLUX RELATED TO THE CONVECTION:
*
FF1 = CONVECTION MOD2 MAT2 T 125. ;
*
*--- IMPOSED TEMPERATURE : USING OF BLOQUE AND DEPI
*
B1 = BLOQUE T D4 ;
E1 = DEPI B1 100. ;
*
*--- GATHERING OF FIRST AND SECOND MEMBERS:
*
RIG1 = CND1 ET CNF1 ET B1 ;
Q1 = FF1 ET E1 ;
*
*--- RESOLUTION:
*
CHTER1 = RESOUDRE RIG1 Q1 ;
TETA3 = EXTR CHTER1 T P3 ;
*
*--- POST-PROCESSING
*
SAUT PAGE ;
SAUT 2 LIGN ;

MESS ' PROBLEM No 1 ' ;
MESS ' TEMPERATURE FOR R = RE
      THEORETICAL      COMPUTED ' ;
MESS '
      T = 0.5715225      T = ' TETA1 ;
SAUT 1 LIGN ;

MESS ' PROBLEM No 2 ' ;
MESS ' TEMPERATURE FOR R = RE
      THEORETICAL      COMPUTED ' ;
MESS '
      T = 1510.923      T = ' TETA2 ;
SAUT 1 LIGN ;
MESS ' PROBLEM No 3 ' ;
MESS ' TEMPERATURE FOR R = RE
      REFERENCE      COMPUTED ' ;
MESS '
      T = 122.68      T = ' TETA3 ;
*
TEMPS ;
*
* VALIDATION CODE FOR THE PROBLEM No 2
*
TREF=1511;RESI=ABS((TETA2-TREF)/TREF);
SI(RESI <EG 5E-2);
  ERRE 0;
SINO;
  ERRE 5;
FINSI;
FIN;

```



Test ther2 (Comments)

1. First problem

- Definition of the thermic model

MOD1 = MODEL SURF11 THERMIQUE ISOTROPE ;
 MAT1 = MATER MOD1 K 130. ;
 CND1 = CONDUCTIVITE MOD1 MAT1 ;

The thermic model (MOD1) is defined with the formulation named THERMIQUE ISOTROPE. This model is described by the geometry SURF11, the conductivity of the model is defined with the operator MATER.

The operator COND(UCTIVITE) builds the matrix of conductivity (CND1).

- Definition of the thermic boundary conditions

B1 = BLOQUE T D1 ;
 E1 = DEPI B1 0.57 ;
 F1 = FLUX MOD1 ;

The operator BLOQUE builds the boundary condition (B1) on the degree of freedom T of the line D1. The operator DEPI(MPOSE) imposes a value of 0.57 on the condition B1.

The operator FLUX imposes a flux of 0.72 on the line D3 which belongs to MOD1.

- Computation

The first and second members of the equation to be solved are gathered and the computation is performed with the operator RESO. The computed object CHTER1 is a CHPOINT type and the values of its component T can be extracted with the operator EXTR(AIRE).

2. Second problem

E3 = SOURCE MOD1 20. SURF1 ;

The model, the imposed temperature and the flux are defined in the same way as in the first configuration. The operator SOURCE defines an imposed volumetric source of heat on the surface SURF1.

The computation is performed like in the first problem by gathering first and second members of the equation and then by solving with the RESO operator.

3. Third problem

MOD1 = MODEL SURF1 THERMIQUE ISOTROPE ;
 MOD2 = MODEL D2 CONVECTION ;
 *
 MAT1 = MATER MOD1 K 8. ;
 MAT2 = MATER MOD2 H 5. ;
 *
 CND1 = CONDUCTIVITE MOD1 MAT1 ;
 CNF1 = CONDUCTIVITE MOD2 MAT2 ;
 *
 FF1 = CONVECTION MOD2 MAT2 T 125. ;

The outer surface is submitted to forced convection : as the problem is axisymmetrical, the outer surface is modelled by the line D2. Along this line is defined a convection model (MOD2) by using the operator MODEL followed by the formulation CONVECTION.

The operator MATER defines the convection exchange coefficient.

The operator CONV(ECTION) imposes a condition of convection on the structure defined in the objects MOD2 and MAT2, it is necessary to indicate the value of the external temperature ($T=125$).



14.2 Test ther7or (Description sheet)

Test name: ther7or

Calculation type: THERMAL STEADY STATE, 3D ORTHOTROPIC

Finite element type: COQ8 COQ6

Topic: Test of the boundary conditions of temperature with orthotropic elements. Three different configurations are tested on a shell:


- an imposed temperature and a forced convection
- an imposed temperature and an imposed flux
- an imposed temperature and a volumetric source of heat

Goal: Find the field of temperatures in the shell for each configuration and compare it with the case of isotropic elements.

Reference CASTEM: Dowlatyari (03/91).

Version: 97' customer version


Model description:




L	$= 6m$
l	$= 3m$
t	$= 0.5m$
K_1	$= 100W/m^o$
K_2	$= 1000W/m^o$
K_3	$= 100W/m^o$
T_{L_inf}	$= 10^oC$
H_{S_COT}	$= 15W/m^{2o}$
T_a	$= 200^oC$
ϕ_{S_COT}	$= 200W/m^2$
Q_{S_COT}	$= 1000W/m^3$

Test ther7or (Results)
CASTEM figures


1st resolution : convection



2nd resolution : imposed flux



3rd resolution : volumetric source



**Test ther7or (Listing CASTEM)**

```

*****
TITRE 'THER7or - ORTHOTROPIC ELEMENTS COQ8 , COQ6 ' ;
OPTION DIME 3 ELEM QUA8 ;
TEMPS ;
*
*---- GEOMETRY : POINTS SUPPORTING THE ELEMENTS ----
*
A1 = 0. 0. 0. ; B1 = 0. 3.0 0. ; AB1 = 0. 1.5 0. ;
A2 = 0. 0. 1. ; B2 = 0. 3. 1. ; AB2 = 0. 1.5 2. ;
A3 = 0. 0. 2. ; B3 = 0. 3. 2. ; AB3 = 0. 1.5 3. ;
A4 = 0. 0. 3. ; B4 = 0. 3. 3. ; AB4 = 0. 1.5 4. ;
A5 = 0. 0. 4. ; B5 = 0. 3. 4. ; AB5 = 0. 1.5 6. ;
A6 = 0. 0. 5. ; B6 = 0. 3. 5. ;
A7 = 0. 0. 6. ; B7 = 0. 3. 6. ;
M = 0. 0. 0. ; N = 2. 0. 0.
*
ANG=20.0 ;
A1=A1 TOUR ANG M N ;A2=A2 TOUR ANG M N ;
A3=A3 TOUR ANG M N ;
A4=A4 TOUR ANG M N ;A5=A5 TOUR ANG M N ;
A6=A6 TOUR ANG M N ;
A7=A7 TOUR ANG M N ;B1=B1 TOUR ANG M N ;
B2=B2 TOUR ANG M N ;
B3=B3 TOUR ANG M N ;B4=B4 TOUR ANG M N ;
B5=B5 TOUR ANG M N ;
B6=B6 TOUR ANG M N ;B7=B7 TOUR ANG M N ;
AB1=AB1 TOUR ANG M N ;
AB2=AB2 TOUR ANG M N ;
AB3=AB3 TOUR ANG M N ;
AB4=AB4 TOUR ANG M N ;
AB5=AB5 TOUR ANG M N ;
*----- GEOMETRY : SURFACES -----
*
QH01 = MANU QUA8 A1 AB1 B1 B2 B3 AB2 A3 A2 ;
QH02 = MANU TRI6 A3 AB2 B3 B4 B5 AB3 ;
QH03 = MANU TRI6 A3 AB3 B5 AB4 A5 A4 ;
QH04 = MANU QUA8 A5 AB4 B5 B6 B7 AB5 A7 A6 ;
QTOT = QH01 ET QH02 ET QH03 ET QH04 ;
*
OEIL = 1000. 0. 0. ;
TRAC OEIL QTOT ;
QSOUR = QH04 ;
*
L_INF = MANU SEG3 A1 AB1 B1 ;
*
S_COT = QH04;
*
* WARNING : it is necessary to orientate the elements
* (with the operator ORIENT) if it is not sure that
* they are orientated coherently
*
*----- DATA FOR THE THERMIC PROBLEM -----
*
*----- MODEL -----
*
MOD1 = MODEL QTOT THERMIQUE ORTHOTROPE COQ8 COQ6 ;
MOD2 = MODEL S_COT CONVECTION SUPERIEURE COQ8 COQ6 ;
*
*----- MATERIAL CHARACTERISTICS -----
*
MAT1 = MATER MOD1 'DIRECTION' A2 'PARALLELE'
      'K1' 100. 'K2' 1000. 'K3' 100. EPAI 0.5 ;
MAT2 = MATER MOD2 'H' 15. ;
*
*----- CONDUCTIVITY MATRICES -----
*
CND1 = CONDUCTIVITE MOD1 MAT1 ;

CND2 = CONDUCTIVITE MOD2 MAT2 ;
*
*--- IMPOSED TEMPERATURES : USING OF BLOQUE AND DEPI ---
*
BB1 = BLOQUE L_INF 'TINF' ;
EE1 = DEPI BB1 10. ;
BB2 = BLOQUE L_INF 'TSUP' ;
EE2 = DEPI BB2 10. ;
BB3 = BLOQUE L_INF 'T ' ;
EE3 = DEPI BB3 10. ;
BB = BB1 ET BB2 ET BB3 ;
*
EE = EE1 ET EE2 ET EE3 ;

*
*----- FLUX RELATED TO THE CONVECTION -----
*
CNV1 = CONVECTION MOD2 MAT2 'T' 200.;
*
*----- IMPOSED FLUX -----
*
FLU1 = FLUX MOD1 200. S_COT SUPE ;
*
*----- SOURCE OF HEAT -----
*
S1 = SOURCE MOD1 MAT1 1000. QSOUR ;
*
*----- GATHERING OF CONDUCTIVITY TERMS -----
*
CCC1 = CND1 ET CND2 ET BB ; CCC2 = CND1 ET BB ;
*
*----- GATHERING OF EQUIVALENT FLUX TERMS -----
*
FFF1 = EE ET CNV1 ; FFF2 = EE ET FLU1 ; FFF3 = EE ET S1 ;
*FFF1 = EE ET CNV1 ; FFF2 = EE ET FLU1 ;
*
*----- 1ST RESOLUTION : CONVECTION -----
*
CHTER1 = RESOUDRE CCC1 FFF1 ;
*
*----- 2ND RESOLUTION : FLUX -----
*
CHTER2 = RESOUDRE CCC2 FFF2 ;
*
*----- 3RD RESOLUTION: SOURCE -----
*
CHTER3 = RESOUDRE CCC2 FFF3 ;
*
*---- POST-TREATMENT : DRAWING OF THE RESULT FIELDS ---
*----- AND EXTRACTING OF THE CHARACTERISTIC VALUES ----
*
*LISTE CHTER1 ;
*LISTE CHTER2 ;
*LISTE CHTER3 ;
*
TITRE 'ELEMENTS COQ6,COQ8 : CONVECTION' ;
CHTER1I = 'EXCO' CHTER1 'TINF' 'NOID' 'TINF' ;
TRACER OEIL QTOT CHTER1I ;
TITRE 'ELEMENTS COQ6,COQ8 : IMPOSED FLUX ' ;
CHTER2I = 'EXCO' CHTER2 'TINF' 'NOID' 'TINF' ;
TRACER OEIL QTOT CHTER2I ;
TITRE 'ELEMENTS COQ6,COQ8 : SOURCE' ;
CHTER3I = 'EXCO' CHTER3 'TINF' 'NOID' 'TINF' ;
TRACER OEIL QTOT CHTER3I ;

T1AB1 = EXTR CHTER1 TSUP AB1 ;

T1AB2 = EXTR CHTER1 TSUP AB2 ;
T1AB3 = EXTR CHTER1 TSUP AB3 ;

```

```

T1AB4 = EXTR CHTER1 TSUP AB4 ;
T1AB5 = EXTR CHTER1 TSUP AB5 ;
T2AB1 = EXTR CHTER2 TSUP AB1 ;
T2AB2 = EXTR CHTER2 TSUP AB2 ;
T2AB3 = EXTR CHTER2 TSUP AB3 ;
T2AB4 = EXTR CHTER2 TSUP AB4 ;
T2AB5 = EXTR CHTER2 TSUP AB5 ;
T3AB1 = EXTR CHTER3 TSUP AB1 ;
T3AB2 = EXTR CHTER3 TSUP AB2 ;
T3AB3 = EXTR CHTER3 TSUP AB3 ;
T3AB4 = EXTR CHTER3 TSUP AB4 ;
T3AB5 = EXTR CHTER3 TSUP AB5 ;
*
DIST = PROG 0. 2. 4. 6. ;
DIST1 = PROG 0. 2. 3. 4. 6. ;
TEMPR1 = PROG 10. 71.44 132.03 158.86 ;
TEMPC1 = PROG T1AB1 T1AB2 T1AB3 T1AB4 T1AB5 ;
EV1R = EVOL MANU 'DISTANCE' DIST 'TEMPERATURE' TEMPR1 ;
EV1C = EVOL MANU 'DISTANCE' DIST1 'TEMPERATURE' TEMPC1 ;
TAB = 'TABLE' ;
TAB.1 = 'MARQ PLUS' ;
TAB.2 = 'MARQ ETOI' ;
TITRE 'TEMPERATURE + CONVECTION. + :ISOTROPIC * :ORTHOT.' ;
DESSIN TAB (EV1R ET EV1C ) ;
TEMPR2 =PROG 10. 26.58 42.878 50.984 ;
TEMPC2 = PROG T2AB1 T2AB2 T2AB3 T2AB4 T2AB5 ;
EV2R = EVOL MANU 'DISTANCE' DIST 'TEMPERATURE' TEMPR2 ;
EV2C = EVOL MANU 'DISTANCE' DIST1 'TEMPERATURE' TEMPC2 ;
TITRE 'TEMPERATURE + IMPOSED FLUX. + :ISOTROPIC * :ORTHOT.' ;
DESSIN TAB (EV2R ET EV2C ) ;
TEMPR3 =PROG 10. 51.43 91.75 111.75 ;
TEMPC3 = PROG T3AB1 T3AB2 T3AB3 T3AB4 T3AB5 ;
EV3R = EVOL MANU 'DISTANCE' DIST 'TEMPERATURE' TEMPR3 ;
EV3C = EVOL MANU 'DISTANCE' DIST1 'TEMPERATURE' TEMPC3 ;
TITRE 'TEMPERATURE + SOURCE. + :ISOTROPIC * :ORTHOT.' ;
DESSIN TAB (EV3R ET EV3C ) ;
TEMPS ;
T1Aq = 10.; T1Bq = 71.44; t1Cq = 132.03; T1Dq = 158.86;
T2Aq = 10.; T2Bq = 26.58; t2Cq = 42.878; T2Dq = 50.984;
T3Aq = 10.; T3Bq = 51.43; T3Cq = 91.75 ; T3Dq = 111.75;

*****
***** VALIDATION CODE *****
*****

ERCA= ABS (( T1AB1-T1Aq)/T1AB1);
ERCB= ABS (( T1AB2-T1Bq)/T1AB2);
ERCC= ABS (( T1AB4-T1Cq)/T1AB4);
ERCD= ABS (( T1AB5-T1Dq)/T1AB5);
ERFA= ABS (( T2AB1-T2Aq)/T2AB1);
ERFB= ABS (( T2AB2-T2Bq)/T2AB2);
ERFC= ABS (( T2AB4-T2Cq)/T2AB4);
ERFD= ABS (( T2AB5-T2Dq)/T2AB5);
ERSA= ABS (( T3AB1-T3Aq)/T3AB1);
ERSB= ABS (( T3AB2-T3Bq)/T3AB2);
ERSC= ABS (( T3AB4-T3Cq)/T3AB4);
ERSD= ABS (( T3AB5-T3Dq)/T3AB5);

ERTOT = PROG ERCA ERCB ERCC ERCD ERFA ERFB ERFC
        ERFD ERSA ERSB ERSC ERSD;
ermax = maxi ertot;
temps;
SI ( ERMAX <EG 0.05 );
    ERRE 0;

```



Test ther7or (Comments)

1. Orthotropic definition

MOD1 = MODEL QTOT THERMIQUE ORTHOTROPE COQ8 COQ6 ;
MAT1 = MATER MOD1 'DIRECTION' A2 'PARALLELE'
'K1' 100. 'K2' 1000. 'K3' 100. EPAI 0.5 ;

The thermal model (MOD1) is defined with the formulation named THERMIQUE ORTHOTROPE. This model is described by the geometry QTOT, the conductivity of the model is defined with the operator MATER : K1, K2 and K3 are the thermal conductivities and the thickness is equal to 0.5 ; the first orthotropic direction is parallel to the vector defined by A2.

2. Model of convection

MOD2 = MODEL S_COT CONVECTION SUPERIEURE COQ8 COQ6 ;
MAT2 = MATER MOD2 'H' 15. ;

The model MOD2 defines a convection formulation. The keyword SUPERIEURE indicates that the convection is located on the upper side of the shell.

3. Degrees of freedom

BB1 = BLOQUE L_INF 'TINF' ;
BB2 = BLOQUE L_INF 'TSUP' ;
BB3 = BLOQUE L_INF 'T' ;

The shells have 3 degrees of freedom : the inside temperature T, the upper temperature TSUP and the lower temperature TINF.

14.3 Test ther-perm (Description sheet)

Test name: ther-perm

Calculation type: THERMAL STEADY STATE, 3D NON LINEAR

Finite element type: QUA4

Topic: Use of the procedure THERMIC in a non linear steady state. An endless shell is submitted to a volumetric source of heat and to an imposed temperature on one side. The conductivity is a linear function of the temperature.


Goal: Find the field of temperatures in the shell and compare it with the analytical solution.

Reference:


- Klaus-Jürgen Bathe & Mohammad R. Khoshgoftaar, Finite element formulation and solution of non-linear heat transfer, Nuclear Engineering and Design, v. 51 (1979), pp. 389-401
- J. Joly, Cas tests non linéaires de validation pour DELFINE, Note technique EMT.SMTS.TTMF 84/29
- V. Arpaci, Conduction Heat Transfer, Adison-Wesley, 1966, pp. 130-132

Version: 97' customer version




Model description:



$$\begin{aligned}
 L &= 5e - 3m \\
 t &= L/10 \\
 T_{li2} &= 100^{\circ}C \\
 T_o &= 100^{\circ}C \\
 Q &= 10^7 W/m^3 \\
 K(T) &= 0.05 * T - 3
 \end{aligned}$$



Test ther-perm (Results)
CASTEM figures





Test ther-perm (Listing CASTEM)

```

*****
*** Options ...
    opti dime 2 elem seg2 echo 1 ;

*** Parameters ...
    L = 5.e-3 ;
    ep = L / 10 ;
    K0 = 2. ;
    beta = 0.025 ;
    T0 = 100. ;
    q = 10000000. ;

*** Points ...
    dens ep ;
    p1 = 0 0 ;
    p2 = 0 ep ;
    vechoriz = L 0 ;

*** Lines ...
    lil = p1 d 1 p2 ;

*** Surface ...
    opti elem qua4 ;
    sul = lil tran vechoriz dini (L/100.) dfin (L/10.) ;
    li2 = cote 3 sul ;
    li3 = cote 2 sul ;
    titr 'Mesh of the shell' ;
    trac sul ;

*** Models ...
    mocnd = model sul thermique ;
    dt1 = -100. ;
    T1 = T0 + dt1 ;
    K1 = K0 * (beta*dt1 + 1) ;
    dt2 = 1000. ;
    T2 = T0 + dt2 ;
    K2 = K0 * (beta*dt2 + 1) ;
    titr 'K(T)' ;
    evk = evol manu 'T' (prog T1 T2) 'K' (prog K1 K2) ;
    dess evk ;
    macnd = mate mocnd 'K' evk ;

*** Imposed temperature ...
    blt = blocq T li2 ;
    til = depi blt T0 ;

*** Volumetric source ...
    fl1 = sour mocnd sul q ;

*** Input of table for THERMIC ...
    tabth = table THERMIQUE ;
    tabth . 'BLOCAGE' = blt ;
    tabth . 'IMPOSE' = til ;
    tabth . 'FLUX' = fl1 ;
    tabth . 'INSTANT(0)' = manu chpo sul 1 T T0
        nature diffus ;
    tabth . 'TABCOND' = table ;
    tabth . 'TABCOND' . mocnd = evk ;
    tabth . 'NIVEAU' = 1 ;

*** Computation ...
    thermic tabth NONLINEAIRE ;

*** Post-treatment ...
    tresu = tabth . TEMPERATURE ;
    titr 'Profile of absolute temperature' ;
    evt = evol chpo tresu T li3 ;

    dess evt ;

    titr 'Profile of relative temperature
        (analytic solution)' ;
    chx = coor 1 li3 ;
    evx = evol chpo chx SCAL li3 ;
    lrx = extr evx ABSC ;
    lrl = prog (dime lrx) * 1. ;
    evl = evol manu lrx lrl ;
    toto1 = evl - ((evx / L)*(evx / L)) ;
    toto2 = (toto1 * (L*L*q*beta/K0)) + evl ;
    kkabsc = extr toto2 ABSC ;
    kkordo = extr toto2 ORDO ;
    toto3 = evol manu kkabsc (kkordo ** 0.5) ;
    toto4 = toto3 - evl ;
    toto5 = (toto4 * (2*K0/(beta*q*L*L))) coul VERT ;
    dess toto5 mima ;

    titr 'Profile of relative temperature
        (Finite Element solution)' ;
    evtrel = ((evt - (T0 * evl)) * (2*K0/(q*L*L)))
        coul ROUG ;
    dess evtrel mima ;

*** Test ...
    ladiff = abs (extr (evtrel - toto5) ORDO) ;
    valtst = maxi ladiff ;
    si(valtst > 1.e-5) ;
        erre 5 ;
    finsi ;

*** Bye ...

    fin ;

```

Test ther-perm (Comments)

1. Material characteristics

```
evk = evol manu 'T' (prog T1 T2) 'K' (prog K1 K2) ;
dess evk ;
macnd = mate mocnd 'K' evk ;
```

The conductivity is a function of the temperature and is defined in the evolution (evk). The declaration of the conductivity is done with the operator mate and the values are given in the shape of the evolution object (evk).

2. Procedure THERMIC

```
tabth = table THERMIQUE ;
tabth . 'BLOCAGE' = blt ;
tabth . 'IMPOSE' = ti1 ;
tabth . 'FLUX' = fl1 ;
tabth . 'INSTANT(0)' = manu chpo su1 1 T T0 nature diffus ;
tabth . 'TABCOND' = table ;
tabth . 'TABCOND' . mocnd = evk ;
tabth . 'NIVEAU' = 1 ;
thermic tabth NONLINEAIRE ;
```

The procedure THERMIC enables to deal with non linear permanent conditions. It requires some input data which are given in a table of subtype THERMIQUE. The results of the procedure is a field of temperature which is contained in the same table.

The table tabth has different indexes :

- 'BLOCAGE' : contains the restraint matrix (RIGIDITE type)
- 'IMPOSE' : contains the imposed values (CHPOINT type)
- 'FLUX' : contains the equivalent flow (CHPOINT type)
- 'INSTANT(0)' : contains the initial temperature field (CHPOINT type)
- 'TABCOND' : contains the tables of conductivities indexed by the model object of the different sub-spaces (here tabth . 'TABCOND' . mocnd contains the evolution of the conductivity evk)
- 'NIVEAU' : indicates the level of message.

3. Post-treatment

```
tresu = tabth . TEMPERATURE ;
titr 'Profile of absolute temperature' ;
evt = evol chpo tresu T li3 ;
dess evt ;
```

The field of temperature generated by the THERMIC procedure is written in tabth . TEMPERATURE (or tresu). It is an object of type CHPOINT. Here the object evt contains the evolution of the component T of the CHPOINT tresu along the line li3.



Chapter 15

THERMO-MECHANICS

15.1 Test thme1 (Description sheet)

Test name: thme1

Calculation type: LINEAR THERMAL ANALYSIS AND MECHANIC

Finite element type: QUA8

Topic: Thermal diffusion in a bar. The structure is an embedded bar subjected to a thermal diffusion. We study its expansion.

Goal: Use of the PASAPAS procedure


Reference: Analytical solution.

Version: 97' customer version

Model description:

Characteristic of the bar:

Length of the bar	l	=	0.1
Isotropic conductivity	K	=	14.6
Specific heat	C	=	460
Specific mass	RHO	=	7800
Thermal expansion coefficient	ALPHA	=	1.10^{-5}
Young's Modulus	YOUNG	=	2.10^{11}
Poisson's Coefficient	NU	=	0.00001



Test thme1 (Listing CASTEM)

```

TITR 'THERMO-MECHANIC CALCULATION' ;
*****
COMPLET = FAUX ; COMM * FULL CALCULATION IF VRAI ;
OPTI ECHO 0 ;
OPTI DIME 2 ELEM QUA8 MODE PLAN DEFO ;
*----- GEOMETRY -----*
DENS 1 ;
PA = 0. 0. ; PB = 0.1 0. ; PC = 0.1 0.01 ;
PD = 0. 0.01 ; PP8 = 0.08 0. ;
D1 = DROI 10 PA PB ; D2 = DROI 1 PB PC ;
D3 = DROI 10 PC PD ; D4 = DROI 1 PD PA ;
L1 = D1 ET D2 ET D3 ET D4 ;
SU1 = SURF L1 PLANE ;
TRAC SU1 ;
PP8 = SU1 POINT PROC PP8 ;
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD1 = MODE SU1 THERMIQUE ISOTROPE CONS UN ;
MOD2 = MODE SU1 MECANIQUE ELASTIQUE ISOTROPE
CONS DEUX ;
MOD = MOD1 ET MOD2 ;
MAT1 = MATE MOD1 K 14.6 'C' 460. RHO 7800. ;
MAT2 = MATE MOD2 ALPHA 1E-5 YOUNG 2E+11 NU 0.00001 ;
MAT = MAT1 ET MAT2 ;
*----- BOUNDARY CONDITIONS -----*
* MECHANIC LOCKING
CL1 = BLOQ UY SU1 ; CL2 = BLOQ UX D4 ;
CL_ME = CL1 ET CL2 ;
* THERMIC LOCKING
CL3 = BLOQ T D2 ; CL4 = BLOQ T D4 ;
CL_TH = CL3 ET CL4 ;
* IMPOSED TEMPERATURES
TEM1 = DEPI CL3 500. ; TEM2 = DEPI CL4 20. ;
EV1 = EVOL MANU X (PROG 0. 200.) Y (PROG 1. 1.) ;
CHA = CHAR TIMP (TEM1 + TEM2) EV1 ;
*----- INITIAL TEMPERATURE -----*
TEM0 = MANU CHPO SU1 1 T 20. ;
* CREATION OF THE TABLE FOR THE PASAPAS PROCEDURE*
TA = TABLE ;
TA.MODELE = MOD ;
TA.CARACTERISTIQUES = MAT ;
TA.BLOCAGES_MECANIQUES = CL_ME ;
TA.BLOCAGES_THERMIQUES = CL_TH ;
TA.TEMPERATURES = TABLE ;
TA.TEMPERATURES . 0 = TEM0 ;
TA.CHARGEMENT = CHA ;
TA.PROCEDURE_THERMIQUE = LINEAIRE ;
TA.TEMPS_CALCULES = PROG 0. PAS 10. 150. ;
PASAPAS TA ;
*----- RESULTS -----*
A = (14.6 / 7800.)/460. ;
X = 0.08 ;
ER1 = 0. ; ER2 = 0. ;
NB = 0 ;
TPR1 = PROG ; TPR2 = PROG ; TPR3 = PROG ;
TEP_TH = PROG ;
DEP_TH = PROG ;
IDIM = (DIME (TA.TEMPS)) - 1 ;
SOLTEENR = PROG
32.779 76.142 116.26 148.47 174.30
195.40 212.98 227.88 240.70 251.85
261.65 270.33 278.08 285.03 291.30 ;
SOLDEENR = PROG
4.51064E-07 3.18778E-06 7.26909E-06
1.18402E-05 1.65525E-05 2.12583E-05
2.58885E-05 3.04069E-05 3.47909E-05
3.90251E-05 4.30987E-05 4.70048E-05

5.07398E-05 5.43030E-05 5.76962E-05 ;
REPETER BOU IDIM ;

ITE = &BOU ;
TE = TA.TEMPS.ITE ;
DEP = TA.DEPLACEMENTS.ITE ;
TEM = TA.TEMPERATURES.ITE ;
TEP1 = EXTR TEM T PP8 ;
DEP1 = EXTR DEP UX PP8 ;
TPR1 = TPR1 ET (PROG TEP1) ;
TPR2 = TPR2 ET (PROG DEP1) ;
TPR3 = TPR3 ET (PROG TE) ;
SI COMPLET ;
CO1 = ERF((0.1 - X)/(2 * ((A * TE)**0.5))) ;
CO2 = ERF(0.1/(2.*((A * TE)**0.5))) ;
SOL_TE = 500. - ((500. - 20.)*CO1/CO2) ;
SINON ;
SOL_TE = EXTR SOLTEENR ITE ;
FINSI ;
TEP_TH = TEP_TH ET (PROG SOL_TE) ;
SI COMPLET ;
MI = PROG ; MI1 = PROG ;
REPETER BOU2 21 ;
ITE2 = &BOU2 - 1 ;
TE1 = X * ITE2 / 20 ;
MI = MI ET (PROG TE1) ;
Y1 = ERF((0.1-TE1)/(2.*((A*TE)**0.5))) ;
MI1 = MI1 ET (PROG Y1) ;
FIN BOU2 ;
EV5 = EVOL MANU XX MI Y MI1 ;
RES = EXTR (SOMM EV5) 1 ;
SOL_DE = 1E-5*(500.-20.)/CO2*((X*CO2)-RES) ;
SINON ;
SOL_DE = EXTR SOLDEENR ITE ;
FINSI ;
DEP_TH = DEP_TH ET (PROG SOL_DE) ;
SI ((FLOT ITE) > ((FLOT IDIM)/2)) ;
ER1 = ER1+(ABS((SOL_TE-TEP1)/SOL_TE)*100.) ;
ER2 = ER2+(ABS((SOL_DE-DEP1)/SOL_DE)*100.) ;
NB = NB + 1 ;
FINSI ;
FIN BOU ;

ER1 = ER1/NB ; ER2 = ER2/NB ;
MESS 'ERROR ON THE TEMPERATURE : 'ER1%' ;
MESS 'ERROR ON THE DISPLACEMENT : 'ER2%' ;

LVAR = TEXT 'EVOLUTION OF THE TEMPERATURE
AT THE POINT'X'M' ;
TITRE LVAR ;
EV_TE = EVOL MANU TIME TPR3 TEMPERATURE TPR1 ;
EV_TET = EVOL MANU TIME TPR3 TEMPERATURE TEP_TH ;
DESS (EV_TE ET EV_TET) ;
LVAR = TEXT 'EVOLUTION OF THE DISPLACEMENT AT
THE POINT'X'M' ;
TITRE LVAR ;
EV_DE = EVOL MANU TIME TPR3 DISPLACEMENT TPR2 ;
EV_DET = EVOL MANU TIME TPR3 DISPLACEMENT DEP_TH ;
DESS (EV_DE ET EV_DET) ;
SI ( ER1 > 2. ) ;
MESS 'INCORRECT RESULT' ;
ERREUR 5 ;
FINSI ;
SI ( ER2 > 5. ) ;
MESS 'INCORRECT RESULT' ;
ERREUR 5 ;
FINSI ;
FIN ;

```



Test thme1 (Comments)

- The THERMIQUE model

```
MOD1 = MODE SU1 THERMIQUE ISOTROPE CONS UN ;
MAT1 = MATE MOD1 K 14.6 'C' 460. RHO 7800. ;
```

The thermic formulation works in three cases: 'ISOTROPE', 'ORTHOTROPE' and 'ANISOTROPE'. Here we study only an isotropic material.

Parameter names for a THERMIQUE ISOTROPE formulation

'K'	isotropic conductivity
'RHO'	specific mass
'C'	specific heat
'H'	exchange coefficient

Parameter names for a THERMIQUE ORTHOTROPE formulation

- Shells (COQ2, COQ3, COQ4, COQ6, COQ8)

'K1','K2','K3'	thermal conductivities
'RHO'	density
'C'	specific heat
'H'	exchange coefficient

- Three-dimensional solid elements

'K1','K2','K3'	thermal conductivities
'RHO'	density
'C'	specific heat
'H'	exchange coefficient

- Two-dimensional solid elements

two-dimensional and axisymmetrical	Fourier's series
'K1','K2'	'K1','K2','K3'
'RHO','H','C'	'RHO','H','C'

Parameter names for a THERMIQUE ANISOTROPE formulation

- Three-dimensional solid elements

'K11','K21','K22'	terms independent of the anisotropic conductivities matrix
'K31','K32','K33'	
'RHO'	density
'H'	exchange coefficient
'C'	specific heat

- Two-dimensional solid elements

two-dimensional and axisymmetrical	Fourier's series
'K11','K21','K22'	'K11','K21','K22'
	'K33'
'RHO','H','C'	'RHO','H','C'

Note: The geometrical properties must be input simultaneously with the material properties when carrying out a thermal analysis shells or a mechanical analysis thick orthotropic shells.

15.2 Test thpl1 (Description sheet)

Test name: thpl1

Calculation type: KINEMATIC ELASTOPLASTIC MATERIAL

Finite element type: QUA4

Topic: A cylindrical bar subjected to a field of temperature. The structure is a 100-mm long and 20-mm diameter cylindrical bar. It is axially embedded at the two tips and subjected to a constant field of temperature (500°C). We solve the problem in axisymetrical mode. We use the PASAPAS non-linear procedure because of the plastic material.


Goal: Calculate the axial stress and the axial plastic deformation.

Reference: Test de la Commission V.P.C.S. Groupe Thermo-plasticite et Statique non-lineaire
Test numero SSNA01/90

Version: 97' customer version

Model description:


Young's modulus	E	=	200 10 ³
Poisson's coefficient	NU	=	0.3
Elastic limit	SYGY	=	300
Stress hardening modulus	H	=	2000
Thermal expansion coefficient	ALPHA	=	1.2 10 ⁻⁵





Test thpl1 (Results)

CASTEM FIGURES



AMPLITUDE
0.00E+00
10.

Test thpl1 (Listing CASTEM)

```

*****
TITR 'A CYLINDRIC BAR SUBJECTED TO A FIELD OF
      TEMPERATURE' ;
*****
OPTI ECHO 0;
OPTI DIME 2 MODE AXIS ELEM QUA4;
TEMPS;
*
*----- GEOMETRY -----*
* Dimensions in millimeters.
RA = 0.0 ; ZA = 0. ; PA = RA ZA ;
RB = 10. ; ZB = 0. ; PB = RB ZB ;
RC = 10. ; ZC = 100.0 ; PC = RC ZC ;
RD = 0.0 ; ZD = 100.0 ; PD = RD ZD ;
DAB = DROI 1 PA PB ; DBC = DROI 10 PB PC ;
DCD = DROI 1 PC PD ; DDA = DROI 10 PD PA ;
SUR = DALL DAB DBC DCD DDA PLAN ;
ELIM SUR 0.001 ;
TITR 'THPL1 : MESH' ;
TRAC QUAL SUR ;
ELSUR = NBEL SUR ;
NOSUR = NBNO SUR ;
*
*----- DESCRIPTION OF THE MODEL OF MATERIAL -----*
MOD = MODE SUR MECANIQUE ELASTIQUE PLASTIQUE
      CINEMATIQUE QUA4 ;
MAT = MATE MOD YOUN 200E3 NU 0.3 SIGY 300 H 2000
      ALPH 1.2E-5 ;
*
*----- BOUNDARY CONDITIONS -----*
CL1 = BLOQUE UZ DAB ; CL2 = BLOQUE UZ DCD ;
CL = CL1 ET CL2 ;
*
*----- DEFINITION OF FIELDS WHICH ARE -----*
*----- RELATIVE TO THEPASAPAS PROCEDURE -----*
* List of pseudo-time where results are expected
T0 = 0. ; T1 = 1. ;
LISTE = PROG T0 T1 ;

* Field-point of temperature at time T0 and T1
TEMP = TABLE;TEMPER = TABLE ;
TEMP . 0 = T0;TEMP . 1 = T1 ;
TEMPER . 0 = MANU CHPO SUR 1 T 0. ;
TEMPER . 1 = MANU CHPO SUR 1 T 500. ;
CHA2 = CHAR T TEMP TEMPER ;

* Mechanic load: any load
LFORCE = PROG 0. 0. ;
FOR = FORCE FR 1000. PC ;
EVOL1 = EVOL MANU TEMPS LISTE FORCE LFORCE ;
CHA1 = CHAR MECA FOR EVOL1 ;

* Initialization of the table TAB1:
* parameters for the PASAPAS procedure
TAB1 = TABLE ;
TAB1.'CHARGEMENT' = (CHA1 ET CHA2) ;
TAB1.'MODELE' = MOD ;
TAB1.'CARACTERISTIQUES' = MAT ;
TAB1.'BLOCAGES_MECANIQUES' = CL ;
TAB1.'TEMPS_CALCULES' = LISTE ;
TAB1.'TEMPS_SAUVES' = LISTE ;

PASAPAS TAB1 ;
*
*----- RESULTS -----*
* Stresses
nn = 1 ;
CHEL1 = TAB1. 'CONTRAINTES'.nn;

CHAM2 = CHAN NOEUD MOD CHEL1 ;
CHPO1 = CHAN CHPO MOD CHAM2 ;

* Extraction of the SMZZ stress
SMZZ1 = EXTR CHPO1 SMZZ PA ;
SMZZ2 = -309.0 ;

* Plastic strains
CHEL10 = TAB1. 'DEFORMATIONS_INELASTIQUES'.nn ;

* Interpolation to the nodes of the mesh
* Utilization of the new chamelems.
CHAM20 = CHAN NOEUD MOD CHEL10 ;
CHPO10 = CHAN CHPO MOD CHAM20 ;

* Extraction of the EPZZ plastic strain
EPZZ1 = EXTR CHPO10 EPZZ PA;
EPZZ2 = -.4455E-2 ;
*
*----- RESULTS PLACARDING -----*
MESS ' RESULTS ' ;
MESS '1) Stresses ' ;
MESS ' Theoretical stress smzz : ' SMZZ2 'Mpa' ;
MESS ' Calculated stress smzz : ' SMZZ1 'Mpa' ;
ERG1 = ( ABS ((SMZZ2 - SMZZ1) / SMZZ2 )) * 100. ;
MESS ' e.g. error of : ' ERG1 '%' ;
MESS '2) Plastic strains ' ;
MESS ' Theoretical plastic straine : ' EPZZ2 ;
MESS ' Calculated plastic strain : ' EPZZ1 ;
ERG2 = ( ABS ((EPZZ2 - EPZZ1) / EPZZ2 )) * 100. ;
MESS ' e.g. error of : ' ERG2 '%' ;
MESS ' Mesh ' ;
MESS ' Number of nodes : ' NOSUR ;
MESS ' Number of elements (qua4) : ' ELSUR ;

DEP1 = TAB1. 'DEPLACEMENTS'.nn ;
DEF0 = DEFO SUR DEP1 0. ;
DEF1 = DEFO SUR DEP1 10. ROUG ;
TITR 'THPL1 : DEFORMED SHAPE' ;
TRAC (DEF0 ET DEF1) ;
*
*----- GOOD WORKING MESSAGE -----*
* Maximum of the relative errors
ERGMAX = MAXI (PROG ERG1 ERG2) ;

SI (ERGMAX <EG 5.) ; ERRE 0 ;
SINON ; ERRE 5 ;
FINSI ;
TEMPS ;
FIN ;
    
```




Test thpl1 (Comments)

- The plastic model with CINEMATIQUE LINEAIRE stress hardening

```
MOD = MODE SUR MECANIQUE ELASTIQUE PLASTIQUE  
      CINEMATIQUE QUA4 ;  
MAT = MATE MOD YOUN 200E3 NU 0.3 SIGY 300 H 2000  
      ALPH 1.2E-5 ;
```

This non-linear behavior works with the Von Mises model with linear kinematic stress hardening. The parameters to be input are the elastic limit (SIGY) and the stress hardening modulus (H).



Chapter 16

MAGNETO-STATIC

16.1 Test Ac2d93 (Description sheet)

Test name: Ac2d93

Calculation type: 2D AXISYMMETRIC VECTOR POTENTIAL, Computation of magnetic field with non linear material

Finite element type: TRI6 QUA8


Topic: A circular coil of rectangular cross section is embeded in an iron box.

Goal: Calculate vector potential and field in the cavity.

Reference: Analytical solution.

Version: 97' customer version




Model description: Soft iron curve for B(H).





Test Ac2d93 (Results)

CASTEM FIGURES



Test Ac2d93 (Listing CASTEM)

```

*****
*
*      2D  AXISYMMETRIC  MAGNETIC FIELD COMPUTATION
*
*      Formulation   : VECTOR POTENTIAL
*
*                      NON LINEAR MATERIAL
*
*****
OPTION DIME 2  ELEM  TRI6  COUL VERT echo 1 ;
GRAPH = 'O' ;
*-----Mesh -----
R1 = 20. ; R2= 25. ; R3= 27. ; R4= 29. ; R5=39. ;
R6 = 130.;Z1= 2.5; Z2= 5. ; Z3= 20. ; Z4= 130.;
Z31=30.;Z32=50.; DI1= 1.; DI2=5.;DI3=60. ;
DI4= 10.;NET2= 6;

DENS DI1 ;
OZ3= 0. Z3; OZ2= 0. Z2;OZ1= 0. Z1;OZ4=0. Z4;
OO= 0. 0.;R0Z0= DI1 0 ;
R1Z0= R1 0. ;R2Z0= R2 0. ;R3Z0 = R3 0. ;
R4Z0 = R4 0.;
R0Z2= DI1 Z2 ; R4Z2 = R4 Z2 ;
DENS DI2 ;
R5Z0= R5 0. ;R5Z3= R5 Z3 ;R0Z3 =DI2 Z3 ;OZ3 = 0. Z3 ;
DENS DI3 ;
R6Z0=R6 0.; R6Z4=R6 Z4 ;
DENS 10 ;
R2Z3= R2 Z3 ; R1Z3= R1 Z3; R4Z1 = R4 Z1;
R3Z1= R3 Z1 ;
DENSITE DI4 ;
R0Z4 = DI4 Z4 ; OZ4 = 0. Z4 ;
NTRA= 1;
FLAN1 =( D OO R0Z0 D R1Z0 ) TRAN DINI DI1 DFIN DI1
(OZ1 MOINS OO ) TRAN DINI DI1 DFIN DI1
(OZ2 MOINS OZ1) COUL VERT ;
*----- COIL SURFACE -----
BOBI =( D R1Z0 R2Z0 ) TRAN DINI DI1 DFIN DI1
(OZ1 MOINS OO ) COUL BLEU ;
*-----
FLAN2 =(INVE( BOBI COTE 3)) TRAN DINI DI1 DFIN DI1
(OZ2 MOINS OZ1) COUL VERT ;
FLAN3 =( D R2Z0 R3Z0 ) TRAN DINI DI1 DFIN DI1
(OZ1 MOINS OO )
TRAN DINI DI1 DFIN DI1 ( OZ2 MOINS OZ1) COUL VERT ;
FLAN4 =( D R3Z0 R4Z0 ) TRAN DINI DI1 DFIN DI
(OZ1 MOINS OO )
TRAN DINI DI1 DFIN DI1 ( OZ2 MOINS OZ1) COUL VERT ;
*-----IRON -----
FER = (D OZ2 R0Z2 D R4Z2 D 6 R4Z0 D R5Z0 D R5Z3 D R0Z3
D OZ3 D OZ2 )
SURF PLANE COUL ROUG ;
*-----EXTERNAL AIR BOX -----
FLAN5 = (D R5Z0 R5Z3 D R6Z4 D R6Z0 D R5Z0)
SURF PLANE ;
*
FLAN6 =(D R0Z3 R5Z3 D R6Z4 D R0Z4 D R0Z3)
SURF PLANE
COUL VERT ;
TUB6 = (D OZ3 R0Z3 D R0Z4 D OZ4 D OZ3)
SURF PLANE COUL vert ;
AIREXT = (FLAN6 ET TUB6 ET FLAN5 ) coul blan ;
AIRIN = (FLAN1 ET FLAN2 ET FLAN3 ET BOBI ET FLAN4 )
coul vert ;
TOUT = AIRIN ET AIREXT ET FER ; ELIM .2 TOUT ;
*
*----- BOUNDARIES -----
*
ENPP = CONTOUR TOUT ;
CEXTR=ENPP POINTS DROITE (130. 0.) (130. 100.) .1 ;
CEXTH=ENPP POINTS DROITE ( 0. 130.) (10. 130.) .1 ;
AXE=ENPP POINTS DROITE ( 0. 0. ) ( 0. 100. ) .1 ;
AX1 =(AIRIN ET FER) POINT DROITE ( 0. 0. ) ( 0. 100. ).1 ;
AX2 = DIFF ( CHAN POI1 AXE) (CHAN POI1 AX1) ;
*
*-----
* FORMER DESCRIPTION IN MILLIMETRES
*----- SHIFT FOR METERS -----
*
deplacer tout homo .001 (0. 0.) ;
*
SI (NEG GRAPH N ) ;
TITRE ' MESH 2D ' ;
TRAC tout ;
FINSI ;
*-----
MU0= 4. * PI * 1.E-7 ;
*
*-----CURENT DESCRIPTION -----
TABCOUR= TABLE ;
DESCOUR TABCOUR 1 BOBI 'AMP' 800.E6 ;
TABB = TABLE ;
*----- AXISYMMETRIC PROBLEM -----
TABB.'AXI' = VRAI ;
NF = FER NBEL ;
FER1 = FER ELEM ( LECT 1 PAS 1 30 ) ;
FER2 = FER ELEM ( LECT 31 PAS 1 NF ) ;
*-----MATERIALS -----
KEVOL = H_B MU0 ;
TABMAT = TABLE ;
OBFER1 = FER1 MODE THERMIQUE ISOTROPE ;
OBFER2 = FER2 MODE THERMIQUE ISOTROPE ;
STN = TABLE ; STN.EV1 = KEVOL ;
TABMAT.OBFER1 = STN ;
STN = TABLE ; STN.EV1 = KEVOL ;
TABMAT.OBFER2 = STN ;
TABB.'TABNUSEC' = TABMAT ;
TABB.'MUAIR' =MU0 ;
* LINEAR MATERIAL CAN BE TRAITED AS A SUPER ELEMENT
* OR NOT
isuper= 1 ;
'SI' ('EGA' isuper 1 ) ;
TABB.'AIRSUP' = AIREXT ;
TABB.'MAITRES' = ( FER CONTOUR) COMPRIS R5Z0 OZ3 ;
TABB.'ENCS' = CEXTR ET CEXTH ET AX2 ;
TABB.'BLOQUE' = BLOQUER 'T' AX1 ;
TABB.'AIR' = AIRIN ;
'SINON' ;
TABB.'AIR' = (AIRIN ET AIREXT) ;
TABB.'BLOQUE' = BLOQUER 'T' (CEXTR ET CEXTH ET AXE ) ;
'FINSI' ;
TABB.'COUR' = TABCOUR ;
*
* POTENTIAL COMPUTATION OR FIRST STEP IF
* NON LINEAR PROBLEM
POT_VECT TABB 'SOLIN' ;
*-----
*-----
SOL1 = (TABB.'POTENTIEL' ) ENLEVER LX ;
RAY = FLAN1 COTE 1 ;
RAY2 = D 10 (0. .0001) ( .020 .0001) ;
* -----SOME POST TREATMENT B COMPUTATION ----
BB = INDUCTIO AIRIN SOL1 VRAI ;
PREF = AIRIN POINT PROCHE ( 0. 0. ) ;
BY0 = EXTR BB 'BY' PREF ;
BY10 = EXTR BB 'BY' ( AIRIN POINT PROCHE ( .010 0.)) ;
TITRE ' COMPOSANTE BY AVANT ET APRES LISSAGE ' ;

```



16.1. TEST AC2D93 (DESCRIPTION SHEET)

```
EVBI = EVOL ROUG CHPO BB 'BY' RAY ;
*----- polynomial smothing if wanted -----
CHLIS = PROI POLY TOUT RAY2 SOL1 1 AXIS ;
BBY = (EXCO CHLIS 'BY' ) NOMC 'BY' ;
BY01 = EXTR BBY 'BY' (RAY2 POINT INITIAL) ;
BAT1 = 800.e6 * 5.e-3 * 5.e-3 * MU0 /10.e-3 ;

option echo 0 ;
MESS '*****';
MESS '* CIRCULAR COIL internal radius 20 mm ' ;
MESS '* total cross section 5*5 mm ' ;
MESS '* SYMMETRY BY HORIZONTAL PLANE ' ;
MESS '* AMPERE mufer >> muo all AMPERE*TURNES' ;
MESS '* in the GAP of 10 mm ' ;
MESS '* ' ;
MESS '* J * EP * HAUT = B/ MU0 * e ' ;
MESS '* WAITED : By = ' bat1 ' TESLAS ' ;
MESS '*****';
MESS ' NUMERICAL DERIVATION BY AU CENTRE ' BY0 ;
MESS ' SMOTHED SOLUTION BY AU CENTRE ' BY01 ;
MESS '*****';
RAP = ABS ((BY01 - bat1) / bat1) ;
SI ( RAP > .01 ) ; ERREUR 5 ; FINSI ;
*****
SI (NEG GRAPH N) ;
TITRE ' POTENTIAL BEFORE AND AFTER SMOTHING ' ;
EVPO1 = EVOL ROUG CHPO SOL1 'T' RAY ;
EVPO2 = EVOL VERT CHPO CHLIS 'A' RAY2 ;
dess (evpo1 et evpo2) ;
TITRE ' BY COMPONENT BEFORE AND AFTER SMOTHING ' ;
EVBI = EVOL ROUG CHPO BB 'BY' RAY ;
EVBI = EVOL VERT CHPO CHLIS 'BY' RAY2 ;
dess (evb1 et evb2) ;
FINSI ;
*
*----- NON LINEAR COMPUTATION -----
* filling
TABB.SOUSTYPE='THERMIQUE' ;
TABB.CRITERE =1.E-4 ;
TABB.NITER =1;
TABB.'OME' = .1 ;
*----- test shortened to 2 iterations -----
TABB.ITERMAX=2;
TABB.NIVEAU =1;
*
*-----
MAG_NLIN TABB ;
*-----
*
SOL2 = enlever (TABB.'POTENTIEL' ) LX ;
BB = INDUCTIO AIRIN SOL2 VRAI ;
BY02 = EXTR BB 'BY' PREF ;
CHLIS = PROI POLY TOUT RAY2 SOL2 1 AXIS ;
BBY = (EXCO CHLIS 'BY' ) NOMC 'BY' ;
BY03 = EXTR BBY 'BY' (RAY2 POINT INITIAL) ;
BAT = 2.4106 ;
MESS '*****';
MESS ' EXPECTED SOLUTIONS AT CENTER BY ' BAT ;
MESS ' NUMERICAL DERIVATION BY ' BY02 ;
MESS ' SMOTHED SOLUTION BY ' BY03 ;
MESS '*****';
itest = 1 ;
*-----GOOD WORKING MESSAGE -----
SI (EGA ITEST 1) ;
RAP = ABS ((BY03 - BAT) / BAT);
SI ( RAP > .01 ) ; ERREUR 5 ; FINSI ;
SINON ;
*COMPUTATION TILL CONVERGENCE OR UP TO 100 MORE ITERATION
TABB.ITERMAX=100;
MAG_NLIN TABB ;
*-----

SOL2 = enlever (TABB.'POTENTIEL' ) LX ;
BB = INDUCTIO AIRIN SOL2 VRAI ;
BY02 = EXTR BB 'BY' PREF ;
CHLIS = PROI POLY TOUT RAY2 SOL2 1 AXIS ;
BBY = (EXCO CHLIS 'BY' ) NOMC 'BY' ;
BY03 = EXTR BBY 'BY' (RAY2 POINT INITIAL) ;
BAT = 2.3468 ;
MESS '*****';
MESS ' EXPECTED SOLUTION ON AXIS BY ' BAT ;
MESS ' NUMERICAL DERIVATION BY AU CENTRE ' BY02 ;
MESS ' SMOTHED SOLUTION BY AU CENTRE ' BY03 ;
MESS '*****';
*
RAP = ABS ((BY03 - BAT) / BAT) ;
SI ( RAP > .01 ) ; ERREUR 5 ; FINSI ;
*****
SI (NEG GRAPH N) ;
TITRE ' POTENTIAL BEFORE AND AFTER SMOTHING ' ;
EVPO1 = EVOL ROUG CHPO SOL2 'T' RAY ;
EVPO2 = EVOL VERT CHPO CHLIS 'A' RAY2 ;
dess (evpo1 et evpo2) ;
TITRE ' BY COMPONENT BEFORE AND AFTER SMOTHING ' ;
EVBI = EVOL ROUG CHPO BB 'BY' RAY ;
EVBI = EVOL VERT CHPO CHLIS 'BY' RAY2 ;
dess (evb1 et evb2) ;
FINSI ;
FINSI ;
FIN ;
```

16.2 Test Ac3d93 (Description sheet)

Test name: Ac3d93

Calculation type: 3D SCALAR POTENTIAL COMPUTATION OF MAGNETIC FIELD Non Linear Material. 2 Scalar potentials method (see Trowbridge - Simkin)

Finite element type: CUB8 or CU20


Topic: A circular coil of rectangular cross section is embeded in an iron box.

Goal: Calculate scalar potential and field in the cavity.

Reference: Analytical solution.

Version: 97' customer version




Model description: Soft iron curve for B(H).





Test Ac3d93 (Results)

CASTEM FIGURES



Test Ac3d93 (Listing CASTEM)

```

*****
*3D MAGNETIC FIELD COMPUTATION 2 POTENTIALS METHOD
*
*REDUCED POTENTIAL (VOLUME CONTAINING INDUCTORS)
*TOTAL POTENTIAL (VOLUME WITHOUT CURRENTS )
*
*****
option echo 1 ;
  TYPEL = 2 ;
GRAPH = 0 ;
  si ( ega typel 2 ) ;
OPTION DIME 2 ELEM QUA8 COUL VERT echo 0 ;
  sinon ;
OPTION DIME 2 ELEM QUA4 COUL VERT echo 0 ;
  finsi ;
*-----MESH -----
R1 =20. ; R2= 25. ; R3= 27. ; R4= 29. ; R5=39.;
R6 = 130.;
Z1= 2.5; Z2= 5. ; Z3= 20. ; Z4= 130.;Z31=30.;
Z32=50.;
DI1= 1.; DI2=5.;DI3=60. ;DI4= 10.;NET2= 6;

  DENS DI1 ;
OZ3= 0. Z3; OZ2= 0. Z2;OZ1= 0. Z1;OZ4=0. Z4;
OO= 0. 0.;R0Z0= DI1 0 ;
R1Z0= R1 0. ;R2Z0= R2 0. ;R3Z0 = R3 0. ;
R4Z0 = R4 0.;
R0Z2= DI1 Z2 ; R4Z2 = R4 Z2 ;
  DENS DI2 ;
R5Z0= R5 0. ;R5Z3= R5 Z3 ;R0Z3 =DI2 Z3 ;
OZ3 = 0. Z3 ;
  DENS DI3 ;
R6Z0=R6 0.; R6Z4=R6 Z4 ;
  DENS I0 ;
R2Z3= R2 Z3 ; R1Z3= R1 Z3; R4Z1 = R4 Z1;
R3Z1= R3 Z1 ;
DENSITE DI4 ;
R0Z4 = DI4 Z4 ; OZ4 = 0. Z4 ;
NTRA= 1;
FLAN1 =( D OO R0Z0 D R1Z0 ) TRAN DINI DI1 DFIN DI1
( OZ1 MOINS OO )
TRAN DINI DI1 DFIN DI1 ( OZ2 MOINS OZ1)
COUL VERT ;
BOBI =( D R1Z0 R2Z0 ) TRAN DINI DI1 DFIN DI1
( OZ1 MOINS OO ) COUL BLEU ;
FLAN2 =(INVE( BOBI COTE 3)) TRAN DINI DI1 DFIN DI1
( OZ2 MOINS OZ1) COUL VERT ;
FLAN3 =( D R2Z0 R3Z0 ) TRAN DINI DI1 DFIN DI1
( OZ1 MOINS OO )
TRAN DINI DI1 DFIN DI1 ( OZ2 MOINS OZ1)
COUL VERT ;
FLAN4 =( D R3Z0 R4Z0 ) TRAN DINI DI1 DFIN DI1
( OZ1 MOINS OO )
TRAN DINI DI1 DFIN DI1 ( OZ2 MOINS OZ1)
COUL VERT ;
FER = (D OZ2 R0Z2 D R4Z2 D 6 R4Z0 D R5Z0 D R5Z3
D R0Z3 D OZ3
D OZ2 )
SURF PLANE COUL ROUG ;

FLAN5 = ( D R5Z0 R5Z3 D R6Z4 D R6Z0 D R5Z0 )
SURF PLANE ;
*
FLAN6 =( D R0Z3 R5Z3 D R6Z4 D R0Z4 D R0Z3 )
SURF PLANE COUL VERT ;
TUB6 = (D OZ3 R0Z3 D R0Z4 D OZ4 D OZ3)
SURF PLANE COUL VERT ;
AIREXT =( FLAN6 ET TUB6 ET FLAN5 ) coul blan ;

AIRIN =(FLAN1 ET FLAN2 ET FLAN3 ET BOBI ET FLAN4)
COUL VERT ;
TFLAN = AIRIN ET FER ET AIREXT ;ELIM .1 TFLAN ;
PTAX = TFLAN POINT DROITE OO ( 0 1. ) .01 ;
TU1 = TFLAN ELEM APPU LARG PTAX ;
TOUT = DIFF TFLAN TU1 ;
*
C1 = (TOUT CONT ) COMPRIS R0Z4 R0Z0 ;
C2 = C1 PROJETER (1 0 ) DROITE OO OZ4 ;
OZ4 = C2 POINT PROCHE OZ4 ;
OO = C2 POINT PROCHE OO ;
TU1 = DALLER C1 ( D 1 R0Z0 OO ) C2 (D OZ4 R0Z4 )
PLAN ;
TUA = (COOR 2 TU1) POINT SUPER Z3 ;
TUA = (TU1 ELEM APPU LARG TUA ) COUL BLAN ;
TUB = (COOR 2 TU1) POINT INFER Z2 ;
TUB= (TU1 ELEM APPU LARG TUB ) COUL VERT ;
TUF = DIFF TU1 ( TUA ET TUB ) COUL ROUG ;
AIRIN = (TOUT ELEM VERT ) ET TUB ;
FER = (TOUT ELEM ROUG ) ET TUF ;
AIREXT = ( TOUT ELEM BLAN) ET TUA ;
* TRAC (AIRIN ET FER ET AIREXT ) FACE ;
*-----
*----- SHIFT TO 3D -----
*-----
OPTION ELEM CUB8 ;
SI ( EGA TYPEL 2 )
OPTION DIME 3 ELEM CU20 ;
FINSI ;
ANG = 20. ;
*
*----- DEFINITION OF THE 2 DOMAINS -----
*----- REDUCED POTENTIAL VOLUME -----
DPHI = AIRIN VOLU 1 ROTA ANG (0 0 0 ) (0 1 0 ) ;
**----- TOTAL POTENTIAL VOLUME -----
DPSI = (AIREXT ET FER) VOLU 1 ROTA ANG (0 0 0 )
(0 1 0 ) ;
*
ELIM .001 DPHI ; DPHI = REGE DPHI ;
ELIM .001 DPSI ; DPSI = REGE DPSI ;
ELIM .001 ( DPHI ET DPSI) ;

*-----SEPARATION SURFACE IDENTIFICATION -----
* --- BELONGS TO REDUCED POTENTIAL VOLUME -----
SEP_PHI = INTE ( DPSI ENVEL ) ( DPHI ENVEL ) ;
*----- SPLITTING OF THE 2 DOMAINS -----
DPSI = DPSI PLUS ( 0 0 0 ) ;
SEP_PSI = SEP_PHI PLUS ( 0 0 0 ) ; ELIM .01 DPSI
SEP_PSI ;
*-----IRON-----
DFER = DPSI ELEM ROUG ;
*
VOL1 = DPHI ET DPSI ;
DAIR = DPSI ELEM BLAN ;
AA = DPHI POINT PLAN (0 0 0 ) ( 10 0 0 ) (0 0 10 )
.00001 ;
AB =( DPHI ENVEL ) ELEM APPU STRIC AA ;
AA = (AB ENVEL ) POINT DROITE (0 0 0 ) ( 10 0 0 )
.00001 ;
AA = ( (AB CONTOUR ) ELEM APPU STRIC AA ) ;
deplacer voll tourner 90 ( 0 0 0 ) ( 1 0 0 ) ;
*
ORIG= SEP_PHI POINT PROCHE (29. 0 0 ) ;
BANTI= SEP_PHI POINT PLAN (0. 0. 0.) (10. 0. 0.)
(0. 10. 0.) .1 ;
geomed = (dpsi et dphi ) point plan (0 0 0) ( 1 0 0)
(0 1 0 ) .1 ;

```



```

MU0 = 4. * PI * 1.E-4 ;
RINT = 20. ;REXT =25. ; HAUT = 5. ; J = 800. ;
C1 = 0. 0. 0. ; ZP1 = 20. 0. 0. ;ZP2 = 0. 20. 0. ;
E = Z1 * 4 ;
UNIT = M ;
  si ( ega UNIT M ) ;
deplacer voll homo .001 (0. 0. 0.) ;
MU0 = 4. * PI * 1.E-7 ;
RINT = RINT/1000; REXT = REXT/1000 ; HAUT = HAUT/1000 ;
ZP1 = ZP1/1000;ZP2 = ZP2 / 1000 ;J = J*1.E6 ;
E = E /1000 ;
  finsi ;
option echo 1 ;
*-----
*-- BIOT and SAVART COMPUTATION of Hs
*-- ON REDUCED POTENTIAL DOMAIN
*-----
BS = BIOT SEP_PHI CERC C1 ZP1 ZP2 RINT REXT HAUT
  J MU0 ;
HS = BS / MU0 ;

  SI (NEG GRAPH 'N' ) ;
OEIL = 1000 1000 500 ;
TITRE ' HS ON SEPARATION SURFACE ' ;
VVVV = VECTEUR BS 1.e-3 'BX' 'BY' 'BZ' ROUG ;
TRAC OEIL SEP_PHI VVVV ;
option donn 5 ;
  FINSI ;
*----- FILLING OF A TABLE -----
*-in this computation there is only one isotropic
* material but it
* is shown how to coope with multi-material
* (even orthotropic )
*-----
TABB= TABLE ;
*
FER1 = DFER ELEM CU20 ;
FER2 = DFER ELEM PR15 ;
*
KEVOL1 = H_B MU0 ;
KEVOL2 = H_B MU0 ;
OBFER1 = FER1 MODE THERMIQUE ISOTROPE ;
OBFER2 = FER2 MODE THERMIQUE ORTHOTROPE ;
TABDEF = TABLE ;
*----- isotropic material -----
STN = TABLE ;
STN.'EV1' = KEVOL1 ;
TABDEF.OBFER1 = STN ;
*----- orthotropic material -----
STN = TABLE ;
DI1 = (0 1 0 ) ;DI2 = (1 0 1 ) ;
STN.'EV1' = KEVOL1 ;STN.'DIR1' = DI1 ;
STN.'EV2' = KEVOL2 ;STN.'DIR2' = DI2 ;
TABDEF.OBFER2 = STN ;
TABB.'TABNUSEC' = TABDEF ;
*-----
TABB.'BIOT'=HS ;
*----- REDUCED POTENTIAL AREA -----
*---
TABB.'DPHI' = DPHI ;
  imet = 0 ;
  si ( ega imet 1 ) ;
MOD1 = DPHI MODE THERMIQUE ISOTROPE ;
MAT1 = MATE MOD1 'K' MU0 ;
RIGC1 = CONduc MOD1 MAT1 ;
TABB.'RIGCON' = RIGC1 ;
  finsi ;
*
TABB.'SEP PHI' = SEP_PHI ;
TABB.'ORIG' = ORIG ;
TABB.'MUAIR' = MU0 ;

*
*----- TOTAL POTENTIAL AREA -----
*-----
TABB.'DPSI' = DPSI ;
TABB.'AIRPSI' = DPSI ELEM BLAN ;
  si ( ega imet 1 ) ;
MOD2 = DAIR MODE THERMIQUE ISOTROPE ;
MAT2 = MATE MOD2 'K' MU0 ;
RIGC2 = CONduc MOD2 MAT2 ;
TABB.'RIGCPHI' = RIGC2 ;
  finsi ;

* TABB.'SEP_PSI' = SEP_PSI ;
*----- GENERAL BOUDARY CONDITIONS -----
TABB.'BLOQUE' = BLOQUER 'T' GEOMED ;
*
*----- BOUNDARY CONDITION LINKED TO POTENTIAL JUMP --
*
TABA = TABLE ;
STN = TABLE ;TABA.1 = STN ;
STN.'LGEO' = BANTI ;STN.'MTYP' = 'TBLOQ' ;
TABB.'POTSYM' = TABA ;
TABB.ISTEP = 1 ;
*----- FIRST STEP POTENTIALS COMPUTATION
POT_SCAL TABB 'SOLIN' ;
*-----
SOL1= TABB.'POTENTIEL' ;
PREF= 0. 0. 0. ;
*----- FIELD COMPUTATION (B) -----
OBJM= MODL DPHI THERMIQUE ISOTROPE ;
BA= CHAN CHPO ( GRAD OBJM SOL1 ) OBJM ;
BA= BA * -1. * MU0 ;
BM= EXTR BA 'T,Z' OO ;
option echo 0 ;
SAUTER 5 LIGNES ;
MESS '*****';
MESS ' CIRCULAR COIL INTERNAL RADIUS' RINT 'mm *';
MESS '*CROSS SECTION ' ( REXT - RINT) '* HAUT 'mm2';
MESS '* CURRENT DENSITY ' J 'A/mm2 *';
MESS '*****';
BS = BIOT AA CERC C1 ZP1 ZP2 RINT REXT HAUT J MU0 ;
MESS '*INDUCTION ON AXIS WITHOUT IRON (BIOT SAVART)';
BS0= EXTR BS 'BZ' OO ;

MESS '*****';
MESS '* AMPERE mufer >> muo ALL AMPERE*TURNS IN THE GAP';
MESS '* GAP = ' E 'mm ' ;
MESS '* J * EP * HAUT = B/ MU0 * e ' ;
MESS '*****';
BTH = MU0 * J * (REXT - RINT) * HAUT / E ;
BCAL = BM + BS0 ;
MESS '*****';
MESS ' BIOT ET SAVART ' BS0 ;
MESS ' TOTAL B on AXIS ( z=0 ) EXPECTED ' BTH;
MESS ' COMPUTED ' BCAL;
MESS '*****';
  SI (NEG GRAPH 'N' ) ;
  TABLE= TABLE ;TATT = TABLE ;
  TABLEG.1 = 'MARQ CROI' ;TATT.1= 'B_COIL' ;
  TABLEG.2 = 'MARQ PLUS' ;TATT.2= 'B_IRON' ;
  TABLEG.3 = 'MARQ CARR' ;TATT.3= 'B_TOTAL' ;
  TABLEG.TITRE = TATT ;
  BRBIO = ( EXCO BS 'BZ' ) NOMC 'SCAL' ;
  BMRAY = EXCO (REDU BA AA) 'T,Z' ;
  EVBI = EVOL ROUG CHPO BRBIO 'SCAL' AA ;
  EVBM = EVOL VERT CHPO BMRAY 'SCAL' AA ;
  EVBT = EVOL TURQ CHPO (BRBIO + BMRAY) 'SCAL' AA ;
  TITRE ' BZ(R) for Z=0 ' ;
  DESS (EVBI ET EVBM ET EVBT ) LEGE TABLEG TITY TESLAS ;
  FINSI ;
  RAP = ABS ((BCAL - BTH) / BTH) ;
  SI ( RAP > .01 ) ; ERREUR 5 ; FINSI ;

```

```

*-----
*                               NON LINEAR  COMPUTATION
*-----
TABB.SOUSTYPE = THERMIQUE ;
TABB.CRITERE =1.E-4 ;
TABB.NITER   =1;
TABB.'OME' = .99 ;
* -----2 it\erations only for the test -----
TABB.ITERMAX=2;
TABB.NIVEAU =1;
*-----
MAG_NLIN TABB ;
*-----
SOL2 = TABB.'POTENTIEL';
BA= CHAN CHPO ( GRAD OBJM SOL2 ) OBJM ;
BA= BA * -1. * MU0 ;
BM= EXTR BA 'T,Z' OO ;
*-----
ITEST = 1 ;
SI ( EGA ITEST 1 ) ;
MESS '*****';
MESS ' VERIFICATION AU CENTRE ' ;
MESS ' CONTRIBUTION FER ATTENDUE 1.9496 CALCULEE ' BM;
*****';
BAT = 1.9496 ;
*****
RAP = ABS ((BM - BAT ) / BAT );
SI ( RAP > .01 ) ; ERREUR 5 ; FINSI ;
*****
FIN ;
FINSI ;
*-----
*- CONTINUATION TILL CONVERGENCE OR UP TO 100 ITERATION
*-----
TABB.ITERMAX=100 ;
*-----
MAG_NLI3 TABB ;
*-----
SOL2 = TABB.'POTENTIEL';
BA= CHAN CHPO ( GRAD OBJM SOL2 ) OBJM ;
BA= BA * -1. * MU0 ;
BM= EXTR BA 'T,Z' OO ;
BCAL = BM + BS0 ;
MESS '*****';
MESS ' CHECK AT CENTER OF THE DEVICE ' ;
MESS ' EXPECTED IRON CONTRIBUTION 1.9059 COMPUTED ' BM;
MESS ' TOTAL BZ COMPUTED AT CENTER ' BCAL;
MESS '*****';
FIN;

```